

Roman Schmied

# Using Mathematica for Quantum Mechanics

A Student's Manual



Springer

# Using Mathematica for Quantum Mechanics

Roman Schmied

# Using Mathematica for Quantum Mechanics

A Student's Manual



Springer

Roman Schmied  
Department of Physics  
University of Basel  
Basel, Switzerland

Additional material to this book can be downloaded from <http://extras.springer.com>.

ISBN 978-981-13-7587-3                      ISBN 978-981-13-7588-0 (eBook)  
<https://doi.org/10.1007/978-981-13-7588-0>

© Springer Nature Singapore Pte Ltd. 2020

This work is subject to copyright. All rights are reserved by the Publisher, whether the whole or part of the material is concerned, specifically the rights of translation, reprinting, reuse of illustrations, recitation, broadcasting, reproduction on microfilms or in any other physical way, and transmission or information storage and retrieval, electronic adaptation, computer software, or by similar or dissimilar methodology now known or hereafter developed.

The use of general descriptive names, registered names, trademarks, service marks, etc. in this publication does not imply, even in the absence of a specific statement, that such names are exempt from the relevant protective laws and regulations and therefore free for general use.

The publisher, the authors and the editors are safe to assume that the advice and information in this book are believed to be true and accurate at the date of publication. Neither the publisher nor the authors or the editors give a warranty, expressed or implied, with respect to the material contained herein or for any errors or omissions that may have been made. The publisher remains neutral with regard to jurisdictional claims in published maps and institutional affiliations.

This Springer imprint is published by the registered company Springer Nature Singapore Pte Ltd.  
The registered company address is: 152 Beach Road, #21-01/04 Gateway East, Singapore 189721, Singapore

# Preface

*The limits of my language mean the limits of my world.*  
Ludwig Wittgenstein

Learning quantum mechanics is difficult and counter-intuitive. The first lectures I heard were filled with strange concepts that had no relationship with the mechanics I knew, and it took me years of solving research problems until I acquired even a semblance of understanding and intuition. This process is much like learning a new language, in which a solid mastery of the concepts and rules is required before new ideas and relationships can be expressed fluently.

The major difficulty in bridging the chasm between introductory quantum lectures, on the one hand, and advanced research topics, on the other hand, was for me the lack of such a language, or of a technical framework in which quantum ideas could be expressed and manipulated. On the one hand, I had the hand tools of algebraic notation, which are understandable but only serve to express very small systems and ideas; on the other hand I had diagrams, circuits, and quasi-phenomenological formulae that describe interesting research problems, but which are difficult to grasp with the mathematical detail I was looking for.

This book is an attempt to help students transform all of the concepts of quantum mechanics into concrete computer representations, which can be constructed, evaluated, analyzed, and hopefully understood at a deeper level than what is possible with more abstract representations. It was written for a Master's and Ph.D. lecture given yearly at the University of Basel, Switzerland. The goal is to give a language to the student in which to speak about quantum physics in more detail, and to start the student on a path of fluency in this language. We will revisit most of the problems encountered in introductory quantum mechanics, focusing on computer implementations for finding analytical as well as numerical solutions and their visualization. On our journey we approach questions such as

- You already know how to calculate the energy eigenstates of a single particle in a simple one-dimensional potential. How can such calculations be generalized to non-trivial potentials, higher dimensions, and interacting particles?

- You have heard that quantum mechanics describes our everyday world just as well as classical mechanics does, but have you ever seen an example where such behavior is calculated in detail and where the transition from classical to quantum physics is evident?
- How can we describe the internal spin structure of particles? How does this internal structure couple to the particles' motion?
- What are qubits and quantum circuits, and how can they be assembled to simulate a future quantum computer?

Most of the calculations necessary to study and visualize such problems are too complicated to be done by hand. Even relatively simple problems, such as two interacting particles in a one-dimensional trap, do not have analytic solutions and require the use of computers for their solution and visualization. More complex problems scale exponentially with the number of degrees of freedom, and make the use of large computer simulations unavoidable.

The methods presented in this book do not pretend to solve large-scale quantum-mechanical problems in an efficient way; the focus here is more on developing a descriptive language. Once this language is established, it will provide the reader with the tools for understanding efficient large-scale calculations better.

## Why Mathematica?

This book is written in the *Wolfram language* of Mathematica (version 11); however, any other language such as MATLAB or Python may be used with suitable translation, as the core ideas presented here are not specific to the Wolfram language.

There are several reasons why Mathematica was chosen over other computer-algebra systems:

- Mathematica is a very high-level programming environment, which allows the user to focus on *what* s/he wants to do instead of *how* it is done. The Wolfram language is extremely expressive and can perform deep calculations with very short and unencumbered programs.
- Mathematica supports a wide range of programming paradigms, which means that you can keep programming in your favorite style. See Sect. 1.9 for a concrete example.
- The *Notebook* interface of Mathematica provides an interactive experience that holds programs, experimental code, results, and graphics in one place.
- Mathematica seamlessly mixes analytic and numerical facilities. For many calculations it allows you to push analytic evaluations as far as possible, and then continue with numerical evaluations by making only minimal changes.
- A very large number of algorithms for analytic and numerical calculations is included in the Mathematica kernel and its libraries.

## Mathematica Source Code

Some sections of this book contain Mathematica source code files that can be downloaded through given hyperlinks. See the section “List of Notebooks in Supplementary Material” for a list of embedded files.

## Outline of Discussed Topics

In five chapters, this book takes the student all the way to relatively complex numerical simulations of quantum circuits and interacting particles with spin:

Chapter 1 gives an introduction to Mathematica and the Wolfram language, with a focus on techniques that will be useful for this book. This chapter can be safely skipped or replaced by an alternative introduction to Mathematica.

Chapter 2 makes the connection between quantum mechanics and vector/matrix algebra. In this chapter, the abstract concepts of quantum mechanics are converted into computer representations, which form the basis for the following chapters.

Chapter 3 discusses quantum systems with finite-dimensional Hilbert spaces, focusing on spin systems and qubits. These are the most basic quantum-mechanical elements and are ideal for making a first concrete use of the tools of Chap. 2.

Chapter 4 discusses the quantum mechanics of particles moving in one- and several-dimensional space. We develop a real-space description of these particles’ motion and interaction, and stay as close as possible to the classical understanding of particle motion in phase space.

Chapter 5 connects the topics of Chaps. 3 and 4, describing particles with spin that move through space.

Basel, Switzerland

Roman Schmied

# Contents

<b>1</b>	<b>Wolfram Language Overview</b>	<b>1</b>
1.1	Introduction	2
1.1.1	Exercises	3
1.2	Variables and Assignments	4
1.2.1	Immediate and Delayed Assignments	4
1.2.2	Exercises	5
1.3	Four Kinds of Bracketing	5
1.4	Prefix and Postfix	6
1.4.1	Exercises	7
1.5	Programming Constructs	7
1.5.1	Procedural Programming	7
1.5.2	Exercises	9
1.5.3	Functional Programming	9
1.5.4	Exercises	10
1.6	Function Definitions	11
1.6.1	Immediate Function Definitions	11
1.6.2	Delayed Function Definitions	12
1.6.3	Memoization: Functions that Remember Their Results	12
1.6.4	Functions with Conditions on Their Arguments	13
1.6.5	Functions with Optional Arguments	14
1.7	Rules and Replacements	15
1.7.1	Immediate and Delayed Rules	16
1.7.2	Repeated Rule Replacement	16
1.8	Debugging and Finding Out How Mathematica Expressions are Evaluated	17
1.8.1	Exercises	18




1.9	Many Ways to Define the Factorial Function . . . . .	19
1.9.1	Exercises . . . . .	21
1.10	Vectors, Matrices, Tensors . . . . .	21
1.10.1	Vectors . . . . .	21
1.10.2	Matrices . . . . .	22
1.10.3	Sparse Vectors and Matrices . . . . .	23
1.10.4	Matrix Diagonalization . . . . .	24
1.10.5	Tensor Operations . . . . .	26
1.10.6	Exercises . . . . .	28
1.11	Complex Numbers . . . . .	28
1.12	Units . . . . .	29
<b>2</b>	<b>Quantum Mechanics: States and Operators . . . . .</b>	<b>33</b>
2.1	Basis Sets and Representations . . . . .	34
2.1.1	Incomplete Basis Sets . . . . .	35
2.1.2	Exercises . . . . .	36
2.2	Time-Independent Schrödinger Equation . . . . .	36
2.2.1	Diagonalization . . . . .	38
2.2.2	Exercises . . . . .	38
2.3	Time-Dependent Schrödinger Equation . . . . .	38
2.3.1	Time-Independent Basis . . . . .	39
2.3.2	Time-Dependent Basis: Interaction Picture . . . . .	40
2.3.3	Special Case: $[\hat{\mathcal{H}}(t), \hat{\mathcal{H}}(t')] = 0 \forall (t, t')$ . . . . .	41
2.3.4	Special Case: Time-Independent Hamiltonian . . . . .	41
2.3.5	Exercises . . . . .	42
2.4	Basis Construction . . . . .	42
2.4.1	Description of a Single Degree of Freedom . . . . .	42
2.4.2	Description of Coupled Degrees of Freedom . . . . .	43
2.4.3	Reduced Density Matrices . . . . .	46
2.4.4	Exercises . . . . .	49
<b>3</b>	<b>Spin and Angular Momentum . . . . .</b>	<b>51</b>
3.1	Quantum-Mechanical Spin and Angular Momentum Operators . . . . .	52
3.1.1	Exercises . . . . .	53
3.2	Spin-1/2 Electron in a DC Magnetic Field . . . . .	54
3.2.1	Time-Independent Schrödinger Equation . . . . .	55
3.2.2	Exercises . . . . .	56
3.3	Coupled Spin Systems: $^{87}\text{Rb}$ Hyperfine Structure . . . . .	56
3.3.1	Eigenstate Analysis . . . . .	58
3.3.2	“Magic” Magnetic Field . . . . .	61

3.3.3	Coupling to an Oscillating Magnetic Field . . . . .	61
3.3.4	Exercises . . . . .	67
3.4	Coupled Spin Systems: Ising Model in a Transverse Field . . . . .	68
3.4.1	Basis Set . . . . .	69
3.4.2	Asymptotic Ground States . . . . .	70
3.4.3	Hamiltonian Diagonalization . . . . .	71
3.4.4	Analysis of the Ground State . . . . .	72
3.4.5	Exercises . . . . .	79
3.5	Coupled Spin Systems: Quantum Circuits . . . . .	80
3.5.1	Quantum Gates . . . . .	80
3.5.2	A Simple Quantum Circuit . . . . .	83
3.5.3	Application: The Quantum Fourier Transform . . . . .	85
3.5.4	Application: Quantum Phase Estimation . . . . .	87
3.5.5	Exercises . . . . .	90
<b>4</b>	<b>Quantum Motion in Real Space . . . . .</b>	<b>91</b>
4.1	One Particle in One Dimension . . . . .	92
4.1.1	Units . . . . .	93
4.1.2	Computational Basis Functions . . . . .	94
4.1.3	The Position Operator . . . . .	99
4.1.4	The Potential-Energy Operator . . . . .	100
4.1.5	The Kinetic-Energy Operator . . . . .	101
4.1.6	The Momentum Operator . . . . .	101
4.1.7	Example: Gravity Well . . . . .	102
4.1.8	The Wigner Quasi-probability Distribution . . . . .	107
4.1.9	1D Dynamics in the Square Well . . . . .	109
4.1.10	1D Dynamics in a Time-Dependent Potential . . . . .	115
4.2	Many Particles in One Dimension: Dynamics with the Non-linear Schrödinger Equation . . . . .	116
4.2.1	Imaginary-Time Propagation for Finding the Ground State of the Non-linear Schrödinger Equation . . . . .	119
4.3	Several Particles in One Dimension: Interactions . . . . .	122
4.3.1	Two Identical Particles in One Dimension with Contact Interaction . . . . .	122
4.3.2	Two Particles in One Dimension with Arbitrary Interaction . . . . .	129
4.4	One Particle in Several Dimensions . . . . .	130
4.4.1	Exercises . . . . .	134
<b>5</b>	<b>Combining Spatial Motion and Spin . . . . .</b>	<b>137</b>
5.1	One Particle in 1D with Spin . . . . .	138
5.1.1	Separable Hamiltonian . . . . .	138
5.1.2	Non-separable Hamiltonian . . . . .	139
5.1.3	Exercises . . . . .	145

5.2	One Particle in 2D with Spin: Rashba Coupling . . . . .	145
5.2.1	Exercises . . . . .	148
5.3	Phase-Space Dynamics in the Jaynes–Cummings Model . . . . .	148
5.3.1	Exercises . . . . .	152
<b>List of Notebooks in Supplementary Material . . . . .</b>		<b>153</b>
<b>Solutions to Exercises . . . . .</b>		<b>155</b>
<b>Index . . . . .</b>		<b>191</b>

# Chapter 1

## Wolfram Language Overview



The Wolfram language is a beautiful and handy tool for expressing a wide variety of technical thoughts. Wolfram Mathematica is the software that implements the Wolfram language. In this chapter, we have a look at the most central parts of this language, without focusing on quantum mechanics yet. Students who are familiar with the Wolfram language may skip this chapter; others may prefer alternative introductions. Wolfram Research, the maker of Mathematica and the Wolfram language, provides many resources for learning:

- <https://www.wolfram.com/mathematica/resources/>—an overview of Mathematica resources to learn at your own pace


---

**Electronic supplementary material** The online version of this chapter ([https://doi.org/10.1007/978-981-13-7588-0\\_1](https://doi.org/10.1007/978-981-13-7588-0_1)) contains supplementary material, which is available to authorized users.

- <https://reference.wolfram.com/language/guide/LanguageOverview.html>—an overview of the Wolfram language
- <https://www.wolfram.com/language/>—the central resource for learning the Wolfram language
- <https://reference.wolfram.com/language/>—the Mathematica documentation.

## 1.1 Introduction

Wolfram Mathematica is an interactive system for mathematical calculations. The Mathematica system is composed of two main components: the *front end*, where you write the input in the Wolfram language, give execution commands, and see the output, and the *kernel*, which does the actual calculations.



This distinction is important to remember because the kernel remembers all the operations in the order they are sent to it, and this order may have nothing to do with the order in which these commands are displayed in the front end.

When you start Mathematica you see an empty “notebook” in which you can write commands. These commands are written in a mixture of text and mathematical symbols and structures, and it takes a bit of practice to master all the special input commands. In the beginning you can write all your input in pure text mode, if you prefer. Let’s try an example: add the numbers  $2 + 3$  by giving the input

```
1 In[1]:= 2+3
```

and, with the cursor anywhere within the “cell” containing this text (look on the right edge of the notebook to see cell limits and groupings) you press “shift-enter”. This sends the contents of this cell to the kernel, which executes it and returns a result that is displayed in the next cell:

```
1 Out[1]= 5
```

If there are many input cells in a notebook, they only get executed in order if you select “Evaluate Notebook” from the “Evaluation” menu; otherwise you can execute the input cells in any order you wish by simply setting the cursor within one cell and pressing “shift-enter”.

The definition of any function or symbol can be called up with the `?`  command:

```
1 In[2]:= ?Factorial
2 n! gives the factorial of n. >>
```

The arrow  $\gg$  that appears at the end of this informative text is a hyperlink into the documentation, where (usually) instructive examples are presented.

### 1.1.1 Exercises

Do the following calculations in Mathematica, and try to understand their structure:

**Q1.1** Calculate the numerical value of the Riemann zeta function  $\zeta(3)$  with

```
1 In[3] := N[Zeta[3]]
```

**Q1.2** Square the previous result (%) with

```
1 In[4] := %^2
```

**Q1.3** Calculate  $\int_0^\infty \sin(x)e^{-x}dx$  with

```
1 In[5] := Integrate[Sin[x]*Exp[-x], {x, 0, Infinity}]
```

**Q1.4** Calculate the first 1000 digits of  $\pi$  with

```
1 In[6] := N[Pi, 1000]
```

or, equivalently, using the Greek symbol  $\pi$ =Pi,

```
1 In[7] := N[ $\pi$ , 1000]
```

**Q1.5** Calculate the analytic and numeric values of the Clebsch–Gordan coefficient  $\langle 100, 10; 200, -12 | 110, -2 \rangle$ :

```
1 In[8] := ClebschGordan[{100, 10}, {200, -12}, {110, -2}]
```

**Q1.6** Calculate the limit  $\lim_{x \rightarrow 0} \frac{\sin x}{x}$  with

```
1 In[9] := Limit[Sin[x]/x, x -> 0]
```

**Q1.7** Make a plot of the above function with

```
1 In[10] := Plot[Sin[x]/x, {x, -20, 20}, PlotRange -> All]
```

**Q1.8** Draw a Mandelbrot set with

```
1 In[11] := F[c_, imax_] := Abs[NestWhile[#^2+c&, 0., Abs[#]<=2&, 1, imax]] <= 2
2 In[12] := With[{n = 100, imax = 1000},
3 Graphics[Raster[Table[Boole[!F[x+I*y, imax]], {y, -2, 2, 1/n}, {x, -2, 2, 1/n}]]]]
```

**Q1.9** Do the same with a built-in function call:

```
1 In[13]:= MandelbrotSetPlot[]
```

## 1.2 Variables and Assignments

<https://reference.wolfram.com/language/howto/WorkWithVariablesAndFunctions.html>.

Variables in the Wolfram language can be letters or words with uppercase or lowercase letters, including Greek symbols. Assigning a value to a variable is done with the `=` symbol,

```
1 In[14]:= a = 5
2 Out[14]= 5
```

If you wish to suppress the output, then you must end the command with a semi-colon:

```
1 In[15]:= a = 5;
```

The variable name can then be used anywhere in an expression:

```
1 In[16]:= a + 2
2 Out[16]= 7
```

### 1.2.1 Immediate and Delayed Assignments

<https://reference.wolfram.com/language/tutorial/ImmediateAndDelayedDefinitions.html>.

Consider the two commands

```
1 In[17]:= a = RandomReal[]
2 Out[17]= 0.38953
3 In[18]:= b := RandomReal[]
```

(your random number will be different).

The first statement `a=...` is an *immediate assignment*, which means that its right-hand side is evaluated when you press shift-enter, produces a specific random value, and is assigned to the variable `a` (and printed out). From now on, every time you use the variable `a`, the *exact same* number will be substituted. In this sense, the variable `a` contains the number 0.389 53 and has no memory of where it got this number from. You can check the definition of `a` with `?a`:

```
1 In[19]:= ?a
2 Global`a
3 a = 0.38953
```

The definition `b:=...` is a *delayed assignment*, which means that when you press shift-enter the right-hand side is not evaluated but merely stored as a definition of `b`. From now on, every time you use the variable `b`, its right-hand-side definition will be substituted and executed, resulting in a new random number each time. You can check the definition of `b` with

```
1 In[20]:= ?b
2          Global`b
3          b := RandomReal[]
```

Let's compare the repeated performance of `a` and `b`:

```
1 In[21]:= {a, b}
2 Out[21]= {0.38953, 0.76226}
3 In[22]:= {a, b}
4 Out[22]= {0.38953, 0.982921}
5 In[23]:= {a, b}
6 Out[23]= {0.38953, 0.516703}
7 In[24]:= {a, b}
8 Out[24]= {0.38953, 0.0865169}
```

If you are familiar with computer file systems, you can think of an immediate assignment as a *hard link* (a direct link to a precomputed inode number) and a delayed assignment as a *soft link* (symbolic link, textual instructions for how to find the linked target).

## 1.2.2 Exercises

**Q1.10** Explain the difference between

```
1 In[25]:= x = u + v
```

and

```
1 In[26]:= y := u + v
```

In particular, distinguish the cases where `u` and `v` are already defined before `x` and `y` are defined, where they are defined only afterwards, and where they are defined before but change values after the definition of `x` and `y`.

## 1.3 Four Kinds of Bracketing

<https://reference.wolfram.com/language/tutorial/TheFourKindsOfBracketingInTheWolframLanguage.html>.

There are four types of brackets in the Wolfram language:

- parentheses for grouping, for example in mathematical expressions:



```
1 In[27]:= 2*(3-7)
```

- square brackets for function calls:

```
1 In[28]:= Sin[0.2]
```

- curly braces for lists:

```
1 In[29]:= v = {a, b, c}
```

- double square brackets for indexing within lists: (see Sect. 1.10)

```
1 In[30]:= v[[2]]
```

## 1.4 Prefix and Postfix

There are several ways of evaluating a function call in the Wolfram language, and we will see most of them in this book. As examples of function calls with a single argument, the main ways in which  $\sin(0.2)$  and  $\sqrt{2+3}$  can be calculated are

**Standard notation** (infinite precedence):

```
1 In[31]:= Sin[0.2]
2 Out[31]= 0.198669
3 In[32]:= Sqrt[2+3]
4 Out[32]= Sqrt[5]
```

**Prefix notation** with @ (quite high precedence, higher than multiplication):

```
1 In[33]:= Sin @ 0.2
2 Out[33]= 0.198669
3 In[34]:= Sqrt @ 2+3
4 Out[34]= 3+Sqrt[2]
```

Notice how the high precedence of the @ operator effectively evaluates  $(\text{Sqrt}@2)+3$ , not  $\text{Sqrt}@ (2+3)$ .

**Postfix notation** with // (quite low precedence, lower than addition):

```
1 In[35]:= 0.2 //Sin
2 Out[35]= 0.198669
3 In[36]:= 2+3 //Sqrt
4 Out[36]= Sqrt[5]
```

Notice how the low precedence of the // operator effectively evaluates  $(2+3) // N$ , not  $2 + (3 // N)$ .

Postfix notation is often used to transform the output of a calculation:

- Adding `//N` to the end of a command will convert the result to decimal representation, if possible.
- Adding `//MatrixForm` to the end of a matrix calculation will display the matrix in a tabular form.
- Adding `//Timing` to the end of a calculation will display the result together with the amount of time it took to execute.

If you are not sure which form is appropriate, for example if you don't know the precedence of the involved operations, then you should use the standard notation or place parentheses where needed.

### 1.4.1 Exercises

**Q1.11** Calculate the decimal value of Euler's constant  $e$  (`E`) using standard, prefix, and postfix notation.

## 1.5 Programming Constructs

When you program in the Wolfram language you can choose between a number of different programming paradigms, and you can mix these as you like. Depending on the chosen style, your program may run much faster or much slower.

### 1.5.1 Procedural Programming

<https://reference.wolfram.com/language/guide/ProceduralProgramming.html>.

A subset of the Wolfram language behaves very similarly to C, Python, Java, or other procedural programming languages. Be very careful to distinguish semi-colons, which separate commands within a single block of code, from commas, which separate different code blocks!

**Looping constructs** behave like in common programming languages:

```
1 In[37]:= For[i = 1, i <= 5, i++,
2           Print[i]]
3           1
4           2
5           3
6           4
7           5
```

Notice that `i` is now a globally defined variable, which you can check with

```
1 In[38]:= ?i
2          Global`i
3          i=6
```

The following, on the other hand, does not define the value of the variable `j` in the global context:

```
1 In[39]:= Do[Print[j], {j, 1, 5}]
2          1
3          2
4          3
5          4
6          5
7 In[40]:= ?j
8          Global`j
```

In this sense, `j` is a local variable in the `Do` context. The following, again, defines `k` as a global variable:

```
1 In[41]:= k = 1;
2          While[k <= 5,
3              Print[k];
4              k++]
5          1
6          2
7          3
8          4
9          5
10 In[42]:= ?k
11          Global`k
12          k=6
```

**Conditional execution:** The conditional statement `If[condition, do-when-true, do-when-false]` follows the same logic as in every other programming language,

```
1 In[43]:= If[5! > 100,
2          Print["larger"],
3          Print["smaller or equal"]]
4          larger
```

Notice that the `If` statement has a return value, similar to the “?” statement of C and Java:

```
1 In[44]:= a = If[5! > 100, 1, -1]
2 Out[44]= 1
```

Apart from *true* and *false*, Mathematica statements can have a third state: *unknown*. For example, the comparison `x==0` evaluates to neither true nor false if `x` is not defined. The fourth slot in the `If` statement covers this case:

```
1 In[45]:= x == 0
2 Out[45]= x == 0
3 In[46]:= If[x == 0, "zero", "nonzero", "unknown"]
4 Out[46]= "unknown"
```

**Modularity:** code can use local variables within a *module*:

```

1 In[47]:= Module[{i},
2           i = 1;
3           While[i > 1/192, i = i/2];
4           i]
5 Out[47]= 1/256

```

After the execution of this code, the variable `i` is still undefined in the global context.

## 1.5.2 Exercises

**Q1.12** Write a program that sums all integers from 123 to 9968. Use only local variables.

**Q1.13** Write a program that sums consecutive integers, starting from 123, until the sum is larger than 10 000. Return the largest integer in this sum. Use only local variables.

## 1.5.3 Functional Programming

<https://reference.wolfram.com/language/guide/FunctionalProgramming.html>.

Functional programming is a very powerful programming technique that can give large speedups in computation because it can often be parallelized over many computers or CPUs. In our context, we often use lists (vectors or matrices, see Sect. 1.10) and want to apply functions to each one of their elements.

The most common functional programming constructs are

**Anonymous functions**<sup>1</sup>: you can quickly define a function with parameters `#1`, `#2`, `#3`, etc., terminated with the `&` symbol: (the symbol `#` is an abbreviation for `#1`)

```

1 In[48]:= f = #^2 &;
2 In[49]:= f[7]
3 Out[49]= 49
4 In[50]:= g = #1-#2 &;
5 In[51]:= g[88, 9]
6 Out[51]= 79

```

Functions and anonymous functions, for example `#^2&`, are first-class objects<sup>2</sup> just like numbers, matrices, etc. You can assign them to variables, as in `In[48]` and `In[50]` above; you can also use them directly as arguments to other functions, as for example in `In[55]` below; or you can use them as return values of other functions, as in `In[478]`.

<sup>1</sup>See [https://en.wikipedia.org/wiki/Anonymous\\_functions](https://en.wikipedia.org/wiki/Anonymous_functions).

<sup>2</sup>See [https://en.wikipedia.org/wiki/First-class\\_citizen](https://en.wikipedia.org/wiki/First-class_citizen).

The symbol `##` stands for the sequence of all parameters of a function:

```
1 In[52]:= f = {1,2,3,##,4,5,6} &;
2 In[53]:= f[7,a,c]
3 Out[53]= {1,2,3,7,a,c,4,5,6}
```

The symbol `#0` stands for the function itself. This is useful for defining recursive anonymous functions (see item 7 of Sect. 1.9).

**Map** `/@`: apply a function to each element of a list.

```
1 In[54]:= a = {1, 2, 3, 4, 5, 6, 7, 8};
2 In[55]:= Map[#^2 &, a]
3 Out[55]= {1, 4, 9, 16, 25, 36, 49, 64}
4 In[56]:= #^2 & /@ a
5 Out[56]= {1, 4, 9, 16, 25, 36, 49, 64}
```

Notice how we have used the anonymous function `#^2&` here without ever giving it a name.

**Apply** `@@`: apply a function to an entire list and generate a single result. For example, applying **Plus** to a list will calculate the sum of the list elements; applying **Times** will calculate their product. This operation is also known as *reduce*.<sup>3</sup>

```
1 In[57]:= a = {1, 2, 3, 4, 5, 6, 7, 8};
2 In[58]:= Apply[Plus, a]
3 Out[58]= 36
4 In[59]:= Plus @@ a
5 Out[59]= 36
6 In[60]:= Apply[Times, a]
7 Out[60]= 40320
8 In[61]:= Times @@ a
9 Out[61]= 40320
```

## 1.5.4 Exercises

**Q1.14** Write an anonymous function with three arguments that returns the product of these arguments.

**Q1.15** Given a list

```
1 In[62]:= a = {0.1, 0.9, 2.25, -1.9};
```

calculate  $x \mapsto \sin(x^2)$  for each element of `a` using the **Map** operation.

**Q1.16** Calculate the sum of all the results of **Q1.15**.

---

<sup>3</sup>See <https://en.wikipedia.org/wiki/MapReduce>.

## 1.6 Function Definitions

<https://reference.wolfram.com/language/tutorial/DefiningFunctions.html>.

Functions are assignments (see Sect. 1.2) with parameters. As for parameter-free assignments, we distinguish between *immediate* and *delayed* function definitions.

### 1.6.1 Immediate Function Definitions

We start with *immediate* definitions: a function  $f(x) = \sin(x)/x$  is defined with

```
1 In[63]:= f[x_] = Sin[x]/x;
```

Notice the underscore `_` symbol after the variable name `x`: this underscore indicates a *pattern* (denoted by `_`) named `x`, not the symbol `x` itself. Whenever this function `f` is called with any parameter value, this parameter value is inserted wherever `x` appears on the right-hand side, as is expected for a function definition. You can find out how `f` is defined with the `?` operator:

```
1 In[64]:= ?f
2          Global`f
3          f[x_] = Sin[x]/x
```

and you can ask for a function evaluation with

```
1 In[65]:= f[0.3]
2 Out[65]= 0.985067
3 In[66]:= f[0]
4          Power::Infinite expression 1/0 encountered.
5          Infinity::Indeterminate expression 0 ComplexInfinity encountered.
6 Out[66]= Indeterminate
```

Apparently the function cannot be evaluated for  $x = 0$ . We can fix this by defining a special function value:

```
1 In[67]:= f[0] = 1;
```

Notice that there is no underscore on the left-hand side, so there is no pattern definition. The full definition of `f` is now

```
1 In[68]:= ?f
2          Global`f
3          f[0] = 1
4          f[x_] = Sin[x]/x
```

If the function `f` is called, then these definitions are checked in order of appearance in this list. For example, if we ask for `f[0]`, then the first entry matches and the value 1 is returned. If we ask for `f[0.3]`, then the first entry does not match (since 0 and 0.3 are not strictly equal), but the second entry matches since anything can be plugged into the pattern named `x`. The result is  $\sin(0.3)/0.3 = 0.985067$ , which is what we expected.

## 1.6.2 Delayed Function Definitions

Just like with delayed assignments (Sect. 1.2.1), we can define delayed function calls. For comparison, we define the two functions

```
1 In[69]:= g1[x_] = x + RandomReal[]
2 Out[69]= 0.949868 + x
3 In[70]:= g2[x_] := x + RandomReal[]
```

Check their effective definitions with `?g1` and `?g2`, and notice that the definition of `g1` was executed immediately when you pressed shift-enter and its result assigned to the function `g1` (with a specific value for the random number, as printed out), whereas the definition of `g2` was left unevaluated and is executed each time anew when you use the function `g2`:

```
1 In[71]:= {g1[2], g2[2]}
2 Out[71]= {2.94987, 2.33811}
3 In[72]:= {g1[2], g2[2]}
4 Out[72]= {2.94987, 2.96273}
5 In[73]:= {g1[2], g2[2]}
6 Out[73]= {2.94987, 2.18215}
```

## 1.6.3 Memoization: Functions that Remember Their Results

<https://reference.wolfram.com/language/tutorial/FunctionsThatRememberValuesTheyHaveFound.html>.

When we define a function that takes a long time to evaluate, we may wish to store its output values such that if the function is called with identical parameter values again, then we do not need to re-evaluate the function but can simply remember the already calculated result.<sup>4</sup> We can make use of the interplay between patterns and values, and between immediate and delayed assignments, to construct such a function that remembers its values from previous function calls.

See if you can understand the following definition.

```
1 In[74]:= F[x_] := F[x] = x^7
```

If you ask for `?F` then you will simply see this definition. Now call

```
1 In[75]:= F[2]
2 Out[75]= 128
```

---

<sup>4</sup>This is technically called *memoization*: <https://en.wikipedia.org/wiki/Memoization>. A similar functionality can be achieved with Mathematica's `Once` operator, which allows fine-grained control over the storage location, conditions, and duration of the persistent result.

and ask for `?F` again. You see that the specific immediate definition of `F[2] = 128` was added to the list of definitions, with the evaluated result 128 (which may have taken a long time to calculate in a more complicated function). The next time you call `F[2]`, the specific definition of `F[2]` will be found earlier in the definitions list than the general definition `F[x_]` and therefore the precomputed value of `F[2]` will be returned.

When you re-define the function `F` after making modifications to it, you must clear the associated remembered values in order for them to be re-computed at the next occasion. It is a good practice to prefix every definition of a memoizing function with a `Clear` command:

```
1 In[76]:= Clear[F];
2 In[77]:= F[x_] := F[x] = x^9
```

For function evaluations that take even longer, we may wish to save the accumulated results to a file in order to read them back at a later time. For the above example, we save all definitions associated with the symbol `F` to the file `Fdef.mx` with

```
1 In[78]:= SetDirectory[NotebookDirectory[]];
2 In[79]:= DumpSave["Fdef.mx", F];
```

The next time we wish to continue the calculation, we define the function `F` and load all of its already known values with

```
1 In[80]:= Clear[F];
2 In[81]:= F[x_] := F[x] = x^9
3 In[82]:= SetDirectory[NotebookDirectory[]];
4 In[83]:= Get["Fdef.mx"];
```

### 1.6.4 Functions with Conditions on Their Arguments

<https://reference.wolfram.com/language/guide/Patterns.html>.

The Wolfram language contains a powerful pattern language that we can use to define functions that only accept certain arguments. For function definitions we will use three main types of patterns:

**Anything-goes:** A function defined as

```
1 In[84]:= f[x_] := x^2
```

can be called with any sort of arguments, since the pattern `x_` can match *anything*:

```
1 In[85]:= f[4]
2 Out[85]= 16
3 In[86]:= f[2.3-0.1I]
4 Out[86]= 5.28-0.46I
5 In[87]:= f[{1,2,3,4}]
6 Out[87]= {1,4,9,16}
7 In[88]:= f[y^2]
8 Out[88]= y^4
```



**Type-restricted:** A pattern like `x_Integer` will only match arguments of integer type. If the function is called with a non-matching argument, then the function is not executed:

```

1 In[89]:= g[x_Integer] := x-3
2 In[90]:= g[x_Rational] := x
3 In[91]:= g[x_Real] := x+3
4 In[92]:= g[x_Complex] := 0
5 In[93]:= g[7]
6 Out[93]= 4
7 In[94]:= g[7.1]
8 Out[94]= 10.1
9 In[95]:= g[2/3]
10 Out[95]= 2/3
11 In[96]:= g[2+3I]
12 Out[96]= 0
13 In[97]:= g[x]
14 Out[97]= g[x]
```

**Conditional:** Complicated conditions can be specified with the `/;` operator:

```

1 In[98]:= h[x_;/x<=3] := x^2
2 In[99]:= h[x_;/x>3] := x-11
3 In[100]:= h[2]
4 Out[100]= 4
5 In[101]:= h[5]
6 Out[101]= -6
```

Conditions involving a single function call returning a Boolean value, for example `x_;/PrimeQ[x]`, can be abbreviated with `x_?PrimeQ`. Other useful “question” functions are `IntegerQ`, `NumericQ`, `EvenQ`, `OddQ`, etc. See <https://reference.wolfram.com/language/tutorial/PuttingConstraintsOnPatterns.html> for more information.

## 1.6.5 Functions with Optional Arguments

<https://reference.wolfram.com/language/tutorial/OptionalAndDefaultArguments.html>.

Function arguments can be optional, indicated with the `:` symbol. For each optional argument, a default value must be defined that is used whenever the function is called without the argument specified. The optional arguments must be the last ones in the arguments list. There can be arbitrarily many optional arguments.

As an example, the function

```

1 In[102]:= f[a_, b_:5] = {a,b}
```

uses the default value  $b = 5$  whenever it is called with only one argument:

```

1 In[103]:= f[7]
2 Out[103]= {7,5}
```

When called with two arguments, the second argument overrides the default value for  $b$ :

```
1 In[104]:=f[7,2]
2 Out[104]={7,2}
```

## 1.7 Rules and Replacements

<https://reference.wolfram.com/language/tutorial/ApplyingTransformationRules.html>.

We will often use replacement rules in the calculations of this course. A replacement rule is an instruction  $x \rightarrow y$  that replaces any occurrence of the symbol (or pattern)  $x$  with the symbol  $y$ . We apply such a rule with the `/.` or `ReplaceAll` operator:

```
1 In[105]:=a + 2 /. a -> 7
2 Out[105]=9
3 In[106]:=ReplaceAll[a + 2, a -> 7]
4 Out[106]=9
5 In[107]:=c - d /. {c -> 2, d -> 8}
6 Out[107]=-6
7 In[108]:=ReplaceAll[c - d, {c -> 2, d -> 8}]
8 Out[108]=-6
```

Rules can contain patterns, in the same way as we use them for defining the parameters of functions (Sect. 1.6):

```
1 In[109]:=a + b /. x_ -> x^2
2 Out[109]=(a + b)^2
```

Notice that here the pattern `x_` matched the entire expression `a + b`, not the subexpressions `a` and `b`. To be more specific and do the replacement only at level 1 of this expression, we can write

```
1 In[110]:=Replace[a + b, x_ -> x^2, {1}]
2 Out[110]=a^2 + b^2
```

Doing the replacement at level 0 gives again

```
1 In[111]:=Replace[a + b, x_ -> x^2, {0}]
2 Out[111]=(a + b)^2
```

At other instances, restricted patterns can be used to achieve a desired result:

```
1 In[112]:=a + 2 /. x_Integer -> x^2
2 Out[112]=4 + a
```

Many Wolfram language functions return their results as replacement rules. For example, the result of solving an equation is a list of rules:

```

1 In[113]:=s = Solve[x^2 - 4 == 0, x]
2 Out[113]={{x -> -2}, {x -> 2}}
```

We can make use of these solutions with the replacement operator `/.`, for example to check the solutions:

```

1 In[114]:=x^2 - 4 /. s
2 Out[114]={0, 0}
```

### 1.7.1 Immediate and Delayed Rules

Just as for assignments (Sect. 1.2.1) and functions (Sect. 1.6), rules can be immediate or delayed. In an immediate rule of the form `x -> y`, the value of `y` is calculated once upon defining the rule. In a delayed rule of the form `x :=> y`, the value of `y` is re-calculated every time the rule is applied. This can be important when the rule is supposed to perform an action. Here is an example: we replace `c` by `f` with

```

1 In[115]:={a, b, c, d, c, a, c, b} /. c -> f
2 Out[115]={a, b, f, d, f, a, f, b}
```

We do the same while counting the number of replacements with

```

1 In[116]:=i = 0;
2 In[117]:={a, b, c, d, c, a, c, b} /. c :=> (i++; Echo[i, "replacement "]; f)
3      >> replacement 1
4      >> replacement 2
5      >> replacement 3
6 Out[117]={a, b, f, d, f, a, f, b}
7 In[118]:=i
8 Out[118]=3
```

In this case, the delayed rule `c :=> (i++; Echo[i, "replacement "]; f)` is a list of commands enclosed in parentheses `()` and separated by semicolons. The first command increments the replacement counter `i`, the second prints a running commentary (see Sect. 1.8), and the third gives the result of the replacement. The result of such a list of commands is always the last expression, in this case `f`.

### 1.7.2 Repeated Rule Replacement

The `/.` operator uses the given list of replacement rules only once:

```

1 In[119]:=a /. {a -> b, b -> c}
2 Out[119]=b
```

The `//.` operator, on the other hand, uses the replacement rules repeatedly until the result no longer changes (in this case, after two applications):

```
1 In[120]:=a //. {a -> b, b -> c}
2 Out[120]=c
```

## 1.8 Debugging and Finding Out How Mathematica Expressions are Evaluated

<https://reference.wolfram.com/language/guide/TuningAndDebugging.html>.  
<https://www.wolfram.com/language/elementary-introduction/2nd-ed/47-debugging-your-code.html>.

The precise way Mathematica evaluates an expression depends on many details and can become very complicated.<sup>5</sup> For finding out more about particular cases, especially when they aren't evaluated in the way that you were expecting, the `Trace` command may be useful. This command gives a list of all intermediate results, which helps in understanding the way that Mathematica arrives at its output:

```
1 In[121]:=Trace[x - 3x + 1]
2 Out[121]={{-(3x), -3x, -3x}, x-3x+1, 1-3x+x, 1-2x}
3 In[122]:=x = 5;
4 In[123]:=Trace[x - 3x + 1]
5 Out[123]={{x, 5}, {{x, 5}, 3x5, 15}, -15, -15}, 5-15+1, -9}
```

A more verbose trace is achieved with `TracePrint`:

```
1 In[124]:=TracePrint[y - 3y + 1]
2      y-3 y+1
3      Plus
4      y
5      -(3 y)
6      Times
7      -1
8      3 y
9      Times
10     3
11     y
12     -3 y
13     -3 y
14     Times
15     -3
16     y
17     1
18     y-3 y+1
19     1-3 y+y
20     1-2 y
21     Plus
22     1
23     -2 y
24     Times
25     -2
26     y
27 Out[124]=1 - 2 y
```

<sup>5</sup>See <https://reference.wolfram.com/language/tutorial/EvaluationOfExpressionsOverview.html>.

It is very useful to print out intermediate results in a long calculation via the `Echo` command, particularly during code development. Calling `Echo[x, label]` prints `x` with the given label, and returns `x`; in this way, the `Echo` command can be simply added to a calculation without perturbing it:

```
1 In[125]:=Table[Echo[i!, "building table: "], {i, 3}]
2           >> building table: 1
3           >> building table: 2
4           >> building table: 6
5 Out[125]={1, 2, 6}
```

In order to run your code “cleanly” after debugging it with `Echo`, you can either remove all instances of `Echo`, or you can re-define `Echo` to do nothing:

```
1 In[126]:=Unprotect[Echo]; Echo = #1 &;
```

Re-running the code of `In[125]` now gives just the result:

```
1 In[127]:=Table[Echo[i!, "building table: "], {i, 3}]
2 Out[127]={1, 2, 6}
```

Finally, it can be very insightful to study the “full form” of expressions, especially when it does not match a pattern that you were expecting to match. For example, the internal full form of ratios depends strongly on the type of numerator or denominator:

```
1 In[128]:=FullForm[a/b]
2 Out[128]=Times[a, Power[b, -1]]
3 In[129]:=FullForm[1/2]
4 Out[129]=Rational[1, 2]
5 In[130]:=FullForm[a/2]
6 Out[130]=Times[Rational[1, 2], a]
7 In[131]:=FullForm[1/b]
8 Out[131]=Power[b, -1]
```

## 1.8.1 Exercises

**Q1.17** Why do we need the `Unprotect` command in `In[126]`?

**Q1.18** To replace a ratio  $a/b$  by the function `ratio[a,b]`, we could enter

```
1 In[132]:=a/b /. {x_/y_ -> ratio[x,y]}
2 Out[132]=ratio[a,b]
```

Why does this not work to replace the ratio  $2/3$  by the function `ratio[2,3]`?

```
1 In[133]:=2/3 /. {x_/y_ -> ratio[x,y]}
2 Out[133]=2/3
```

## 1.9 Many Ways to Define the Factorial Function

### [Supplementary Material 1]

The following list of definitions of the factorial function is based on the Wolfram demo <https://www.wolfram.com/training/videos/EDU002/>. Try to understand as many of these definitions as possible. What this means in practice is that for most problems you can pick the programming paradigm that suits your way of thinking best, instead of being forced into one way or another. The different paradigms have different advantages and disadvantages, which may become clearer to you as you become more familiar with them.

You must call `Clear[f]` between different definitions!

1. Define the function `f` to be an alias of the built-in function `Factorial`: calling `f[5]` is now strictly the same thing as calling `Factorial[5]`, which in turn is the same thing as calling `5!`.

```
1 In[134]:=f = Factorial;
```

2. A call to `f` is forwarded to the function “!”: calling `f[5]` triggers the evaluation of `5!`.

```
1 In[135]:=f[n_] := n!
```

3. Use the mathematical definition  $n! = \Gamma(n + 1)$ :

```
1 In[136]:=f[n_] := Gamma[n+1]
```

4. Use the mathematical definition  $n! = \prod_{i=1}^n i$ :

```
1 In[137]:=f[n_] := Product[i, {i,n}]
```

5. Rule-based recursion, using the Wolfram language’s built-in pattern-matching capabilities: calling `f[5]` leads to a call of `f[4]`, which leads to a call of `f[3]`, and so on until `f[1]` immediately returns the result 1, after which the program unrolls the recursion stack and does the necessary multiplications:

```
1 In[138]:=f[1] = 1;
2 In[139]:=f[n_] := n*f[n-1]
```

6. The same recursion but without rules (no pattern-matching):

```
1 In[140]:=f[n_] := If[n == 1, 1, n*f[n-1]]
```

7. Define the same recursion through functional programming: `f` is a function whose name is `#0` and whose first (and only) argument is `#1`. The end of the function definition is marked with `&`.

```
1 In[141]:=f = If[#1 == 1, 1, #1*#0[#1-1]]&;
```

8. procedural programming with a `Do` loop:

```

1 In[142]:=f[n_] := Module[{t = 1},
2           Do[t = t*i, {i, n}];
3           t]

```

9. procedural programming with a **For** loop: this is how you would compute factorials in procedural programming languages like C. It is a very precise step-by-step prescription of how exactly the computer is supposed to do the calculation.

```

1 In[143]:=f[n_] := Module[{t = 1, i},
2           For[i = 1, i <= n, i++,
3               t *= i];
4           t]

```

10. Make a list of the numbers  $1 \dots n$  (with **Range[n]**) and then multiply them together at once, by applying the function **Times** to this list. This is the most elegant way of multiplying all these numbers together, because both the generation of the list of integers and their multiplication are done with internally optimized methods. The programmer merely specifies *what* he would like the computer to do, and not *how* it is to be done.

```

1 In[144]:=f[n_] := Times @@ Range[n]

```

11. Make a list of the numbers  $1 \dots n$  and then multiply them together one after the other.

```

1 In[145]:=f[n_] := Fold[Times, 1, Range[n]]

```

12. Functional programming: make a list of functions  $\{t \mapsto t, t \mapsto 2t, t \mapsto 3t, \dots, t \mapsto nt\}$ , and then, starting with the number 1, apply each of these functions once.

```

1 In[146]:=f[n_] := Fold[#2[#1]&, 1, Array[Function[t, #1*t]&, n]]

```

13. Construct a list whose length we know to be  $n!$ :

```

1 In[147]:=f[n_] := Length[Permutations[Range[n]]]

```

14. Use repeated pattern-based replacement (**//.**, see Sect. 1.7.2) to find the factorial: start with the object  $\{1, n\}$  and apply the given rule until the result no longer changes because the pattern no longer matches.

```

1 In[148]:=f[n_] := First[{1,n} //. {a_,b_;/b>0} :> {b*a,b-1}]

```

15. Build a string whose length is  $n!$ :

```

1 In[149]:=f[n_] := StringLength[Fold[StringJoin[Table[#1, {#2}]]&, "A", Range[n]]]

```

16. Starting from the number  $n$ , repeatedly replace each number  $m$  by a list containing  $m$  times the number  $m - 1$ . At the end, we have a list of lists of ... of lists

that overall contains  $n!$  times the number 1. Flatten it out and count the number of elements.

```
1 In[150]:=f[n_] := Length[Flatten[n /. m_ /; m > 1 :> Table[m - 1, {m}]]]
```

17. Analytically calculate  $\frac{d^n(x^n)}{dx^n}$ , the  $n$ th derivative of  $x^n$ :

```
1 In[151]:=f[n_] := D[x^n, {x, n}]
```

### 1.9.1 Exercises

**Q1.19** In which ones of the definitions of Sect. 1.9 can you replace a delayed assignment (`:=`) with an immediate assignment (`=`) or vice-versa? What changes if you do this replacement? (see Sect. 1.2.1).

**Q1.20** In which ones of the definitions of Sect. 1.9 can you replace a delayed rule (`:->`) with an immediate rule (`->`) or vice-versa? What changes if you do this replacement? (see Sect. 1.7.1).

**Q1.21** Can you use the trick of Sect. 1.6.3 for any of the definitions of Sect. 1.9?

**Q1.22** Write two very different programs that calculate the first hundred Fibonacci numbers  $\{1, 1, 2, 3, 5, 8, \dots\}$ , where each number is the sum of the two preceding ones.

## 1.10 Vectors, Matrices, Tensors

In this book we will use vectors and matrices to represent quantum states and operators, respectively.

### 1.10.1 Vectors

<https://reference.wolfram.com/language/tutorial/VectorOperations.html>.

In the Wolfram language, vectors are represented as lists of objects, for example lists of real or complex numbers:

```
1 In[152]:=v = {1,2,3,2,1,7+I};
2 In[153]:=Length[v]
3 Out[153]=6
```

You can access any element by its index, using double brackets, with the first element having index 1 (as in Fortran or Matlab), *not* 0 (as in C, Java, or Python):



```

1 In[154]:=v[[4]]
2 Out[154]=2

```

Negative indices count from the end of the list:

```

1 In[155]:=v[[-1]]
2 Out[155]=7+I

```

Lists can contain arbitrary elements (for example strings, graphics, expressions, lists, functions, etc.).

If two vectors  $\vec{a}$  and  $\vec{b}$  of equal length are defined, then their scalar product  $\vec{a}^* \cdot \vec{b}$  is calculated with

```

1 In[156]:=a = {0.1, 0.2, 0.3 + 2I};
2 In[157]:=b = {-0.27I, 0, 2};
3 In[158]:=Conjugate[a].b
4 Out[158]=0.6 - 4.027I

```

Vectors of equal length can be element-wise added, subtracted, multiplied etc. with the usual operators:

```

1 In[159]:=a + b
2 Out[159]={0.1 - 0.27I, 0.2, 2.3 + 2.I}
3 In[160]:=2 a
4 Out[160]={0.2, 0.4, 0.6 + 4.I}

```

## 1.10.2 Matrices

<https://reference.wolfram.com/language/tutorial/BasicMatrixOperations.html>.

Matrices are lists of lists, where each sublist describes a row of the matrix:

```

1 In[161]:=M = {{3,2,7},{1,1,2},{0,-1,5},{2,2,1}};
2 In[162]:=Dimensions[M]
3 Out[162]={4, 3}

```

In this example, **M** is a  $4 \times 3$  matrix. Pretty-printing a matrix is done with the `MatrixForm` wrapper,

```

1 In[163]:=MatrixForm[M]

```

Accessing matrix elements is analogous to accessing vector elements:

```

1 In[164]:=M[[1,3]]
2 Out[164]=7
3 In[165]:=M[[2]]
4 Out[165]={1, 1, 2}

```

Matrices can be transposed with `Transpose[M]`.

Matrix–vector and matrix–matrix multiplications are done with the `.` operator:

```
1 In[166]:=M.a
2 Out[166]={2.8 + 14.I, 0.9 + 4.I, 1.3 + 10.I, 0.9 + 2.I}
```

### 1.10.3 Sparse Vectors and Matrices

<https://reference.wolfram.com/language/guide/SparseArrays.html>.

Large matrices can take up enormous amounts of computer memory. In practical situations we are often dealing with matrices that are “sparse”, meaning that most of their entries are zero. A much more efficient way of storing them is therefore as a list of only their nonzero elements, using the `SparseArray` function.

A given vector or matrix is converted to sparse representation with

```
1 In[167]:=M = {{0,3,0,0,0,0,0,0,0,0},
2             {0,0,0,-1,0,0,0,0,0,0},
3             {0,0,0,0,0,0,0,0,0,0}};
4 In[168]:=Ms = SparseArray[M]
5 Out[168]=SparseArray[<2>, {3, 10}]
```

where the output shows that `Ms` is a  $3 \times 10$  sparse matrix with 2 non-zero entries. We could have entered this matrix more easily by giving the list of non-zero entries,

```
1 In[169]:=Ms = SparseArray[{{1, 2} -> 3, {2, 4} -> -1, {3, 10}}];
```

which we can find out from

```
1 In[170]:=ArrayRules[Ms]
2 Out[170]={{1, 2} -> 3, {2, 4} -> -1, {_, _} -> 0}
```

which includes a specification of the default pattern `{_, _}`. This sparse array is converted back into a normal array with

```
1 In[171]:=Normal[Ms]
2 Out[171]={{0,3,0,0,0,0,0,0,0,0},
3           {0,0,0,-1,0,0,0,0,0,0},
4           {0,0,0,0,0,0,0,0,0,0}}
```

Sparse arrays and vectors can be used just like full arrays and vectors (they are internally converted automatically whenever necessary). But for some linear algebra operations they can be much more efficient. A matrix multiplication of two sparse matrices, for example, scales only with the number of non-zero elements of the matrices, not with their size.

## 1.10.4 Matrix Diagonalization

“Solving” the time-independent Schrödinger equation, as we will be doing in Sect. 2.2, involves calculating the eigenvalues and eigenvectors of Hermitian<sup>6</sup> matrices.

In what follows it is assumed that we have defined  $H$  as a Hermitian matrix. As an example we will use

```
1 In[172]:=H = {{0, 0.3, I, 0},
2             {0.3, 1, 0, 0},
3             {-I, 0, 1, -0.2},
4             {0, 0, -0.2, 3}};
```

### Eigenvalues

The eigenvalues of a matrix  $H$  are computed with

```
1 In[173]:=Eigenvalues[H]
2 Out[173]={3.0237, 1.63842, 0.998322, -0.660442}
```

Notice that these eigenvalues (energy values) are not necessarily sorted, even though in this example they appear in descending order. For a sorted list we use

```
1 In[174]:=Sort[Eigenvalues[H]]
2 Out[174]={-0.660442, 0.998322, 1.63842, 3.0237}
```

For very large matrices  $H$ , and in particular for sparse matrices (see Sect. 1.10.3), it is computationally inefficient to calculate all eigenvalues. Further, we are often only interested in the lowest-energy eigenvalues and eigenvectors. There are very efficient algorithms for calculating extremal eigenvalues,<sup>7</sup> which can be used by specifying options to the `Eigenvalues` function: if we only need the largest two eigenvalue, for example, we call

```
1 In[175]:=Eigenvalues[H, 2, Method -> {"Arnoldi",
2             "Criteria" -> "RealPart",
3             MaxIterations -> 10^6}]
4 Out[175]={3.0237, 1.63842}
```

There is no direct way to calculate the *smallest* eigenvalues; but since the smallest eigenvalues of  $H$  are the largest eigenvalues of  $-H$  we can use

```
1 In[176]:=-Eigenvalues[-H, 2, Method -> {"Arnoldi",
2             "Criteria" -> "RealPart",
3             MaxIterations -> 10^6}]
4 Out[176]={0.998322, -0.660442}
```

### Eigenvectors

The eigenvectors of a matrix  $H$  are computed with

<sup>6</sup>A complex matrix  $H$  is *Hermitian* if  $H = H^\dagger$ . See [https://en.wikipedia.org/wiki/Hermitian\\_matrix](https://en.wikipedia.org/wiki/Hermitian_matrix).

<sup>7</sup>Arnoldi–Lanczos algorithm: [https://en.wikipedia.org/wiki/Lanczos\\_algorithm](https://en.wikipedia.org/wiki/Lanczos_algorithm).

```

1 In[177]:= Eigenvectors[H]
2 Out[177]= {{0.-0.0394613I, 0.-0.00584989I, -0.117564, 0.992264},
3            {0.+0.533642I, 0.+0.250762I, 0.799103, 0.117379},
4            {0.-0.0053472I, 0.+0.955923I, -0.292115, -0.029187},
5            {0.-0.844772I, 0.+0.152629I, 0.512134, 0.0279821}}

```

In this case of a  $4 \times 4$  matrix, this generates a list of four ortho-normal 4-vectors.

Usually we are interested in calculating the eigenvalues and eigenvectors at the same time:

```

1 In[178]:= Eigensystem[H]
2 Out[178]= {{3.0237, 1.63842, 0.998322, -0.660442},
3            {{0.-0.0394613I, 0.-0.00584989I, -0.117564, 0.992264},
4             {0.+0.533642I, 0.+0.250762I, 0.799103, 0.117379},
5             {0.-0.0053472I, 0.+0.955923I, -0.292115, -0.029187},
6             {0.-0.844772I, 0.+0.152629I, 0.512134, 0.0279821}}}

```

which generates a list containing the eigenvalues and the eigenvectors. The ordering of the elements in the eigenvalues list corresponds to the ordering in the eigenvectors list; but the sorting order is generally undefined. To generate a list of (eigenvalue, eigenvector) pairs in ascending order of eigenvalues, we calculate

```

1 In[179]:= Sort[Transpose[Eigensystem[H]]]
2 Out[179]= {{-0.660442, {0.-0.844772I, 0.+0.152629I, 0.512134, 0.0279821}},
3            {0.998322, {0.-0.0053472I, 0.+0.955923I, -0.292115, -0.029187}},
4            {1.63842, {0.+0.533642I, 0.+0.250762I, 0.799103, 0.117379}},
5            {3.0237, {0.-0.0394613I, 0.-0.00584989I, -0.117564, 0.992264}}}

```

To generate a sorted list of eigenvalues `eval` and a corresponding list of eigenvectors `evvec` we calculate

```

1 In[180]:= {eval, evvec} = Transpose[Sort[Transpose[Eigensystem[H]]]];
2 In[181]:= eval
3 Out[181]= {-0.660442, 0.998322, 1.63842, 3.0237}
4 In[182]:= evvec
5 Out[182]= {{0.-0.844772I, 0.+0.152629I, 0.512134, 0.0279821},
6            {0.-0.0053472I, 0.+0.955923I, -0.292115, -0.029187},
7            {0.+0.533642I, 0.+0.250762I, 0.799103, 0.117379},
8            {0.-0.0394613I, 0.-0.00584989I, -0.117564, 0.992264}}

```

The trick with calculating only the lowest-energy eigenvalues can be applied to eigenvalue calculations as well, since the eigenvectors of  $-H$  and  $H$  are the same:

```

1 In[183]:= {eval, evvec} = Transpose[Sort[Transpose[-Eigensystem[-H, 2,
2             Method -> {"Arnoldi", "Criteria" -> "RealPart", MaxIterations -> 10^6}]]]];
3 In[184]:= eval
4 Out[184]= {-0.660442, 0.998322}
5 In[185]:= evvec
6 Out[185]= {{-0.733656+0.418794I, 0.132553-0.0756656I,
7            -0.253889-0.444771I, -0.0138721-0.0243015 I},
8            {-0.000575666-0.00531612I, 0.102912+0.950367I,
9            -0.290417+0.0314484I, -0.0290174+0.00314221I}}

```

Notice that these eigenvectors are not the same as those calculated further above! This difference is due to arbitrary multiplications of the eigenvectors with phase factors  $e^{i\varphi}$ .

To check that the vectors in `evvec` are ortho-normalized, we calculate the matrix product

```
1 In[186]:=Conjugate[evect].Transpose[evect] //Chop //MatrixForm
```

and verify that the matrix of scalar products is indeed equal to the unit matrix.

To check that the vectors in `evect` are indeed eigenvectors of `H`, we calculate all matrix elements of `H` in this basis of eigenvectors:

```
1 In[187]:=Conjugate[evect].H.Transpose[evect] //Chop //MatrixForm
```

and verify that the result is a diagonal matrix whose diagonal elements are exactly the eigenvalues `eval`.

### 1.10.5 Tensor Operations

<https://reference.wolfram.com/language/guide/RearrangingAndRestructuringLists.html>.

We have seen above that in the Wolfram language, a vector is a list of numbers (Sect. 1.10.1) and a matrix is a list of lists of numbers (Sect. 1.10.2). Higher-rank tensors are correspondingly represented as lists of lists of ...of lists of numbers. In this section we describe general tools for working with tensors, which extend the methods used for vectors and matrices. See Sect. 2.4.3 for a concrete application of higher-rank tensors. We note that the sparse techniques of Sect. 1.10.3 naturally extend to higher-rank tensors.

As an example, we start by defining a list (i.e., a vector) containing 24 elements:

```
1 In[188]:=v = Range[24]
2 Out[188]= {1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20,21,22,23,24}
3 In[189]:=Dimensions[v]
4 Out[189]= {24}
```

We have chosen the elements in this vector to indicate their position in order to make the following transformations easier to understand.

#### Reshaping

We reshape the list `v` into a  $2 \times 3 \times 4$  tensor with

```
1 In[190]:=t = ArrayReshape[v, {2,3,4}]
2 Out[190]= {{{1,2,3,4},{5,6,7,8},{9,10,11,12}},
3           {{13,14,15,16},{17,18,19,20},{21,22,23,24}}}
4 In[191]:=Dimensions[t]
5 Out[191]= {2, 3, 4}
```

Notice that the order of the elements has not changed; but they are now arranged as a list of lists of lists of numbers. Alternatively, we could reshape `v` into a  $2 \times 2 \times 3 \times 2$  tensor with

```

1 In[192]:=u = ArrayReshape[v, {2,2,3,2}]
2 Out[192]= {{{{1,2},{3,4},{5,6}},{7,8},{9,10},{11,12}}},
3           {{{13,14},{15,16},{17,18}},{19,20},{21,22},{23,24}}}}
4 In[193]:=Dimensions[u]
5 Out[193]= {2, 2, 3, 2}

```

## Flattening

The reverse operation is called flattening:

```

1 In[194]:=Flatten[t] == Flatten[u] == v
2 Out[194]= True

```

Tensor flattening can be applied more specifically, without flattening the entire structure into a single list. As an example, in `u` we flatten indices 1&2 together and indices 3&4 together, to find a  $4 \times 6$  matrix that we could have calculated directly with `ArrayReshape[v, {4,6}]`:

```

1 In[195]:=Flatten[u, {{1,2}, {3,4}}]
2 Out[195]= {{1,2,3,4,5,6},{7,8,9,10,11,12},{13,14,15,16,17,18},{19,20,21,22,23,24}}
3 In[196]:= % == ArrayReshape[v, {4,6}]
4 Out[196]= True

```

We sometimes use the `ArrayFlatten` command, which is just a special case of `Flatten` with fixed arguments, flattening indices 1&3 together and indices 2&4 together:

```

1 In[197]:=ArrayFlatten[u] == Flatten[u, {{1,3}, {2,4}}]
2 Out[197]= True

```

## Transposing

A tensor transposition is a re-ordering of a tensor's indices. For example,

```

1 In[198]:=tt = Transpose[t, {2,3,1}]
2 Out[198]= {{{{1,5,9},{13,17,21}},{2,6,10},{14,18,22}},
3           {{{3,7,11},{15,19,23}},{4,8,12},{16,20,24}}}}
4 In[199]:=Dimensions[tt]
5 Out[199]= {4, 2, 3}

```

generates a  $4 \times 2 \times 3$ -tensor `tt`, where the first index of `t` is the second index of `tt`, the second index of `t` is the third index of `tt`, and the third index of `t` is the first index of `tt`; this order of index shuffling is given in the parameter list `{2,3,1}` meaning  $\{1\text{st}, 2\text{nd}, 3\text{rd}\} \mapsto \{2\text{nd}, 3\text{rd}, 1\text{st}\}$ . More explicitly,

```

1 In[200]:=Table[t[[i,j,k]] == tt[[k,i,j]], {i,2}, {j,3}, {k,4}]
2 Out[200]= {{{{True,True,True,True},{True,True,True,True}},
3           {{{True,True,True,True}},{True,True,True,True}},
4           {{{True,True,True,True},{True,True,True,True}}}}

```

## Contracting

As a generalization of a scalar product, indices of equal length of a tensor can be contracted. This is the operation of summing over an index that appears twice in

the list of indices. For example, contracting indices 2 and 5 of the rank-6 tensor  $X_{a,b,c,d,e,f}$  yields the rank-4 tensor with elements  $Y_{a,c,d,f} = \sum_i X_{a,i,c,d,i,f}$ .

For example, we can either contract indices 1&2 in `u`, or indices 1&4, or indices 2&4, since they are all of length 2:

```

1 In[201]:=TensorContract[u, {1, 2}]
2 Out[201]={{20, 22}, {24, 26}, {28, 30}}
3 In[202]:=TensorContract[u, {1, 4}]
4 Out[202]={{15, 19, 23}, {27, 31, 35}}
5 In[203]:=TensorContract[u, {2, 4}]
6 Out[203]={{9, 13, 17}, {33, 37, 41}}
```

### 1.10.6 Exercises

**Q1.23** Calculate the eigenvalues and eigenvectors of the Pauli matrices:

[https://en.wikipedia.org/wiki/Pauli\\_matrices](https://en.wikipedia.org/wiki/Pauli_matrices)

Are the eigenvectors ortho-normal? If not, find an ortho-normal set.

**Q1.24** After `In[203]`, try to contract indices 3&4 in the tensor `u`. What went wrong?

## 1.11 Complex Numbers

By default all variables in the Wolfram language are assumed to be complex numbers, unless otherwise specified. All mathematical functions can take complex numbers as their input, often by analytic continuation.<sup>8</sup>

The most commonly used functions on complex numbers are `Conjugate`, `Re`, `Im`, `Abs`, and `Arg`. When applied to numerical arguments they do what we expect:

```

1 In[204]:=Conjugate[2 + 3I]
2 Out[204]=2 - 3I
3 In[205]:=Im[0.7]
4 Out[205]=0
```

When applied to variable arguments, however, they fail and frustrate the inexperienced user:

```

1 In[206]:=Conjugate[x+I*y]
2 Out[206]=Conjugate[x] - I*Conjugate[y]
3 In[207]:=Im[a]
4 Out[207]=Im[a]
```

This behavior is due to Mathematica not knowing that `x`, `y`, and `a` in these examples are real-valued. There are several ways around this, all involving *assumptions*. The

<sup>8</sup>See [https://en.wikipedia.org/wiki/Analytic\\_continuation](https://en.wikipedia.org/wiki/Analytic_continuation).

first is to use the `ComplexExpand` function, which assumes that all variables are *real*:

```
1 In[208]:=Conjugate[x+I*y] //ComplexExpand
2 Out[208]=x - I*y
3 In[209]:=Im[a] //ComplexExpand
4 Out[209]=0
```

The second is to use explicit *local* assumptions, which may be more specific than assuming that all variables are real-valued:

```
1 In[210]:=Assuming[Element[x, Reals] && Element[y, Reals],
2 Conjugate[x + I*y] //FullSimplify]
3 Out[210]=x - I*y
4 In[211]:=Assuming[Element[a, Reals], Im[a] //FullSimplify]
5 Out[211]=0
```

The third is to use *global* assumptions (in general, global system variables start with the `$` sign):

```
1 In[212]:= $Assumptions = Element[x, Reals] && Element[y, Reals] && Element[a, Reals];
2 In[213]:=Conjugate[x+I*y] //FullSimplify
3 Out[213]=x - I*y
4 In[214]:=Im[a] //FullSimplify
5 Out[214]=0
```

## 1.12 Units

<https://reference.wolfram.com/language/tutorial/UnitsOverview.html>.

The Wolfram language is capable of dealing with units of measure, as required for physical calculations. For example, we can make the assignment

```
1 In[215]:= s = Quantity[3, "m"];
```

to specify that `s` should be three meters. A large number of units can be used, as well as physical constants:

```
1 In[216]:= kB = Quantity["BoltzmannConstant"];
```

will define the variable `kB` to be Boltzmann's constant. Take note that complicated or slightly unusual quantities are evaluated through the online service *Wolfram Alpha*<sup>®</sup>, which means that you need an internet connection in order to evaluate them. For this and other reasons, unit calculations are very slow and to be avoided whenever possible.

If you are unsure whether your expression has been interpreted correctly, the full internal form



```

1 In[217]:=FullForm[kB]
2 Out[217]=Quantity[1, "BoltzmannConstant"]

```

usually helps. Alternatively, converting to SI units can often clarify a definition:

```

1 In[218]:=UnitConvert[kB]
2 Out[218]=Quantity[1.38065*10^-23, "kg m^2/(s^2 K)"]

```

In principle, we can use this mechanism to do all the calculations in this book with units; however, for the sake of generality (as many other computer programs cannot deal with units) when we do numerical calculations, we will convert every quantity into dimensionless form in what follows.

In order to eliminate units from a calculation, we must determine a set of units in which to express the relevant quantities. This means that every physical quantity  $x$  is expressed as the product of a *unit* and a *dimensionless multiplier*. The actual calculations are performed only with the dimensionless multipliers. A smart choice of units can help in implementing a problem.

As an example we calculate the acceleration of an A380 airplane ( $m = 560$  t) due to its jet engines ( $F = 4 \times 311$  kN). The easiest way is to use the Wolfram language's built-in unit processing:

```

1 In[219]:=F = Quantity[4*311, "kN"];
2 In[220]:=m = Quantity[560, "t"];
3 In[221]:=a = UnitConvert[F/m, "m/s^2"] //N
4 Out[221]=2.22143 m/s^2

```

This method is, however, much slower than using purely numerical calculations, and furthermore cannot be generalized to matrix and vector algebra.

Now we do the same calculation with dimensionless multipliers only. For this, we first set up a consistent set of units, for example the SI units:

```

1 In[222]:=ForceUnit = Quantity["Newtons"];
2 In[223]:=MassUnit = Quantity["Kilograms"];
3 In[224]:=AccelerationUnit = UnitConvert[ForceUnit/MassUnit]
4 Out[224]=1 m/s^2

```

It is important that these units are consistent with each other, i.e., that the product of the mass and acceleration units gives the force unit. The calculation is now effected with a simple numerical division  $a = F/m$ :

```

1 In[225]:=F = Quantity[4*311, "kN"] / ForceUnit
2 Out[225]=1244000
3 In[226]:=m = Quantity[560, "t"] / MassUnit
4 Out[226]=560000
5 In[227]:=a = F/m //N
6 Out[227]=2.22143

```

This result of 2.221 43 acceleration units, meaning 2.221 43 m/s<sup>2</sup>, is the same as [Out\[221\]](#).

We can do this type of calculation in any consistent unit system: as a second example, we use the unit definitions

```

1 In[228]:=ForceUnit = Quantity["KiloNewtons"];
2 In[229]:=MassUnit = Quantity["AtomicMassUnit"];
3 In[230]:=AccelerationUnit = UnitConvert[ForceUnit/MassUnit]
4 Out[230]=6.022141*10^29 m/s^2

```

and calculate

```

1 In[231]:=F = Quantity[4*311, "kN"] / ForceUnit
2 Out[231]=1244
3 In[232]:=m = Quantity[560, "t"] / MassUnit
4 Out[232]=3.3723989*10^32
5 In[233]:=a = F/m //N
6 Out[233]=3.68877*10^-30


```

This result is again the same as [Out \[221\]](#), because  $3.688\,77 \times 10^{-30}$  acceleration units are  $3.688\,77 \times 10^{-30} \times 6.022\,141 \times 10^{29} \text{ m/s}^2$ .

It is not important which unit system we use. In practice, it is often convenient to use a system of units that yields dimensionless multipliers that are on the order of unity; but this is not a strict requirement.

## Chapter 2

# Quantum Mechanics: States and Operators



If you are like most students of quantum mechanics, then you have begun your quantum studies by hearing stories about experiments such as Young's double slit,<sup>1</sup> the Stern–Gerlach spin quantization,<sup>2</sup> and Heisenberg's uncertainty principle.<sup>3</sup> Many concepts and analogies are introduced to get an idea of what quantum mechanics is about and to begin to develop an intuition for it. Yet there is a large gap between this kind of qualitative understanding and being able to solve even the simplest quantum-

---

**Electronic supplementary material** The online version of this chapter ([https://doi.org/10.1007/978-981-13-7588-0\\_2](https://doi.org/10.1007/978-981-13-7588-0_2)) contains supplementary material, which is available to authorized users.

<sup>1</sup> See [https://en.wikipedia.org/wiki/Double-slit\\_experiment](https://en.wikipedia.org/wiki/Double-slit_experiment).

<sup>2</sup> See [https://en.wikipedia.org/wiki/Stern-Gerlach\\_experiment](https://en.wikipedia.org/wiki/Stern-Gerlach_experiment).

<sup>3</sup> See [https://en.wikipedia.org/wiki/Uncertainty\\_principle](https://en.wikipedia.org/wiki/Uncertainty_principle).

mechanical problems on a computer, essentially because a computer only works with numbers, not with stories, analogies, or visualizations.

The goal of this chapter is to connect the fundamental quantum-mechanical concepts to representations that a computer can understand. We develop the tools that will be used in the remaining chapters to express and solve interesting quantum-mechanical problems.

## 2.1 Basis Sets and Representations

Quantum-mechanical problems are usually specified in terms of operators and quantum states. The quantum states are elements of a Hilbert space; the operators act on such vectors. How can these objects be represented on a computer, which only understands numbers but not Hilbert spaces?

In order to find a computer-representable form of these abstract objects, we assume that we know an ortho-normal<sup>4</sup> basis  $\{|i\rangle\}_i$  of this Hilbert space, with scalar product  $\langle i|j\rangle = \delta_{ij}$ . In Sect. 2.4 we will talk about how to construct such bases. For now we make the assumption that this basis is complete, such that  $\sum_i |i\rangle\langle i| = \mathbb{1}$ . We will see in Sect. 2.1.1 how to deal with incomplete basis sets.

Given any operator  $\hat{A}$  acting on this Hilbert space, we use the completeness relation twice to find

$$\hat{A} = \mathbb{1} \cdot \hat{A} \cdot \mathbb{1} = \left[ \sum_i |i\rangle\langle i| \right] \cdot \hat{A} \cdot \left[ \sum_j |j\rangle\langle j| \right] = \sum_{ij} \underbrace{\langle i|\hat{A}|j\rangle}_{A_{ij}} |i\rangle\langle j|. \quad (2.1)$$

We define a numerical matrix  $A$  with elements  $A_{ij} = \langle i|\hat{A}|j\rangle \in \mathbb{C}$  to rewrite this as

$$\hat{A} = \sum_{ij} A_{ij} |i\rangle\langle j|. \quad (2.2)$$

The same can be done with a state vector  $|\psi\rangle$ : using the completeness relation,

$$|\psi\rangle = \mathbb{1} \cdot |\psi\rangle = \left[ \sum_i |i\rangle\langle i| \right] \cdot |\psi\rangle = \sum_i \underbrace{\langle i|\psi\rangle}_{\psi_i} |i\rangle, \quad (2.3)$$

and by defining a numerical vector  $\vec{\psi}$  with elements  $\psi_i = \langle i|\psi\rangle \in \mathbb{C}$  the state vector is

$$|\psi\rangle = \sum_i \psi_i |i\rangle. \quad (2.4)$$

---

<sup>4</sup>The following calculations can be extended to situations where the basis is not ortho-normal. For the scope of this book we are however not interested in this complication.

Both the matrix  $A$  and the vector  $\vec{\psi}$  are complex-valued objects which can be represented in any computer system. Equations (2.2) and (2.4) serve to convert between Hilbert-space representations and number-based (matrix/vector-based) representations. These equations are at the center of what it means to find a computer representation of a quantum-mechanical problem.

### 2.1.1 Incomplete Basis Sets

For infinite-dimensional Hilbert spaces we must usually content ourselves with finite basis sets that approximate the low-energy physics (or, more generally, the physically relevant dynamics) of the problem. In practice this means that an orthonormal basis set may not be complete,

$$\sum_i |i\rangle\langle i| = \hat{P}, \quad (2.5)$$

which is the projector onto that subspace of the full Hilbert space which the basis is capable of describing. We denote  $\hat{Q} = \mathbb{1} - \hat{P}$  as the complement of this projector:  $\hat{Q}$  is the projector onto the remainder of the Hilbert space that is left out of this truncated description. The equivalent of Eq. (2.1) is then

$$\begin{aligned} \hat{A} &= \mathbb{1} \cdot \hat{A} \cdot \mathbb{1} = (\hat{P} + \hat{Q}) \cdot \hat{A} \cdot (\hat{P} + \hat{Q}) = \hat{P} \cdot \hat{A} \cdot \hat{P} + \hat{P} \cdot \hat{A} \cdot \hat{Q} + \hat{Q} \cdot \hat{A} \cdot \hat{P} + \hat{Q} \cdot \hat{A} \cdot \hat{Q} \\ &= \underbrace{\sum_{ij} A_{ij} |i\rangle\langle j|}_{\text{within described subspace}} + \underbrace{\hat{P} \cdot \hat{A} \cdot \hat{Q} + \hat{Q} \cdot \hat{A} \cdot \hat{P}}_{\text{neglected coupling to (high-energy) part}} + \underbrace{\hat{Q} \cdot \hat{A} \cdot \hat{Q}}_{\text{neglected (high-energy) part}} \end{aligned} \quad (2.6)$$

In the same way, the equivalent of Eq. (2.3) is

$$|\psi\rangle = \mathbb{1} \cdot |\psi\rangle = (\hat{P} + \hat{Q}) \cdot |\psi\rangle = \underbrace{\sum_i \psi_i |i\rangle}_{\text{within described subspace}} + \underbrace{\hat{Q}|\psi\rangle}_{\text{neglected (high-energy) part}} \quad (2.7)$$

Since  $\hat{Q}$  is the projector onto the neglected subspace, the component  $\hat{Q}|\psi\rangle$  of Eq. (2.7) is the part of the quantum state  $|\psi\rangle$  that is left out of the description in the truncated basis. In specific situations we will need to make sure that all terms involving  $\hat{Q}$  in Eqs. (2.6) and (2.7) can be safely neglected. See Eq. (4.28) for a problematic example of an operator expressed in a truncated basis.

### Variational Ground-State Calculations

Calculating the ground state of a Hamiltonian in an incomplete basis set is a special case of the variational method.<sup>5</sup> As we will see for example in Sect. 4.1.7, the

<sup>5</sup>See [https://en.wikipedia.org/wiki/Variational\\_method\\_\(quantum\\_mechanics\)](https://en.wikipedia.org/wiki/Variational_method_(quantum_mechanics)).

variational ground-state energy is always larger than the true ground-state energy. When we add more basis functions, the numerically calculated ground-state energy decreases monotonically. At the same time, the overlap (scalar product) of the numerically calculated ground state with the true ground state monotonically increases to unity. These convergence properties often allow us to judge whether or not a chosen computational basis set is sufficiently complete.

### 2.1.2 Exercises

**Q2.1** We describe a spin-1/2 system in the basis  $\mathcal{B}$  containing the two states

$$\begin{aligned} |\uparrow_{\vartheta,\varphi}\rangle &= \cos\left(\frac{\vartheta}{2}\right) |\uparrow\rangle + e^{i\varphi} \sin\left(\frac{\vartheta}{2}\right) |\downarrow\rangle \\ |\downarrow_{\vartheta,\varphi}\rangle &= -e^{-i\varphi} \sin\left(\frac{\vartheta}{2}\right) |\uparrow\rangle + \cos\left(\frac{\vartheta}{2}\right) |\downarrow\rangle \end{aligned} \quad (2.8)$$

1. Show that the basis  $\mathcal{B} = \{|\uparrow_{\vartheta,\varphi}\rangle, |\downarrow_{\vartheta,\varphi}\rangle\}$  is orthonormal.
2. Show that the basis  $\mathcal{B}$  is complete:  $|\uparrow_{\vartheta,\varphi}\rangle\langle\uparrow_{\vartheta,\varphi}| + |\downarrow_{\vartheta,\varphi}\rangle\langle\downarrow_{\vartheta,\varphi}| = \mathbf{1}$ .
3. Express the states  $|\uparrow\rangle$  and  $|\downarrow\rangle$  as vectors in the basis  $\mathcal{B}$ .
4. Express the Pauli operators  $\hat{\sigma}_x$ ,  $\hat{\sigma}_y$ ,  $\hat{\sigma}_z$  as matrices in the basis  $\mathcal{B}$ .
5. Show that  $|\uparrow_{\vartheta,\varphi}\rangle$  and  $|\downarrow_{\vartheta,\varphi}\rangle$  are eigenvectors of  $\hat{\sigma}(\vartheta, \varphi) = \hat{\sigma}_x \sin(\vartheta) \cos(\varphi) + \hat{\sigma}_y \sin(\vartheta) \sin(\varphi) + \hat{\sigma}_z \cos(\vartheta)$ . What are the eigenvalues?

**Q2.2** The eigenstate basis for the description of the infinite square well of unit width is made up of the ortho-normalized functions

$$\langle x|n\rangle = \phi_n(x) = \sqrt{2} \sin(n\pi x) \quad (2.9)$$

defined on the interval  $[0, 1]$ , with  $n \in \{1, 2, 3, \dots\}$ .

1. Calculate the function  $P_\infty(x, y) = \langle x| \left[ \sum_{n=1}^{\infty} |n\rangle\langle n| \right] |y\rangle$ .
2. In computer-based calculations we limit the basis set to  $n \in \{1, 2, 3, \dots, n_{\max}\}$  for some large value of  $n_{\max}$ . Using Mathematica, calculate the function  $P_{n_{\max}}(x, y) = \langle x| \left[ \sum_{n=1}^{n_{\max}} |n\rangle\langle n| \right] |y\rangle$  (use the [Sum](#) function). Make a plot for  $n_{\max} = 10$  (use the [DensityPlot](#) function).
3. What does the function  $P$  represent?

## 2.2 Time-Independent Schrödinger Equation

The time-independent Schrödinger equation is

$$\hat{\mathcal{H}}|\psi\rangle = E|\psi\rangle. \quad (2.10)$$

As in Sect. 2.1 we use a computational basis to express the Hamiltonian operator  $\hat{\mathcal{H}}$  and the quantum state  $\psi$  as

$$\hat{\mathcal{H}} = \sum_{ij} H_{ij} |i\rangle\langle j|, \quad |\psi\rangle = \sum_i \psi_i |i\rangle. \quad (2.11)$$

With these substitutions the Schrödinger equation becomes

$$\begin{aligned} \left[ \sum_{ij} H_{ij} |i\rangle\langle j| \right] \left[ \sum_k \psi_k |k\rangle \right] &= E \left[ \sum_\ell \psi_\ell |\ell\rangle \right] \\ \sum_{ijk} H_{ij} \psi_k \underbrace{\langle j|k\rangle}_{=\delta_{jk}} |i\rangle &= \sum_\ell E \psi_\ell |\ell\rangle \\ \sum_{ij} H_{ij} \psi_j |i\rangle &= \sum_\ell E \psi_\ell |\ell\rangle \end{aligned} \quad (2.12)$$

Multiplying this equation by  $\langle m|$  from the left, and using the orthonormality of the basis set, gives

$$\begin{aligned} \langle m| \sum_{ij} H_{ij} \psi_j |i\rangle &= \langle m| \sum_\ell E \psi_\ell |\ell\rangle \\ \sum_{ij} H_{ij} \psi_j \underbrace{\langle m|i\rangle}_{=\delta_{mi}} &= \sum_\ell E \psi_\ell \underbrace{\langle m|\ell\rangle}_{=\delta_{m\ell}} \\ \sum_j H_{mj} \psi_j &= E \psi_m \end{aligned} \quad (2.13)$$

In matrix notation this can be written as

$$\boxed{\mathbf{H} \cdot \vec{\psi} = E \vec{\psi}}. \quad (2.14)$$

This is the central equation of this book. It is the time-independent Schrödinger equation in a form that computers can understand, namely an eigenvalue equation in terms of numerical (complex) matrices and vectors.

If you think that there is no difference between Eqs. (2.10) and (2.14), then I invite you to re-read this section as I consider it extremely important for what follows in this course. You can think of Eq. (2.10) as an abstract relationship between operators and vectors in Hilbert space, while Eq. (2.14) is a *numerical representation* of this relationship in a concrete basis set  $\{|i\rangle\}_i$ . They both contain the exact same information (since we converted one to the other in a few lines of mathematics) but they are conceptually very different, as one is understandable by a computer and the other is not.

### 2.2.1 Diagonalization

The matrix form of Eq. (2.14) of the Schrödinger equation is an eigenvalue equation as you know from linear algebra. Given a matrix of complex numbers  $\mathbf{H}$  we can find the eigenvalues  $E_i$  and eigenvectors  $\vec{\psi}_i$  using Mathematica's built-in procedures, as described in Sect. 1.10.4.

### 2.2.2 Exercises

**Q2.3** Express the spin-1/2 Hamiltonian

$$\hat{\mathcal{H}} = \sin(\vartheta) \cos(\varphi) \hat{\sigma}_x + \sin(\vartheta) \sin(\varphi) \hat{\sigma}_y + \cos(\vartheta) \hat{\sigma}_z \quad (2.15)$$

in the basis  $\{|\uparrow\rangle, |\downarrow\rangle\}$ , and calculate its eigenvalues and eigenvectors. NB:  $\hat{\sigma}_{x,y,z}$  are the Pauli operators.

## 2.3 Time-Dependent Schrödinger Equation

The time-dependent Schrödinger equation is

$$i\hbar \frac{d}{dt} |\psi(t)\rangle = \hat{\mathcal{H}}(t) |\psi(t)\rangle, \quad (2.16)$$

where the Hamiltonian  $\hat{\mathcal{H}}$  can have an explicit time dependence. This differential equation has the formal solution

$$|\psi(t)\rangle = \hat{\mathcal{U}}(t_0; t) |\psi(t_0)\rangle \quad (2.17)$$

in terms of the *propagator*

$$\begin{aligned} \hat{\mathcal{U}}(t_0; t) = & \mathbb{1} - \frac{i}{\hbar} \int_{t_0}^t dt_1 \hat{\mathcal{H}}(t_1) - \frac{1}{\hbar^2} \int_{t_0}^t dt_1 \int_{t_0}^{t_1} dt_2 \hat{\mathcal{H}}(t_1) \hat{\mathcal{H}}(t_2) + \frac{i}{\hbar^3} \int_{t_0}^t dt_1 \int_{t_0}^{t_1} dt_2 \int_{t_0}^{t_2} dt_3 \hat{\mathcal{H}}(t_1) \hat{\mathcal{H}}(t_2) \hat{\mathcal{H}}(t_3) \\ & + \frac{1}{\hbar^4} \int_{t_0}^t dt_1 \int_{t_0}^{t_1} dt_2 \int_{t_0}^{t_2} dt_3 \int_{t_0}^{t_3} dt_4 \hat{\mathcal{H}}(t_1) \hat{\mathcal{H}}(t_2) \hat{\mathcal{H}}(t_3) \hat{\mathcal{H}}(t_4) + \dots \end{aligned} \quad (2.18)$$

that propagates any state from time  $t_0$  to time  $t$ . An alternative form is given by the *Magnus expansion*<sup>6</sup>

$$\hat{\mathcal{U}}(t_0; t) = \exp \left[ \sum_{k=1}^{\infty} \hat{\Omega}_k(t_0; t) \right] \quad (2.19)$$

---

<sup>6</sup>See [https://en.wikipedia.org/wiki/Magnus\\_expansion](https://en.wikipedia.org/wiki/Magnus_expansion).



with the contributions

$$\begin{aligned}
 \hat{\Omega}_1(t_0; t) &= -\frac{i}{\hbar} \int_{t_0}^t dt_1 \hat{\mathcal{H}}(t_1) \\
 \hat{\Omega}_2(t_0; t) &= -\frac{1}{2\hbar^2} \int_{t_0}^t dt_1 \int_{t_0}^{t_1} dt_2 [\hat{\mathcal{H}}(t_1), \hat{\mathcal{H}}(t_2)] \\
 \hat{\Omega}_3(t_0; t) &= \frac{i}{6\hbar^3} \int_{t_0}^t dt_1 \int_{t_0}^{t_1} dt_2 \int_{t_0}^{t_2} dt_3 \left( [\hat{\mathcal{H}}(t_1), [\hat{\mathcal{H}}(t_2), \hat{\mathcal{H}}(t_3)]] + [\hat{\mathcal{H}}(t_3), [\hat{\mathcal{H}}(t_2), \hat{\mathcal{H}}(t_1)]] \right) \\
 &\dots
 \end{aligned} \tag{2.20}$$

This expansion in terms of different-time commutators is often easier to evaluate than Eq. (2.18), especially when the contributions vanish for  $k > k_{\max}$  (see Sect. 2.3.3 for the case  $k_{\max} = 1$ ). Even if higher-order contributions do not vanish entirely, they (usually) decrease in importance much more rapidly with increasing  $k$  than those of Eq. (2.18). Also, even if the Magnus expansion is artificially truncated (neglecting higher-order terms), the quantum-mechanical evolution is still unitary; this is not the case for Eq. (2.18).

Notice that the exponential in Eq. (2.19) has an operator or a matrix as their argument: in Mathematica this matrix exponentiation is done with the `MatrixExp` function. It does not calculate the exponential element-by-element, but instead calculates

$$e^{\hat{A}} = \sum_{n=0}^{\infty} \frac{\hat{A}^n}{n!}, \quad e^A = \sum_{n=0}^{\infty} \frac{A^n}{n!}. \tag{2.21}$$

### 2.3.1 Time-Independent Basis

We express the quantum state again in terms of the chosen basis, which is assumed to be time-independent. This leaves the time-dependence in the expansion coefficients,

$$\hat{\mathcal{H}}(t) = \sum_{ij} H_{ij}(t) |i\rangle\langle j|, \quad |\psi(t)\rangle = \sum_i \psi_i(t) |i\rangle. \tag{2.22}$$

Inserting these expressions into the time-dependent Schrödinger equation (2.16) gives

$$i\hbar \sum_i \dot{\psi}_i(t) |i\rangle = \left[ \sum_{jk} H_{jk}(t) |j\rangle\langle k| \right] \sum_{\ell} \psi_{\ell}(t) |\ell\rangle = \sum_{jk} H_{jk}(t) \psi_k(t) |j\rangle. \tag{2.23}$$

Multiplying with  $\langle m|$  from the left:

$$i\hbar \dot{\psi}_m(t) = \sum_k H_{mk}(t) \psi_k(t) \tag{2.24}$$

or, in matrix notation,

$$\boxed{i\hbar\dot{\vec{\psi}}(t) = \mathbf{H}(t) \cdot \vec{\psi}(t).} \quad (2.25)$$

Since the matrix  $\mathbf{H}(t)$  is supposedly known, this equation represents a system of coupled complex differential equations for the vector  $\vec{\psi}(t)$ , which can be solved on a computer.

### 2.3.2 Time-Dependent Basis: Interaction Picture

It can be advantageous to use a time-dependent basis. The most frequently used such basis is given by the interaction picture of quantum mechanics, where the Hamiltonian can be split into a time-independent principal part  $\hat{\mathcal{H}}_0$  and a small time-dependent part  $\hat{\mathcal{H}}_1$ :

$$\hat{\mathcal{H}}(t) = \hat{\mathcal{H}}_0 + \hat{\mathcal{H}}_1(t). \quad (2.26)$$

Assuming that we can diagonalize  $\hat{\mathcal{H}}_0$ , possibly numerically, such that the eigenfunctions satisfy  $\hat{\mathcal{H}}_0|i\rangle = E_i|i\rangle$ , we propose the time-dependent basis

$$|i(t)\rangle = e^{-iE_it/\hbar}|i\rangle. \quad (2.27)$$

If we express any quantum state in this basis as

$$|\psi(t)\rangle = \sum_i \psi_i(t) |i(t)\rangle = \sum_i \psi_i(t) e^{-iE_it/\hbar} |i\rangle, \quad (2.28)$$

the time-dependent Schrödinger equation becomes

$$\begin{aligned} \sum_i [i\hbar\dot{\psi}_i(t) + E_i\psi_i(t)] e^{-iE_it/\hbar} |i\rangle &= \sum_j \psi_j(t) e^{-iE_jt/\hbar} E_j |j\rangle + \sum_j \psi_j(t) e^{-iE_jt/\hbar} \hat{\mathcal{H}}_1(t) |j\rangle \\ \sum_i i\hbar\dot{\psi}_i(t) e^{-iE_it/\hbar} |i\rangle &= \sum_j \psi_j(t) e^{-iE_jt/\hbar} \hat{\mathcal{H}}_1(t) |j\rangle \end{aligned} \quad (2.29)$$

Multiply by  $\langle k|$  from the left:

$$\begin{aligned} \langle k| \sum_i i\hbar\dot{\psi}_i(t) e^{-iE_it/\hbar} |i\rangle &= \langle k| \sum_j \psi_j(t) e^{-iE_jt/\hbar} \hat{\mathcal{H}}_1(t) |j\rangle \\ \sum_i i\hbar\dot{\psi}_i(t) e^{-iE_it/\hbar} \underbrace{\langle k|i\rangle}_{=\delta_{ki}} &= \sum_j \psi_j(t) e^{-iE_jt/\hbar} \langle k|\hat{\mathcal{H}}_1(t)|j\rangle \\ i\hbar\dot{\psi}_k(t) &= \sum_j \psi_j(t) e^{-i(E_j-E_k)t/\hbar} \langle k|\hat{\mathcal{H}}_1(t)|j\rangle. \end{aligned} \quad (2.30)$$

This is the same matrix/vector evolution expression as Eq. (2.25), except that here the Hamiltonian matrix elements must be defined as

$$H_{ij}(t) = \langle i | \hat{\mathcal{H}}_1(t) | j \rangle e^{-i(E_j - E_i)t/\hbar}. \quad (2.31)$$

We see immediately that if the interaction Hamiltonian vanishes [ $\hat{\mathcal{H}}_1(t) = 0$ ], then the expansion coefficients  $\psi_i(t)$  become time-independent, as expected since they are the coefficients of the eigenfunctions of the time-independent Schrödinger equation.

When a quantum-mechanical system is composed of different parts that have vastly different energy scales of their internal evolution  $\hat{\mathcal{H}}_0$ , then the use of Eq. (2.31) can have great numerical advantages. It turns out that the relevant interaction terms  $H_{ij}(t)$  in the interaction picture will have relatively slowly evolving phases  $\exp[-i(E_j - E_i)t/\hbar]$ , on a time scale given by relative energy *differences* and not by *absolute* energies; this makes it possible to solve the coupled differential equations of Eq. (2.25) numerically without using an absurdly small time step.

### 2.3.3 Special Case: $[\hat{\mathcal{H}}(t), \hat{\mathcal{H}}(t')] = 0 \forall(t, t')$

If the Hamiltonian commutes with itself at different times,  $[\hat{\mathcal{H}}(t), \hat{\mathcal{H}}(t')] = 0 \forall(t, t')$ , the propagator (2.19) of Eq. (2.16) can be simplified to

$$\hat{\mathcal{U}}(t_0; t) = \exp \left[ -\frac{i}{\hbar} \int_{t_0}^t \hat{\mathcal{H}}(s) ds \right], \quad (2.32)$$

and the corresponding solution of Eq. (2.25) is

$$\vec{\psi}(t) = \exp \left[ -\frac{i}{\hbar} \int_{t_0}^t \mathbf{H}(s) ds \right] \cdot \vec{\psi}(t_0). \quad (2.33)$$

Again, these matrix exponentials are calculated with `MatrixExp` in Mathematica.

### 2.3.4 Special Case: Time-Independent Hamiltonian

In the special (but common) case where the Hamiltonian is time-independent, the integral in Eq. (2.33) can be evaluated immediately, and the solution is

$$\vec{\psi}(t) = \exp \left[ -\frac{i(t - t_0)}{\hbar} \mathbf{H} \right] \cdot \vec{\psi}(t_0). \quad (2.34)$$

If we have a specific Hamiltonian matrix  $\mathbf{H}$  defined, for example the matrix of Sect. 1.10.4, we can calculate the propagator  $\mathbf{U}(\Delta t) = \exp[-i\mathbf{H}\Delta t/\hbar]$  for  $\Delta t = t - t_0$  with

```
1 In[234]:=U[Δt_] = MatrixExp[-I*H*Δt/ħ]
```

The resulting expression for  $\mathbf{U}[\Delta t]$  will in general be very long, and slow to compute. A more efficient definition is to matrix-exponentiate a numerical matrix for specific values of the propagation interval  $\Delta t$ , using a delayed assignment:

```
1 In[235]:=U[Δt_?NumericQ] := MatrixExp[-I*H*N[Δt]/ħ]
```

### 2.3.5 Exercises

**Q2.4** Demonstrate that the propagator (2.32) gives a quantum state (2.17) that satisfies Eq. (2.16).

**Q2.5** Calculate the propagator of the Hamiltonian of Q2.3.

**Q2.6** After In[234] and In[235], check ?U. Which definition of U comes first? Why?

## 2.4 Basis Construction

In principle, the choice of basis set  $\{|i\rangle\}_i$  does not influence the way a computer program like Mathematica solves a quantum-mechanical problem. In practice, however, we always need a *constructive* way to find some basis for a given quantum-mechanical problem. A basis that takes the system's Hamiltonian into account may give a computationally simpler description; but in complicated systems it is often more important to find *any* way of constructing a usable basis set than finding the perfect one.

### 2.4.1 Description of a Single Degree of Freedom

When we describe a single quantum-mechanical degree of freedom, it is often possible to deduce a useful basis set from knowledge of the Hilbert space itself. This is what we will be doing in Chap. 3 for spin systems, where the well-known Dicke basis  $\{|S, M_S\rangle\}_{M_S=-S}^S$  turns out to be very useful.

For more complicated degrees of freedom, we can find inspiration for a basis choice from an associated Hamiltonian. Such Hamiltonians describing a single degree of freedom are often so simple that they can be diagonalized by hand. If this

is not the case, real-world Hamiltonians  $\hat{\mathcal{H}}$  can often be decomposed like Eq. (2.26) into a “simple” part  $\hat{\mathcal{H}}_0$  that is time-independent and can be diagonalized easily, and a “difficult” part  $\hat{\mathcal{H}}_1$  that usually contains complicated interactions and/or time-dependent terms but is of smaller magnitude. A natural choice of basis set is the set of eigenstates of  $\hat{\mathcal{H}}_0$ , or at least those eigenstates below a certain cutoff energy since they will be optimally suited to describe the complete low-energy behavior of the degree of freedom in question. This latter point is especially important for infinite-dimensional systems (Chap. 4), where any computer representation will necessarily truncate the dimensionality, as discussed in Sect. 2.1.1.

### Examples of Basis Sets for Single Degrees of Freedom:

- *spin degree of freedom*: Dicke states  $|S, M_S\rangle$  (see Chap. 3)
- *translational degree of freedom*: square-well eigenstates, harmonic oscillator eigenstates (see Chap. 4)
- *rotational degree of freedom*: spherical harmonics
- *atomic system*: hydrogen-like orbitals
- *translation-invariant system*: periodic plane waves
- *periodic system (crystal)*: periodic plane waves on the reciprocal lattice.

### 2.4.2 Description of Coupled Degrees of Freedom

A broad range of quantum-mechanical systems of interest are governed by Hamiltonians of the form

$$\hat{\mathcal{H}}(t) = \left( \sum_{k=1}^N \hat{\mathcal{H}}^{(k)}(t) \right) + \hat{\mathcal{H}}_{\text{int}}(t), \quad (2.35)$$

where  $N$  individual degrees of freedom are governed by their individual Hamiltonians  $\hat{\mathcal{H}}^{(k)}(t)$ , while their interactions are described by  $\hat{\mathcal{H}}_{\text{int}}(t)$ . This is a situation we will encounter repeatedly as we construct more complicated quantum-mechanical problems from simpler parts. A few simple examples are:

- A set of  $N$  interacting particles: the Hamiltonians  $\hat{\mathcal{H}}^{(k)}$  describe the individual particles, while  $\hat{\mathcal{H}}_{\text{int}}$  describes their interactions (see Sect. 3.4).
- A single particle moving in three spatial degrees of freedom: the three Hamiltonians  $\hat{\mathcal{H}}^{(x)} = -\frac{\hbar^2}{2m} \frac{\partial^2}{\partial x^2}$ ,  $\hat{\mathcal{H}}^{(y)} = -\frac{\hbar^2}{2m} \frac{\partial^2}{\partial y^2}$ ,  $\hat{\mathcal{H}}^{(z)} = -\frac{\hbar^2}{2m} \frac{\partial^2}{\partial z^2}$  describe the kinetic energy in the three directions, while  $\hat{\mathcal{H}}_{\text{int}}$  contains the potential energy, which usually couples these three degrees of freedom (see Sect. 4.4).
- A single particle with internal (spin) and external (motional) degrees of freedom, which are coupled through a state-dependent potential in  $\hat{\mathcal{H}}_{\text{int}}$  (see Chap. 5).

The existence of individual Hamiltonians  $\hat{\mathcal{H}}^{(k)}$  assumes that the Hilbert space of the complete system has a tensor-product structure

$$V = V^{(1)} \otimes V^{(2)} \otimes \dots \otimes V^{(N)}, \quad (2.36)$$

where each Hamiltonian  $\hat{\mathcal{H}}^{(k)}$  acts only in a single component space,

$$\hat{\mathcal{H}}^{(k)} = \mathbb{1}^{(1)} \otimes \mathbb{1}^{(2)} \otimes \dots \otimes \mathbb{1}^{(k-1)} \otimes \hat{h}^{(k)} \otimes \mathbb{1}^{(k+1)} \otimes \dots \otimes \mathbb{1}^{(N)}. \quad (2.37)$$

Further, if we are able to construct bases  $\{|i_k\rangle^{(k)}\}_{i_k=1}^{n_k}$  for all of the component Hilbert spaces  $V^{(k)}$ , as in Sect. 2.4.1, then we can construct a basis for the full Hilbert space  $V$  by taking all possible tensor products of basis functions:

$$|i_1, i_2, \dots, i_N\rangle = |i_1\rangle^{(1)} \otimes |i_2\rangle^{(2)} \otimes \dots \otimes |i_N\rangle^{(N)}. \quad (2.38)$$

This basis will have  $\prod_{k=1}^N n_k$  elements, which can easily become a very large number for composite systems.

## Quantum States

A product state of the complete system

$$|\psi\rangle = |\psi_1\rangle^{(1)} \otimes |\psi_2\rangle^{(2)} \otimes \dots \otimes |\psi_N\rangle^{(N)} \quad (2.39)$$

can be described in the following way. First, each single-particle state is decomposed in its own basis as in Eq. (2.4),

$$|\psi_k\rangle^{(k)} = \sum_{i_k=1}^{n_k} \psi_{i_k}^{(k)} |i_k\rangle^{(k)}. \quad (2.40)$$

Inserting these expansions into Eq. (2.39) gives the expansion into the basis functions 2.38 of the full system,

$$\begin{aligned} |\psi\rangle &= \left[ \sum_{i_1=1}^{n_1} \psi_{i_1}^{(1)} |i_1\rangle^{(1)} \right] \otimes \left[ \sum_{i_2=1}^{n_2} \psi_{i_2}^{(2)} |i_2\rangle^{(2)} \right] \otimes \dots \otimes \left[ \sum_{i_N=1}^{n_N} \psi_{i_N}^{(N)} |i_N\rangle^{(N)} \right] \\ &= \sum_{i_1=1}^{n_1} \sum_{i_2=1}^{n_2} \dots \sum_{i_N=1}^{n_N} \left[ \psi_{i_1}^{(1)} \psi_{i_2}^{(2)} \dots \psi_{i_N}^{(N)} \right] |i_1, i_2, \dots, i_N\rangle \end{aligned} \quad (2.41)$$

In Mathematica, such a state tensor product can be calculated as follows. For example, assume that  $\psi_1$  is a vector containing the expansion of  $|\psi_1\rangle^{(1)}$  in its basis, and similarly for  $\psi_2$  and  $\psi_3$ . The vector  $\psi$  of expansion coefficients of the full state  $|\psi\rangle = |\psi_1\rangle^{(1)} \otimes |\psi_2\rangle^{(2)} \otimes |\psi_3\rangle^{(3)}$  is calculated with

```
1 In[236]:= ψ = Flatten[KroneckerProduct[ψ1, ψ2, ψ3]]
```

See Q2.10 for a numerical example.

More generally, any state can be written as

$$|\psi\rangle = \sum_{i_1=1}^{n_1} \sum_{i_2=1}^{n_2} \cdots \sum_{i_N=1}^{n_N} \psi_{i_1, i_2, \dots, i_N} |i_1, i_2, \dots, i_N\rangle, \quad (2.42)$$

of which Eq. (2.41) is a special case with  $\psi_{i_1, i_2, \dots, i_N} = \psi_{i_1}^{(1)} \psi_{i_2}^{(2)} \cdots \psi_{i_N}^{(N)}$ .

## Operators

If the Hilbert space has the tensor-product structure of Eq. (2.36), then the operators acting on this full space are often given as tensor products as well,

$$\hat{A} = \hat{a}_1^{(1)} \otimes \hat{a}_2^{(2)} \otimes \cdots \otimes \hat{a}_N^{(N)}, \quad (2.43)$$

or as a sum over such products. If every single-particle operator is decomposed in its own basis as in Eq. (2.2),

$$\hat{a}_k^{(k)} = \sum_{i_k=1}^{n_k} \sum_{j_k=1}^{n_k} a_{i_k, j_k}^{(k)} |i_k\rangle^{(k)} \langle j_k|^{(k)}, \quad (2.44)$$

inserting these expressions into Eq. (2.43) gives the expansion into the basis functions 2.38 of the full system,

$$\begin{aligned} \hat{A} &= \left[ \sum_{i_1=1}^{n_1} \sum_{j_1=1}^{n_1} a_{i_1, j_1}^{(1)} |i_1\rangle^{(1)} \langle j_1|^{(1)} \right] \otimes \left[ \sum_{i_2=1}^{n_2} \sum_{j_2=1}^{n_2} a_{i_2, j_2}^{(2)} |i_2\rangle^{(2)} \langle j_2|^{(2)} \right] \otimes \cdots \otimes \left[ \sum_{i_N=1}^{n_N} \sum_{j_N=1}^{n_N} a_{i_N, j_N}^{(N)} |i_N\rangle^{(N)} \langle j_N|^{(N)} \right] \\ &= \sum_{i_1=1}^{n_1} \sum_{j_1=1}^{n_1} \sum_{i_2=1}^{n_2} \sum_{j_2=1}^{n_2} \cdots \sum_{i_N=1}^{n_N} \sum_{j_N=1}^{n_N} \left[ a_{i_1, j_1}^{(1)} a_{i_2, j_2}^{(2)} \cdots a_{i_N, j_N}^{(N)} \right] |i_1, i_2, \dots, i_N\rangle \langle j_1, j_2, \dots, j_N|. \end{aligned} \quad (2.45)$$

In Mathematica, such an operator tensor product can be calculated similarly to In [236] above. For example, assume that `a1` is a matrix containing the expansion of  $\hat{a}_1^{(1)}$  in its basis, and similarly for `a2` and `a3`. The matrix `A` of expansion coefficients of the full operator  $\hat{A} = \hat{a}_1^{(1)} \otimes \hat{a}_2^{(2)} \otimes \hat{a}_3^{(3)}$  is calculated with

```
1 In[237]:= A = KroneckerProduct[a1, a2, a3]
```

Often we need to construct operators which act only on one of the component spaces, as in Eq. (2.37). For example, in a 3-composite system the subsystem Hamiltonians  $\hat{h}^{(1)}$ ,  $\hat{h}^{(2)}$ , and  $\hat{h}^{(3)}$  are first expanded to the full Hilbert space,

```

1 In[238]:=H1 = KroneckerProduct[h1,
2           IdentityMatrix[Dimensions[h2]],
3           IdentityMatrix[Dimensions[h3]]];
4 In[239]:=H2 = KroneckerProduct[IdentityMatrix[Dimensions[h1]],
5           h2,
6           IdentityMatrix[Dimensions[h3]]];
7 In[240]:=H3 = KroneckerProduct[IdentityMatrix[Dimensions[h1]],
8           IdentityMatrix[Dimensions[h2]],
9           h3];

```

where `IdentityMatrix[Dimensions[h1]]` generates a unit matrix of size equal to that of `h1`. In this way, the matrices `H1`, `H2`, `H3` are of equal size and can be added together, even if `h1`, `h2`, `h3` all have different sizes (expressed in Hilbert spaces of different dimensions):

```

1 In[241]:=H = H1 + H2 + H3;

```

More generally, any operator can be written as

$$\hat{A} = \sum_{i_1=1}^{n_1} \sum_{j_1=1}^{n_1} \sum_{i_2=1}^{n_2} \sum_{j_2=1}^{n_2} \cdots \sum_{i_N=1}^{n_N} \sum_{j_N=1}^{n_N} a_{i_1, j_1, i_2, j_2, \dots, i_N, j_N} |i_1, i_2, \dots, i_N\rangle \langle j_1, j_2, \dots, j_N|, \quad (2.46)$$

of which Eq. (2.45) is a special case with  $a_{i_1, j_1, i_2, j_2, \dots, i_N, j_N} = a_{i_1, j_1}^{(1)} a_{i_2, j_2}^{(2)} \cdots a_{i_N, j_N}^{(N)}$ .

### 2.4.3 Reduced Density Matrices [\[Supplementary Material 1\]](#)

In this section we calculate reduced density matrices by partial tracing. We start with the most general tripartite case, and then specialize to the more common bipartite case.

Assume that our quantum-mechanical system is composed of three parts A, B, C, and that its Hilbert space is a tensor product of the three associated Hilbert spaces with dimensions  $d_A, d_B, d_C$ :  $V = V^{(A)} \otimes V^{(B)} \otimes V^{(C)}$ . Similar to Eq. (2.46), any state of this system can be written as a density matrix

$$\hat{\rho}_{ABC} = \sum_{i, i'=1}^{d_A} \sum_{j, j'=1}^{d_B} \sum_{k, k'=1}^{d_C} \rho_{i, j, k, i', j', k'} |i_A, j_B, k_C\rangle \langle i'_A, j'_B, k'_C|, \quad (2.47)$$

where we use the basis states  $|i_A, j_B, k_C\rangle = |i\rangle^{(A)} \otimes |j\rangle^{(B)} \otimes |k\rangle^{(C)}$  defined in terms of the three basis sets of the three component Hilbert spaces.

We calculate a reduced density matrix  $\hat{\rho}_{AC} = \text{Tr}_B \hat{\rho}_{ABC}$ , which describes what happens to our knowledge of the subsystems A and C when we forget about subsystem B. For example, we could be studying a system of three particles, and take an interest in the state of particles A and C after we have lost particle B. This reduced density matrix is defined as a partial trace,



$$\begin{aligned}
\hat{\rho}_{AC} &= \sum_{j''=1}^{d_B} \langle j''_B | \hat{\rho}_{ABC} | j''_B \rangle = \sum_{j''=1}^{d_B} \langle j''_B | \left[ \sum_{i,i'=1}^{d_A} \sum_{j,j'=1}^{d_B} \sum_{k,k'=1}^{d_C} \rho_{i,j,k,i',j',k'} |i_A, j_B, k_C\rangle \langle i'_A, j'_B, k'_C| \right] |j''_B\rangle \\
&= \sum_{j''=1}^{d_B} \sum_{i,i'=1}^{d_A} \sum_{j,j'=1}^{d_B} \sum_{k,k'=1}^{d_C} \rho_{i,j,k,i',j',k'} \langle j''_B | i_A, j_B, k_C \rangle \langle i'_A, j'_B, k'_C | j''_B \rangle \\
&= \sum_{j''=1}^{d_B} \sum_{i,i'=1}^{d_A} \sum_{j,j'=1}^{d_B} \sum_{k,k'=1}^{d_C} \rho_{i,j,k,i',j',k'} [\delta_{j'',j} |i_A, k_C\rangle] [\delta_{j'',j'} \langle i'_A, k'_C|] \\
&= \sum_{i,i'=1}^{d_A} \sum_{k,k'=1}^{d_C} \left[ \sum_{j=1}^{d_B} \rho_{i,j,k,i',j,k'} \right] |i_A, k_C\rangle \langle i'_A, k'_C|,
\end{aligned} \tag{2.48}$$

which makes no reference to subsystem B. It only describes the joint system AC that is left after forgetting about subsystem B.

In Mathematica, we mostly use flattened basis sets, that is, our basis set for the joint Hilbert space of subsystems A, B, C is a flat list of length  $d = d_A d_B d_C$ :

$$\{|1_A, 1_B, 1_C\rangle, |1_A, 1_B, 2_C\rangle, \dots, |1_A, 1_B, d_C\rangle, |1_A, 2_B, 1_C\rangle, |1_A, 2_B, 2_C\rangle, \dots, |1_A, 2_B, d_C\rangle, \dots, |d_A, d_B, d_C\rangle\}. \tag{2.49}$$

In Sect. 1.10.5 we have seen how lists and tensors can be re-shaped. As we will see below, these tools are used to switch between representations involving indices  $(i, j, k)$  (i.e., lists with three indices, rank-three tensors) corresponding to Eq. (2.47), and lists involving a single flattened-out index corresponding more to Eq. (2.49).

In practical calculations, any density matrix  $\rho_{ABC}$  of the joint system is given as a  $d \times d$  matrix whose element  $(u, v)$  is the prefactor of the contribution  $|u\rangle\langle v|$  with the indices  $u$  and  $v$  addressing elements in the flat list of Eq. (2.49). In order to calculate a reduced density matrix, we first reshape this  $d \times d$  density matrix  $\rho_{ABC}$  into a rank-six tensor  $\mathbf{R}$  with dimensions  $d_A \times d_B \times d_C \times d_A \times d_B \times d_C$ , and with elements  $r_{i,j,k,i',j',k'}$  of Eq. (2.47):

```
1 In[242] := R = ArrayReshape[ρABC, {dA,dB,dC,dA,dB,dC}]
```

Next, we contract indices 2 and 5 of  $\mathbf{R}$  in order to do the partial trace over subsystem B, as is done in Eq. (2.48) (effectively setting  $j = j'$  and summing over  $j$ ). We find a rank-4 tensor  $\mathbf{S}$  with dimensions  $d_A \times d_C \times d_A \times d_C$ :

```
1 In[243] := S = TensorContract[R, {2,5}]
```

Finally, we flatten out this tensor again (simultaneously combining indices 1&2 and 3&4) to find the  $d_A d_C \times d_A d_C$  reduced density matrix  $\rho_{AC}$ :

```
1 In[244] := ρAC = Flatten[S, {{1,2}, {3,4}}]
```

We assemble all of these steps into a generally usable function:

```

1 In[245]:=rdm[pABC_?MatrixQ, {dA_Integer /; dA >= 1,
2           dB_Integer /; dB >= 1,
3           dC_Integer /; dC >= 1}] /;
4     Dimensions[pABC] == {dA*dB*dC, dA*dB*dC} :=
5     Flatten[TensorContract[ArrayReshape[pABC, {dA,dB,dC,dA,dB,dC}], {2,5}],
6           {{1,2}, {3,4}}]

```

When our system is in a pure state,  $\hat{\rho}_{ABC} = |\psi\rangle\langle\psi|$ , this procedure can be simplified greatly. This is particularly important for large system dimensions, where calculating the full density matrix  $\hat{\rho}_{ABC}$  may be impossible due to memory constraints. For this, we assume that  $|\psi\rangle = \sum_{i=1}^{d_A} \sum_{j=1}^{d_B} \sum_{k=1}^{d_C} \psi_{i,j,k} |i_A, j_B, k_C\rangle$ , and therefore  $\rho_{i,j,k,i',j',k'} = \psi_{i,j,k} \psi_{i',j',k'}^*$ . Again, in Mathematica the coefficients of a state vector  $\psi_{ABC}$  are a flat list referring to the elements of the flat basis of Eq. (2.49), and so we start by constructing a rank-3 tensor  $P$  with dimensions  $d_A \times d_B \times d_C$ , whose elements are exactly the  $\psi_{i,j,k}$ , similar to In[242]:

```

1 In[246]:=P = ArrayReshape[ψABC, {dA,dB,dC}]

```

We transpose this rank-three tensor into a  $d_A \times d_C \times d_B$  tensor  $P1$  and a  $d_B \times d_A \times d_C$  tensor  $P2$  by changing the order of the indices:

```

1 In[247]:=P1 = Transpose[P, {1, 3, 2}]
2 In[248]:=P2 = Transpose[P, {2, 1, 3}]

```

Now we can contract the index  $j_B$  by a dot product, to find a rank-4 tensor  $Q$  with dimensions  $d_A \times d_C \times d_A \times d_C$ :

```

1 In[249]:=Q = P1 . Conjugate[P2]

```

Finally we flatten  $Q$  into the  $d_A d_C \times d_A d_C$  reduced density matrix  $\rho_{AC}$  by combining indices 1&2 and 3&4:

```

1 In[250]:=ρAC = Flatten[Q, {{1,2}, {3,4}}]

```

We assemble all of these steps into a generally usable function that extends the definition of In[245]:

```

1 In[251]:=rdm[ψABC_?VectorQ, {dA_Integer /; dA >= 1,
2           dB_Integer /; dB >= 1,
3           dC_Integer /; dC >= 1}] /;
4     Length[ψABC] == dA*dB*dC :=
5     With[{P = ArrayReshape[ψABC, {dA,dB,dC}]},
6           Flatten[Transpose[P, {1,3,2}].ConjugateTranspose[P], {{1,2}, {3,4}}]]

```

Notice that we have merged the transposition of In[248] and the complex-conjugation of In[249] into a single call of the `ConjugateTranspose` function.

## Bipartite Systems

Consider now the more common case of a bipartite system composed of only two subsystems A and B. We can still use the definitions developed above for tripartite (ABC) structures by introducing a trivial third subsystem with dimension  $d_C = 1$ . This trivial subsystem will not change anything since it must always be in its one and only possible state. Therefore, given a density matrix  $\rho_{AB}$  of the joint system AB, we calculate the reduced density matrices of subsystems A and B with

```
1 In[252]:= ρA = rdm[ρAB, {dA,dB,1}];
2 In[253]:= ρB = rdm[ρAB, {1,dA,dB}];
```

respectively, since it is always the middle subsystem of a given list of three subsystems that is eliminated through partial tracing. In typical Mathematica fashion, we define a `traceout` function that traces out the first  $d$  dimensions if  $d > 0$  and the last  $d$  dimensions if  $d < 0$ :

```
1 In[254]:= traceout[ρ_?MatrixQ, d_Integer /; d >= 1] /;
2   Length[ρ] == Length[Transpose[ρ]] && Divisible[Length[ρ], d] :=
3   rdm[ρ, {1, d, Length[ρ]/d}]
4 In[255]:= traceout[ρ_?MatrixQ, d_Integer /; d <= -1] /;
5   Length[ρ] == Length[Transpose[ρ]] && Divisible[Length[ρ], -d] :=
6   rdm[ρ, {Length[ρ]/(-d), -d, 1}]
7 In[256]:= traceout[ψ_?VectorQ, d_Integer /; d >= 1] /; Divisible[Length[ψ], d] :=
8   rdm[ψ, {1, d, Length[ψ]/d}]
9 In[257]:= traceout[ψ_?VectorQ, d_Integer /; d <= -1] /; Divisible[Length[ψ], -d] :=
10  rdm[ψ, {Length[ψ]/(-d), -d, 1}]
```

### 2.4.4 Exercises

**Q2.7** Two particles of mass  $m$  are moving in a three-dimensional harmonic potential  $V(r) = \frac{1}{2}m\omega^2 r^2$  with  $r = \sqrt{x^2 + y^2 + z^2}$ , and interacting via  $s$ -wave scattering  $V_{\text{int}} = g\delta^3(\vec{r}_1 - \vec{r}_2)$ .

1. Write down the Hamiltonian of this system.
2. Propose a basis set in which we can describe the quantum mechanics of this system.
3. Calculate the matrix elements of the Hamiltonian in this basis set.

**Q2.8** Calculate  $\psi$  in In[236] without using `KroneckerProduct`, but using the `Table` command instead.

**Q2.9** Calculate  $A$  in In[237] without using `KroneckerProduct`, but using the `Table` command instead.

**Q2.10** Given two spin-1/2 particles in states

$$|\psi\rangle^{(1)} = 0.8|\uparrow\rangle - 0.6|\downarrow\rangle, \quad |\psi\rangle^{(2)} = 0.6i|\uparrow\rangle + 0.8|\downarrow\rangle, \quad (2.50)$$


use the `KroneckerProduct` function to calculate the joint state  $|\psi\rangle = |\psi\rangle^{(1)} \otimes |\psi\rangle^{(2)}$ , and compare the result to a manual calculation. In which order do the coefficients appear in the result of `KroneckerProduct`?

**Q2.11** For the state of Eq. (2.50), calculate the reduced density matrices  $\rho^{(1)}$  and  $\rho^{(2)}$  by tracing out the other subsystem. Compare them to the density matrices  $|\psi\rangle^{(1)}\langle\psi|^{(1)}$  and  $|\psi\rangle^{(2)}\langle\psi|^{(2)}$ . What do you notice?

See also **Q3.19** and **Q3.20**.

## Chapter 3

# Spin and Angular Momentum



In this chapter we put together everything we have studied so far—Mathematica, quantum mechanics, computational bases, units—to study simple quantum systems. We start our explorations of quantum mechanics with the description of angular momentum. The reason for this choice is that, in contrast to the mechanically more intuitive linear motion (Chap. 4), rotational motion is described with finite-dimensional Hilbert spaces and thus lends itself as a relatively simple starting point. As applications we look at the hyperfine structure of alkali atoms, lattice spin models, and quantum circuits.

---

**Electronic supplementary material** The online version of this chapter ([https://doi.org/10.1007/978-981-13-7588-0\\_3](https://doi.org/10.1007/978-981-13-7588-0_3)) contains supplementary material, which is available to authorized users.

### 3.1 Quantum-Mechanical Spin and Angular Momentum Operators [Supplementary Material 1]

A classical rotational motion is described by its angular momentum, which is a three-dimensional pseudovector<sup>1</sup> whose direction indicates the rotation axis and whose length gives the rotational momentum. For an isolated system, the angular momentum is conserved and is thus very useful in the description of the system's state.

In quantum mechanics, angular momentum is equally described by a three-dimensional pseudovector operator  $\hat{\hat{S}}$ , with operator elements (in Cartesian coordinates)  $\hat{\hat{S}} = (\hat{S}_x, \hat{S}_y, \hat{S}_z)$ . The joint eigenstates of the squared angular momentum magnitude  $\|\hat{\hat{S}}\|^2 = \hat{S}^2 = \hat{S}_x^2 + \hat{S}_y^2 + \hat{S}_z^2$  and of the  $z$ -component  $\hat{S}_z$  are called the Dicke states  $|S, M\rangle$ , and satisfy

$$\hat{S}^2 |S, M\rangle = S(S+1) |S, M\rangle \quad (3.1a)$$

$$\hat{S}_z |S, M\rangle = M |S, M\rangle \quad (3.1b)$$

For every integer or half-integer value of the angular momentum  $S \in \{0, \frac{1}{2}, 1, \frac{3}{2}, 2, \dots\}$ , there is a set of  $2S+1$  Dicke states  $|S, M\rangle$  with  $M \in \{-S, -S+1, \dots, S-1, S\}$  that form a basis for the description of the rotation axis orientation. These states also satisfy the following relationships with respect to the  $x$ - and  $y$ -components of the angular momentum:

$$\hat{S}_+ |S, M\rangle = \sqrt{S(S+1) - M(M+1)} |S, M+1\rangle \quad \text{raising operator} \quad (3.2a)$$

$$\hat{S}_- |S, M\rangle = \sqrt{S(S+1) - M(M-1)} |S, M-1\rangle \quad \text{lowering operator} \quad (3.2b)$$

$$\hat{S}_\pm = \hat{S}_x \pm i\hat{S}_y \quad \text{Cartesian components} \quad (3.2c)$$

As you know, quantum mechanics is not limited to spins or angular momenta of length  $S = 1/2$ .

In Mathematica we represent these operators in the Dicke basis as follows, with the elements of the basis set ordered with decreasing projection quantum number  $M$ :

```

1 In[258]:=SpinQ[S_] := IntegerQ[2S] && S>=0
2 In[259]:=splus[0] = {{0}} //SparseArray;
3 In[260]:=splus[S_?SpinQ] := splus[S] =
4   SparseArray[Band[{1,2}] -> Table[Sqrt[S(S+1)-M(M+1)],
5     {M,S-1,-S,-1}], {2S+1,2S+1}]
6 In[261]:=sminus[S_?SpinQ] := Transpose[splus[S]]
7 In[262]:=sx[S_?SpinQ] := sx[S] = (splus[S]+sminus[S])/2
8 In[263]:=sy[S_?SpinQ] := sy[S] = (splus[S]-sminus[S])/(2I)
9 In[264]:=sz[S_?SpinQ] := sz[S] = SparseArray[Band[{1,1}]->Range[S,-S,-1], {2S+1,2S+1}]
10 In[265]:=id[S_?SpinQ] := id[S] = IdentityMatrix[2S+1, SparseArray]
```

<sup>1</sup>See <https://en.wikipedia.org/wiki/Pseudovector>.

- Notice that we have defined all these matrix representations as *sparse* matrices (see Sect. 1.10.3), which will make larger calculations much more efficient later on. Further, all definitions are memoizing (see Sect. 1.6.3) to reduce execution time when they are used repeatedly.
- The function `SpinQ[S]` yields `True` only if `S` is a nonnegative half-integer value and can therefore represent a physically valid spin. In general, functions ending in `...Q` are *questions* on the character of an argument (see Sect. 1.6.4).
- The operator  $\hat{S}_+$ , defined with `splus[S]`, contains only one off-diagonal band of non-zero values. The `SparseArray` matrix constructor allows building such banded matrices by simply specifying the starting point of the band and a vector with the elements of the nonzero band.
- The operator  $\hat{S}_z$ , defined with `sz[S]`, shows you the ordering of the basis elements since it has the projection quantum numbers on the diagonal.
- The last operator `id[S]` is the unit operator operating on a spin of length `S`, and will be used below for tensor-product definitions. Note that the `IdentityMatrix` function usually returns a full matrix, which is not suitable for large-scale calculations. By giving it a `SparseArray` option, it returns a sparse identity matrix of desired size.
- All these matrices can be displayed with, for example,

```

1 In[266]:=sx[3/2] //Normal
2 Out[266]={{0, Sqrt[3]/2, 0, 0},
3           {Sqrt[3]/2, 0, 1, 0},
4           {0, 1, 0, Sqrt[3]/2},
5           {0, 0, Sqrt[3]/2, 0}}
```

or, for a more traditional view,

```

1 In[267]:=sx[3/2] //MatrixForm
```

### 3.1.1 Exercises

- Q3.1** Verify that for  $S = 1/2$  the above Mathematica definitions give the Pauli matrices:  $\hat{S}_i = \frac{1}{2}\hat{\sigma}_i$  for  $i = x, y, z$ .
- Q3.2** Verify in Mathematica that for given integer or half-integer  $S$ , the three operators (matrices)  $\hat{\mathbf{S}} = \{\hat{S}_x, \hat{S}_y, \hat{S}_z\}$  behave like a quantum-mechanical pseudovector of length  $\|\hat{\mathbf{S}}\| = \sqrt{S(S+1)}$ :
1. Show that  $[\hat{S}_x, \hat{S}_y] = i\hat{S}_z$ ,  $[\hat{S}_y, \hat{S}_z] = i\hat{S}_x$ , and  $[\hat{S}_z, \hat{S}_x] = i\hat{S}_y$ .
  2. Show that  $\hat{S}_x^2 + \hat{S}_y^2 + \hat{S}_z^2 = S(S+1)\mathbb{1}$ .
  3. What is the largest value of  $S$  for which you can do these verifications within one minute (each) on your computer? *Hint:* use the `Timing` function.
- Q3.3** The operators  $\hat{S}_{x,y,z}$  are the generators of rotations: a rotation by an angle  $\alpha$  around the axis given by a normalized vector  $\vec{n}$  is done with the operator

$\hat{R}_{\vec{n}}(\alpha) = \exp(-i\alpha\vec{n} \cdot \hat{\vec{S}})$ . Set  $\vec{n} = \{\sin(\vartheta) \cos(\varphi), \sin(\vartheta) \sin(\varphi), \cos(\vartheta)\}$  and calculate the operator  $\hat{R}_{\vec{n}}(\alpha)$  explicitly for  $S = 0$ ,  $S = 1/2$ , and  $S = 1$ . Check that for  $\alpha = 0$  you find the unit operator.

## 3.2 Spin-1/2 Electron in a DC Magnetic Field

### [Supplementary Material 2]

As a first example we look at a single spin  $S = 1/2$ . We use the basis containing the two states  $|\uparrow\rangle = |\frac{1}{2}, \frac{1}{2}\rangle$  and  $|\downarrow\rangle = |\frac{1}{2}, -\frac{1}{2}\rangle$ , which we know to be eigenstates of the operators  $\hat{S}^2$  and  $\hat{S}_z$ . The matrix expressions of the operators relevant for this system are given by the Pauli matrices divided by two,

$$S_x = \frac{1}{2} \begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix} = \frac{1}{2} \sigma_x \quad S_y = \frac{1}{2} \begin{pmatrix} 0 & -i \\ i & 0 \end{pmatrix} = \frac{1}{2} \sigma_y \quad S_z = \frac{1}{2} \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix} = \frac{1}{2} \sigma_z \quad (3.3)$$

In Mathematica we enter these as

```
1 In[268]:=Sx = sx[1/2]; Sy = sy[1/2]; Sz = sz[1/2];
```

using the general definitions of angular momentum operators given in Sect. 3.1. Alternatively, we can write

```
1 In[269]:= {Sx, Sy, Sz} = (1/2) * Table[PauliMatrix[i], {i, 1, 3}];
```

As a Hamiltonian we use the coupling of this electron spin to an external magnetic field,  $\hat{\mathcal{H}} = -\hat{\vec{\mu}} \cdot \vec{B}$ . The magnetic moment of the electron is  $\hat{\vec{\mu}} = \mu_B g_e \hat{\vec{S}}$  in terms of its spin  $\hat{\vec{S}}$ , the Bohr magneton  $\mu_B = 9.274\,009\,68(20) \times 10^{-24} \text{ J/T}$ , and the electron's  $g$ -factor  $g_e = -2.002\,319\,304\,362\,2(15)$ .<sup>2</sup> The Hamiltonian is therefore

$$\hat{\mathcal{H}} = -\mu_B g_e (\hat{S}_x B_x + \hat{S}_y B_y + \hat{S}_z B_z). \quad (3.4)$$

In our chosen matrix representation this Hamiltonian is

$$\mathbf{H} = -\mu_B g_e (S_x B_x + S_y B_y + S_z B_z) = -\frac{1}{2} \mu_B g_e \begin{pmatrix} B_z & B_x - i B_y \\ B_x + i B_y & -B_z \end{pmatrix}. \quad (3.5)$$

In order to implement this Hamiltonian, we first define a system of units. Here we express magnetic field strengths in Gauss and energies in MHz times Planck's

<sup>2</sup>Notice that the magnetic moment of the electron is anti-parallel to its spin ( $g_e < 0$ ). The reason for this is the electron's negative electric charge. When the electron spin is parallel to the magnetic field, the electron's energy is higher than when they are anti-parallel.



constant (it is common to express energies in units of frequency, where the conversion is sometimes implicitly done via Planck's constant):

```
1 In[270]:=MagneticFieldUnit = Quantity["Gausses"];
2 In[271]:=EnergyUnit = Quantity["PlanckConstant"]*Quantity["MHz"] //UnitConvert;
```

In this unit system, the Bohr magneton is approximately 1.4 MHz/G:

```
1 In[272]:=μB = Quantity["BohrMagneton"]/(EnergyUnit/MagneticFieldUnit) //UnitConvert
2 Out[272]=1.3996245
```

We define the electron's  $g$ -factor with

```
1 In[273]:=ge = UnitConvert["ElectronGFactor"]
2 Out[273]=-2.00231930436
```

The Hamiltonian of Eq. (3.4) is then

```
1 In[274]:=H[Bx_, By_, Bz_] = -μB * ge * (Sx*Bx+Sy*By+Sz*Bz)
```

### Natural Units

An alternative choice of units, called *natural units*, is designed to simplify a calculation by making the numerical value of the largest possible number of quantities equal to 1. In the present case, this would be achieved by relating the field and energy units to each other in such a way that the Bohr magneton becomes equal to 1:

```
1 In[275]:=MagneticFieldUnit = Quantity["Gausses"];
2 In[276]:=EnergyUnit = MagneticFieldUnit * Quantity["BohrMagneton"] //UnitConvert;
3 In[277]:=μB = Quantity["BohrMagneton"]/(EnergyUnit/MagneticFieldUnit) //UnitConvert
4 Out[277]=1.0000000
```

In this way, calculations can often be simplified substantially because the Hamiltonian effectively becomes much simpler than it looks in other unit systems. We will be coming back to this point in future calculations.

#### 3.2.1 Time-Independent Schrödinger Equation

The time-independent Schrödinger equation for our spin-1/2 problem is, from Eq. (2.14),

$$-\frac{1}{2}\mu_B g_e \begin{pmatrix} B_z & B_x - iB_y \\ B_x + iB_y & -B_z \end{pmatrix} \cdot \vec{\psi} = E \vec{\psi} \quad (3.6)$$

The eigenvalues of the Hamiltonian (in our chosen energy units) and eigenvectors are calculated with:

```
1 In[278]:=Eigensystem[H[Bx,By,Bz]]
```

As described in Sect. 1.10.4 the output is a list with two entries, the first being a list of eigenvalues and the second a list of associated eigenvectors. As long as the Hamiltonian matrix is Hermitian, the eigenvalues will all be real-valued; but the eigenvectors can be complex. Since the Hilbert space of this spin problem has dimension 2, and the basis contains two vectors, there are necessarily two eigenvalues and two associated eigenvectors of length 2. The eigenvalues can be called  $E_{\pm} = \pm \frac{1}{2} \mu_B g_e \|\vec{B}\|$ . The list of eigenvalues is given in the Mathematica output as  $\{E'_-, E'_+\}$ . Notice that these eigenvalues only depend on the magnitude of the magnetic field, and not on its direction. This is to be expected: since there is no preferred axis in this system, there cannot be any directional dependence. The choice of the basis as the eigenstates of the  $\hat{S}_z$  operator was entirely arbitrary, and therefore the energy eigenvalues cannot depend on the orientation of the magnetic field with respect to this quantization axis.

The associated eigenvectors are

$$\vec{\psi}_{\pm} = \left\{ \frac{B_z \pm \|\vec{B}\|}{B_x + iB_y}, 1 \right\}, \quad (3.7)$$

which Mathematica returns as a list of lists,  $\{\vec{\psi}_-, \vec{\psi}_+\}$ . Notice that these eigenvectors are not normalized.

### 3.2.2 Exercises

**Q3.4** Calculate the eigenvalues (in units of J) and eigenvectors (ortho-normalized) of an electron spin in a magnetic field of 1 T in the  $x$ -direction.

**Q3.5** Set  $\vec{B} = B[\vec{e}_x \sin(\vartheta) \cos(\varphi) + \vec{e}_y \sin(\vartheta) \sin(\varphi) + \vec{e}_z \cos(\vartheta)]$  and calculate the eigenvalues and normalized eigenvectors of the electron spin Hamiltonian.

## 3.3 Coupled Spin Systems: $^{87}\text{Rb}$ Hyperfine Structure [Supplementary Material 3]

Ground-state Rubidium-87 atoms consist of a nucleus with spin  $I = 3/2$ , a single valence electron (spin  $S = 1/2$ , orbital angular momentum  $L = 0$ , and therefore total spin  $J = 1/2$ ), and 36 core electrons that do not contribute any angular momentum. In a magnetic field along the  $z$ -axis, the effective Hamiltonian of this system is<sup>3</sup>

$$\hat{\mathcal{H}} = \hat{\mathcal{H}}_0 + h A_{\text{hfs}} \hat{\mathbf{I}} \cdot \hat{\mathbf{J}} - \mu_B B_z (g_I \hat{I}_z + g_S \hat{S}_z + g_L \hat{L}_z), \quad (3.8)$$

<sup>3</sup>See <http://steck.us/alkalidata/rubidium87numbers.pdf>.

where  $h$  is Planck's constant,  $\mu_B$  is the Bohr magneton,  $A_{\text{hfs}} = 3.417\,341\,305\,452\,145(45)$  GHz is the spin–spin coupling constant in the ground state of  $^{87}\text{Rb}$ ,  $g_I = +0.000\,995\,141\,4(10)$  is the nuclear  $g$ -factor,  $g_S = -2.002\,319\,304\,362\,2(15)$  is the electron spin  $g$ -factor, and  $g_L = -0.999\,993\,69$  is the electron orbital  $g$ -factor.

The first part  $\hat{\mathcal{H}}_0$  of Eq. (3.8) contains all electrostatic interactions, core electrons, nuclear interactions etc. We will assume that the system is in the ground state of  $\hat{\mathcal{H}}_0$ , which means that the valence electron is in the  $5^2\text{S}_{1/2}$  state and the nucleus is deexcited. This ground state is eight-fold degenerate and consists of the four magnetic sublevels of the  $I = 3/2$  nuclear spin, the two sublevels of the  $S = 1/2$  electronic spin, and the single level of the  $L = 0$  angular momentum. The basis for the description of this atom is therefore the tensor product basis of a spin-3/2, a spin-1/2, and a spin-0.<sup>4</sup>

The spin operators acting on this composite system are defined as in Sect. 2.4.2. For example, the nuclear-spin operator  $\hat{I}_x$  is extended to the composite system by acting trivially on the electron spin and orbital angular momenta,  $\hat{I}_x \mapsto \hat{I}_x \otimes \mathbb{1} \otimes \mathbb{1}$ . The electron-spin operators are defined accordingly, for example  $\hat{S}_x \mapsto \mathbb{1} \otimes \hat{S}_x \otimes \mathbb{1}$ . The electron orbital angular momentum operators are, for example,  $\hat{L}_x \mapsto \mathbb{1} \otimes \mathbb{1} \otimes \hat{L}_x$ . In Mathematica these operators are defined with

```

1 In[279]:=Ix = KroneckerProduct[sx[3/2], id[1/2], id[0]];
2 In[280]:=Iy = KroneckerProduct[sy[3/2], id[1/2], id[0]];
3 In[281]:=Iz = KroneckerProduct[sz[3/2], id[1/2], id[0]];
4 In[282]:=Sx = KroneckerProduct[id[3/2], sx[1/2], id[0]];
5 In[283]:=Sy = KroneckerProduct[id[3/2], sy[1/2], id[0]];
6 In[284]:=Sz = KroneckerProduct[id[3/2], sz[1/2], id[0]];
7 In[285]:=Lx = KroneckerProduct[id[3/2], id[1/2], sx[0]];
8 In[286]:=Ly = KroneckerProduct[id[3/2], id[1/2], sy[0]];
9 In[287]:=Lz = KroneckerProduct[id[3/2], id[1/2], sz[0]];

```

The total electron angular momentum is  $\hat{\mathbf{J}} = \hat{\mathbf{S}} + \hat{\mathbf{L}}$ :

```

1 In[288]:=Jx = Sx + Lx; Jy = Sy + Ly; Jz = Sz + Lz;

```

The total angular momentum of the  $^{87}\text{Rb}$  atom is  $\hat{\mathbf{F}} = \hat{\mathbf{I}} + \hat{\mathbf{J}}$ :

```

1 In[289]:=Fx = Ix + Jx; Fy = Iy + Jy; Fz = Iz + Jz;

```

Before defining the system's Hamiltonian, we declare a system of units. Any system will work here, so we stay with units commonly used in atomic physics: magnetic fields are expressed in Gauss, while energies are expressed in MHz times Planck's constant. As time unit we choose the microsecond:

```

1 In[290]:=MagneticFieldUnit = Quantity["Gausses"];
2 In[291]:=EnergyUnit = Quantity["PlanckConstant"] * Quantity["Megahertz"];
3 In[292]:=TimeUnit = Quantity["Microseconds"];

```

<sup>4</sup>The spin-0 subsystem is trivial and could be left out in principle. It is included here to show the method in a more general way.

The numerical values of the Bohr Magneton and the reduced Planck constant in these units are

```

1 In[293]:=μBn = Quantity["BohrMagneton"]/(EnergyUnit/MagneticFieldUnit)
2 Out[293]=1.3996245
3 In[294]:=ħn = Quantity["ReducedPlanckConstant"]/(EnergyUnit*TimeUnit)
4 Out[294]=0.15915494

```

Using these definitions we define the hyperfine Hamiltonian with magnetic field in the  $z$ -direction as

```

1 In[295]:=Hhf = A(Ix.Jx+Iy.Jy+Iz.Jz) - μB*Bz*(gI*Iz+gS*Sz+gL*Lz);
2 In[296]:=hfc = {μB -> μBn, ħ -> ħn,
3   A->Quantity["PlanckConstant"]*Quantity[3.417341305452145,"GHz"]/EnergyUnit,
4   gS -> -2.0023193043622,
5   gL -> -0.99999369,
6   gI -> +0.0009951414};

```

This yields the Hamiltonian as an  $8 \times 8$  matrix, and we can calculate its eigenvalues and eigenvectors with

```

1 In[297]:= {eval, evec} = Eigensystem[Hhf] //FullSimplify;


```

We plot the energy eigenvalues with

```

1 In[298]:= Plot[Evaluate[eval /. hfc], {Bz, 0, 3000},
2   Frame -> True, FrameLabel -> {"Bz / G", "E / MHz"}]

```



### 3.3.1 Eigenstate Analysis

In this section we analyze the results `eval` and `evec` from the Hamiltonian diagonalization above. For this we first need to define *ortho-normalized* eigenvectors since in general we cannot assume `evec` to be ortho-normalized.

In general we can always define an ortho-normalized eigenvector set with

```

1 In[299]:= nevec = Orthogonalize[evec]

```

The problem with this definition is, however, immediately apparent if you look at the output given by Mathematica: since no assumptions on the reality of the variables

were made, the orthogonalization is done in too much generality and quickly becomes unwieldy. Even using `Assuming` and `ComplexExpand`, as in Sect. 1.11, does not give satisfactory results. But if we notice that the eigenvectors in `nevec` are all purely real-valued and are already orthogonal, then a simple vector-by-vector normalization is sufficient for calculating an ortho-normalized eigenvector set:

```
1 In[300]:=nevec = #/Sqrt[#, #] & /@ nevec;
2 In[301]:=nevec . Transpose[nevec] //FullSimplify
```

The fact that `In[301]` finds a unit matrix implies that the vectors in `nevec` are ortho-normal.

### Field-Free Limit

In the field-free limit  $B_z = 0$  the energy levels are

```
1 In[302]:=Assuming[A > 0, Limit[eval, Bz -> 0]]
2 Out[302]={3A/4, 3A/4, -5A/4, 3A/4, -5A/4, 3A/4, -5A/4, 3A/4}
```

We see that the level with energy  $-\frac{5}{4}A$  is three-fold degenerate while the level with energy  $\frac{3}{4}A$  is five-fold degenerate. This is also visible in the eigenvalue plot above. Considering that we have coupled two spins of lengths  $I = \frac{3}{2}$  and  $J = \frac{1}{2}$ , we expect the composite system to have either total spin  $F = 1$  (three sublevels) or  $F = 2$  (five sublevels); we can make the tentative assignment that the  $F = 1$  level is at energy  $E_1 = -\frac{5}{4}A$  and the  $F = 2$  level at  $E_2 = \frac{3}{4}A$ .

In order to demonstrate this assignment we express the matrix elements of the operators  $\hat{F}^2$  and  $\hat{F}_z$  in the field-free eigenstates, making sure to normalize these eigenstates before taking the limit  $B_z \rightarrow 0$ :

```
1 In[303]:=nevec0 = Assuming[A > 0, Limit[nevec, Bz -> 0]];
2 In[304]:=nevec0 . (Fx.Fx+Fy.Fy+Fz.Fz) . Transpose[nevec0]
3 In[305]:=nevec0 . Fz . Transpose[nevec0]
```

Notice that in this calculations we have used the fact that all eigenvectors are real, which may not always be the case for other Hamiltonians. We see that the field-free normalized eigenvectors `nevec0` are eigenvectors of both  $\hat{F}^2$  and  $\hat{F}_z$ , and from looking at the eigenvalues we can identify them as

$$\{|2, 2\rangle, |2, -2\rangle, |1, 0\rangle, |2, 0\rangle, |1, 1\rangle, |2, 1\rangle, |1, -1\rangle, |2, -1\rangle\} \quad (3.9)$$

in the notation  $|F, M_F\rangle$ . These labels are often used to identify the energy eigenstates even for small  $B_z \neq 0$ .

### Low-Field Limit

For small magnetic fields, we series-expand the energy eigenvalues to first order in  $B_z$ :

```
1 In[306]:=Assuming[A > 0, Series[eval, {Bz, 0, 1}] //FullSimplify]
```

From these low-field terms, in combination with the field-free level assignment, we see that the  $F = 1$  and  $F = 2$  levels have effective  $g$ -factors of  $g_1 = (-g_S + 5g_I)/4 \approx 0.501\,824$  and  $g_2 = -(-g_S - 3g_I)/4 \approx -0.499\,833$ , respectively, so that their energy eigenvalues follow the form

$$E_{F,M_F}(B_z) = E_F(0) - \mu_B M_F g_F B_z + \mathcal{O}(B_z^2). \quad (3.10)$$

These energy shifts due to the magnetic field are called *Zeeman shifts*.

### High-Field Limit

The energy eigenvalues in the high-field limit are infinite; but we can calculate their lowest-order series expansions with

```
1 In[307]:=Assuming[μB > 0 && gS < -gI < 0,
2           Series[eval, {Bz, Infinity, 0}] //FullSimplify]
```

From these expansions we can already identify the states in the eigenvalue plot above.

In order to calculate the eigenstates in the high-field limit we must again make sure to normalize the states *before* taking the limit  $B_z \rightarrow \infty$ <sup>5</sup>:

```
1 In[308]:=nevecinf = Assuming[μB > 0 && gS < -gI < 0,
2           FullSimplify[Limit[nevec, Bz -> Infinity], A > 0]]
3 Out[308]={{1, 0, 0, 0, 0, 0, 0, 0},
4           {0, 0, 0, 0, 0, 0, 0, 1},
5           {0, 0, 0, -1, 0, 0, 0, 0},
6           {0, 0, 0, 0, 1, 0, 0, 0},
7           {0, -1, 0, 0, 0, 0, 0, 0},
8           {0, 0, 1, 0, 0, 0, 0, 0},
9           {0, 0, 0, 0, 0, -1, 0, 0},
10          {0, 0, 0, 0, 0, 0, 1, 0}}
```

From this we immediately identify the high-field eigenstates as our basis states in a different order,

$$\{| \frac{3}{2}, \frac{1}{2} \rangle, | -\frac{3}{2}, -\frac{1}{2} \rangle, | \frac{1}{2}, -\frac{1}{2} \rangle, | -\frac{1}{2}, \frac{1}{2} \rangle, | \frac{3}{2}, -\frac{1}{2} \rangle, | \frac{1}{2}, \frac{1}{2} \rangle, | -\frac{1}{2}, -\frac{1}{2} \rangle, | -\frac{3}{2}, \frac{1}{2} \rangle\} \quad (3.11)$$

where we have used the abbreviation  $|M_I, M_J\rangle = | \frac{3}{2}, M_I \rangle \otimes | \frac{1}{2}, M_J \rangle$ . You can verify this assignment by looking at the matrix elements of the  $\hat{I}_z$  and  $\hat{J}_z$  operators with


```
1 In[309]:=nevecinf . Iz . Transpose[nevecinf]
2 In[310]:=nevecinf . Jz . Transpose[nevecinf]
```

<sup>5</sup>Note that in [In\[308\]](#) we use two stages of assumptions, using the assumption  $A > 0$  only in [FullSimplify](#) but not in [Limit](#). This is done in order to work around an inconsistency in Mathematica 11.3.0.0, and may be simplified in a future edition.

### 3.3.2 “Magic” Magnetic Field

The energy eigenvalues of the low-field states  $|1, -1\rangle$  and  $|2, 1\rangle$  have almost the same first-order magnetic field dependence since  $g_1 \approx -g_2$  (see low-field limit above). If we plot their energy difference as a function of magnetic field we find an extremal point:

```
In[311]:=Plot[eval[[6]]-eval[[7]]-2A /. hfc, {Bz, 0, 6}]
```



At the “magic” field strength  $B_0 = 3.22896 \text{ G}$  the energy difference is independent of the magnetic field (to first order):

```
In[312]:=NMinimize[eval[[6]] - eval[[7]] - 2 A /. hfc, Bz]
Out[312]={{-0.00449737, {Bz -> 3.22896}}}
```

This is an important discovery for quantum information science with  $^{87}\text{Rb}$  atoms. If we store a qubit in the state  $|\vartheta, \varphi\rangle = \cos(\vartheta/2)|1, -1\rangle + e^{i\varphi} \sin(\vartheta/2)|2, 1\rangle$  and tune the magnetic field exactly to the magic value, then the experimentally unavoidable magnetic-field fluctuations will not lead to fluctuations of the energy difference between the two atomic levels and thus will not lead to qubit decoherence. Very long qubit coherence times can be achieved in this way.

For the present case where  $|g_I| \ll |g_S|$ , the magic field is approximately  $B_z \approx \frac{16A_{qI}}{3\mu_B g_S^2}$ .

### 3.3.3 Coupling to an Oscillating Magnetic Field

In this section we study the coupling of a  $^{87}\text{Rb}$  atom to a weak oscillating magnetic field. Such a field could be the magnetic part of an electromagnetic wave, whose electric field does not couple to our atom in the electronic ground state. This calculation is a template for more general situations where a quantum-mechanical system is driven by an oscillating field.

The  $^{87}\text{Rb}$  hyperfine Hamiltonian in the presence of an oscillating magnetic field is

$$\hat{\mathcal{H}}(t) = \underbrace{hA_{\text{hfs}} \hat{\mathbf{I}} \cdot \hat{\mathbf{J}} - \mu_B B_z (g_I \hat{I}_z + g_S \hat{S}_z + g_L \hat{L}_z)}_{\hat{\mathcal{H}}_0} - \cos(\omega t) \times \underbrace{\mu_B \vec{\mathbf{B}}^{\text{ac}} \cdot (g_I \hat{\mathbf{I}} + g_S \hat{\mathbf{S}} + g_L \hat{\mathbf{L}})}_{-\hat{\mathcal{H}}_1} \quad (3.12)$$

where the static magnetic field is assumed to be in the  $z$  direction, as before. Unfortunately,  $[\hat{\mathcal{H}}(t), \hat{\mathcal{H}}(t')] = [\hat{\mathcal{H}}_1, \hat{\mathcal{H}}_0] (\cos(\omega t) - \cos(\omega t')) \neq 0$  in general, so we cannot use the exact solution of Eq. (2.33) of the time-dependent Schrödinger equation. In fact, the time-dependent Schrödinger equation of this system has no analytic solution at all. In what follows we will calculate approximate solutions.

Since we have diagonalized the time-independent Hamiltonian  $\hat{\mathcal{H}}_0$  already, we use its eigenstates as a basis for calculating the effect of the oscillating perturbation  $\hat{\mathcal{H}}_1(t)$ . In general, calling  $\{|i\rangle\}_{i=1}^8$  the set of eigenstates of  $\hat{\mathcal{H}}_0$ , with  $\hat{\mathcal{H}}_0|i\rangle = E_i|i\rangle$  for  $i \in \{1 \dots 8\}$ , we expand the general hyperfine state as in Eq. (2.28),

$$|\psi(t)\rangle = \sum_{i=1}^8 \psi_i(t) e^{-iE_i t/\hbar} |i\rangle. \quad (3.13)$$

The time-dependent Schrödinger equation for the expansion coefficients  $\psi_i(t)$  in this interaction picture is given in Eq. (2.30): for  $i = 1 \dots 8$  we have

$$i\hbar \dot{\psi}_i(t) = \sum_{j=1}^8 \psi_j(t) e^{-i(E_j - E_i)t/\hbar} \cos(\omega t) \langle i|\hat{\mathcal{H}}_1|j\rangle = \frac{1}{2} \sum_{j=1}^8 \psi_j(t) \left[ e^{-i\left(\frac{E_j - E_i}{\hbar} - \omega\right)t} + e^{i\left(\frac{E_j - E_i}{\hbar} - \omega\right)t} \right] T_{ij}, \quad (3.14)$$

where we have replaced  $\cos(\omega t) = \frac{1}{2}e^{i\omega t} + \frac{1}{2}e^{-i\omega t}$  and defined

$$T_{ij} = \langle i|\hat{\mathcal{H}}_1|j\rangle = -\langle i|\left[\mu_B \vec{\mathbf{B}}^{\text{ac}} \cdot (g_I \hat{\mathbf{I}} + g_S \hat{\mathbf{S}} + g_L \hat{\mathbf{L}})\right]|j\rangle. \quad (3.15)$$

From Eq. (3.14) we can proceed in various ways:

**Transition matrix elements:** The time-independent matrix elements  $T_{ij}$  of the perturbation Hamiltonian are called the *transition matrix elements* and describe how the populations of the different eigenstates of  $\hat{\mathcal{H}}_0$  are coupled through the oscillating field. We calculate them in Mathematica as follows:

```

1 In[313]:=H0 = A*(Ix.Jx + Iy.Jy + Iz.Jz) - μB*Bz*(gS*Sz + gL*Lz + gI*Iz);
2 In[314]:=H1 = -μB*(gS*(Bacx*Sx + Bacy*Sy + Bacz*Sz)
3           + gI*(Bacx*Ix + Bacy*Iy + Bacz*Iz)
4           + gL*(Bacx*Lx + Bacy*Ly + Bacz*Lz));
5 In[315]:=H[t_] = H0 + H1*Cos[ω*t];
6 In[316]:= {eval, evec} = Eigensystem[H0] //FullSimplify;
7 In[317]:= nevec = Map[#,Sqrt[#. #] &, evec];
8 In[318]:= T = Assuming[A > 0, nevec.H1.Transpose[nevec] //FullSimplify];

```



Looking at this matrix  $T$  we see that not all energy levels are directly coupled by an oscillating magnetic field. For example,  $T_{1,2} = 0$  indicates that the populations of the states  $|1\rangle$  and  $|2\rangle$  can only be coupled indirectly through other states, but not directly (hint: check  $T[[1,2]]$ ).

**Numerical solution:** Equation (3.14) is a series of linear coupled differential equations, which we write down explicitly in Mathematica with

```
1 In[319]:=deqs = Table[I*\h*Subscript[\psi,i]'\[t] ==
2             Sum[Subscript[\psi,j]\[t]*Exp[-I*(eval[[j]]-eval[[i]])*t/\h]
3             *Cos[\omega*t]*T[[i,j]], {j, 8}], {i,8}];
```


Assuming concrete conditions, for example the initial state  $|\psi(t=0)\rangle = |F=2, M_F=-2\rangle$  which is the second eigenstate `nevec[[2]]` [see Eq. (3.9)], and magnetic fields  $B_z = 3.22896\text{ G}$ ,  $B_x^{\text{ac}} = 100\text{ mG}$ ,  $B_y^{\text{ac}} = B_z^{\text{ac}} = 0$ , and an ac field angular frequency of  $\omega = 2\pi \times 6827.9\text{ MHz}$ , we can find the time-dependent state  $|\psi(t)\rangle$  with

```
1 In[320]:=S = NDSolve[Join[deqs /. hfc /. {Bz->3.22896, Bax->0.1, Bacy->0, Bacz->0,
2             \omega->2*\pi*6827.9},
3             {Subscript[\psi,1][0]==0, Subscript[\psi,2][0]==1,
4             Subscript[\psi,3][0]==0, Subscript[\psi,4][0]==0,
5             Subscript[\psi,5][0]==0, Subscript[\psi,6][0]==0,
6             Subscript[\psi,7][0]==0, Subscript[\psi,8][0]==0}],
7             Table[Subscript[\psi,i]\[t],{i,8}], {t, 0, 30},
8             MaxStepSize->10^(-5), MaxSteps->10^7]
```


Notice that the maximum step size in this numerical solution is very small ( $10^{-5}$  time units or 10ps), since it needs to capture the fast oscillations of more than 6.8 GHz. As a result, a large number of numerical steps is required, which makes this way of studying the evolution very difficult in practice.

We plot the resulting populations with

```
1 In[321]:=Plot[Evaluate[Abs[Subscript[\psi,2]\[t] /. S[[1]]]^2], {t, 0, 30}]
```



```
1 In[322]:=Plot[Evaluate[Abs[Subscript[\psi,7]\[t] /. S[[1]]]^2], {t, 0, 30}]
```



We see that the population is mostly sloshing between  $\hat{\mathcal{H}}_0$ -eigenstates  $|2\rangle \approx |F=2, M_F=-2\rangle$  and  $|7\rangle \approx |F=1, M_F=-1\rangle$  [see Eq. (3.9)]. Each population oscillation takes about  $8.2\,\mu\text{s}$  (the Rabi period), and we say that the Rabi frequency is about  $120\,\text{kHz}$ .

**Rotating-wave approximation:** The time-dependent prefactor  $\exp\left[-i\left(\frac{E_j-E_i}{\hbar}-\omega\right)t\right] + \exp\left[i\left(\frac{E_i-E_j}{\hbar}-\omega\right)t\right]$  of Eq. (3.14) oscillates very rapidly unless either  $\frac{E_j-E_i}{\hbar} - \omega \approx 0$  or  $\frac{E_i-E_j}{\hbar} - \omega \approx 0$ , where one of its terms changes slowly in time. The *rotating-wave approximation* (RWA) consists of neglecting all rapidly rotating terms in Eq. (3.14). Assume that there is a single<sup>6</sup> pair of states  $|i\rangle$  and  $|j\rangle$  such that  $E_i - E_j \approx \hbar\omega$ , with  $E_i > E_j$ , while all other states have an energy difference far from  $\hbar\omega$ . The RWA thus consists of simplifying Eq. (3.14) to

$$\begin{aligned} i\hbar\dot{\psi}_i(t) &\approx \frac{1}{2}\psi_j(t)e^{i\left(\frac{E_i-E_j}{\hbar}-\omega\right)t}T_{ij} = \frac{1}{2}\psi_j(t)T_{ij}e^{-i\Delta t} \\ i\hbar\dot{\psi}_j(t) &\approx \frac{1}{2}\psi_i(t)e^{-i\left(\frac{E_i-E_j}{\hbar}-\omega\right)t}T_{ji} = \frac{1}{2}\psi_i(t)T_{ji}e^{i\Delta t} \\ i\hbar\dot{\psi}_k(t) &\approx 0 \text{ for } k \notin \{i, j\} \end{aligned} \quad (3.16)$$

with  $T_{ji} = T_{ij}^*$  and the detuning  $\Delta = \omega - (E_i - E_j)/\hbar$ . All other terms in Eq. (3.14) have been neglected because they rotate so fast in time that they “average out” to zero. This approximate system of differential equations has the exact solution

$$\begin{aligned} \psi_i(t) &= e^{-\frac{i}{2}\Delta t} \left[ \psi_i(0) \cos\left(\frac{\Omega t}{2}\right) + i \left( \frac{\Delta}{\Omega} \psi_i(0) - \frac{T_{ij}}{\hbar\Omega} \psi_j(0) \right) \sin\left(\frac{\Omega t}{2}\right) \right] \\ \psi_j(t) &= e^{\frac{i}{2}\Delta t} \left[ \psi_j(0) \cos\left(\frac{\Omega t}{2}\right) - i \left( \frac{\Delta}{\Omega} \psi_j(0) + \frac{T_{ij}^*}{\hbar\Omega} \psi_i(0) \right) \sin\left(\frac{\Omega t}{2}\right) \right] \\ \psi_k(t) &= \psi_k(0) \text{ for } k \notin \{i, j\} \end{aligned} \quad (3.17)$$

<sup>6</sup>The following derivation is readily extended to situations where several pairs of states have an energy difference approximately equal to  $\hbar\omega$ . In such a case we need to solve a larger system of coupled differential equations.

in terms of the generalized Rabi frequency  $\Omega = \sqrt{|T_{ij}|^2/\hbar^2 + \Delta^2}$ . We can see that the population sloshes back and forth (“Rabi oscillation”) between the two levels  $|i\rangle$  and  $|j\rangle$  with angular frequency  $\Omega$ , as we had seen numerically above. We can verify this solution in Mathematica as follows. First we define

```
1 In[323] := Δ = ω - (Ei-Ej)/ħ;
2 In[324] := Ω = Sqrt[Tij*Tji/ħ^2 + Δ^2];
```

and the solutions

```
1 In[325] := Ψi[t_] = E^(-I*Δ*t/2) * (Ψi0*Cos[Ω*t/2] + I*(Δ/Ω*Ψi0 - Tij/(ħ*Ω)*Ψj0)
2           * Sin[Ω*t/2]);
3 In[326] := Ψj[t_] = E^(I*Δ*t/2) * (Ψj0*Cos[Ω*t/2] - I*(Δ/Ω*Ψj0 + Tji/(ħ*Ω)*Ψi0)
4           * Sin[Ω*t/2]);
```

With these definitions, we can check the Schrödinger equations (3.16):

```
1 In[327] := FullSimplify[I*ħ*Ψi'[t] == (1/2) * Ψj[t] * Exp[-I*Δ*t]*Tij]
2 Out[327] = True
3 In[328] := FullSimplify[I*ħ*Ψj'[t] == (1/2) * Ψi[t] * Exp[I*Δ*t]*Tji]
4 Out[328] = True
```

as well as the initial conditions

```
1 In[329] := Ψi[0]
2 Out[329] = Ψi0
3 In[330] := Ψj[0]
4 Out[330] = Ψj0
```

**Dressed states:** If we insert the RWA solutions, Eq. (3.17), into the definition of the general hyperfine state, Eq. (3.13), and set all coefficients  $\psi_k = 0$  for  $k \notin \{i, j\}$ , and then write  $\sin(z) = (e^{iz} - e^{-iz})/(2i)$  and  $\cos(z) = (e^{iz} + e^{-iz})/2$ , we find the state

$$\begin{aligned}
 |\psi(t)\rangle &\approx \psi_i(t)e^{-iE_i t/\hbar}|i\rangle + \psi_j(t)e^{-iE_j t/\hbar}|j\rangle \\
 &= \frac{1}{2}e^{-i\left(E_i - \frac{\hbar(\Omega - \Delta)}{2}\right)t/\hbar} \left\{ \left[ \psi_i(0)\left(1 + \frac{\Delta}{\Omega}\right) - \psi_j(0)\frac{T_{ij}}{\hbar\Omega} \right]|i\rangle + \left[ \psi_j(0)\left(1 - \frac{\Delta}{\Omega}\right) - \psi_i(0)\frac{T_{ij}^*}{\hbar\Omega} \right]e^{i\omega t}|j\rangle \right\} \\
 &\quad + \frac{1}{2}e^{-i\left(E_i + \frac{\hbar(\Omega + \Delta)}{2}\right)t/\hbar} \left\{ \left[ \psi_i(0)\left(1 - \frac{\Delta}{\Omega}\right) + \psi_j(0)\frac{T_{ij}}{\hbar\Omega} \right]|i\rangle + \left[ \psi_j(0)\left(1 + \frac{\Delta}{\Omega}\right) + \psi_i(0)\frac{T_{ij}^*}{\hbar\Omega} \right]e^{i\omega t}|j\rangle \right\}.
 \end{aligned} \tag{3.18}$$

In order to interpret this state more clearly, we need to expand our view of the problem to include the quantized driving field. For this we assume that the driving mode of the field (for example, the used mode of the electromagnetic field) in state  $|n\rangle$  contains  $n$  quanta of vibration (for example, photons), and has an energy of  $E_n = n\hbar\omega$ . The two states  $|i\rangle$  and  $|j\rangle$  describing our system, with  $E_i - E_j \approx \hbar\omega$ , actually correspond to states in the larger system containing the driving field. In this sense, we can say that the state  $|i, n\rangle$ , with the system in state  $|i\rangle$  and the driving field containing  $n$  quanta, is approximately resonant with the state  $|j, n+1\rangle$ , with the system in state  $|j\rangle$  and the driving field containing  $n+1$  quanta. A transition from  $|i\rangle$  to  $|j\rangle$  is actually a transition from

$|i, n\rangle$  to  $|j, n+1\rangle$ , where one quantum is added simultaneously to the driving field in order to conserve energy (approximately). A transition from  $|j\rangle$  to  $|i\rangle$  corresponds to the system absorbing one quantum from the driving field. The energy of the quantized driving field contributes an additional time dependence

$$|i\rangle \mapsto |i, n\rangle e^{-in\omega t}, \quad |j\rangle \mapsto |j, n+1\rangle e^{-i(n+1)\omega t}, \quad (3.19)$$

and Eq. (3.18) thus becomes

$$\begin{aligned} |\psi(t)\rangle \approx & \frac{1}{2} e^{-i(E_i + n\hbar\omega + \frac{\hbar(\Delta - \Omega)}{2})t/\hbar} \left\{ \left[ \psi_i(0) \left( 1 + \frac{\Delta}{\Omega} \right) - \psi_j(0) \frac{T_{ij}}{\hbar\Omega} \right] |i, n\rangle + \left[ \psi_j(0) \left( 1 - \frac{\Delta}{\Omega} \right) - \psi_i(0) \frac{T_{ij}^*}{\hbar\Omega} \right] |j, n+1\rangle \right\} \\ & + \frac{1}{2} e^{-i(E_i + n\hbar\omega + \frac{\hbar(\Delta + \Omega)}{2})t/\hbar} \left\{ \left[ \psi_i(0) \left( 1 - \frac{\Delta}{\Omega} \right) + \psi_j(0) \frac{T_{ij}}{\hbar\Omega} \right] |i, n\rangle + \left[ \psi_j(0) \left( 1 + \frac{\Delta}{\Omega} \right) + \psi_i(0) \frac{T_{ij}^*}{\hbar\Omega} \right] |j, n+1\rangle \right\} \\ & = \frac{1}{2} e^{-iE_- t/\hbar} |-\rangle + \frac{1}{2} e^{-iE_+ t/\hbar} |+\rangle \end{aligned} \quad (3.20)$$

With this substitution, the state consists of two components, called *dressed states*,

$$|\pm\rangle = \left[ \psi_i(0) \left( 1 \mp \frac{\Delta}{\Omega} \right) \pm \psi_j(0) \frac{T_{ij}}{\hbar\Omega} \right] |i, n\rangle + \left[ \psi_j(0) \left( 1 \pm \frac{\Delta}{\Omega} \right) \pm \psi_i(0) \frac{T_{ij}^*}{\hbar\Omega} \right] |j, n+1\rangle. \quad (3.21)$$

that are time-invariant apart from their energy (phase) prefactors. These energy prefactors correspond to the effective energy of the dressed states in the presence of the oscillating field,<sup>7</sup>

$$E_{\pm} = E_i + n\hbar\omega + \frac{\hbar(\Delta \pm \Omega)}{2} = E_j + (n+1)\hbar\omega + \frac{\hbar(-\Delta \pm \Omega)}{2}. \quad (3.22)$$

We look at these dressed states in two limits:

- On resonance ( $\Delta = 0$ ), we have  $\hbar\Omega = |T_{ij}|$ , and the dressed states of Eq. (3.21) become

$$\begin{aligned} |\pm\rangle &= \left[ \psi_i(0) \pm \psi_j(0) \frac{T_{ij}}{|T_{ij}|} \right] |i, n\rangle + \left[ \psi_j(0) \pm \psi_i(0) \frac{T_{ij}^*}{|T_{ij}|} \right] |j, n+1\rangle \\ &= \left[ \psi_i(0) \pm \psi_j(0) \frac{T_{ij}}{|T_{ij}|} \right] \left( |i, n\rangle \pm \frac{T_{ij}^*}{|T_{ij}|} |j, n+1\rangle \right), \end{aligned} \quad (3.23)$$

which are equal mixtures of the original states  $|i, n\rangle$  and  $|j, n+1\rangle$ . They have energies

<sup>7</sup>The instantaneous energy of a state is defined as  $E = \langle \hat{H} \rangle = i\hbar \langle \frac{\partial}{\partial t} \rangle$ . For a state  $|\psi(t)\rangle = e^{-i\omega t} |\phi\rangle$  the energy is  $E = i\hbar \langle \psi(t) | \frac{\partial}{\partial t} | \psi(t) \rangle = i\hbar \langle \phi | e^{i\omega t} \frac{\partial}{\partial t} e^{-i\omega t} | \phi \rangle = \hbar\omega$ .

$$E_{\pm} = E_i + n\hbar\omega \pm \frac{1}{2}|T_{ij}| = E_j + (n+1)\hbar\omega \pm \frac{1}{2}|T_{ij}| \quad (3.24)$$

in the presence of a resonant ac coupling field: the degeneracy of the levels  $|i, n\rangle$  and  $|j, n+1\rangle$  is lifted, and the dressed states are split by  $E_+ - E_- = |T_{ij}|$ .

- Far off-resonance ( $\Delta \rightarrow \pm\infty$ ) we have  $\Omega \approx |\Delta| + \frac{|T_{ij}|^2}{2\hbar^2|\Delta|}$ , and Eq. (3.20) becomes

$$|\psi(t)\rangle \approx e^{-i\left(E_i + n\hbar\omega - \frac{|T_{ij}|^2}{4\hbar\Delta}\right)t/\hbar} \psi_i(0)|i, n\rangle + e^{-i\left(E_j + (n+1)\hbar\omega + \frac{|T_{ij}|^2}{4\hbar\Delta}\right)t/\hbar} \psi_j(0)|j, n+1\rangle. \quad (3.25)$$

(Hint: to verify this, look at the cases  $\Delta \rightarrow +\infty$  and  $\Delta \rightarrow -\infty$  separately).

The energy levels  $|i, n\rangle$  and  $|j, n+1\rangle$  are thus shifted by  $\mp \frac{|T_{ij}|^2}{4\hbar\Delta}$ , respectively, and there is no population transfer between the levels. That is, the dressed states become equal to the original states. Remember that we had assumed  $E_i > E_j$ :

- For a blue-detuned drive ( $\Delta \rightarrow +\infty$ ), the upper level  $|i\rangle$  is *lowered* in energy by  $\Delta E = \frac{|T_{ij}|^2}{4\hbar\Delta}$  while the lower level  $|j\rangle$  is *raised* in energy by  $\Delta E$ .
- For a red-detuned drive ( $\Delta \rightarrow -\infty$ ), the upper level  $|i\rangle$  is *raised* in energy by  $\Delta E = \frac{|T_{ij}|^2}{4\hbar|\Delta|}$  while the lower level  $|j\rangle$  is *lowered* in energy by  $\Delta E$ .

These shifts are called *ac Zeeman shifts* in this case, or *level shifts* more generally. When the oscillating field is a light field, level shifts are often called *light shifts* or *ac Stark shifts*.

### 3.3.4 Exercises

**Q3.6** Take two angular momenta, for example  $I = 3$  and  $J = 5$ , and calculate the eigenvalues of the operators  $\hat{I}^2$ ,  $\hat{I}_z$ ,  $\hat{J}^2$ ,  $\hat{J}_z$ ,  $\hat{F}^2$ , and  $\hat{F}_z$ , where  $\hat{\mathbf{F}} = \hat{\mathbf{I}} + \hat{\mathbf{J}}$ .

**Q3.7** In Q3.6 you have coupled two angular momenta but you have not used any Clebsch–Gordan coefficients. Why not? Where do these coefficients appear?

**Q3.8** For a spin of a certain length, for example  $S = 100$ , take the state  $|S, S\rangle$  (a spin pointing in the  $+z$  direction) and calculate the expectation values  $\langle \hat{S}_x \rangle$ ,  $\langle \hat{S}_y \rangle$ ,  $\langle \hat{S}_z \rangle$ ,  $\langle \hat{S}_x^2 \rangle - \langle \hat{S}_x \rangle^2$ ,  $\langle \hat{S}_y^2 \rangle - \langle \hat{S}_y \rangle^2$ ,  $\langle \hat{S}_z^2 \rangle - \langle \hat{S}_z \rangle^2$ . *Hint:* the expectation value of an operator  $\hat{A}$  is  $\langle S, S | \hat{A} | S, S \rangle$ .

**Q3.9** Use In[323] and In[324] to calculate the detuning  $\Delta$  and the generalized Rabi frequency  $\Omega$  for the  $^{87}\text{Rb}$  solution of In[320], where the population oscillates between the levels  $i = 2$  and  $j = 7$ . What is the oscillation period corresponding to  $\Omega$ ? Does it match the plots of In[321] and In[322]?

**Q3.10** Do the presented alkali atom calculation for  $^{23}\text{Na}$ : are there any magic field values?

<http://steck.us/alkalidata/sodiumnumbers.pdf>

**Q3.11** Do the presented alkali atom calculation for  $^{85}\text{Rb}$ : are there any magic field values?

<http://steck.us/alkalidata/rubidium85numbers.pdf>

**Q3.12** Do the presented alkali atom calculation for  $^{133}\text{Cs}$ : are there any magic field values?

<http://steck.us/alkalidata/cesiumnumbers.pdf>

**Q3.13** Set  $\vec{B} = 0$  and  $\vec{B}^{\text{ac}} = B(\vec{e}_x + i\vec{e}_y)$  in the expression for **T** in [In\[318\]](#). Which transitions are allowed for such circularly-polarized light around the quantization axis? *Hint*: use Eq. (3.9) to identify the states.

**Q3.14** Set  $\vec{B} = 0$  and  $\vec{B}^{\text{ac}} = B\vec{e}_z$  in the expression for **T** in [In\[318\]](#). Which transitions are allowed for such linearly-polarized light along the quantization axis? *Hint*: use Eq. (3.9) to identify the states.

### 3.4 Coupled Spin Systems: Ising Model in a Transverse Field [\[Supplementary Material 4\]](#)

We now turn to larger numbers of coupled quantum-mechanical spins. A large class of such coupled spin systems can be described with Hamiltonians of the form

$$\hat{\mathcal{H}} = \sum_{k=1}^N \hat{\mathcal{H}}^{(k)} + \sum_{k=1}^{N-1} \sum_{k'=k+1}^N \hat{\mathcal{H}}_{\text{int}}^{(k,k')}, \quad (3.26)$$

where the  $\hat{\mathcal{H}}^{(k)}$  are single-spin Hamiltonians (for example couplings to a magnetic field) and the  $\hat{\mathcal{H}}_{\text{int}}^{(k,k')}$  are coupling Hamiltonians between two spins. Direct couplings between three or more spins can usually be neglected.

As an example we study the dimensionless “transverse Ising” Hamiltonian

$$\hat{\mathcal{H}} = -\frac{b}{2} \sum_{k=1}^N \hat{S}_x^{(k)} - \sum_{k=1}^N \hat{S}_z^{(k)} \hat{S}_z^{(k+1)} \quad (3.27)$$

acting on a ring of  $N$  spin- $S$  systems where the  $(N+1)$ st spin is identified with the first spin. We can read off three limits from this Hamiltonian:

- For  $b \rightarrow \pm\infty$  the spin–spin coupling Hamiltonian can be neglected, and the ground state will have all spins aligned with the  $\pm x$  direction,

$$|\psi_{+\infty}\rangle = |+\rangle^{\otimes N}, \quad |\psi_{-\infty}\rangle = |-\rangle^{\otimes N}. \quad (3.28)$$

The system is therefore in a product state for  $b \rightarrow \pm\infty$ , which means that there is no entanglement between spins. In the basis of  $|S, M\rangle$  Dicke states, Eqs. (3.1) and (3.2), the single-spin states making up these product states are

$$|+x\rangle = 2^{-S} \sum_{M=-S}^S \sqrt{\binom{2S}{M+S}} |S, M\rangle, \quad (3.29a)$$

$$|-x\rangle = 2^{-S} \sum_{M=-S}^S (-1)^{M+S} \sqrt{\binom{2S}{M+S}} |S, M\rangle, \quad (3.29b)$$

which are aligned with the  $x$ -axis in the sense that  $\hat{S}_x|+x\rangle = S|+x\rangle$  and  $\hat{S}_x|-x\rangle = -S|-x\rangle$ .

- For  $b = 0$  the Hamiltonian contains only nearest-neighbor ferromagnetic spin-spin couplings  $-\hat{S}_z^{(k)}\hat{S}_z^{(k+1)}$ . We know that this Hamiltonian has two degenerate ground states: all spins pointing up or all spins pointing down,

$$|\psi_{0\uparrow}\rangle = |+z\rangle^{\otimes N}, \quad |\psi_{0\downarrow}\rangle = |-z\rangle^{\otimes N}, \quad (3.30)$$

where in the Dicke-state representation of Eq. (3.1) we have  $|+z\rangle = |S, +S\rangle$  and  $|-z\rangle = |S, -S\rangle$ . While these two states are product states, for  $|b| \ll 1$  the perturbing Hamiltonian  $-\frac{b}{2} \sum_{k=1}^N \hat{S}_x^{(k)}$  is diagonal in the states  $\frac{|\psi_{0\uparrow}\rangle \pm |\psi_{0\downarrow}\rangle}{\sqrt{2}}$ , which are not product states. The exact ground state for  $0 < b \ll 1$  is close to  $\frac{|\psi_{0\uparrow}\rangle + |\psi_{0\downarrow}\rangle}{\sqrt{2}}$ , and for  $-1 \ll b < 0$  it is close to  $\frac{|\psi_{0\uparrow}\rangle - |\psi_{0\downarrow}\rangle}{\sqrt{2}}$ . These are both maximally entangled states (“Schrödinger cat states”).

Now we calculate the ground state  $|\psi_b\rangle$  as a function of the parameter  $b$ , and compare the results to the above asymptotic limits.

### 3.4.1 Basis Set

The natural basis set for describing a set of  $N$  coupled spins is the tensor-product basis (see Sect. 2.4.2). In this basis, the spin operators  $\hat{S}_{x,y,z}^{(k)}$  acting only on spin  $k$  are defined as having a trivial action on all other spins, for example

$$\hat{S}_x^{(k)} \mapsto \underbrace{\mathbb{1} \otimes \mathbb{1} \otimes \cdots \otimes \mathbb{1}}_{(k-1)} \otimes \hat{S}_x \otimes \underbrace{\mathbb{1} \otimes \cdots \otimes \mathbb{1}}_{(N-k)}. \quad (3.31)$$

In Mathematica such single-spin- $S$  operators acting on spin  $k$  out of a set of  $N$  spins are defined as follows. First we define the operator acting as  $\hat{a} = \mathbf{a}$  on the  $k$ th spin out of a set of  $n$  spins, and trivially on all others:

```

1 In[331]:= op[S_?SpinQ, n_Integer, k_Integer, a_?MatrixQ] /;
2 1<=k<=n && Dimensions[a] == {2S+1, 2S+1} :=
3 KroneckerProduct[IdentityMatrix[(2S+1)^(k-1)], SparseArray],
4 a,
5 IdentityMatrix[(2S+1)^(n-k)], SparseArray]]

```

Next, we specialize this to  $\hat{a} = \hat{S}_x, \hat{S}_y, \hat{S}_z$ :

```

1 In[332]:=sx[S_?SpinQ, n_Integer, k_Integer] /; 1<=k<=n := op[S, n, k, sx[S]]
2 In[333]:=sy[S_?SpinQ, n_Integer, k_Integer] /; 1<=k<=n := op[S, n, k, sy[S]]
3 In[334]:=sz[S_?SpinQ, n_Integer, k_Integer] /; 1<=k<=n := op[S, n, k, sz[S]]

```

Notice that we have used  $n = N$  because the symbol  $N$  is already used internally in Mathematica.

From these we assemble the Hamiltonian:

```

1 In[335]:=H[S_?SpinQ, n_Integer;/n>=3, b_] := -b/2*Sum[sx[S, n, k], {k, n}] -
2          Sum[sz[S, n, k].sz[S, n, Mod[k+1,n,1]], {k, n}]

```

The modulus `Mod[k+1, n, 1]` represents the periodicity of the spin ring and ensures that the index remains within  $1 \dots N$  (i.e., a modulus with offset 1).

### 3.4.2 Asymptotic Ground States

The asymptotic ground states for  $b = 0$  and  $b \rightarrow \pm\infty$  mentioned above are all product states of the form  $|\psi\rangle = |\theta\rangle^{\otimes N}$  where  $|\theta\rangle$  is the state of a single spin. We form an  $N$ -particle tensor product state of such single-spin states with

```

1 In[336]:=productstate[0_?VectorQ, 1] = 0;
2 In[337]:=productstate[0_?VectorQ, n_Integer;/n>=2] :=
3          Flatten[KroneckerProduct @@ Table[0, n]]

```

in accordance with [In \[236\]](#); notice that the case  $N = 1$  requires special attention.

The particular single-spin states  $|+x\rangle, |-x\rangle, |+z\rangle, |-z\rangle$  we will be using are

```

1 In[338]:=xup[S_?SpinQ] := 2^(-S)*Table[Sqrt[Binomial[2S,M+S]],{M,S,-S,-1}]
2 In[339]:=xdn[S_?SpinQ] := 2^(-S)*Table[(-1)^(M+S)*Sqrt[Binomial[2S,M+S]],{M,S,-S,-1}]
3 In[340]:=zup[S_?SpinQ] := SparseArray[1 -> 1, 2S+1]
4 In[341]:=zdn[S_?SpinQ] := SparseArray[-1 -> 1, 2S+1]

```

We can check that these are correct with

```

1 In[342]:=Table[sx[S].xup[S] == S*xup[S], {S, 0, 4, 1/2}]
2 Out[342]={True, True, True, True, True}
3 In[343]:=Table[sx[S].xdn[S] == -S*xdn[S], {S, 0, 4, 1/2}]
4 Out[343]={True, True, True, True, True}
5 In[344]:=Table[sz[S].zup[S] == S*zup[S], {S, 0, 4, 1/2}]
6 Out[344]={True, True, True, True, True}
7 In[345]:=Table[sz[S].zdn[S] == -S*zdn[S], {S, 0, 4, 1/2}]
8 Out[345]={True, True, True, True, True}

```

From these we construct the product states

```

1 In[346]:=allxup[S_?SpinQ,n_Integer;/n>=1] := productstate[xup[S],n]
2 In[347]:=alldn[S_?SpinQ,n_Integer;/n>=1] := productstate[xdn[S],n]
3 In[348]:=allzup[S_?SpinQ,n_Integer;/n>=1] := productstate[zup[S],n]
4 In[349]:=allzdn[S_?SpinQ,n_Integer;/n>=1] := productstate[zdn[S],n]

```



### 3.4.3 Hamiltonian Diagonalization

We find the  $m$  lowest-energy eigenstates of this Hamiltonian with the procedures described in Sect. 1.10.4: for example, with  $S = 1/2$  and  $N = 20$ ,<sup>8</sup>

```

1 In[350]:=With[{S = 1/2, n = 20},
2   (* Hamiltonian *)
3   h[b_] = H[S, n, b];
4   (* two degenerate ground states for b=0 *)
5   gs0up = allzup[S, n];
6   gs0dn = allzdn[S, n];
7   (* ground state for b=+Infinity *)
8   gsplusinf = allxup[S, n];
9   (* ground state for b=-Infinity *)
10  gsminusinf = allxdn[S, n];
11  (* numerically calculate lowest m eigenstates *)
12  Clear[gs];
13  gs[b_?NumericQ, m_Integer /; m>=1] := gs[b, m] = -Eigensystem[-h[N[b]], m,
14    Method -> {"Arnoldi", "Criteria" -> "RealPart", MaxIterations -> 10^6}] //
15    Transpose //Sort //Transpose;
16 ]

```

Comments:

- $gs0up = |\psi_{0\uparrow}\rangle$  and  $gs0dn = |\psi_{0\downarrow}\rangle$  are the exact degenerate ground states for  $b = 0$ ;  $gsplusinf = |\psi_{+\infty}\rangle$  and  $gsminusinf = |\psi_{-\infty}\rangle$  are the exact nondegenerate ground states for  $b = \pm\infty$ .
- The function `gs`, which calculates the  $m$  lowest-lying eigenstates of the Hamiltonian, remembers its calculated values (see Sect. 1.6.3): this is important here because such eigenstate calculations can take a long time when  $n$  is large.
- The function `gs` numerically calculates the eigenvalues using `h[N[b]]` as a Hamiltonian, which ensures that the Hamiltonian contains floating-point machine-precision numbers instead of exact numbers in case  $b$  is given as an exact number. Calculating the eigenvalues and eigenvectors of a matrix of exact numbers takes extremely long (please try: on line 13 of `In[350]` replace `-Eigensystem[-h[N[b]], ...` with `-Eigensystem[-h[b], ...` and compare the run time of `gs[1, 2]` with that of `gs[1.0, 2]`).
- The operations `//Transpose //Sort //Transpose` on line 15 of `In[350]` ensure that the eigenvalues (and associated eigenvectors) are sorted in ascending energy order (see `In[150]`).
- When the ground state is degenerate, which happens here for  $b \approx 0$ , the Arnoldi algorithm has some difficulty finding the correct degeneracy. This means that `gs[0, 2]` may return two non-degenerate eigenstates instead of the (correct) two degenerate ground states. This is a well-known problem that can be circumvented by calculating more eigenstates.
- A problem involving  $N$  spin- $S$  systems leads to matrices of size  $(2S + 1)^N \times (2S + 1)^N$ . This scaling quickly becomes very problematic (even if we use sparse matrices) and is at the center of why quantum mechanics is difficult. Imagine a system composed of  $N = 1000$  spins  $S = 1/2$ : its state vector is a list of

<sup>8</sup>The attached Mathematica code uses  $N = 14$  instead, since calculations with  $N = 20$  take a long time.

$2^{1000} = 1.07 \times 10^{301}$  complex numbers! Comparing this to the fact that there are only about  $10^{80}$  particles in the universe, we conclude that such a state vector could never be written down and therefore the Hilbert space method of quantum mechanics we are using here is fundamentally flawed. But as this is an introductory course, we will stick to this classical matrix-mechanics formalism and let the computer bear the weight of its complexity. Keep in mind, though, that this is not a viable strategy for large systems, as each doubling of computer capacity only allows us to add a single spin to the system, which, using Moore's law, allows us to add one spin every two years.<sup>9</sup>

There are alternative formulations of quantum mechanics, notably the path-integral formalism, which partly circumvent this problem; but the computational difficulty is not eliminated, it is merely shifted. Modern developments such as tensor networks<sup>10</sup> try to limit the accessible Hilbert space by restricting calculations to a subspace where the entanglement between particles is bounded. This makes sense since almost all states of the huge Hilbert space are so complex and carry such complicated quantum-mechanical entanglement that (i) they would be extremely difficult to generate with realistic Hamiltonians, and (ii) they would decohere within very short time.

### 3.4.4 Analysis of the Ground State

#### Energy Gap

Much of the behavior of our Ising spin chain can be seen in a plot of the *energy gap*, which is the energy difference between the ground state and the first excited state. With  $m = 2$  we calculate the two lowest-lying energy levels and plot their energy difference as a function of the parameter  $b$ :

```


1 In[351]:=With[{bmax = 3, db = 1/64, m = 2},
2   ListLinePlot[Table[{b, gs[b,m][[1,2]]-gs[b,m][[1,1]]},
3     {b, -bmax, bmax, db}]]

```

Notice how the fact that the `gs` function remembers its own results speeds up this calculation by a factor of 2 (see Sect. 1.6.3).

<sup>9</sup>Moore's law is the observation that over the history of computing hardware, the number of transistors on integrated circuits doubles approximately every two years. From [https://en.wikipedia.org/wiki/Moore's\\_law](https://en.wikipedia.org/wiki/Moore's_law).

<sup>10</sup>Matrix product states and tensor networks: [https://en.wikipedia.org/wiki/Matrix\\_product\\_state](https://en.wikipedia.org/wiki/Matrix_product_state).



Even in this small 20-spin simulation we can see that this gap is approximately

$$E_1 - E_0 \approx \begin{cases} 0 & \text{if } |b| < 1, \\ \frac{|b|-1}{2} & \text{if } |b| > 1. \end{cases} \quad (3.32)$$

This observation of a qualitative change in the excitation gap suggests that at  $b = \pm 1$  the system undergoes a *quantum phase transition* (i.e., a phase transition induced by quantum fluctuations instead of thermal fluctuations). We note that the gap of Eq. (3.32) is independent of the particle number  $N$  and is therefore a *global* property of the Ising spin ring, not a property of each individual spin (in which case it would scale with  $N$ ).


### Overlap with Asymptotic States

Once a ground state  $|\psi_b\rangle$  has been calculated, we compute its overlap with the asymptotically known states using scalar products. Notice that for  $b = 0$  we calculate the scalar products with the states  $\frac{|\psi_{0\uparrow}\rangle \pm |\psi_{0\downarrow}\rangle}{\sqrt{2}}$  as they are the approximate ground states for  $|b| \ll 1$ .

```

1 In[352]:=With[{bmax = 3, db = 1/64, m = 2},
2   ListLinePlot[
3     Table[{b, Abs[gsminusinf.gs[b,m][[2,1]]]^2},
4           {b, Abs[gsplusinf.gs[b, m][[2,1]]]^2},
5           {b, Abs[((gs0up-gs0dn)/Sqrt[2]).gs[b,m][[2,1]]^2},
6           {b, Abs[((gs0up+gs0dn)/Sqrt[2]).gs[b,m][[2,1]]^2},
7           {b, Abs[((gs0up-gs0dn)/Sqrt[2]).gs[b,m][[2,1]]^2 +
8             Abs[((gs0up+gs0dn)/Sqrt[2]).gs[b,m][[2,1]]^2}},
9           {b, -bmax, bmax, db}] //Transpose]

```



Observations:

- The overlap  $|\langle\psi_b|\psi_{-\infty}\rangle|^2$  (red) approaches 1 as  $b \rightarrow -\infty$ .
- The overlap  $|\langle\psi_b|\psi_{+\infty}\rangle|^2$  (green) approaches 1 as  $b \rightarrow +\infty$ .
- The overlap  $\left|\langle\psi_b|\frac{|\psi_{0\uparrow}\rangle-|\psi_{0\downarrow}\rangle}{\sqrt{2}}\right|^2$  (cyan) is mostly negligible.
- The overlap  $\left|\langle\psi_b|\frac{|\psi_{0\uparrow}\rangle+|\psi_{0\downarrow}\rangle}{\sqrt{2}}\right|^2$  (orange) approaches 1 as  $b \rightarrow 0$ .
- The sum of these last two,  $\left|\langle\psi_b|\frac{|\psi_{0\uparrow}\rangle-|\psi_{0\downarrow}\rangle}{\sqrt{2}}\right|^2 + \left|\langle\psi_b|\frac{|\psi_{0\uparrow}\rangle+|\psi_{0\downarrow}\rangle}{\sqrt{2}}\right|^2 = |\langle\psi_b|\psi_{0\uparrow}\rangle|^2 + |\langle\psi_b|\psi_{0\downarrow}\rangle|^2$  (thin black), approaches 1 as  $b \rightarrow 0$  and is less prone to numerical noise.
- If you redo this calculation with an *odd* number of spins, you may find different overlaps with the  $\frac{|\psi_{0\uparrow}\rangle\pm|\psi_{0\downarrow}\rangle}{\sqrt{2}}$  asymptotic states. Their sum, however, drawn in black, should be insensitive to the parity of  $N$ .
- For  $|b| \lesssim 0.2$  the excitation gap (see above) is so small that the calculated ground-state eigenvector is no longer truly the ground state but becomes mixed with the first excited state due to numerical inaccuracies. This leads to the jumps in the orange and cyan curves (notice, however, that their sum, shown in black, is stable). If you redo this calculation with larger values for  $m$ , you may get better results.

## Magnetization


Studying the ground state coefficients list directly is of limited use because of the large amount of information contained in its numerical representation. We gain more insight by studying specific observables, for example the magnetizations  $\langle\hat{S}_x^{(k)}\rangle$ ,  $\langle\hat{S}_y^{(k)}\rangle$ , and  $\langle\hat{S}_z^{(k)}\rangle$ . We add the following definition to the `With[]` clause in [In\[350\]](#):

```

16 (* spin components expectation values *)
17 Clear[mx,my,mz];
18 mx[b_?NumericQ, m_Integer /; m >= 1, k_Integer] :=
19   mx[b, m, k] = With[{g = gs[b,m][[2,1]]},
20     Re[Conjugate[g].(sx[S, n, Mod[k, n, 1]].g)]];
21 my[b_?NumericQ, m_Integer /; m >= 1, k_Integer] :=
22   my[b, m, k] = With[{g = gs[b,m][[2,1]]},
23     Re[Conjugate[g].(sy[S, n, Mod[k, n, 1]].g)]];
24 mz[b_?NumericQ, m_Integer /; m >= 1, k_Integer] :=
25   mz[b, m, k] = With[{g = gs[b,m][[2,1]]},
26     Re[Conjugate[g].(sz[S, n, Mod[k, n, 1]].g)]];
27 ]

```

In our transverse Ising model only the  $x$ -component of the magnetization is nonzero. Due to the translational symmetry of the system we can look at the magnetization of any spin, for example the first one ( $k = 1$ ):  $m_x(b)$  (blue) and  $m_z(b)$  (orange, non-zero due to numerical inaccuracies)



We see that in the phases of large  $|b|$ , the spins are almost entirely polarized, while in the phase  $|b| < 1$  the  $x$ -magnetization is roughly proportional to  $b$ .

### Spin–Spin Fluctuation Correlations

Quantum-mechanical spins always fluctuate around their mean direction. In the example of Q3.8, the state  $|S, S\rangle$  points on average along the  $+z$  direction in the sense that  $\langle \hat{\mathbf{S}} \rangle = \langle S, S | \hat{\mathbf{S}} | S, S \rangle = \{0, 0, S\}$ ; but it fluctuates away from this axis as  $\langle \hat{S}_x^2 \rangle = \langle \hat{S}_y^2 \rangle = S/2$ .

By introducing the fluctuation operator  $\hat{\delta\mathbf{S}} = \hat{\mathbf{S}} - \langle \hat{\mathbf{S}} \rangle$ , we can interpret spin fluctuations through the expectation values  $\langle \hat{\delta\mathbf{S}} \rangle = \{0, 0, 0\}$  (fluctuations always average to zero) and  $\langle (\hat{\delta\mathbf{S}})^2 \rangle = \langle \hat{\delta\mathbf{S}} \cdot \hat{\delta\mathbf{S}} \rangle = \langle \hat{\mathbf{S}} \cdot \hat{\mathbf{S}} \rangle - \langle \hat{\mathbf{S}} \rangle \cdot \langle \hat{\mathbf{S}} \rangle = S(S+1) - \|\langle \hat{\mathbf{S}} \rangle\|^2$ . Since the spin magnetization has length  $0 \leq \|\langle \hat{\mathbf{S}} \rangle\| \leq S$ , these fluctuations satisfy  $S \leq \langle (\hat{\delta\mathbf{S}})^2 \rangle \leq S(S+1)$ : they are positive for every spin state.

When two (or more) spins are present, their quantum-mechanical fluctuations can become correlated. We quantify such spin–spin fluctuation correlations between two spins  $k$  and  $k'$  with the measure

$$C_{k,k'} = \langle \hat{\delta\mathbf{S}}^{(k)} \cdot \hat{\delta\mathbf{S}}^{(k')} \rangle = \langle \hat{\mathbf{S}}^{(k)} \cdot \hat{\mathbf{S}}^{(k')} \rangle - \langle \hat{\mathbf{S}}^{(k)} \rangle \cdot \langle \hat{\mathbf{S}}^{(k')} \rangle, \quad (3.33)$$

which has the form of a statistical covariance.<sup>11</sup> For any spin length  $S$  (assuming  $S^{(k)} = S^{(k')}$ ), the first term of Eq. (3.33) can be written as

$$\langle \hat{\mathbf{S}}^{(k)} \cdot \hat{\mathbf{S}}^{(k')} \rangle = \frac{\langle (\hat{\mathbf{S}}^{(k)} + \hat{\mathbf{S}}^{(k')})^2 \rangle - \langle (\hat{\mathbf{S}}^{(k)})^2 \rangle - \langle (\hat{\mathbf{S}}^{(k')})^2 \rangle}{2} = \frac{1}{2} \langle (\hat{\mathbf{S}}^{(k)} + \hat{\mathbf{S}}^{(k')})^2 \rangle - S(S+1), \quad (3.34)$$

<sup>11</sup> See <https://en.wikipedia.org/wiki/Covariance>.

which allows us to predict its expectation value as a function of the total-spin quantum number describing the two spins- $S$ . As this quantum number can be anywhere between 0 and  $2S$ , we have  $0 \leq \langle (\hat{\mathbf{S}}^{(k)} + \hat{\mathbf{S}}^{(k')})^2 \rangle \leq 2S(2S+1)$ . This expectation value is not restricted to integer values. As a result we make the following observations:

- $-S(S+1) \leq C_{k,k'} \leq S^2$ : spin fluctuations can be correlated ( $C_{k,k'} > 0$ ), anti-correlated ( $C_{k,k'} < 0$ ), or uncorrelated ( $C_{k,k'} = 0$ ).
- The strongest correlations  $C_{k,k'} = S^2$  are found when the two spins- $S$  form a joint spin- $2S$  and at the same time are unaligned ( $\langle \hat{\mathbf{S}}^{(k)} \rangle \cdot \langle \hat{\mathbf{S}}^{(k')} \rangle = 0$ ).
- The strongest anti-correlations  $C_{k,k'} = -S(S+1)$  are found when the two spins- $S$  form a joint spin-0 (i.e., a spin-singlet). In this case, the magnetizations always vanish:  $\langle \hat{\mathbf{S}}^{(k)} \rangle = \langle \hat{\mathbf{S}}^{(k')} \rangle = \{0, 0, 0\}$ .

For the specific case  $S = 1/2$ , which we use in the present calculations, two spins can form a joint singlet (total spin 0;  $\langle (\hat{\mathbf{S}}^{(k)} + \hat{\mathbf{S}}^{(k')})^2 \rangle = 0$ ), a joint triplet (total spin 1;  $\langle (\hat{\mathbf{S}}^{(k)} + \hat{\mathbf{S}}^{(k')})^2 \rangle = 2$ ), or a mixture of these ( $0 \leq \langle (\hat{\mathbf{S}}^{(k)} + \hat{\mathbf{S}}^{(k')})^2 \rangle \leq 2$ ), and the correlation is restricted to the values  $-\frac{3}{4} \leq C_{k,k'} \leq +\frac{1}{4}$  for all states. Specific cases are:

- In the pure joint singlet state  $\frac{|\uparrow\downarrow\rangle - |\downarrow\uparrow\rangle}{\sqrt{2}}$  the correlation is precisely  $C_{k,k'} = -\frac{3}{4}$ . A fluctuation of one spin implies a counter-fluctuation of the other in order to keep them anti-aligned and in a spin-0 joint state. Remember that the spin monogamy theorem states that if spins  $k$  and  $k'$  form a joint singlet, then both must be uncorrelated with all other spins in the system.
- In a pure joint triplet state, i.e., any mixture of the states  $|\uparrow\uparrow\rangle$ ,  $|\downarrow\downarrow\rangle$ , and  $\frac{|\uparrow\downarrow\rangle + |\downarrow\uparrow\rangle}{\sqrt{2}}$ , the correlation is  $0 \leq C_{k,k'} \leq +\frac{1}{4}$ . A fluctuation of one spin implies a similar fluctuation of the other in order to keep them aligned and in a spin-1 joint state.
- The maximum correlation  $C_{k,k'} = +\frac{1}{4}$  is reached for unaligned triplet states, i.e., when  $\langle \hat{\mathbf{S}}^{(k)} \rangle \cdot \langle \hat{\mathbf{S}}^{(k')} \rangle = 0$ . Examples include the states  $\frac{|\uparrow\uparrow\rangle + |\downarrow\downarrow\rangle}{\sqrt{2}}$ ,  $\frac{|\uparrow\uparrow\rangle - |\downarrow\downarrow\rangle}{\sqrt{2}}$ , and  $\frac{|\uparrow\downarrow\rangle + |\downarrow\uparrow\rangle}{\sqrt{2}}$ .
- In the fully parallel triplet states  $|\uparrow\uparrow\rangle$  and  $|\downarrow\downarrow\rangle$ , the magnetizations are aligned but their fluctuations are uncorrelated:  $C_{k,k'} = 0$ , and hence  $\langle \hat{\mathbf{S}}^{(k)} \cdot \hat{\mathbf{S}}^{(k')} \rangle = \langle \hat{\mathbf{S}}^{(k)} \rangle \cdot \langle \hat{\mathbf{S}}^{(k')} \rangle$ .


In order to estimate these spin fluctuation correlations, we add the following definition to the `With[]` clause in [In\[350\]](#):

```

27 (* spin-spin correlation operator *)
28 Clear[Cop];
29 Cop[k1_Integer, k2_Integer] := Cop[k1, k2] =
30   With[{q1 = Mod[k1, n, 1], q2 = Mod[k2, n, 1]},
31     sx[S, n, q1].sx[S, n, q2] + sy[S, n, q1].sy[S, n, q2]
32     + sz[S, n, q1].sz[S, n, q2]];
33 (* spin-spin correlations *)
34 Clear[c];
35 c[b_?NumericQ, m_Integer /; m >= 1, {k1_Integer, k2_Integer}] :=
36   c[b, m, {k1, k2}] = With[{g = gs[b, m][[2, 1]]},
37     Re[Conjugate[g].(Cop[k1, k2].g)] - (mx[b, m, k1]*mx[b, m, k2]
38       + my[b, m, k1]*my[b, m, k2] + mz[b, m, k1]*mz[b, m, k2])];
39 ]

```

Since our spin ring is translationally invariant, we can simply plot  $C_\delta = C_{1,1+\delta}$ : for  $N = 20$  and  $\delta = 1 \dots 10$  (top to bottom),



Observations:

- The spin fluctuations are maximally correlated ( $C = +\frac{1}{4}$ ) for  $b = 0$ , in the ferromagnetic phase. They are all either pointing up or pointing down, so every spin is correlated with every other spin; keep in mind that the magnetization vanishes at the same time (Sect. 3.4.4). It is only the spin–spin interactions that correlate the spins’ directions and therefore their fluctuations.
- The spin fluctuations are uncorrelated ( $C \rightarrow 0$ ) for  $b \rightarrow \pm\infty$ , in the paramagnetic phases. They are all pointing in the  $+x$  direction for  $b \gg 1$  or in the  $-x$  direction for  $b \ll -1$ , but they are doing so in an independent way and would keep pointing in that direction even if the spin–spin interactions were switched off. This means that the fluctuations of the spins’ directions are uncorrelated.

### Entropy of Entanglement

We know now that in the limits  $b \rightarrow \pm\infty$  the spins are polarized (magnetized) but their fluctuations are uncorrelated, while close to  $b = 0$  they are unpolarized (unmagnetized) but their fluctuations are maximally correlated. Here we quantify these correlations with the *entropy of entanglement*, which measures the entanglement of a single spin with the rest of the spin chain.

In a system composed of two subsystems A and B, the entropy of entanglement is defined as the *von Neumann entropy* of the reduced density matrix (see Sect. 2.4.3),

$$S_{AB} = -\text{Tr}(\hat{\rho}_A \log_2 \hat{\rho}_A) = -\sum_i \lambda_i \log_2 \lambda_i \quad (3.35)$$

where the  $\lambda_i$  are the eigenvalues of  $\hat{\rho}_A$  (or of  $\hat{\rho}_B$ ; the result is the same). Care must be taken with the case  $\lambda_i = 0$ : we find  $\lim_{\lambda \rightarrow 0} \lambda \log_2 \lambda = 0$ . For this we define the function

```
1 In[353]:=s[0|0.] = 0;
2 In[354]:=s[x_] = -x*Log[2, x];
```

that uses Mathematica's pattern matching to separate out the special case  $x = 0$ . Note that we use an alternative pattern<sup>12</sup> `0|0.` that matches both an analytic zero `0` and a numeric zero `0.`, which Mathematica distinguishes carefully.<sup>13</sup>

We define the entropy of entanglement of the first spin with the rest of the spin ring using the definition of `In[257]`, tracing out the last  $(2S+1)^{N-1}$  degrees of freedom and leaving only the first  $2S+1$  degrees of freedom of the first spin:

```
1 In[355]:=EE[S_?SpinQ,  $\psi$ ] :=
2 Total[s /@ Re[Eigenvalues[traceout[ $\psi$ , -Length[ $\psi$ ]/(2S+1)]]]
```


Observations:

- Entanglement entropies of the known asymptotic ground states:

```
1 In[356]:=EE[1/2, (gs0up+gs0dn)/Sqrt[2]]
2 Out[356]=1
3 In[357]:=EE[1/2, (gs0up-gs0dn)/Sqrt[2]]
4 Out[357]=1
5 In[358]:=EE[1/2, gsplusinf]
6 Out[358]=0
7 In[359]:=EE[1/2, gsminusinf]
8 Out[359]=0
```

- Entanglement entropy as a function of  $b$ : again the calculation is numerically difficult around  $b \approx 0$  because of the quasi-degeneracy.

```
1 In[360]:=With[{bmax = 3, db = 1/64, m = 2},
2 ListLinePlot[Table[{b, EE[1/2, gs[b,m][[2,1]]}],
3 {b, -bmax, bmax, db}], PlotRange -> {0, 1}]]
```



Notice that the quantum phase transitions at  $b = \pm 1$  are not visible in this plot.

<sup>12</sup>See <https://reference.wolfram.com/language/tutorial/PatternsInvolvingAlternatives.html>.

<sup>13</sup>Experiment: `0==0.` yields `True` (testing for semantic identity), whereas `0===0.` yields `False` (testing for symbolic identity).



### 3.4.5 Exercises

**Q3.15** For  $S = 1/2$ , what is the largest value of  $N$  for which you can calculate the ground state of the transverse Ising model at the critical point  $b = 1$ ?

**Q3.16** Study the transverse Ising model with  $S = 1$ :

1. At which values of  $b$  do you find quantum phase transitions?
2. Characterize the ground state in terms of magnetization, spin–spin correlations, and entanglement entropy.

**Q3.17** Study the transverse XY model for  $S = 1/2$ :

$$\hat{\mathcal{H}} = -\frac{b}{2} \sum_{k=1}^N \hat{S}_z^{(k)} - \sum_{k=1}^N \left( \hat{S}_x^{(k)} \hat{S}_x^{(k+1)} + \hat{S}_y^{(k)} \hat{S}_y^{(k+1)} \right) \quad (3.36)$$

1. Guess the shape of the ground states for  $b \pm \infty$  [notice that the first term in the Hamiltonian of Eq. (3.36) is in the  $z$ -direction!] and compare to the numerical calculations.
2. At which values of  $b$  do you find quantum phase transitions?
3. Characterize the ground state in terms of magnetization, spin–spin correlations, and entanglement entropy.

**Q3.18** Study the Heisenberg model for  $S = 1/2$ :

$$\hat{\mathcal{H}} = -\frac{b}{2} \sum_{k=1}^N \hat{S}_z^{(k)} - \sum_{k=1}^N \hat{\mathbf{S}}^{(k)} \cdot \hat{\mathbf{S}}^{(k+1)} \quad (3.37)$$

1. Guess the shape of the ground states for  $b \pm \infty$  [notice that the first term in the Hamiltonian of Eq. (3.37) is in the  $z$ -direction!] and compare to the numerical calculations.
2. What is the ground-state degeneracy for  $b = 0$ ?
3. At which values of  $b$  do you find quantum phase transitions?
4. Characterize the ground state in terms of magnetization, spin–spin correlations, and entanglement entropy.

**Q3.19** Consider two spin-1/2 particles in the triplet state  $|\psi\rangle = |\uparrow\uparrow\rangle$ . Subsystem A is the first spin, and subsystem B is the second spin.

1. What is the density matrix  $\hat{\rho}_{AB}$  of this system?
2. What is the reduced density matrix  $\hat{\rho}_A$  of subsystem A (the first spin)? Is this a pure state? If yes, what state?
3. What is the reduced density matrix  $\hat{\rho}_B$  of subsystem B (the second spin)? Is this a pure state? If yes, what state?
4. Calculate the von Neumann entropies of  $\hat{\rho}_{AB}$ ,  $\hat{\rho}_A$ , and  $\hat{\rho}_B$ .

**Q3.20** Consider two spin-1/2 particles in the singlet state  $|\psi\rangle = \frac{|\uparrow\downarrow\rangle - |\downarrow\uparrow\rangle}{\sqrt{2}}$ . Subsystem A is the first spin, and subsystem B is the second spin.

1. What is the density matrix  $\hat{\rho}_{AB}$  of this system?
2. What is the reduced density matrix  $\hat{\rho}_A$  of subsystem A (the first spin)? Is this a pure state? If yes, what state?
3. What is the reduced density matrix  $\hat{\rho}_B$  of subsystem B (the second spin)? Is this a pure state? If yes, what state?
4. Calculate the von Neumann entropies of  $\hat{\rho}_{AB}$ ,  $\hat{\rho}_A$ , and  $\hat{\rho}_B$ .

### 3.5 Coupled Spin Systems: Quantum Circuits [Supplementary Material 5]

The computational structure developed so far in this chapter can be used to simulate quantum circuits, such as they are used to run quantum algorithms leading all the way to quantum computers. In its simplest form, a quantum circuit contains a set of  $N$  spin-1/2 quantum objects called *qubits*, on which a sequence of operations called *quantum gates* is executed. In analogy to classical binary logic, the basis states of the qubits' Hilbert space are usually denoted as  $|0\rangle$  (replacing the spin-1/2 state  $|\uparrow\rangle$ ) and  $|1\rangle$  (replacing  $|\downarrow\rangle$ ).

In this section, we go through the steps of assembling quantum circuits and simulating their behavior on a classical computer. Naturally, the matrix representation of quantum gates and circuits constructed here is neither efficient nor desirable for building an actual quantum computer. It is merely useful for acquiring a detailed understanding of the workings of quantum circuits and algorithms.

In what follows, we adhere strictly to Chapter 5 of Nielsen&Chuang,<sup>14</sup> which provides many more details of the calculations, as well as further reading for the interested student.

#### 3.5.1 Quantum Gates

Any quantum circuit can be constructed from a set of simple building blocks, similarly to a classical digital circuit. These building blocks are canonical quantum gates,<sup>15</sup> of which we implement a useful subset here.

##### Single-Qubit Gates

Single-qubit gates act on one specific qubit in a set:

<sup>14</sup>Michael A. Nielsen and Isaac L. Chuang: *Quantum Computation and Quantum Information*, 10th Anniversary Edition, Cambridge University Press, Cambridge, UK (2010).

<sup>15</sup>See [https://en.wikipedia.org/wiki/Quantum\\_logic\\_gate](https://en.wikipedia.org/wiki/Quantum_logic_gate).

- The Pauli-X gate  $\boxed{X}$  acts like  $\hat{\sigma}_x = |1\rangle\langle 0| + |0\rangle\langle 1|$  on the desired qubit, and has no effect on all other qubits. A single-qubit input state  $|\psi_{\text{in}}\rangle$  entering the gate from the left is transformed into the output state  $|\psi_{\text{out}}\rangle = \hat{\sigma}_x |\psi_{\text{in}}\rangle$  exiting the gate towards the right.
- The Pauli-Y gate  $\boxed{Y}$  acts like  $\hat{\sigma}_y = i|1\rangle\langle 0| - i|0\rangle\langle 1|$  on the desired qubit, and has no effect on all other qubits.
- The Pauli-Z gate  $\boxed{Z}$  acts like  $\hat{\sigma}_z = |0\rangle\langle 0| - |1\rangle\langle 1|$  on the desired qubit, and has no effect on all other qubits.
- The Hadamard gate  $\boxed{H}$  acts like  $\frac{\hat{\sigma}_x + \hat{\sigma}_z}{\sqrt{2}} = \frac{|0\rangle\langle 0| + |0\rangle\langle 1| + |1\rangle\langle 0| - |1\rangle\langle 1|}{\sqrt{2}}$  on the desired qubit, and has no effect on all other qubits.

To implement these single-qubit gates in a general way, we proceed as in [In\[331\]](#) by defining a matrix that represents the operator  $\hat{a}$  acting on the  $k$ th qubit in a set of  $n$  qubits:

```
1 In[361]:= op[n_Integer, k_Integer, a_] /; 1<=k<=n && Dimensions[a]=={2,2} :=
2           KroneckerProduct[IdentityMatrix[2^(k-1)], SparseArray],
3           a,
4           IdentityMatrix[2^(n-k), SparseArray]]
```

This allows us to define the single-qubit Pauli and Hadamard gates with

```
1 In[362]:= {id, sx, sy, sz} = Table[SparseArray[PauliMatrix[i]], {i, 0, 3}];
2 In[363]:= X[n_Integer, k_Integer] /; 1<=k<=n := op[n, k, sx]
3 In[364]:= Y[n_Integer, k_Integer] /; 1<=k<=n := op[n, k, sy]
4 In[365]:= Z[n_Integer, k_Integer] /; 1<=k<=n := op[n, k, sz]
5 In[366]:= H[n_Integer, k_Integer] /; 1<=k<=n := op[n, k, (sx+sz)/Sqrt[2]]
```

as well as the corresponding rotation operators  $\hat{R}_x(\phi) = \frac{1+e^{i\phi}}{2}\mathbb{1} + \frac{1-e^{i\phi}}{2}\sigma_x = e^{i\phi/2} e^{-i\phi\hat{\sigma}_x/2}$  etc. that are also known as phase gates,

```
1 In[367]:= RX[n_Integer, k_Integer, phi_] /; 1<=k<=n :=
2           op[n, k, (1+Exp[I*phi])/2*id + (1-Exp[I*phi])/2*sx]
3 In[368]:= RY[n_Integer, k_Integer, phi_] /; 1<=k<=n :=
4           op[n, k, (1+Exp[I*phi])/2*id + (1-Exp[I*phi])/2*sy]
5 In[369]:= RZ[n_Integer, k_Integer, phi_] /; 1<=k<=n :=
6           op[n, k, (1+Exp[I*phi])/2*id + (1-Exp[I*phi])/2*sz]
```

## Two-Qubit Gates

Interesting quantum circuits require operations that involve more than one qubit.

The SWAP gate exchanges the state of qubits  $j$  and  $k$  in a set of  $n$  qubits:

$$\begin{array}{c} j \text{ --- } \times \\ k \text{ --- } \times \end{array}$$

Without going through complicated considerations over basis-set indices, we construct it through the definition  $\text{SWAP}^{(jk)} = (\mathbb{1}^{(j)} \otimes \mathbb{1}^{(k)} + \hat{\sigma}_x^{(j)} \otimes \hat{\sigma}_x^{(k)} + \hat{\sigma}_y^{(j)} \otimes \hat{\sigma}_y^{(k)} + \hat{\sigma}_z^{(j)} \otimes \hat{\sigma}_z^{(k)})/2$  and building on the above Pauli gates:

```
1 In[370]:= SWAP[n_Integer, {j_Integer, k_Integer}] /; 1<=j<=n && 1<=k<=n && j!=k :=
2           (IdentityMatrix[2^n, SparseArray] +
3           X[n, j].X[n, k] + Y[n, j].Y[n, k] + Z[n, j].Z[n, k])/2
```

The matrix representation of a two-qubit SWAP takes on the familiar form

```

1 In[371]:=SWAP[2, {1,2}] //Normal
2 Out[371]={{1, 0, 0, 0},
3           {0, 0, 1, 0},
4           {0, 1, 0, 0},
5           {0, 0, 0, 1}}
```

The square root of the SWAP gate is also sometimes used, and is defined similarly:


```

1 In[372]:=SQRTSWAP[n_Integer, {j_Integer, k_Integer}] /; 1<=j<=n && 1<=k<=n && j!=k :=
2 (3+I)/4 * IdentityMatrix[2^n, SparseArray] +
3 (1-I)/4 * (X[n,j].X[n,k] + Y[n,j].Y[n,k] + Z[n,j].Z[n,k])
```

To define the controlled-NOT or CNOT gate, we first make a general definition for controlled gates. The  $n$ -qubit operator `CTRL[n, λ, A]` acts like the operator  $\hat{A}$  if all qubits in the list  $\lambda = \{i_1, i_2, \dots, i_k\}$  are in the  $|1\rangle$  state, and has no action (acts like the identity operator on  $n$  qubits) if any of the qubits in the list  $\lambda$  are in the  $|0\rangle$  state:

$$\text{CTRL} = \left[ \bigotimes_{j=1}^k |1\rangle\langle 1|^{(i_j)} \right] \cdot \hat{A} + \left[ \mathbb{1} - \bigotimes_{j=1}^k |1\rangle\langle 1|^{(i_j)} \right] \cdot \mathbb{1} = \mathbb{1} + \left[ \bigotimes_{j=1}^k |1\rangle\langle 1|^{(i_j)} \right] \cdot (\hat{A} - \mathbb{1}) \quad (3.38)$$

Its circuit representation is



The bracket in the last expression of Eq. (3.38) is constructed with `Apply[Dot, op[n, #, P1]&/@λ]` that first constructs a list of projection operators  $|1\rangle\langle 1|^{(i_j)}$  for the control qubits, and then applies the `Dot` operator to assemble them into the product  $\bigotimes_{j=1}^k |1\rangle\langle 1|^{(i_j)}$ .

```

1 In[373]:=P0 = (id + σz)/2 //SparseArray; (* qubit projector |0⟩⟨0| *)
2 In[374]:=P1 = (id - σz)/2 //SparseArray; (* qubit projector |1⟩⟨1| *)
3 In[375]:=CTRL[n_Integer, λ_ /; VectorQ[λ, IntegerQ], A_] /;
4 (Unequal@@λ) && Min[λ]>=1 && Max[λ]<=n && Dimensions[A]=={2^n, 2^n} :=
5 IdentityMatrix[2^n, SparseArray] +
6 Apply[Dot, op[n, #, P1]&/@λ].(A - IdentityMatrix[2^n, SparseArray])
```

With this definition, the CNOT operator  $\text{CNOT}^{(jk)} = |0\rangle\langle 0|^{(j)} \otimes \mathbb{1}^{(k)} + |1\rangle\langle 1|^{(j)} \otimes \sigma_x^{(k)}$



is simply the CTRL operator with a single element in the list  $\lambda = \{j\}$  and a single-qubit  $\hat{A} = \hat{\sigma}_x^{(k)}$  operator,

```

1 In[376]:=CNOT[n_Integer, j_Integer -> k_Integer] /; 1<=j<=n && 1<=k<=n && j!=k :=
2   CTRL[n, {j}, op[n, k, σx]]

```

Notice that here we use the notation `CNOT[n, j->k]` to indicate that qubit  $j$  controls qubit  $k$ : this arrow notation `->` is purely for syntactic beauty and has no further effects (it is a pattern like any other, with no unintended side effects). The matrix representation of a two-qubit CNOT takes on the familiar form

```

1 In[377]:=CNOT[2, 1->2] //Normal
2 Out[377]={{1, 0, 0, 0},
3           {0, 1, 0, 0},
4           {0, 0, 0, 1},
5           {0, 0, 1, 0}}

```

### Three-Qubit Gates

For completeness, we define three-qubit gates that are sometimes useful in the construction of general quantum circuits.

The CCNOT gate or Toffoli gate is a controlled-NOT gate `CCNOT[n, {i, j}->k]` with two controlling qubits,  $i$  and  $j$ , and is defined in analogy to the CNOT gate:



```

1 In[378]:=CCNOT[n_Integer, {i_Integer, j_Integer} -> k_Integer] /;
2   1<=i<=n && 1<=j<=n && 1<=k<=n && Unequal[i,j,k] :=
3   CTRL[n, {i,j}, op[n, k, σx]]

```

The controlled-SWAP gate or Fredkin gate `CSWAP[n, i->{j,k}]` conditionally swaps two qubits,  $j$  and  $k$ :




```

1 In[379]:=CSWAP[n_Integer, i_Integer -> {j_Integer, k_Integer}] /;
2   1<=i<=n && 1<=j<=n && 1<=k<=n && Unequal[i,j,k] :=
3   CTRL[n, {i}, SWAP[n, {j, k}]]

```

### 3.5.2 A Simple Quantum Circuit

As a simple example, we study the quantum circuit



The unitary operation corresponding to this circuit is a Hadamard gate on qubit 1, followed by a controlled-NOT gate where qubit 1 controls the inversion of qubit 2. In Mathematica this gate sequence needs to be written from right to left, because the gates are represented by matrices that will be applied to a state vector on their right:

```

1 In[380]:=S = CNOT[2, 1->2] . H[2, 1];
2 In[381]:=Normal[S]
3 Out[381]={{1/Sqrt[2], 0, 1/Sqrt[2], 0},
4           {0, 1/Sqrt[2], 0, 1/Sqrt[2]},
5           {0, 1/Sqrt[2], 0, -1/Sqrt[2]},
6           {1/Sqrt[2], 0, -1/Sqrt[2], 0}}

```

The matrix representation of `Out[381]` refers to the two-qubit basis set  $\mathcal{B}_2 = \{|00\rangle, |01\rangle, |10\rangle, |11\rangle\}$ , which we can inspect for any number of qubits with

```

1 In[382]:=B[n_Integer /; n>=1] := Tuples[{0, 1}, n]
2 In[383]:=B[2]
3 Out[383]={{0,0}, {0,1}, {1,0}, {1,1}}

```

The input state of our circuit is the product state  $|\psi_{\text{in}}\rangle = |0\rangle \otimes |0\rangle = |00\rangle$ , which is the first element of  $\mathcal{B}_2$ :

```

1 In[384]:=ψin = {1,0,0,0};

```


The output state of our circuit follows from the application of  $S$ ,

```

1 In[385]:=ψout = S . ψin
2 Out[385]={1/Sqrt[2], 0, 0, 1/Sqrt[2]}

```

Looking at the basis set  $\mathcal{B}_2$  we identify this output state with the maximally entangled state  $|\psi_{\text{out}}\rangle = \frac{|00\rangle + |11\rangle}{\sqrt{2}}$ . Projective measurements on the two qubits,



give 50% probability of finding the classical result “00” and 50% probability of finding “11”, whereas the bit combinations “01” and “10” never occur:

```

1 In[386]:=Abs[ψout]^2
2 Out[386]={1/2, 0, 0, 1/2}

```

It is important to recognize that these four probabilities are insufficient to identify the state  $|\psi_{\text{out}}\rangle$ , even if many measurements are made, because any state whose diagonal density-matrix elements match `Out[386]` gives these probabilities. Generally, in order to identify a two-qubit output state fully, a quantum-state tomography (QST)<sup>16</sup> must be performed, which involves applying further phase gates (qubit rotations) before the projective measurements and measuring all sixteen (fifteen non-trivial) expectation values  $\langle \psi_{\text{out}} | \hat{\sigma}_1 \otimes \hat{\sigma}_2 | \psi_{\text{out}} \rangle$  for  $\hat{\sigma}_1, \hat{\sigma}_2 \in \{\mathbb{1}, \hat{\sigma}_x, \hat{\sigma}_y, \hat{\sigma}_z\}$ , followed by an inversion procedure to estimate the density matrix<sup>17</sup>:

<sup>16</sup>See [https://en.wikipedia.org/wiki/Quantum\\_tomography](https://en.wikipedia.org/wiki/Quantum_tomography).

<sup>17</sup>Equation (3.39) is a direct inversion that may not result in a positive semi-definite density matrix if experimental noise is present. In such cases, more elaborate inversion procedures are available.

$$\begin{aligned}
\hat{\rho} &= \frac{1}{4} \sum_{\hat{\sigma}_1 \in \{\mathbb{1}, \hat{\sigma}_x, \hat{\sigma}_y, \hat{\sigma}_z\}} \sum_{\hat{\sigma}_2 \in \{\mathbb{1}, \hat{\sigma}_x, \hat{\sigma}_y, \hat{\sigma}_z\}} \langle \psi_{\text{out}} | \hat{\sigma}_1 \otimes \hat{\sigma}_2 | \psi_{\text{out}} \rangle \cdot \hat{\sigma}_1 \otimes \hat{\sigma}_2 \\
&= \frac{1}{4} \begin{bmatrix} \mathbb{1}\mathbb{1} + \mathbb{1}z + z\mathbb{1} + zz & \mathbb{1}x - i\mathbb{1}y + zx - izy & x\mathbb{1} + xz - iy\mathbb{1} - izy & xx - ix - iyx - yy \\ \mathbb{1}x + i\mathbb{1}y + zx + izy & \mathbb{1}\mathbb{1} - \mathbb{1}z + z\mathbb{1} - zz & xx + ix - iyx + yy & x\mathbb{1} - xz - iy\mathbb{1} + izy \\ x\mathbb{1} + xz + iy\mathbb{1} + izy & xx - ix + iyx + yy & \mathbb{1}\mathbb{1} + \mathbb{1}z - z\mathbb{1} - zz & \mathbb{1}x - i\mathbb{1}y - zx + izy \\ xx + ix + iyx - yy & x\mathbb{1} - xz + iy\mathbb{1} - izy & \mathbb{1}x + i\mathbb{1}y - zx - izy & \mathbb{1}\mathbb{1} - \mathbb{1}z - z\mathbb{1} + zz \end{bmatrix}
\end{aligned} \tag{3.39}$$

(abbreviating  $xy = \langle \psi_{\text{out}} | \hat{\sigma}_x \otimes \hat{\sigma}_y | \psi_{\text{out}} \rangle$  etc.) A full QST on  $n$  qubits requires measuring  $4^n - 1$  such expectation values, which makes the QST infeasible in general.

### 3.5.3 Application: The Quantum Fourier Transform

The discrete classical Fourier transform<sup>18</sup> (CFT) of a list of  $N$  complex numbers  $\vec{x} = \{x_0, x_1, \dots, x_{N-1}\}$  is given by the list  $\vec{y} = \{y_0, y_1, \dots, y_{N-1}\}$  with elements

$$y_j = \frac{1}{\sqrt{N}} \sum_{k=0}^{N-1} x_k e^{2\pi i jk/N}. \tag{3.40}$$

It can be seen as a unitary matrix operation

$$\vec{y} = \mathbf{F} \cdot \vec{x} \quad \text{with } F_{jk} = e^{2\pi i jk/N} / \sqrt{N}. \tag{3.41}$$

With the Fast Fourier Transform (FFT) algorithm,<sup>19</sup> the computational effort of evaluating Eq. (3.41) is of order  $\mathcal{O}[N \log(N)]$ .

The discrete Quantum Fourier Transform (QFT) is precisely the same transformation, except that the vectors  $\vec{x}$  and  $\vec{y}$  are encoded into quantum states. For this, a quantum system with Hilbert space dimension  $N$  is described by a basis set  $\{|0\rangle, |1\rangle, \dots, |N-1\rangle\}$ , and the states  $|x\rangle = \sum_{j=0}^{N-1} x_j |j\rangle$  and  $|y\rangle = \sum_{j=0}^{N-1} y_j |j\rangle$  are seen as related by the unitary QFT operator  $\hat{\mathcal{F}}$  such that

$$|y\rangle = \hat{\mathcal{F}}|x\rangle \quad \text{with } \langle j | \hat{\mathcal{F}} | k \rangle = F_{jk}, \tag{3.42}$$

in analogy to Eq. (3.41). The idea of this section is that the QFT can be evaluated much faster than the CFT, even though both are mathematically equivalent.

We assume that  $N = 2^n$  is an integer power of two.<sup>20</sup> The Hilbert space of  $n$  qubits has exactly  $2^n = N$  dimensions, and therefore we use these  $n$  qubits to encode the states  $|x\rangle$  and  $|y\rangle$  in the following way. The  $2^n$  basis states  $\mathcal{B}_n = \{|00 \dots 00\rangle, |00 \dots 01\rangle, |00 \dots 10\rangle, \dots, |11 \dots 11\rangle\}$  are, in our usual construction


<sup>18</sup>See [https://en.wikipedia.org/wiki/Discrete\\_Fourier\\_transform](https://en.wikipedia.org/wiki/Discrete_Fourier_transform).

<sup>19</sup>See [https://en.wikipedia.org/wiki/Fast\\_Fourier\\_transform](https://en.wikipedia.org/wiki/Fast_Fourier_transform).

<sup>20</sup>For all other cases, choose  $n$  as the smallest integer  $\geq \log_2(N)$  and set  $x_N \dots x_{2^n-1}$  to zero.

through tensor products (Sect. 2.4.2), listed in increasing order when interpreted as binary numbers (see In[382]). We give each basis state a new label equal to this binary number:  $|00 \dots 00\rangle = |0\rangle$ ,  $|00 \dots 01\rangle = |1\rangle$ ,  $|00 \dots 10\rangle = |2\rangle$ , ...,  $|11 \dots 11\rangle = |2^n - 1\rangle$ , such that the state of the first qubit is the most significant bit (MSB) of the binary representation of the basis state's index, and the state of the  $n$ th qubit is the least significant bit (LSB) of the binary representation of the basis state's index. What follows below is a quantum circuit operating on these  $n$  qubits that has the effect of the QFT operator  $\hat{\mathcal{F}}$ , as expressed in this binary basis.

The Quantum Fourier Transform circuit is assembled from single-qubit Hadamard gates and two-qubit controlled Z-phase gates, where  $\hat{R}_k = \hat{R}_z(2\pi/2^k) = |0\rangle\langle 0| + e^{2\pi i/2^k}|1\rangle\langle 1|$  using In[369]:



To construct the  $i$ th dashed block consisting of a Hadamard gate on qubit  $i$  followed by  $n - i$  controlled Z-phase gates, we remember that the application of matrix operators happens from right to left, in the reverse order from that shown in the circuit diagram above. We first construct a list of the controlled RZ operators and contract it by applying Dot:

```

1 In[387]:=QFTblock[n_Integer, i_Integer] /; 1<=i<=n :=
2   Apply[Dot, Table[CTRL[n, {j}, RZ[n, i, 2π/2^(j+1-i)]], {j, n, i+1, -1}]].
3   H[n,i]

```

We assemble the  $n$ -qubit QFT operator from these dashed QFTblock blocks and a set of SWAP operations that reverses the qubit order,

```

1 In[388]:=QFT[n_Integer] /; n>=1 :=
2   Apply[Dot, Table[SWAP[n, {i, n+1-i}], {i, 1, n/2}]].
3   Apply[Dot, Table[QFTblock[n, i], {i, n, 1, -1}]]

```

The matrix representation of this QFT operator is a  $2^n \times 2^n$  matrix with element  $(j, k)$  given by  $2^{-n/2} e^{2\pi i j k / 2^n}$ , precisely as expected from Eq. (3.42) with  $N = 2^n$ . We check this relation for  $n = 1 \dots 6$  with

```

1 In[389]:=Table[QFT[n] == 2^(-n/2)*Table[Exp[2π*I*j*k/2^n], {j,0,2^n-1}, {k,0,2^n-1}],
2   {n, 6}] //FullSimplify
3 Out[389]={True, True, True, True, True, True}

```

The resources used to construct this quantum circuit are

- $n$  Hadamard gates,
- $\frac{n(n-1)}{2}$  controlled Z-phase gates, and
- $n/2$  swap gates.




In the present classical simulation of quantum circuits, each quantum gate is a sparse  $2^n \times 2^n$  matrix, usually containing  $\mathcal{O}(2^n)$  nonzero matrix elements; applying such a simulated gate to a state therefore takes  $\mathcal{O}(2^n)$  time, which makes the simulated QFT no faster than the classical FFT, which scales as  $\mathcal{O}(2^n n)$ . However, if we can construct a physical system in which these gates can be applied in a time that scales at most polynomially with  $n$ , then the QFT is a massive improvement over the scaling of the classical FFT. The development of such physical qubit/gate systems is the focus of much ongoing scientific research.

### 3.5.4 Application: Quantum Phase Estimation

The QFT circuit of Sect. 3.5.3 cannot be used by itself in practice, because it requires the preparation of an arbitrary quantum state containing an exponential number of parameters  $x_j$ , as well as a full quantum-state tomography to read out an exponential number of parameters  $y_j$  describing the final state (see Sect. 3.5.2). In this section we study a quantum circuit that uses the QFT as a component, and circumvents these exponential input/output bottlenecks.

Unitary matrices have eigenvalues that are of unit norm, and can be written as  $e^{2\pi i\varphi}$  with  $\varphi \in \mathbb{R}$ . The question addressed here is: given a unitary operator  $\hat{U}_\varphi$  and an eigenstate  $|u\rangle$  such that  $\hat{U}_\varphi|u\rangle = e^{2\pi i\varphi}|u\rangle$ , can we estimate  $\varphi$  efficiently, that is, to  $t$  binary digits with an effort that scales polynomially with  $t$ ?

The answer is yes, using the following quantum circuit that makes use of the Quantum Fourier Transform of Sect. 3.5.3 but (i) starts with an initial product state that can be prepared with  $\mathcal{O}(t)$  effort, and (ii) does not require a full quantum state tomography, but instead finishes with a simple projective measurement that takes  $\mathcal{O}(t)$  effort.



To set up a quantum phase estimation in Mathematica, we begin by defining the unitary operator  $\hat{U}_\varphi$  and its eigenstate  $|u\rangle$  with

```
1 In[390]:=u = {1};
2 In[391]:=U[φ_] = {{Exp[2π*I*φ]}};
```

and check that they satisfy  $\hat{U}_\varphi|u\rangle = e^{2\pi i\varphi}|u\rangle$  and  $\langle u|u\rangle = 1$ :

```

1 In[392]:= {U[φ].u == E^(2π*I*φ)*u, Norm[u] == 1}
2 Out[392]= {True, True}

```

Here we use a one-dimensional quantum system: the operator  $\hat{U}_\varphi$  is a  $1 \times 1$  matrix, and the state  $|u\rangle$  is a list of length 1. More generally, the Hilbert space of the system under test (SUT) can be arbitrarily large (see Q3.22), and more complex quantum circuits can be substituted for  $\hat{U}$  in more elaborate experiments.

In order to construct the phase estimation circuit, we will also need the unit operator acting on the SUT:

```

1 In[393]:= U0 = IdentityMatrix[Length[u], SparseArray];

```

The controlled version of the  $\hat{U}_\varphi$  operator, where the  $i$ th qubit out of a set of  $n$  qubits controls the application of  $\hat{U}_\varphi$  to the SUT, is  $\hat{U}_\varphi^{(i)} = |0\rangle\langle 0|^{(i)} \otimes \mathbb{1} + |1\rangle\langle 1|^{(i)} \otimes \hat{U}_\varphi$ . We use the tensor-product techniques of Sect. 2.4.2 to couple the qubits to the SUT:

```

1 In[394]:= CTRLU[n_Integer, i_Integer, φ_] /; 1<=i<=n :=
2 KroneckerProduct[op[n,i,P0], U0] + KroneckerProduct[op[n,i,P1], U[φ]]

```

The initial state of the phase estimation circuit is  $|\psi_0\rangle = |0\rangle^{\otimes t} \otimes |u\rangle$ . We know that the state  $|0\rangle^{\otimes t} = |00\dots 00\rangle$  is the first basis state in the computational basis  $\mathcal{B}_n$  (eigen-basis of  $\hat{\sigma}_z$ ), and construct it with `SparseArray[1->1, 2^t]`. As an example, we work with  $t = 4$  qubits here:

```

1 In[395]:= t = 4;
2 In[396]:= ψ0 = Flatten[KroneckerProduct[SparseArray[1->1, 2^t], u]] //Normal;

```

Applying a Hadamard gate to each qubit gives the state

```

1 In[397]:= ψ1 = KroneckerProduct[Apply[Dot, Table[H[t, i], {i, t}]], U0] . ψ0;

```

Applying the controlled  $\hat{U}_\varphi^m = \hat{U}_{m\varphi}$  operations sequentially then gives the state

```

1 In[398]:= ψ2[φ_] = Apply[Dot, Table[CTRLU[t, i, 2^(t-i)*φ], {i, t, 1, -1}]] . ψ1;

```

Finally, an inverse QFT yields the phase-estimation state  $|\varepsilon_\varphi\rangle$ . Remember that the QFT is a unitary operation, and therefore its inverse is its Hermitian conjugate:

```

1 In[399]:= ε[φ_] = KroneckerProduct[ConjugateTranspose[QFT[t]], U0] . ψ2[φ];

```

We use the techniques of Sect. 2.4.3, in particular In[257], to drop the component  $|u\rangle$  at the end of the quantum circuit and find the reduced density matrix of the qubits. The diagonal elements of this reduced density matrix are the probabilities of finding the various basis states of  $\mathcal{B}_n$  in a projective measurement as shown on the right of the above circuit:

```

1 In[400]:= prob[φ_?NumericQ] := Re[Diagonal[traceout[ε[N[φ]], -Length[u]]]]

```

The first element of `prob[φ]` gives the probability of measurement outcomes  $\{0, 0, 0, 0\}$ , that is, the probability that the qubits are in the joint state  $|0000\rangle = |0\rangle \otimes |0\rangle \otimes |0\rangle \otimes |0\rangle$ . The second element of `prob[φ]` gives the probability of measurement outcomes  $\{0, 0, 0, 1\}$ , that is, the probability that the qubits are in the joint state  $|0001\rangle = |0\rangle \otimes |0\rangle \otimes |0\rangle \otimes |1\rangle$ . And so forth: the  $j$ th element of `prob[φ]` gives the probability of measurement outcomes corresponding to the binary representation of  $j - 1$ .

The trick of this phase-estimation quantum circuit is that the information on  $\varphi$  is contained in the state  $|\varepsilon_\varphi\rangle$  in a way that can be extracted from these probabilities without doing a full quantum-state tomography. We get an idea of what this means by looking at the probabilities for the different measurement outcomes when  $\varphi$  is an integer multiple of  $2^{-t}$ :

```

1 In[401]:=Table[prob[φ], {φ, 0, 1, 2^(-t)}] //Chop
2 Out[401]={{1., 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
3           {0, 1., 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
4           {0, 0, 1., 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
5           {0, 0, 0, 1., 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
6           {0, 0, 0, 0, 1., 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
7           {0, 0, 0, 0, 0, 1., 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
8           {0, 0, 0, 0, 0, 0, 1., 0, 0, 0, 0, 0, 0, 0, 0, 0},
9           {0, 0, 0, 0, 0, 0, 0, 1., 0, 0, 0, 0, 0, 0, 0, 0},
10          {0, 0, 0, 0, 0, 0, 0, 0, 1., 0, 0, 0, 0, 0, 0, 0},
11          {0, 0, 0, 0, 0, 0, 0, 0, 0, 1., 0, 0, 0, 0, 0, 0},
12          {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1., 0, 0, 0, 0, 0},
13          {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1., 0, 0, 0, 0},
14          {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1., 0, 0, 0},
15          {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1., 0, 0},
16          {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1., 0},
17          {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1.},
18          {1., 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}}
```

Whenever  $\varphi$  is an integer multiple of  $2^{-t} = 1/16$ , we find that only one basis state is occupied, and therefore the outcomes of the projective measurements on the 4 qubits always give the same results, with no quantum fluctuation. A single projective measurement of all 4 qubits can be interpreted as a binary number  $j \in \{0, 1, 2, \dots, 15\}$  that is related to the phase estimate as  $\varphi = j/16$ ; no quantum-state tomography is required.

What happens when  $\varphi$  is not an integer multiple of  $2^{-t}$ ? It turns out that the basis state corresponding to the nearest integer multiple of  $2^{-t}$  will be found most frequently in the projective measurements. For example, for  $\varphi = 0.2$  the probabilities for projecting  $|\varepsilon_{0.2}\rangle$  into the 16 basis states are

```

1 In[402]:=prob[0.2]
2 Out[402]={0., 0.01, 0.02, 0.88, 0.06, 0.01, 0., 0., 0., 0., 0., 0., 0., 0., 0.}
```

(rounded here to two decimals). The fourth basis state, which is  $|0011\rangle$  corresponding to  $\varphi = 3/16$ , will be found in about 88% of all experiments, and so a plurality vote most likely yields  $\varphi \approx 3/16 = 0.1875$  as a fair estimate of the phase, with an upper bound on the error of  $2^{-t-1} = 1/32$ ; no quantum-state tomography required. We extract the expected plurality-vote winner of a large number of experiments with

```

1 In[403]:=mostprobable[φ_?NumericQ] := (Ordering[prob[φ], -1][[1]] - 1)/2^t
```

where the `Ordering` function is used to give the position of the largest element:

```
1 In[404]:=mostprobable[0.2]
2 Out[404]=3/16
```


It can be shown that `mostprobable[ $\varphi$ ]==Mod[Round[ $\varphi$ , 2^(-t)], 1]`. Increasing the number of qubits  $t$  results in more precise estimates, while keeping the circuit complexity at  $\mathcal{O}(t^2)$ .

### 3.5.5 Exercises

- Q3.21** For the output state  $|\psi_{\text{out}}\rangle$  of [Out\[386\]](#), calculate all expectation values necessary to fill in Eq. (3.39).
- Q3.22** Multi-dimensional phase estimation: set  $u = \{1, 1\}/\sqrt{2}$  and  $\hat{U}_\varphi = e^{2\pi i\varphi} \{\{1, 0\}, \{0, 1\}\}$  (two-dimensional system under test) and show that the phase-estimation algorithm still works.
- Q3.23** What happens if  $|u\rangle$  is not an eigenstate of  $\hat{U}_\varphi$ ? Set  $u = \{1, 1\}/\sqrt{2}$  and  $\hat{U}_\varphi = \{\{e^{2\pi i\varphi}, 0\}, \{0, e^{4\pi i\varphi}\}\}$  (two-dimensional system with two different evolution frequencies) and re-evaluate the attached Mathematica script. Plot `prob[ $\varphi$ ]` for a range of frequencies  $\varphi$  using `ListDensityPlot` and interpret the resulting figure.

## Chapter 4

# Quantum Motion in Real Space



So far we have studied the quantum formalism in the abstract (Chap. 2) and in the context of rotational dynamics (Chap. 3). In this chapter we work with the spatial motion of point particles, which represents a kind of mechanics that is much closer to our everyday experience. Here, quantum states are called *wavefunctions* and depend

---

**Electronic supplementary material** The online version of this chapter ([https://doi.org/10.1007/978-981-13-7588-0\\_4](https://doi.org/10.1007/978-981-13-7588-0_4)) contains supplementary material, which is available to authorized users.

on the spatial coordinate(s). This apparent difference to the material covered in the previous chapters disappears when we express all wavefunctions in a basis set. We develop numerical methods for studying spatial dynamics that stay as close to a real-space description as quantum mechanics allows.

## 4.1 One Particle in One Dimension

A single particle moving in one dimension is governed by a Hamiltonian of the form

$$\hat{\mathcal{H}} = \hat{T} + \hat{V} \quad (4.1)$$

in terms of the kinetic operator  $\hat{T}$  and the potential operator  $\hat{V}$ . These operators are usually expressed in the Dirac position basis set  $\{|x\rangle\}_{x \in \mathbb{R}}$ ,<sup>1</sup> which diagonalizes the position operator in the sense that  $\hat{x}|x\rangle = x|x\rangle$ ,<sup>2</sup> is ortho-normalized  $\langle x|y\rangle = \delta(x - y)$ , and complete  $\int_{-\infty}^{\infty} |x\rangle\langle x| dx = \mathbb{1}$ . Using this Dirac basis, the explicit expressions for the operators in the Hamiltonian are

$$\hat{T} = -\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} dx |x\rangle \frac{d^2}{dx^2} \langle x|, \quad \hat{V} = \int_{-\infty}^{\infty} dx |x\rangle V(x) \langle x|, \quad (4.2)$$

where  $m$  is the particle's mass and  $V(x)$  is its potential. Single-particle states  $|\psi\rangle$ , on the other hand, are written in this basis as

$$|\psi\rangle = \int_{-\infty}^{\infty} dx \psi(x) |x\rangle, \quad (4.3)$$

where  $\psi(x) = \langle x|\psi\rangle$  is the wavefunction.

In what follows we restrict the freedom of the particle to a domain  $x \in \Omega = [0, a]$ , where  $a$  can be very large in order to approximately describe infinite systems (example: Sect. 4.1.7). This assumes the potential to be

$$V(x) = \begin{cases} \infty & \text{for } x \leq 0 \\ W(x) & \text{for } 0 < x < a \\ \infty & \text{for } x \geq a \end{cases} \quad (4.4)$$

---

<sup>1</sup>To be exact, the Dirac position basis set spans a space that is much larger than the Hilbert space of square-integrable smooth functions used in quantum mechanics. This can be seen by noting that this basis set has an uncountably infinite number of elements  $|x\rangle$ , while the dimension of the Hilbert space in question is only countably infinite [see Eq. (4.5) for a countably infinite basis set]. The underlying problem of the *continuum*, which quantum mechanics attempts to resolve, is discussed with some of its philosophical origins and implications by Erwin Schrödinger in his essay “Science and Humanism” (Cambridge University Press, 1951, ISBN 978-0521575508).

<sup>2</sup>This eigenvalue equation is tricky: remember that  $\hat{x}$  is an operator,  $|x\rangle$  is a state, and  $x$  is a real number.

This restriction is necessary in order to achieve a finite representation of the system in a computer.

### Exercises

- Q4.1** Insert Eqs. (4.2) and (4.3) into the time-independent Schrödinger equation  $\hat{\mathcal{H}}|\psi\rangle = E|\psi\rangle$ . Use the ortho-normality of the Dirac basis to derive the usual form of the Schrödinger equation for a particle's wavefunction in 1D:  $-\frac{\hbar^2}{2m}\psi''(x) + V(x)\psi(x) = E\psi(x)$ .
- Q4.2** Use Eq. (4.3) to show that the scalar product between two states is given by the usual formula  $\langle\psi|\chi\rangle = \int_{-\infty}^{\infty} \psi^*(x)\chi(x)dx$ .

### 4.1.1 Units

In order to proceed with implementing the Hamiltonian (4.1), we first need a consistent set of units (see Sect. 1.12) in which to express length, time, mass, and energy. Of these four units, only three are independent: expressions like the classical kinetic energy  $E = \frac{1}{2}mv^2$  indicate a fixed relationship between these four units.

A popular system of units is the International System of Units (SI),<sup>3</sup> in which this consistency is built in:

```
1 In[405]:=LengthUnit = Quantity["Meters"];      (* choose freely *)
2 In[406]:=TimeUnit = Quantity["Seconds"];      (* choose freely *)
3 In[407]:=MassUnit = Quantity["Kilograms"];    (* choose freely *)
4 In[408]:=EnergyUnit = MassUnit*LengthUnit^2/TimeUnit^2 //UnitConvert;
```

The consistency of this set of definitions is seen in In[408], making the energy unit depend on the other units, which in turn can be chosen freely. Many other combinations are possible, as long as this consistency remains.

Another popular choice is to additionally couple the time and energy units through Planck's constant  $\hbar$ , and make both dependent on the length and mass units (thus reducing the system of units to only two degrees of freedom):

```
1 In[409]:=LengthUnit = Quantity["Meters"];      (* choose freely *)
2 In[410]:=MassUnit = Quantity["Kilograms"];    (* choose freely *)
3 In[411]:=TimeUnit =
4     MassUnit*LengthUnit^2/Quantity["ReducedPlanckConstant"] //UnitConvert;
5 In[412]:=EnergyUnit = Quantity["ReducedPlanckConstant"]/TimeUnit //UnitConvert;
```

This latter set of units is what we will be using in what follows, without restriction of generality. We express the reduced Planck constant in these units with

```
1 In[413]:=ħ = Quantity["ReducedPlanckConstant"]/(EnergyUnit*TimeUnit) //UnitConvert //N
2 Out[413]=1.
```

which is equal to unity because of our chosen coupling between energy and time units; in other unit systems the value will be different. Note the use of `//N` at the

<sup>3</sup>See [https://en.wikipedia.org/wiki/International\\_System\\_of\\_Units](https://en.wikipedia.org/wiki/International_System_of_Units).

end of `In[413]` to force the result to be a pure machine-precision number instead of a variable-precision number that tracks the accuracy of the involved physical quantities.

To set the physical size  $a$  of the computational box, for example to  $a = 5 \mu\text{m}$ , we execute

```
1 In[414]:= a = Quantity[5, "Micrometers"]/LengthUnit //UnitConvert //N;
```

and to set the particle's mass  $m$ , for example to the neutron's mass,

```
1 In[415]:= m = Quantity["NeutronMass"]/MassUnit //UnitConvert //N;
```

In the calculations that follow, we will not be explicit about the system of units and the physical quantities. Instead, we will use direct dimensionless definitions such as

```
1 In[416]:= a = 30; (* calculation box size in units of length *)
2 In[417]:= m = 1; (* particle mass in units of mass *)
3 In[418]:= h = 1; (* value of h assuming In[412] *)
```

These are to be replaced by `In[414]`, `In[415]`, and `In[413]` in a more concrete physical situation.

### 4.1.2 Computational Basis Functions

In order to perform quantum-mechanical calculations of a particle moving in one dimension, we need a basis set that is more practical than the Dirac basis used to define the relevant operators and states above. Indeed, Dirac states  $|x\rangle$  are difficult to represent in a computer because they are uncountable, densely spaced, and highly singular.

The most generally useful basis sets for computations are the *momentum basis* and the *finite-resolution position basis*, which we will look at in turn, and which will be shown to be related to each other by a type-I discrete sine transform.

#### Momentum Basis

The simplest one-dimensional quantum-mechanical system of the type of Eq. (4.1) is the infinite square well with  $W(x) = 0$ . Its energy eigenstates  $\phi_n(x)$  for  $n = 1, 2, 3, \dots$  satisfy the Schrödinger equation  $-\frac{\hbar^2}{2m}\phi_n''(x) = E_n\phi_n(x)$  (see Q4.1) and the boundary conditions  $\phi_n(0) = \phi_n(a) = 0$  necessitated by Eq. (4.4). Their explicit normalized forms are

$$\langle x|n\rangle = \phi_n(x) = \sqrt{\frac{2}{a}} \sin\left(\frac{n\pi x}{a}\right) \quad (4.5)$$

with eigen-energies

$$E_n = \frac{n^2\pi^2\hbar^2}{2ma^2}. \quad (4.6)$$



We know from the Sturm–Liouville theorem<sup>4</sup> that these functions form a complete set (see Q2.2); further, we can use Mathematica to show that they are ortho-normalized:

```
1 In[419]:= φ[a_, n_, x_] = Sqrt[2/a]*Sin[n*π*x/a];
2 In[420]:= Table[Integrate[φ[a,n1,x]*φ[a,n2,x], {x, 0, a}],
3           {n1, 10}, {n2, 10}] //MatrixForm
```

They are eigenstates of the squared momentum operator  $\hat{p}^2 = (-i\hbar \frac{d}{dx})^2 = -\hbar^2 \frac{d^2}{dx^2}$ :

$$\hat{p}^2|n\rangle = \frac{n^2\pi^2\hbar^2}{a^2}|n\rangle, \quad (4.7)$$

which we verify with


```
1 In[421]:= -ħ^2*D[φ[a,n,x], {x,2}] == (n^2*π^2*ħ^2)/a^2*φ[a,n,x]
2 Out[421]= True
```

This makes the kinetic operator  $\hat{T} = \hat{p}^2/(2m)$  diagonal in this basis:

$$\langle n|\hat{T}|n'\rangle = E_n\delta_{nn'}, \quad \hat{T} = \sum_{n=1}^{\infty}|n\rangle E_n \langle n|. \quad (4.8)$$

However, in general the potential energy, and most other operators that will appear later, are difficult to express in this momentum basis.

The momentum basis of Eq. (4.5) contains a countably infinite number of basis functions, which is a great advantage over the uncountably infinite cardinality of the Dirac basis set. In practical calculations, we restrict the computational basis to  $n \in \{1 \dots n_{\max}\}$ , which means that we only consider physical phenomena with excitation energies below  $E_{n_{\max}} = \frac{\pi^2\hbar^2}{2ma^2}n_{\max}^2$  (see Sect. 2.1.1). Here is an example of what these position-basis functions look like for  $n_{\max} = 10$ :



Using the approximate completeness of the momentum basis,  $\sum_{n=1}^{n_{\max}}|n\rangle\langle n| \approx \mathbb{1}$  (see Sect. 2.1.1), the kinetic Hamiltonian thus becomes

$$\hat{T} \approx \left[ \sum_{n=1}^{n_{\max}}|n\rangle\langle n| \right] \hat{T} \left[ \sum_{n'=1}^{n_{\max}}|n'\rangle\langle n'| \right] = \sum_{n,n'=1}^{n_{\max}}|n\rangle\langle n|\hat{T}|n'\rangle\langle n'| = \sum_{n=1}^{n_{\max}}|n\rangle E_n \langle n|. \quad (4.9)$$

<sup>4</sup>See [https://en.wikipedia.org/wiki/Sturm-Liouville\\_theory](https://en.wikipedia.org/wiki/Sturm-Liouville_theory).

We set up the kinetic Hamiltonian operator as a sparse diagonal matrix in the momentum basis with

```
1 In[422]:=nmax = 100;
2 In[423]:=TM = SparseArray[Band[{1,1}] -> Range[nmax]^2*\pi^2*\hbar^2/(2*m*a^2)];
```

where  $n_{\max} = 100$  was chosen as an example.

### Finite-resolution Position Basis

Given an energy-limited momentum basis set  $\{|n\rangle\}_{n=1}^{n_{\max}}$  from above, we define a set of  $n_{\max}$  equally-spaced points

$$x_j = j \cdot \Delta \quad (4.10)$$


for  $j \in \{1 \dots n_{\max}\}$ , with spacing  $\Delta = a/(n_{\max} + 1)$ . These grid points fill the calculation range  $x \in [0, a]$  uniformly without covering the end points. We then define a new basis set as the closest possible representations of delta-functions at these points: for  $j \in \{1 \dots n_{\max}\}$ ,

$$|j\rangle = \sqrt{\Delta} \sum_{n=1}^{n_{\max}} \phi_n(x_j) |n\rangle. \quad (4.11)$$

The spatial wavefunctions of these basis states are

$$\langle x|j\rangle = \vartheta_j(x) = \sqrt{\Delta} \sum_{n=1}^{n_{\max}} \phi_n(x_j) \phi_n(x). \quad (4.12)$$

Here is an example of what these position-basis functions look like for  $n_{\max} = 10$ :



This new basis set is also ortho-normal,  $\langle j|j'\rangle = \delta_{jj'}$ , and it is strongly local in the sense that only the basis function  $\vartheta_j(x)$  is nonzero at  $x_j$ , while all others vanish:

$$\langle x_{j'}|j\rangle = \vartheta_j(x_{j'}) = \delta_{jj'}/\sqrt{\Delta}. \quad (4.13)$$

We define these basis functions in Mathematica with

```
1 In[424]:=nmax = 10;
2 In[425]:=Δ = a/(nmax+1);
3 In[426]:=xx[j_] = j*Δ;
4 In[427]:=θ[j_, x_] = Sqrt[Δ]*Sum[φ[n,xx[j]]*φ[n,x], {n, nmax}];
```

Since the basis function  $\vartheta_j(x) = \theta[j, x]$  is the only one which is nonzero at  $x_j = \text{xx}[j]$ , and it is close to zero everywhere else (exactly zero at the  $x_{j' \neq j}$ ), we can usually make several approximations:

- If a wavefunction is given as a vector  $\vec{v} = \mathbf{v}$  in the position basis,  $|\psi\rangle = \sum_{j=1}^{n_{\max}} v_j |j\rangle$ , then by Eq. (4.13) the wavefunction is known at the grid points:

$$\psi(x_j) = \langle x_j | \psi \rangle = \langle x_j | \sum_{j'=1}^{n_{\max}} v_{j'} |j'\rangle = \sum_{j'=1}^{n_{\max}} v_{j'} \langle x_j | j' \rangle = \sum_{j'=1}^{n_{\max}} v_{j'} \delta_{jj'} / \sqrt{\Delta} = \frac{v_j}{\sqrt{\Delta}}. \quad (4.14)$$

The density profile is thus given by the values of  $\rho(x_j) = |\psi(x_j)|^2 = |v_j|^2 / \Delta$ . This allows for very easy plotting of wavefunctions and densities by linearly interpolating between these grid points (i.e., an interpolation of order 1):

```
1 In[428] := ListLinePlot[Transpose[{Table[xx[j], {j, nmax}], Abs[v]^2/Δ}]]
```

By the truncation of the basis at  $n_{\max}$ , the wavefunction has no frequency components faster than one half-wave per grid-point spacing, and therefore we can be sure that this linear interpolation is a reasonably accurate representation of the wavefunction  $\psi(x)$  and the density  $\rho(x) = |\langle x | \psi \rangle|^2$ , in particular as  $n_{\max} \rightarrow \infty$ .

- An even simpler interpolation of order zero assumes that the wavefunction is constant over intervals  $[-\Delta/2, \Delta/2]$  centered at each grid point. This primitive interpolation is used, for example, to calculate the Wigner quasi-probability distribution in Sect. 4.1.8.
- Similarly, if a density operator is given by  $\hat{\rho} = \sum_{j,j'=1}^{n_{\max}} R_{j,j'} |j\rangle \langle j'|$ , then the value of the density operator at a grid point  $(x_j, x_{j'})$  is given by

$$\begin{aligned} \rho(x_j, x_{j'}) &= \langle x_j | \hat{\rho} | x_{j'} \rangle = \langle x_j | \left[ \sum_{j'', j'''=1}^{n_{\max}} R_{j'', j'''} |j''\rangle \langle j'''| \right] | x_{j'} \rangle = \sum_{j'', j'''=1}^{n_{\max}} R_{j'', j'''} \langle x_j | j'' \rangle \langle j'''| x_{j'} \rangle \\ &= \sum_{j'', j'''=1}^{n_{\max}} R_{j'', j'''} \frac{\delta_{j,j''}}{\sqrt{\Delta}} \frac{\delta_{j',j'''}}{\sqrt{\Delta}} = \frac{R_{j,j'}}{\Delta}. \end{aligned} \quad (4.15)$$

That is, the coefficients  $R_{j,j'}$  and the density values  $\rho(x_j, x_{j'})$  are very closely related. The diagonal elements of this expression ( $j = j'$ ) give the spatial density profile.

- For any function  $f(x)$  that varies slowly (smoothly) over length scales of the grid spacing  $\Delta$ , we can make the approximation

$$f(x) \vartheta_j(x) \approx f(x_j) \vartheta_j(x). \quad (4.16)$$

This approximation becomes exact on every grid point according to Eq. (4.13), and the assumed smoothness of  $f(x)$  makes it a good estimate for any  $x$ .

### Conversion Between Basis Sets

Within the approximation of a truncation at maximum energy  $E_{n_{\max}}$ , we can express any wavefunction  $|\psi\rangle$  in both basis sets of Eqs. (4.5) and (4.12):

$$|\psi\rangle = \sum_{n=1}^{n_{\max}} u_n |n\rangle = \sum_{j=1}^{n_{\max}} v_j |j\rangle \quad (4.17)$$

Inserting the definition of Eq. (4.11) into Eq. (4.17) we find

$$\sum_{n=1}^{n_{\max}} u_n |n\rangle = \sum_{j=1}^{n_{\max}} v_j \left[ \sqrt{\Delta} \sum_{n'=1}^{n_{\max}} \phi_{n'}(x_j) |n'\rangle \right] = \sum_{n'=1}^{n_{\max}} \left[ \sqrt{\Delta} \sum_{j=1}^{n_{\max}} v_j \phi_{n'}(x_j) \right] |n'\rangle \quad (4.18)$$

and therefore, since the basis set  $\{|n\rangle\}$  is ortho-normalized,

$$u_n = \sqrt{\Delta} \sum_{j=1}^{n_{\max}} v_j \phi_n(x_j) = \sum_{j=1}^{n_{\max}} X_{nj} v_j \quad (4.19)$$

with the basis conversion coefficients

$$X_{nj} = \langle n | j \rangle = \sqrt{\Delta} \phi_n(x_j) = \sqrt{\frac{a}{n_{\max} + 1}} \sqrt{\frac{2}{a}} \sin\left(\frac{n\pi x_j}{a}\right) = \sqrt{\frac{2}{n_{\max} + 1}} \sin\left(\frac{\pi n j}{n_{\max} + 1}\right). \quad (4.20)$$

The inverse transformation is found from  $|n\rangle = \sum_{j=1}^{n_{\max}} \langle j | n \rangle |j\rangle$  inserted into Eq. (4.17), giving

$$v_j = \sum_{n=1}^{n_{\max}} X_{nj} u_n \quad (4.21)$$

in terms of the same coefficients of Eq. (4.20). Thus the transformations relating the vectors  $\vec{u}$  (with components  $u_n$ ) and  $\vec{v}$  (with components  $v_j$ ) are  $\vec{v} = \mathbf{X} \cdot \vec{u}$  and  $\vec{u} = \mathbf{X} \cdot \vec{v}$  in terms of the same symmetric orthogonal matrix  $\mathbf{X}$  with coefficients  $X_{nj}$ .

We could calculate these coefficients with

```
In[429]:= X = Table[Sqrt[2/(nmax+1)]*Sin[π*n*j/(nmax+1)], {n, nmax}, {j, nmax}]/N;
```

but this is not very efficient, especially for large  $n_{\max}$ .

It turns out that Eqs. (4.19) and (4.21) relate the vectors  $\vec{u}$  and  $\vec{v}$  by a *type-I discrete sine transform* (DST-I), which Mathematica can evaluate very efficiently via a fast Fourier transform.<sup>5</sup> Since the DST-I is its own inverse, we can use

<sup>5</sup>See [https://en.wikipedia.org/wiki/Discrete\\_sine\\_transform](https://en.wikipedia.org/wiki/Discrete_sine_transform) and [https://en.wikipedia.org/wiki/Fast\\_Fourier\\_transform](https://en.wikipedia.org/wiki/Fast_Fourier_transform). The precise meaning of the DST-I can be seen from its equivalent definition through a standard discrete Fourier transform of doubled length: for a complex vector  $\mathbf{v}$ , we can substitute `FourierDST[v,1]` by `DST1[v_?VectorQ]:=-I*Fourier[Join[{0},v,{0}],Reverse[-v]]][[2;;Length[v]+1]]`. In this sense it is the discrete Fourier transform of a list  $\mathbf{v}$  augmented with (i) zero boundary conditions and (ii) reflection anti-symmetry at the boundaries. Remember that the `Fourier[]` transform assumes periodic boundary conditions, which are incorrect in the present setup, and need to be modified into a DST-I.

```

1 In[430]:=v = FourierDST[u, 1];
2 In[431]:=u = FourierDST[v, 1];

```

to effect such conversions. We will see a very useful application of this transformation when we study the time-dependent behavior of a particle in a potential (“split-step method”, Sect. 4.1.9).

The matrix  $\mathbf{X}$  is also useful for converting *operator* representations between the basis sets: the momentum representation  $\mathbf{U}$  and the position representation  $\mathbf{V}$  of the same operator satisfy  $\mathbf{V} = \mathbf{X} \cdot \mathbf{U} \cdot \mathbf{X}$  and  $\mathbf{U} = \mathbf{X} \cdot \mathbf{V} \cdot \mathbf{X}$ . In practice, as above we can convert operators between the position and momentum representation with a two-dimensional type-I discrete sine transform:

```

1 In[432]:=V = FourierDST[U, 1];
2 In[433]:=U = FourierDST[V, 1];

```

This easy conversion is very useful for the construction of the matrix representations of Hamiltonian operators, since the kinetic energy is diagonal in the momentum basis, Eq. (4.7), while the potential energy operator is approximately diagonal in the position basis, Eq. (4.25).

### 4.1.3 The Position Operator

The position operator  $\hat{x} = \int_{-\infty}^{\infty} dx |x\rangle x \langle x|$  is one of the basic operators that is used frequently to construct Hamiltonians of moving particles. The exact expressions for the matrix elements of this operator in the momentum basis are

$$\langle n | \hat{x} | n' \rangle = \int_0^a dx \sqrt{\frac{2}{a}} \sin\left(\frac{n\pi x}{a}\right) x \sqrt{\frac{2}{a}} \sin\left(\frac{n'\pi x}{a}\right) = \begin{cases} \frac{a}{2} & \text{if } n = n' \\ -\frac{8ann'}{\pi^2(n^2 - n'^2)^2} & \text{if } n - n' \text{ is odd} \\ 0 & \text{otherwise} \end{cases} \quad (4.22)$$

This allows us to construct the exact matrix representations of the operator  $\hat{x}$  in both the momentum ( $\mathbf{xM}$ ) and the position ( $\mathbf{xP}$ ) bases:

```

1 In[434]:=xM = SparseArray[{
2     Band[{1,1}] -> a/2,
3     {n1_,n2_} /; OddQ[n1-n2] -> -8*a*n1*n2/(π^2*(n1^2-n2^2)^2)},
4     {nmax,nmax}];
5 In[435]:=xP = FourierDST[xM, 1];

```

A simple approximation of the position operator, which will be extremely useful in what follows, is found by observing that  $\mathbf{xP}$  is almost a diagonal matrix, with approximately the grid coordinates  $x_1, x_2, \dots, x_{n_{\max}}$  on the diagonal. This approximate form can be proved by using the locality of the position basis functions, Eq. (4.16):

$$\begin{aligned}
 x_{jj'} &= \langle j | \hat{x} | j' \rangle = \langle j | \left[ \int_{-\infty}^{\infty} dx |x\rangle x \langle x| \right] | j' \rangle = \int_0^a dx \langle j | x \rangle x \langle x | j' \rangle = \int_0^a dx \vartheta_j^*(x) x \vartheta_{j'}(x) \\
 &\approx x_{j'} \int_0^a dx \vartheta_j^*(x) \vartheta_{j'}(x) = \delta_{jj'} x_{j'}. \quad (4.23)
 \end{aligned}$$

The resulting approximate diagonal form of the position operator in the position basis, found from the approximate completeness relation  $\sum_{j=1}^{n_{\max}} |j\rangle \langle j| \approx \mathbb{1}$ , is

$$\hat{x} \approx \left[ \sum_{j=1}^{n_{\max}} |j\rangle \langle j| \right] \hat{x} \left[ \sum_{j'=1}^{n_{\max}} |j'\rangle \langle j'| \right] = \sum_{j,j'=1}^{n_{\max}} |j\rangle \langle j | \hat{x} | j' \rangle \langle j'| \approx \sum_{j,j'=1}^{n_{\max}} |j\rangle \delta_{jj'} x_j \langle j'| = \sum_{j=1}^{n_{\max}} |j\rangle x_j \langle j|. \quad (4.24)$$

```

1 In[436]:= Δ = a/(nmax+1); (* the grid spacing *)
2 In[437]:= xgrid = Range[nmax]*Δ; (* the computational grid *)
3 In[438]:= xP = SparseArray[Band[{1,1}] -> xgrid]; (* x operator, position basis *)
4 In[439]:= xM = FourierDST[xP, 1]; (* x operator, momentum basis *)

```

#### 4.1.4 The Potential-Energy Operator

If a potential energy function  $W(x)$  varies smoothly over length scales of the grid spacing  $\Delta$ , then the trick of Sect. 4.1.3 allows us to approximate the matrix elements of this potential energy in the position basis,

$$\begin{aligned}
 V_{jj'} &= \langle j | \hat{V} | j' \rangle = \langle j | \left[ \int_0^a dx |x\rangle W(x) \langle x| \right] | j' \rangle = \int_0^a dx \langle j | x \rangle W(x) \langle x | j' \rangle = \int_0^a dx \vartheta_j^*(x) W(x) \vartheta_{j'}(x) \\
 &\approx W(x_{j'}) \int_0^a dx \vartheta_j^*(x) \vartheta_{j'}(x) = \delta_{jj'} W(x_j), \quad (4.25)
 \end{aligned}$$

where we have used the definitions of Eqs. (4.2) and (4.4). This is a massive simplification compared to the explicit evaluation of potential integrals for each specific potential energy function. The potential-energy operator thus becomes approximately

$$\hat{V} \approx \left[ \sum_{j=1}^{n_{\max}} |j\rangle \langle j| \right] \hat{V} \left[ \sum_{j'=1}^{n_{\max}} |j'\rangle \langle j'| \right] = \sum_{j,j'=1}^{n_{\max}} |j\rangle \langle j | \hat{V} | j' \rangle \langle j'| \approx \sum_{j,j'=1}^{n_{\max}} |j\rangle \delta_{jj'} W(x_j) \langle j'| = \sum_{j=1}^{n_{\max}} |j\rangle W(x_j) \langle j|. \quad (4.26)$$

```

1 In[440]:= Wgrid = Map[W, xgrid]; (* the potential on the computational grid *)
2 In[441]:= VP = SparseArray[Band[{1,1}]->Wgrid]; (* potential operator, position basis *)
3 In[442]:= VM = FourierDST[VP, 1]; (* potential operator, momentum basis *)

```

### 4.1.5 The Kinetic-Energy Operator

The representation of the kinetic energy operator can be calculated very accurately with the description given above. We transform the definition of `In[423]` to the finite-resolution position basis with

```
1 In[443]:=TP = FourierDST[TM, 1]; (* kinetic operator, position basis *)
```

For large  $n_{\max}$  and small excitation energies the exact kinetic-energy operator can be replaced by the position-basis form

$$\langle j | \hat{T} | j' \rangle \approx \frac{\hbar^2}{2m\Delta^2} \times \begin{cases} 2 & \text{if } j = j', \\ -1 & \text{if } |j - j'| = 1, \\ 0 & \text{if } |j - j'| \geq 2, \end{cases} \quad (4.27)$$

which corresponds to replacing the second derivative in the kinetic operator by the finite-differences expression  $\psi''(x) \approx -[2\psi(x) - \psi(x - \Delta) - \psi(x + \Delta)] / \Delta^2$ . While Eq. (4.27) looks simple, it is ill suited for the calculations that will follow because (i) any matrix exponentials involving  $\hat{T}$  will be difficult to calculate, and (ii) it is not very accurate (higher-order finite-differences expressions<sup>6</sup> are not much better). Thus we will not be using such approximations in what follows, and prefer the more useful and more accurate definition through `In[423]` and `In[443]`.

### 4.1.6 The Momentum Operator

The discussion has so far been conducted in terms of the kinetic-energy operator  $\hat{T} = \hat{p}^2 / (2m)$  without explicitly talking about the momentum operator  $\hat{p} = -i\hbar \frac{d}{dx}$ . This was done because the matrix representation of the momentum operator is problematic. A direct calculation of the matrix elements in the momentum basis yields

$$\langle n | \hat{p} | n' \rangle = -i\hbar \int_0^a dx \phi_n(x) \frac{d\phi_{n'}(x)}{dx} = \frac{\hbar}{a} \times \begin{cases} \frac{4inn'}{n^2 - n'^2} & \text{if } n - n' \text{ is odd,} \\ 0 & \text{if } n - n' \text{ is even.} \end{cases} \quad (4.28)$$

In Mathematica, this is implemented with the definition

```
1 In[444]:=pM = SparseArray[{n1_,n2_}/;OddQ[n1-n2]->(4*I*h*n1*n2)/(a*(n2^2-n1^2)),
2 {nmax,nmax}]; (* momentum operator, momentum basis *)
3 In[445]:=pP = FourierDST[pM, 1]; (* momentum operator, position basis *)
```

This result is, however, unsatisfactory, since (i) it generates a matrix that is not sparse, and (ii) for a finite basis size  $n \leq n_{\max} < \infty$  it does not exactly generate the

<sup>6</sup>See [https://en.wikipedia.org/wiki/Finite\\_difference\\_coefficient](https://en.wikipedia.org/wiki/Finite_difference_coefficient) for explicit forms of higher-order finite-differences expressions that can be used to approximate derivatives.

kinetic-energy operator  $\hat{T} = \hat{p}^2/(2m)$  (see [Q4.3](#)). We will avoid using the momentum operator whenever possible, and use the kinetic-energy operator  $\hat{T}$  instead (see above). An example of the direct use of  $\hat{p}$  is given in [Sect. 5.2](#).

For large  $n_{\max}$  and small excitation energies the exact momentum operator can be replaced by the position-basis form

$$\langle j|\hat{p}|j'\rangle \approx \frac{i\hbar}{2\Delta} \times \begin{cases} -1 & \text{if } j - j' = -1, \\ +1 & \text{if } j - j' = +1, \\ 0 & \text{if } |j - j'| \neq 1, \end{cases} \quad (4.29)$$

which corresponds to replacing the first derivative in the momentum operator by the finite-differences expression  $\psi'(x) \approx [\psi(x + \Delta) - \psi(x - \Delta)]/(2\Delta)$ . While [Eq. \(4.29\)](#) looks simple, it is ill suited for the calculations that will follow because any matrix exponentials involving  $\hat{p}$  will still be difficult to calculate; further, the same finite-differences caveats as in [Sect. 4.1.5](#) apply. Thus we will not be using such approximations in what follows, and prefer the more accurate definition through [In \[444\]](#).

### Exercises

- Q4.3** Using  $n_{\max} = 100$ , calculate the matrix representations of the kinetic-energy operator  $\hat{T}$  and the momentum operator  $\hat{p}$  in the momentum basis. Compare the spectra of  $\hat{T}$  and  $\hat{p}^2/(2m)$  and notice the glaring differences, even at low energies. *Hint:* use natural units such that [a=m=ħ=1](#) for simplicity.
- Q4.4** Using  $n_{\max} = 20$ , calculate the matrix representations of the position operator  $\hat{x}$  and the momentum operator  $\hat{p}$  in the momentum basis. To what extent is the commutation relation  $[\hat{x}, \hat{p}] = i\hbar$  satisfied? *Hint:* use natural units such that [a=m=ħ=1](#) for simplicity.

### 4.1.7 Example: Gravity Well [\[Supplementary Material 1\]](#)

As an example of a single particle moving in one spatial dimension, we study the gravity well. This problem can be solved analytically, which helps us to determine the accuracy of our numerical methods.

We assume that a particle of mass  $m$  is free to move in the vertical direction  $x$ , where  $x = 0$  is the earth's surface and  $x > 0$  is up; the particle is forbidden from travelling below the earth's surface (i.e., it is restricted to  $x > 0$  at all times). There is no dissipation or friction. The Hamiltonian of the particle's motion is

$$\hat{\mathcal{H}} = -\frac{\hbar^2}{2m} \frac{d^2}{dx^2} + mg\hat{x}. \quad (4.30)$$

The wavefunction  $\psi(x)$  of this particle must satisfy the boundary condition  $\psi(x) = 0 \forall x \leq 0$ .



In what follows we use the length unit  $\mathcal{L} = \left(\frac{\hbar^2}{m^2 g}\right)^{1/3}$ , which is proportional to the size of the ground state of Eq. (4.30), as well as the mass unit  $\mathcal{M} = m$ . Following Sect. 4.1.1 we then define the time unit  $\mathcal{T} = \mathcal{M}\mathcal{L}^2/\hbar = \left(\frac{\hbar}{mg^2}\right)^{1/3}$  and the energy unit  $\mathcal{E} = \hbar/\mathcal{T} = (mg^2\hbar^2)^{1/3}$ . These natural units lead to simple expressions for the parameters:  $\mathbf{m} = \frac{m}{\mathcal{M}} = 1$ ,  $\mathbf{h} = \frac{\hbar}{\mathcal{E}\mathcal{T}} = 1$ , and  $\mathbf{g} = \frac{g}{\mathcal{L}/\mathcal{T}^2} = 1$ . As a result, we can set up the Hamiltonian (4.30) without fixing the particle's mass and gravitational acceleration explicitly:

```
1 In[446]:=m = h = g = 1;
```

Other systems of units can be used in the same way by using the tools of Sect. 4.1.1: first define a consistent set of units, and then express the physical quantities in terms of these units. The gravitational acceleration in particular would be set with

```
1 In[447]:=g = Quantity["StandardAccelerationOfGravity"]/(LengthUnit/TimeUnit^2);
```

### Analytic Quantum Energy Eigenstates

The exact normalized eigenstates and associated energy eigenvalues of Eq. (4.30) are

$$\psi_k(x) = \begin{cases} \left(\frac{2m^2g}{\hbar^2}\right)^{1/6} \cdot \frac{\text{Ai}\left[\alpha_k + x \cdot \left(\frac{2m^2g}{\hbar^2}\right)^{1/3}\right]}{\text{Ai}'(\alpha_k)} & \text{if } x > 0 \\ 0 & \text{if } x \leq 0 \end{cases} \quad E_k = -\alpha_k \cdot \left(\frac{mg^2\hbar^2}{2}\right)^{1/3} \quad (4.31)$$

for  $k = 1, 2, 3, \dots$ , where  $\text{Ai}(z) = \text{AiryAi}[z]$  is the Airy function,  $\text{Ai}'(z)$  its first derivative, and  $\alpha_k = \text{AiryAiZero}[k]$  its zeros:  $\alpha_1 \approx -2.33811$ ,  $\alpha_2 \approx -4.08795$ ,  $\alpha_3 \approx -5.52056$ , etc.


For comparison to numerical calculations below, we define the exact eigenstates and eigen-energies with

```
1 In[448]:=ψ[k_,x_] = (2*m^2*g/h^2)^(1/6)*AiryAi[AiryAiZero[k]+x*(2*m^2*g/h^2)^(1/3)]/
2               AiryAi'[AiryAiZero[k]];
3 In[449]:=ε[k_] = -AiryAiZero[k]*(m*g^2*h^2/2)^(1/3);
```

The ground-state energy is, in our chosen energy unit  $\mathcal{E}$ ,

```
1 In[450]:=N[ε[1]]
2 Out[450]=1.85576
```

The lowest three energy eigenstates look thus:



## Numerical Solution (I): Momentum Basis

Our first numerical attempt to find the ground state of the gravity well relies on the momentum basis of states  $|n\rangle$ . For this approach we treat the Hamiltonian of the problem in the same way as discussed in Chaps. 2 and 3: we express each term of the Hamiltonian as a matrix in a fixed basis set.

Since our calculation will take place in a finite box  $x \in [0, a]$ , we must choose the box size  $a$  large enough to contain most of the ground-state probability if we want to calculate it accurately. For the present calculation we choose  $a = 10\mathcal{L}$ , which is sufficient (see figure above) since we picked the length unit  $\mathcal{L}$  similar to the ground-state size:

```
1 In[451]:= a = 10;
```

We only use a small number of basis functions here, to illustrate the method:

```
2 In[452]:= nmax = 12;
```

The matrix elements of the kinetic energy are set up following In[423]

```
3 In[453]:= TM = SparseArray[Band[{1,1}] -> Range[nmax]^2*\pi^2*\hbar^2/(2*m*a^2)];
```

The matrix elements of the potential energy of Eq. (4.30) are, from In[434],

```
4 In[454]:= xM = SparseArray[{
5     Band[{1,1}] -> a/2,
6     {n1_,n2_} /; OddQ[n1-n2] -> -8*a*n1*n2/(pi^2*(n1^2-n2^2)^2)},
7     {nmax,nmax}];
8 In[455]:= VM = m*g*xM;
```

The full Hamiltonian in the momentum representation is therefore

```
9 In[456]:= HM = TM + VM;
```

and the ground-state energy and wavefunction coefficients in the momentum representation

```
10 In[457]:= gsM = -Eigensystem[-N[HM], 1,
11     Method -> {"Arnoldi", "Criteria" -> "RealPart", MaxIterations -> 10^6}];
```

The ground state energy is

```
12 In[458]:= gsM[[1, 1]]
13 Out[458]= 1.85608
```

very close to the exact result of Out[450].

The ground state wavefunction is defined as a sum over basis functions,

```
14 In[459]:= phi[n_, x_] = Sqrt[2/a]*Sin[n*\pi*x/a];
15 In[460]:= psi0[x_] = gsM[[2,1]] . Table[phi[n, x], {n, nmax}];
```

We can calculate the overlap of this numerical ground state with the exact one given in In[448],  $|\langle\psi_0|\psi_1\rangle|^2$ :

```
16 In[461]:= Abs[NIntegrate[psi0[x]*psi[1,x], {x, 0, a}]]^2
17 Out[461]= 0.999965
```

Even for  $n_{\max} = 12$  this overlap is already very close to unity in magnitude. It quickly approaches unity as  $n_{\max}$  increases, with the mismatch decreasing as  $n_{\max}^{-9}$  for this specific system. The numerically calculated ground-state energy approaches the exact result from above, with the mismatch decreasing as  $n_{\max}^{-7}$  for this specific system. These convergence properties, discussed in Sect. 2.1.1, are very general and allow us to extrapolate many quantities to  $n_{\max} \rightarrow \infty$  by polynomial fits of numerically calculated quantities as functions of  $n_{\max}$ .

### Numerical Solution (II): Mixed Basis

The numerical method outlined above only works because we have an analytic expression for the matrix elements of the potential operator  $\hat{V} = mg\hat{x}$ , given in Eq. (4.22). For a more general potential, the method of Eq. (4.26) is more useful, albeit less accurate. Here we re-do the numerical ground-state calculation in the position basis. The computation is set up in the same way as above,

```
1 In[462]:=a = 10;
2 In[463]:=nmax = 12;
3 In[464]:=Δ = a/(nmax+1); (* grid spacing *)
4 In[465]:=xgrid = Range[nmax]*Δ; (* the computational grid *)
```

The matrix elements of the kinetic-energy operator in the position basis are calculated with a discrete sine transform,

```
5 In[466]:=TM = SparseArray[Band[{1,1}] -> Range[nmax]^2*π^2*ħ^2/(2*m*a^2)];
6 In[467]:=TP = FourierDST[TM, 1];
```

The matrix elements of the potential energy of in Eq. (4.30) are, from In[440],

```
7 In[468]:=W[x_] = m*g*x; (* the potential function *)
8 In[469]:=Wgrid = Map[W, xgrid]; (* the potential on the computational grid *)
9 In[470]:=VP = SparseArray[Band[{1,1}] -> Wgrid];
```

The full Hamiltonian in the position representation is therefore

```
10 In[471]:=HP = TP + VP;
```

and the ground-state energy and wavefunction coefficients in the position representation

```
11 In[472]:=gsP = -Eigensystem[-N[HP], 1,
12 Method -> {"Arnoldi", "Criteria" -> "RealPart", MaxIterations -> 10^6}];
```

The ground state energy is now less close to the exact value than before, due to the additional approximation of Eq. (4.26):


```
13 In[473]:=gsP[[1, 1]]
14 Out[473]=1.86372
```

We therefore need a larger  $n_{\max}$  to achieve the same accuracy as in the first numerical calculation. The great advantage of the present calculation is, however, that it is easily generalized to arbitrary potential-energy functions in In[468].

As shown in In[428], the wavefunction can be plotted approximately with

```
15 In[474]:=γ = Join[{0,0}], Transpose[{xgrid, gsP[[2,1]]/Sqrt[Δ]], {{a,0}}];
16 In[475]:=ListLinePlot[γ]
```

where we have “manually” added the known boundary values  $\gamma(0) = \gamma(a) = 0$  to the list of numerically calculated wave-function values.



You can see that even with  $n_{\max} = 12$  grid points this ground-state wavefunction (blue lines interpolating between blue calculated points) looks remarkably close to the exact one (orange line, see plot on page 103).

If we need to go beyond linear interpolation, the precise wavefunction is calculated by converting to the momentum representation as in [In\[431\]](#) and multiplying with the basis functions as in [In\[460\]](#):

```
17 In[476]:= φ[n_, x_] = Sqrt[2/a]*Sin[n*π*x/a];
18 In[477]:= ψ0[x_] = FourierDST[gs[[2,1]],1] . Table[φ[n, x], {n, nmax}];
```

## Exercises

- Q4.5** What is the probability to find the particle below  $x = 1$  (i.e., below  $x = \mathcal{L}$ ) when it is in the ground state of the gravity well, Eq. 4.30? Calculate analytically, with numerical method I, and with numerical method II.
- Q4.6** Calculate the mean height  $\langle \hat{x} \rangle$  in the ground state of the gravity well. How large is this quantity for a neutron in earth’s gravitational field? *Hint:* see *Quantum states of neutrons in the Earth’s gravitational field* by Valery V. Nesvizhevsky *et al.*, Nature 415, pages 297–299 (2002).
- Q4.7** Calculate the energy levels and energy eigenstates of a particle in a harmonic potential, described by the Hamiltonian

$$\hat{\mathcal{H}} = -\frac{\hbar^2}{2m} \frac{d^2}{dx^2} + \frac{1}{2} m \omega^2 \hat{x}^2. \quad (4.32)$$

Do the calculated energy levels match the analytically known values? *Hint:* use the system of units given in [In\[409\]](#) ff with a length unit of  $\mathcal{L} = \sqrt{\hbar/(m\omega)}$ , a mass unit  $\mathcal{M} = m$ , and an energy unit  $\mathcal{E} = \hbar\omega$  (i.e., the natural units). Choose the calculation box with size  $a = 10\mathcal{L}$  and shift the minimum of the harmonic potential to the center of the calculation box.

### 4.1.8 The Wigner Quasi-probability Distribution [Supplementary Material 2]

The Wigner quasi-probability distribution<sup>7</sup> of a wavefunction  $\psi(x)$  is a real-valued distribution in phase space defined as

$$W(x, k) = \frac{1}{\pi} \int_{-\infty}^{\infty} dy \psi(x - y) \psi^*(x + y) e^{2iky}, \quad (4.33)$$

where  $k = p/\hbar$  is the wavenumber, closely related to the momentum but in units of inverse length.  $W$  often makes it easier to interpret wavefunctions than simply plotting  $\psi(x)$ , especially when  $\psi(x)$  is complex-valued. Time-dependent wavefunctions are often plotted as Wigner distribution movies, which makes it easier to track a particle as it moves through phase space. In the classical limit, the time-dependent Wigner distribution becomes the classical phase-space density that satisfies the Liouville equation.

For a quick and easy evaluation of the Wigner distribution, we approximate the wavefunction as piecewise constant, using Eq. (4.13):  $\psi(x) \approx \psi(x_{[x/\Delta]}) = v_{[x/\Delta]}/\sqrt{\Delta}$ , where we have used the calculation grid spacing  $\Delta = a/(n_{\max} + 1)$  and the nearest-integer rounding function  $[z] = \text{round}(z)$ . This approximation will be valid as long as  $|k| \ll \pi/\Delta$ . Inserting it into Eq. (4.33), and assuming that  $x = x_j = j\Delta$  is a grid point (i.e., we will only sample the Wigner function on the spatial grid of the calculation), we can split the integral over  $y$  into integrals over segments of length  $\Delta$ ,

$$\begin{aligned} W(x_j, k) &= \frac{1}{\pi} \sum_{m=-\infty}^{\infty} \int_{-\Delta/2}^{\Delta/2} dy \psi(x_j - (m\Delta + y)) \psi^*(x_j + (m\Delta + y)) e^{2ik(m\Delta + y)} \\ &\approx \frac{1}{\pi} \sum_{m=-\infty}^{\infty} \int_{-\Delta/2}^{\Delta/2} dy \psi(x_{j-m}) \psi^*(x_{j+m}) e^{2ik(m\Delta + y)} = \frac{1}{\pi} \sum_{m=-\infty}^{\infty} \underbrace{\psi(x_{j-m})}_{\frac{v_{j-m}}{\sqrt{\Delta}}} \underbrace{\psi^*(x_{j+m})}_{\frac{v_{j+m}^*}{\sqrt{\Delta}}} e^{2ikm\Delta} \underbrace{\int_{-\Delta/2}^{\Delta/2} dy e^{2iky}}_{\text{sinc}(k\Delta)/k} \\ &= \frac{\text{sinc}(k\Delta)}{\pi} \sum_{m=-\min(j-1, n_{\max}-j)}^{\min(j-1, n_{\max}-j)} v_{j-m} v_{j+m}^* e^{2ikm\Delta}, \end{aligned} \quad (4.34)$$

where  $\text{sinc}(z) = \sin(z)/z$ . The following Mathematica code converts a coefficient vector  $\vec{v}$  of length  $n_{\max}$  into a function of the dimensionless momentum  $\kappa = a \cdot k$  that calculates  $W(x_j, k)$  for every grid point  $j = 0, 1, \dots, n_{\max} + 1$  (including the boundary grid points that are usually left out of our calculations):

```
In[478]:=WignerDistribution[v_?VectorQ] := With[{nmax = Length[v]},
Function[κ, Evaluate[Sinc[κ/(nmax+1)]]/π*Table[
Sum[v[[j-m]]*Conjugate[v[[j+m]]]*Exp[2*I*κ*m/(nmax+1)],
{m,-Min[j-1,nmax-j],Min[j-1,nmax-j]}]]/Re[ComplexExpand, {j,0,nmax+1}]]]
```

<sup>7</sup>See [https://en.wikipedia.org/wiki/Wigner\\_quasiprobability\\_distribution](https://en.wikipedia.org/wiki/Wigner_quasiprobability_distribution).

Notice that this function `WignerDistribution` returns an anonymous function (see Sect. 1.5.3) of one parameter, which in turn returns a list of values. As an example of its use, we make a 2D plot of the Wigner distribution on the interval  $x \in [x_{\min}, x_{\max}]$ <sup>8</sup>:

```


1 In[479]:=WignerDistributionPlot[Y_,
2           {xmin_?NumericQ, xmax_?NumericQ} /; xmax > xmin] :=
3           Module[{nmax, qmax, w, W},
4             (* number of grid points *)
5             nmax = Length[Y];
6             (* calculate the Wigner distribution *)
7             w = WignerDistribution[Y];
8             (* evaluate it on the natural dimensionless momentum grid *)
9             qmax = Floor[nmax/2];
10            W = Table[w[q*π], {q, -qmax, qmax}];
11            (* make a plot *)
12            ArrayPlot[W, FrameTicks->Automatic, AspectRatio -> 1/GoldenRatio,
13                      DataRange->{{xmin,xmax},qmax*π/(xmax-xmin)*{-1,1}},
14                      ColorFunctionScaling -> False,
15                      ColorFunction -> (Blend[{Blue, White, Red}, (π*#+1)/2]&)]]
```

Notice that we evaluate the Wigner distribution only up to momenta  $\pm n_{\max}\pi/(2a)$ , which is the Nyquist limit in this finite-resolution system.<sup>9</sup> The color scheme is chosen such that the Wigner distribution values range  $[-\frac{1}{\pi}, +\frac{1}{\pi}]$  is mapped onto the colors blended from blue, white, and red, such that negative Wigner values are shown in shades of blue while positive values are shown in shades of red.

As an example, we plot the Wigner distribution of the numerical ground-state wavefunction shown on page 106: on the left, the exact distribution from Eq. (4.33); on the right, the grid evaluation of `In[478]` and `In[479]` (calculated with  $n_{\max} = 40$ ) with

```

1 In[480]:=WignerDistributionPlot[gsP[[2, 1]], {0, a}]
```



### Extension to Density Operators

If the state of the system is not pure, but given as a density matrix  $\rho(x, x') = \langle x | \hat{\rho} | x' \rangle$  instead of as a wavefunction  $\psi(x) = \langle x | \psi \rangle$ , then we do not have the option of plotting the wavefunction and we can only resort to the Wigner distribution for a graphical representation.

Noticing that Eq. (4.33) contains the term

<sup>8</sup>This procedure works for situations other than the usual  $x_{\min} = 0$  and  $x_{\max} = a$ .

<sup>9</sup>See [https://en.wikipedia.org/wiki/Nyquist\\_frequency](https://en.wikipedia.org/wiki/Nyquist_frequency).

$$\psi(x-y)\psi^*(x+y) = \langle x-y|\psi\rangle\langle\psi|x+y\rangle = \langle x-y|\hat{\rho}|x+y\rangle = \rho(x-y, x+y) \quad (4.35)$$

in terms of the pure-state density operator  $\hat{\rho} = |\psi\rangle\langle\psi|$ , the definition of the Wigner distribution is generalized to

$$W(x, k) = \frac{1}{\pi} \int_{-\infty}^{\infty} dy \rho(x-y, x+y) e^{2iky}. \quad (4.36)$$

We can make the same approximations as in Eq. (4.34) to calculate the Wigner function on a spatial grid point  $x = x_j$  [see Eq. (4.15)]:

$$\begin{aligned} W(x_j, k) &= \frac{1}{\pi} \sum_{m=-\infty}^{\infty} \int_{-\Delta/2}^{\Delta/2} dy \rho[x_j - (m\Delta + y), x_j + (m\Delta + y)] e^{2ik(m\Delta + y)} \\ &\approx \frac{1}{\pi} \sum_{m=-\infty}^{\infty} \int_{-\Delta/2}^{\Delta/2} dy \rho(x_{j-m}, x_{j+m}) e^{2ik(m\Delta + y)} \\ &= \frac{\text{sinc}(k\Delta)}{\pi} \sum_{m=-\min(j-1, n_{\max}-j)}^{\min(j-1, n_{\max}-j)} R_{j-m, j+m} e^{2ikm\Delta}. \end{aligned} \quad (4.37)$$

In analogy to In[478] we define

```

1 In[481]:=WignerDistribution[R_ /; MatrixQ[R, NumericQ] &&
2   Length[R] == Length[Transpose[R]]] :=
3   With[{n = Length[R]},
4     Function[k, Evaluate[Sinc[k/(n+1)]/π*Table[
5       Sum[R[[j-m, j+m]]*Exp[2*I*k*m/(n+1)],
6       {m, -Min[j-1, n-j], Min[j-1, n-j]}]]/Re//ComplexExpand, {j, 0, n+1}]]]]

```

For a pure state, the density matrix has the coefficients  $R_{j,j'} = v_j v_{j'}^*$ , and the definitions of In[478] and In[481] thus give exactly the same result if we use

```

1 In[482]:=R = KroneckerProduct[v, Conjugate[v]]

```

In addition, the 2D plotting function of In[479] also works when called with a density matrix as first parameter.

## Exercises

**Q4.8** Plot the Wigner distribution of the first excited state of the gravity well. What do you notice?

### 4.1.9 1D Dynamics in the Square Well [Supplementary Material 3]

Assume again a single particle of mass  $m$  moving in a one-dimensional potential, with the time-independent Hamiltonian given in Eq. (4.1). The motion is again restricted

to  $x \in [0, a]$ . We want to study the time-dependent wavefunction  $\psi(x, t) = \langle x | \psi(t) \rangle$  given in Eq. (2.34),

$$|\psi(t)\rangle = \exp\left[-\frac{i(t-t_0)}{\hbar}\hat{\mathcal{H}}\right]|\psi(t_0)\rangle. \quad (4.38)$$

The simplest way of computing this propagation is to express the wavefunction and the Hamiltonian in a particular basis and use a matrix exponentiation to find the time dependence of the expansion coefficients of the wavefunction. For example, if we use the finite-resolution position basis, we have seen on page 105 how to find the matrix representation of the Hamiltonian, [HP](#). For a given initial wavefunction represented by a position-basis coefficient vector [v0](#) we can then define

```
In[483]:=v[Δt_?NumericQ] := MatrixExp[-I*HP*Δt/ħ].v0
```

as the propagation over a time interval  $\Delta t = t - t_0$ . If you try this, you will see that calculating  $|\psi(t)\rangle$  in this way is not very efficient, because the matrix exponentiation is a numerically difficult operation.

A much more efficient method can be found by using the Trotter expansion

$$e^{\lambda(X+Y)} = e^{\frac{\lambda}{2}X} e^{\lambda Y} e^{\frac{\lambda}{2}X} \times e^{\frac{\lambda^3}{24}[X,[X,Y]] + \frac{\lambda^3}{12}[Y,[X,Y]]} \times e^{-\frac{\lambda^4}{48}[X,[X,[X,Y]]] - \frac{\lambda^4}{16}[X,[Y,[X,Y]]] - \frac{\lambda^4}{24}[Y,[Y,[X,Y]]]} \dots$$

$$\approx e^{\frac{\lambda}{2}X} e^{\lambda Y} e^{\frac{\lambda}{2}X}, \quad (4.39)$$

where the approximation is valid for small  $\lambda$  since the neglected terms are of third and higher orders in  $\lambda$  (notice that there is no second-order term in  $\lambda$ !). Setting  $\lambda = -\frac{i(t-t_0)}{M\hbar}$  for some large integer  $M$ , as well as  $X = \hat{V}$  and  $Y = \hat{T}$ , we find

$$|\psi(t)\rangle = e^{M\lambda\hat{\mathcal{H}}}|\psi(t_0)\rangle = \left[e^{\lambda\hat{\mathcal{H}}}\right]^M |\psi(t_0)\rangle = \left[e^{\lambda(\hat{T}+\hat{V})}\right]^M |\psi(t_0)\rangle \stackrel{\text{Trotter Eq. (4.39)}}{\underset{\text{}}{=}} \lim_{M \rightarrow \infty} \left[e^{\frac{\lambda}{2}\hat{V}} e^{\lambda\hat{T}} e^{\frac{\lambda}{2}\hat{V}}\right]^M |\psi(t_0)\rangle. \quad (4.40)$$

This can be evaluated very efficiently. We express the potential Hamiltonian in the finite-resolution position basis, Eq. (4.26), the kinetic Hamiltonian in the momentum basis, Eq. (4.9), and the time-dependent wavefunction in both bases of Eq. (4.17):

$$|\psi(t)\rangle = \sum_{n=1}^{n_{\max}} u_n(t)|n\rangle = \sum_{j=1}^{n_{\max}} v_j(t)|j\rangle \quad (4.41a)$$

$$\hat{V} \approx \sum_{j=1}^{n_{\max}} W(x_j)|j\rangle\langle j| \quad (4.41b)$$

$$\hat{T} \approx \sum_{n=1}^{n_{\max}} \frac{n^2 \pi^2 \hbar^2}{2ma^2}|n\rangle\langle n| \quad (4.41c)$$



The expansion coefficients of the wavefunction are related by a type-I discrete sine transform, see Eq. (4.19), Eq. (4.21), In[430], and In[431].

The great advantage of the diagonal matrices of Eq. (4.41b) and Eq. (4.41c) is that algebra with diagonal matrices is as simple as algebra with scalars, but applied to the diagonal elements one-by-one. In particular, for any diagonal matrix  $D = \sum_j d_j |j\rangle\langle j|$  the integer matrix powers are  $D^k = \sum_j d_j^k |j\rangle\langle j|$ , and matrix exponentials are calculated by exponentiating each diagonal element separately:  $\exp(D) = \sum_{k=0}^{\infty} D^k/k! = \sum_{k=0}^{\infty} (\sum_j d_j^k |j\rangle\langle j|)/k! = \sum_j (\sum_{k=0}^{\infty} d_j^k/k!) |j\rangle\langle j| = \sum_j \exp(d_j) |j\rangle\langle j|$ . As a result,

$$e^{\frac{\lambda}{2}\hat{V}} = \sum_{j=1}^{n_{\max}} e^{\frac{\lambda}{2}W(x_j)} |j\rangle\langle j|, \quad (4.42)$$

and the action of the potential Hamiltonian thus becomes straightforward:

$$e^{\frac{\lambda}{2}\hat{V}} |\psi(t)\rangle = \left[ \sum_{j=1}^{n_{\max}} e^{\frac{\lambda}{2}W(x_j)} |j\rangle\langle j| \right] \left[ \sum_{j'=1}^{n_{\max}} v_{j'}(t) |j'\rangle \right] = \sum_{j,j'=1}^{n_{\max}} e^{\frac{\lambda}{2}W(x_j)} v_{j'}(t) |j\rangle\langle j'| = \sum_{j=1}^{n_{\max}} \underbrace{\left[ e^{\frac{\lambda}{2}W(x_j)} v_j(t) \right]}_{v'_j} |j\rangle, \quad (4.43)$$

which is an element-by-element multiplication of the coefficients of the wavefunction with the exponentials of the potential—no matrix operations are required. The expansion coefficients (position basis) after propagation with the potential Hamiltonian for a “time” step  $\lambda/2$  are therefore

$$v'_j = e^{\frac{\lambda}{2}W(x_j)} v_j. \quad (4.44)$$

The action of the kinetic Hamiltonian in the momentum representation is found in exactly the same way:

$$e^{\lambda\hat{T}} |\psi(t)\rangle = \left[ \sum_{n=1}^{n_{\max}} e^{\lambda \frac{n^2\pi^2\hbar^2}{2ma^2}} |n\rangle\langle n| \right] \left[ \sum_{n'=1}^{n_{\max}} u_{n'}(t) |n'\rangle \right] = \sum_{n=1}^{n_{\max}} \underbrace{\left[ e^{\lambda \frac{n^2\pi^2\hbar^2}{2ma^2}} u_n(t) \right]}_{u'_n} |n\rangle. \quad (4.45)$$

The expansion coefficients (momentum basis) after propagation with the kinetic Hamiltonian for a “time” step  $\lambda$  are therefore

$$u'_n = e^{\lambda \frac{n^2\pi^2\hbar^2}{2ma^2}} u_n. \quad (4.46)$$

We know that a type-I discrete sine transform brings the wavefunction from the finite-resolution position basis to the momentum basis and vice-versa. The propagation under the kinetic Hamiltonian thus consists of

1. a type-I discrete sine transform to calculate the coefficients  $v_j \mapsto u_n$ ,
2. an element-by-element multiplication, Eq. (4.46), to find the coefficients  $u_n \mapsto u'_n$ ,
3. and a second type-I discrete sine transform to calculate the coefficients  $u'_n \mapsto v'_j$ .

Here we assemble all these pieces into a program that propagates a state  $|\psi(t_0)\rangle$ , which is given as a coefficient vector  $\vec{v}$  in the finite-resolution position basis, forward in time to  $t = t_0 + \Delta t$ . First, for reference, a procedure for the exact propagation by matrix exponentiation and matrix–vector multiplication, as in [In\[483\]](#):

```

1 In[484]:=VP = SparseArray[Band[{1,1}]->Wgrid];
2 In[485]:=TM = SparseArray[Band[{1,1}]->Range[nmax]^2*\pi^2*\hbar^2/(2*m*a^2)];
3 In[486]:=TP = FourierDST[TM, 1];
4 In[487]:=HP = TP + VP;
5 In[488]:=propExact[\Delta_?NumericQ, v0_ /; VectorQ[v0, NumericQ]] :=
6     MatrixExp[-I*HP*N[\Delta/\hbar]].v0

```

Next, an iterative procedure that propagates by  $M$  small steps via the Trotter approximation, Eq. (4.39):

```

1 In[489]:=propApprox[\Delta_?NumericQ, M_Integer /; M >= 1,
2     v0_ /; VectorQ[v0, NumericQ]] :=
3     Module[{λ, Ke, Pe2, propKin, propPot2, prop},
4         (* compute the λ constant *)
5         λ = -I*N[\Delta/(M*\hbar)];
6         (* compute the diagonal elements of exp[λ*T] *)
7         Ke = Exp[λ*Range[nmax]^2*\pi^2*\hbar^2/(2*m*a^2)];
8         (* propagate by a full time-step with T *)
9         propKin[v_] := FourierDST[Ke*FourierDST[v, 1], 1];
10        (* compute the diagonal elements of exp[λ*V/2] *)
11        Pe2 = Exp[λ/2*Wgrid];
12        (* propagate by a half time-step with V *)
13        propPot2[v_] := Pe2*v;
14        (* propagate by a full time-step by H=T+V *)
15        (* using the Trotter approximation *)
16        prop[v_] := propPot2[propKin[propPot2[v]]];
17        (* step-by-step propagation *)
18        Nest[prop, v0, M]]

```

Notice that there are no basis functions, integrals, etc. involved in this calculation; everything is done in terms of the values of the wavefunction on the grid  $x_1 \dots x_{n_{\max}}$ . This efficient method is called *split-step propagation*.

The `Nest` command “nests” a function call: for example, `Nest[f, x, 3]` calculates  $f(f(f(x)))$ . We use this on line 18 of [In\[489\]](#) to repeatedly propagate by small time steps via the Trotter approximation. Since this algorithm internally calculates the wavefunction at all the intermediate times  $t = t_0 + \frac{m}{M}(t - t_0)$  for  $m = 1, 2, 3, \dots, M$ , we can modify our program in order to follow this time evolution. To achieve this we simply replace the `Nest` command with `NestList`, which is similar to `Nest` but returns all intermediate results: for example, `NestList[f, x, 3]` returns the list  $\{x, f(x), f(f(x)), f(f(f(x)))\}$ . We replace last line of the code above with

```

18 Transpose[{Range[0, M]/M*\Delta, NestList[prop, v0, M]}]

```

which now returns a list of pairs containing (i) the time and (ii) the wavefunction at the corresponding time.


### Example: Bouncing in the Gravity Well

As an example of particle dynamics, we return to the gravity well of Sect. 4.1.7. Classically, if we drop a particle from height  $x_0$  at  $t = 0$  under the influence of gravity, then its trajectory is  $x(t) = x_0 - \frac{1}{2}gt^2$ , until it reaches the earth's surface ( $x = 0$ ) at time  $t_1 = \sqrt{2x_0/g}$ . We plot this classical bouncing trajectory for a scaled starting height  $x_0 = 15$  with

```
1 In[490]:=With[{x0 = 15, Δt = 50}, {t1 = Sqrt[2*x0/g]},
2          Plot[x0 - Mod[t, 2*t1, -t1]^2/2, {t, 0, Δt}]]
```


In order to simulate a quantum particle bouncing along this trajectory, we start at the same height  $x_0 = 15$  but assume that the particle initially has a wavefunction of root-mean-square width  $\sigma = 1$ : the initial state in the position basis is

```
1 In[491]:=x0 = 15; (* starting height *)
2 In[492]:=σ = 1; (* starting width *)
3 In[493]:=t1 = Sqrt[2*x0/g]; (* classical bounce time *)
4 In[494]:=vv = Normalize[N[Exp[-((xgrid-x0)/(2*σ))^2]]]; (* starting state *)
5 In[495]:=ListLinePlot[Join[{0,0}],Transpose[{xgrid,vv/Sqrt[Δ]}],{a,0}},
6          PlotRange->All]
```



We propagate this particle in time for  $\Delta t = 50$  time units, using  $M = 1000$  time steps, and plot the time-dependent density  $\rho(x, t) = |\psi(x, t)|^2 = |\langle x | \psi(t) \rangle|^2$  using the trick of Eq. (4.13):

```
1 In[496]:=With[{Δt = 50, M = 1000},
2          ρ = ArrayPad[Abs[propApprox[Δt, M, vv][[All,2]]]^2/Δt, {{0, 0}, {1, 1}}],
3          ArrayPlot[Reverse[Transpose[ρ]], DataRange -> {{0, Δt}, {0, a}}]]
```



The orange overlaid curve shows the classical particle trajectory from [In\[490\]](#), which the quantum particle follows approximately while self-interfering during the reflections.


To study the correspondence between classical and quantum-mechanical motion more quantitatively, we can calculate and plot time-dependent quantities such as the time-dependent mean position: using Eq. (4.24),

$$\langle \hat{x} \rangle(t) = \langle \psi(t) | \hat{x} | \psi(t) \rangle \approx \langle \psi(t) | \left[ \sum_{j=1}^{n_{\max}} |j\rangle x_j \langle j| \right] | \psi(t) \rangle = \sum_{j=1}^{n_{\max}} x_j |v_j(t)|^2. \quad (4.47)$$

```

1 In[497]:=With[{Δt = 50, M = 1000},
2   ListLinePlot[{{#[[1]], Abs[#[[2]]]^2.xgrid} & /@ propApprox[Δt, M, vv]]]

```



Here the quantum deviations from the classical trajectory (orange) become apparent.

### 4.1.10 1D Dynamics in a Time-Dependent Potential

While the direct propagation of Eq. 2.34 only works for time-independent Hamiltonians, the split-step method of In[489] can be extended to time-dependent Hamiltonians, in particular to time-dependent potentials  $W(x, t)$ . For this, we assume that the potential varies slowly enough in time that it is almost constant during a Trotter step  $\Delta t/M$ ; this assumption usually becomes exact as  $M \rightarrow \infty$ .

```

1 In[498]:=propApprox[Wt_,
2           Δt_?NumericQ, M_Integer /; M >= 1,
3           v0_ /; VectorQ[v0, NumericQ]] :=
4           Module[{λ, Ke, propKin, propPot2, prop},
5             (* compute the λ constant *)
6             λ = -I*N[Δt/(M*ħ)];
7             (* compute the diagonal elements of exp[λ*T] *)
8             Ke = Exp[λ*Range[nmax]^2*π^2/2];
9             (* propagate by a full time-step with T *)
10            propKin[v_] := FourierDST[Ke*FourierDST[v, 1], 1];
11            (* propagate by a half time-step with V *)
12            (* evaluating the potential at time t *)
13            propPot2[t_, v_] := Exp[λ/2*(Wt[#,&t]/@xgrid)]*v;
14            (* propagate by a full time-step by H=T+V *)
15            (* using the Trotter approximation *)
16            (* starting at time t *)
17            prop[v_, t_] := propPot2[t+3Δt/(4M), propKin[propPot2[t+Δt/(4M), v]]];
18            (* step-by-step propagation *)
19            Transpose[{Range[0, M]/M*Δt, FoldList[prop, v0, Range[0,M-1]/M*Δt]}]

```


- The definition of `propApprox` now needs a time-dependent potential `Wt[x, t]` that it can evaluate as the propagation proceeds. This potential must be specified as a pure function with two arguments, as in the example below.
- The exponentials for the potential propagation, calculated once-and-for-all on line 11 of In[489], are now re-calculated in each call of the `propPot2` function.
- In the Trotter propagation step of Eq. 4.40 we evaluate the potential twice in each propagation interval  $[t, t + \Delta t/M]$ : once at  $t + \frac{1}{4}\Delta t/M$  for the first half-step with the potential operator  $\hat{V}$ , and once at  $t + \frac{3}{4}\Delta t/M$  for the second half-step.
- On line 19 of In[498] we have replaced `NestList` by `FoldList`, which is more flexible: for example, `FoldList[f, x, {a, b, c}]` calculates the list  $\{x, f(x, a), f(f(x, a), b), f(f(f(x, a), b), c)\}$ . By giving the list of propagation interval starting times as the last argument of `FoldList`, the `prop` function is called repeatedly, with the current interval starting time as the second argument.

As an example, we calculate the time-dependent density profile under the same conditions as above, except that the gravitational acceleration is modulated periodically:  $W(x, t) = W(x) \cdot [1 + A \cdot \sin(\omega t)]$ . The oscillation frequency  $\omega = \pi/t_1$  is chosen to drive the bouncing particle resonantly and enhance its amplitude. This time-dependent potential is passed as the first argument to `propApprox`:

```

1 In[499]:=With[{A = 0.1, ω = π/t1, Δt = 50, M = 1000},
2   Wt[x_, t_] = W[x]*(1 + A*Sin[ω*t]);
3   ρ = ArrayPad[Abs[propApprox[Wt, Δt, M, vv][[All, 2]]]^2/Δ, {{0, 0}, {1, 1}}];
4   ArrayPlot[Reverse[Transpose[ρ]], DataRange -> {{0, Δt}, {0, a}}]

```



The increase in bouncing amplitude can be seen clearly in this density plot.

### Exercises

**Q4.9** Convince yourself that the Trotter expansion of Eq. 4.39 is really necessary, i.e., that  $e^{X+Y} \neq e^X e^Y$  if  $X$  and  $Y$  do not commute. *Hint:* use two concrete non-commuting objects  $X$  and  $Y$ , for example two random  $2 \times 2$  matrices as generated with `RandomReal[{0, 1}, {2, 2}]`.

**Q4.10** Given a particle moving in the range  $x \in [0, a]$  with the Hamiltonian

$$\hat{\mathcal{H}} = -\frac{\hbar^2}{2m} \frac{d^2}{dx^2} + W_0 \sin(10\pi x/a), \quad (4.48)$$

compute its time-dependent wavefunction starting from a “moving Gaussian”  $\psi(t=0) \propto e^{-\frac{(x-a/2)^2}{4\sigma^2}} e^{ikx}$  with  $\sigma = 0.05a$  and  $k = 100/a$ . Study  $\langle \hat{x} \rangle(t)$  using first  $W_0 = 0$  and then  $W_0 = 5000 \frac{\hbar^2}{ma^2}$ . *Hint:* use natural units such that  $a=m=\hbar=1$  for simplicity.

## 4.2 Many Particles in One Dimension: Dynamics with the Non-linear Schrödinger Equation

The advantage of the split-step evolution of Eq. 4.40 becomes particularly clear when the system’s energy depends on the wavefunction in a more complicated way than in the usual time-independent Schrödinger equation. A widely used example is the

nonlinear energy functional<sup>10</sup>

$$E[\psi] = \underbrace{-\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} dx \psi^*(x) \psi''(x)}_{E_{\text{kin}}[\psi]} + \underbrace{\int_{-\infty}^{\infty} dx V(x) |\psi(x)|^2}_{E_{\text{pot}}[\psi]} + \underbrace{\frac{\kappa}{2} \int_{-\infty}^{\infty} dx |\psi(x)|^4}_{E_{\text{int}}[\psi]}, \quad (4.49)$$

in which the last term describes the *mean-field interactions* between  $N$  particles that are all in wavefunction  $\psi(x)$  (normalized to  $\int_{-\infty}^{\infty} dx |\psi(x)|^2 = 1$ ), and which are therefore in a joint product wavefunction  $\psi(x)^{\otimes N}$  (see Eq. 2.39). Each particle sees a potential  $V_{\text{int}}(x) = \frac{\kappa}{2} |\psi(x)|^2$  generated by the average density  $(N-1)|\psi(x)|^2$  of other particles with the same wavefunction, usually through collisional interactions. In three dimensions, the coefficient  $\kappa = (N-1) \times 4\pi\hbar^2 a_s / m$  approximates the mean-field  $s$ -wave scattering between a particle and the  $(N-1)$  other particles, with  $s$ -wave scattering length  $a_s$  (see Sect. 4.4); in the present one-dimensional example, no such identification is made.

In order to find the ground state (energy minimum) of Eq. 4.49 under the constraint of wavefunction normalization  $\int_{-\infty}^{\infty} dx |\psi(x)|^2 = 1$ , we use the Lagrange multiplier<sup>11</sup> method: using the Lagrange multiplier  $\mu$  called the *chemical potential*, we conditionally minimize the energy with respect to the wavefunction by setting its functional derivative<sup>12</sup>

$$\frac{\delta}{\delta \psi^*(x)} \left( E[\psi] - \mu \int_{-\infty}^{\infty} dx |\psi(x)|^2 \right) = -\frac{\hbar^2}{m} \psi''(x) + 2V(x)\psi(x) + 2\kappa |\psi(x)|^2 \psi(x) - 2\mu \psi(x) = 0 \quad (4.50)$$

to zero. This yields the non-linear Schrödinger equation

$$\left[ -\frac{\hbar^2}{2m} \frac{d^2}{dx^2} + \underbrace{V(x) + \kappa |\psi(x)|^2}_{V_{\text{eff}}(x)} \right] \psi(x) = \mu \psi(x), \quad (4.51)$$

also called the *Gross–Pitaevskii equation* for the description of dilute Bose–Einstein condensates. By analogy to the linear Schrödinger equation, it also has a time-dependent form for the description of Bose–Einstein condensate dynamics,

$$i\hbar \frac{\partial \psi(x, t)}{\partial t} = \left[ -\frac{\hbar^2}{2m} \frac{\partial^2}{\partial x^2} + \underbrace{V(x, t) + \kappa |\psi(x, t)|^2}_{V_{\text{eff}}(x, t)} \right] \psi(x, t). \quad (4.52)$$

For any  $\kappa \neq 0$  there is no solution of the form of Eq. 4.38. But the split-step method of Eq. 4.40 can still be used to simulate Eq. 4.52 because the

<sup>10</sup>A functional is an operation that calculates a number from a given function. For example,  $E[\psi] : L^2 \rightarrow \mathbb{R}$  converts a wavefunction  $\psi \in L^2$  into an energy  $E \in \mathbb{R}$ . See [https://en.wikipedia.org/wiki/Functional\\_\(mathematics\)](https://en.wikipedia.org/wiki/Functional_(mathematics)).

<sup>11</sup>See [https://en.wikipedia.org/wiki/Lagrange\\_multiplier](https://en.wikipedia.org/wiki/Lagrange_multiplier).

<sup>12</sup>See [https://en.wikipedia.org/wiki/Functional\\_derivative](https://en.wikipedia.org/wiki/Functional_derivative).

wavefunction-dependent effective potential  $V_{\text{eff}}(x, t)$  is still diagonal in the position representation. We extend the Mathematica code of [In\[498\]](#) by modifying the `propPot2` method to include a non-linear term with prefactor  $\kappa$  (added as an additional argument to the `propApprox` function), and do not forget that the wavefunction at grid point  $x_j$  is  $\psi(x_j) = v_j/\sqrt{\Delta}$ :


```
1 In[500]:=propApprox[Wt_, κ_?NumericQ, Δt_?NumericQ, M_Integer /; M >= 1,
2           v0_ /; VectorQ[v0, NumericQ]] :=
```

and

```
13 propPot2[t_, v_] := Exp[λ/2*((Wt[#1,t]&/@xgrid) + κ*Abs[v]^2/Δ)]*v;
```

As an example, we plot the time-dependent density for the time-independent gravitational well  $W(x, t) = mgx$  and  $\kappa = -3 \cdot (g\hbar^4/m)^{1/3}$  (attractive interaction),  $\kappa = 0$  (no interaction),  $\kappa = +3 \cdot (g\hbar^4/m)^{1/3}$  (repulsive interaction):

```
1 In[501]:=With[{κ = -3 * (g*ħ^4/m)^(1/3), Δt = 50, M = 10^3},
2   ρ = ArrayPad[Abs[propApprox[W[#1]&, κ, Δt, M, vv][[All, 2]]^2/Δ, {{0, 0}, {1, 1}}];
3   ArrayPlot[Reverse[Transpose[ρ]], DataRange -> {{0, Δt}, {0, a}}]
```



Observations:

- The noninteractive case ( $\kappa = 0$ ) shows a slow broadening and decoherence of the wavepacket.
- Attractive interactions ( $\kappa < 0$ ) make the wavepacket collapse to a tight spot and bounce almost like a classical particle.
- Repulsive interactions ( $\kappa > 0$ ) make the wavepacket broader, which slows down its decoherence.

## Exercises

**Q4.11** Dimensionless problem ( $a=m=\hbar=1$ ): Given a particle moving in the range  $x \in [0, 1]$  with the non-linear Hamiltonian

$$\hat{\mathcal{H}} = -\frac{1}{2} \frac{d^2}{dx^2} + \underbrace{\Omega \left[ \left( \frac{x - \frac{1}{2}}{\delta} \right)^2 - 1 \right]}_{W(x)} + \kappa |\psi(x)|^2, \quad (4.53)$$

do the following calculations:



1. Plot the potential  $W(x)$  for  $\Omega = 1$  and  $\delta = \frac{1}{4}$  (use  $\kappa = 0$ ). What are the main characteristics of this potential? *Hint:* compute  $W(\frac{1}{2})$ ,  $W'(\frac{1}{2})$ ,  $W(\frac{1}{2} \pm \delta)$ ,  $W'(\frac{1}{2} \pm \delta)$ .
2. Calculate and plot the time-dependent density  $|\psi(x, t)|^2$  for  $\Omega = 250$ ,  $\delta = \frac{1}{4}$ , and  $\kappa = 0$ , starting from  $\psi_0(x) \propto \exp\left[-\left(\frac{x-x_0}{2\sigma}\right)^2\right]$  with  $x_0 = 0.2694$  and  $\sigma = 0.0554$ . Calculate the probabilities for finding the particle in the left half ( $x < \frac{1}{2}$ ) and in the right half ( $x > \frac{1}{2}$ ) up to  $t = 20$ . What do you observe?
3. What do you observe for  $\kappa = 0.5$ ? Why?

### 4.2.1 Imaginary-Time Propagation for Finding the Ground State of the Non-linear Schrödinger Equation

*[Supplementary Material 4]*

In the previous section we have looked at the dynamical evolution of a Bose–Einstein condensate with the time-dependent nonlinear Schrödinger Eq. 4.52, which could be performed with minimal modifications to previous Mathematica code. The time-independent nonlinear Schrödinger Eq. 4.51, on the other hand, seems at first sight inaccessible to our established methods: it is an operator eigenvalue equation, with the operator acting non-linearly on the wavefunction and thus invalidating the matrix diagonalization method of Sect. 2.2. How can we determine the ground state of Eq. 4.51?

You may remember from statistical mechanics that at temperature  $T$ , the density operator of a system governed by a Hamiltonian  $\hat{\mathcal{H}}$  is

$$\hat{\rho}(\beta) = \frac{e^{-\beta\hat{\mathcal{H}}}}{Z(\beta)} \quad (4.54)$$

with  $\beta = 1/(k_B T)$  the reciprocal temperature in terms of the Boltzmann constant  $k_B = 1.3806488(13) \times 10^{-23} \text{ J/K}$ . The partition function  $Z(\beta) = \text{Tr } e^{-\beta\hat{\mathcal{H}}}$  ensures that the density operator has the correct norm,  $\text{Tr } \hat{\rho}(\beta) = 1$ .

We know that at zero temperature the system will be in its ground state  $|\gamma\rangle$ ,<sup>13</sup>

$$\lim_{\beta \rightarrow \infty} \hat{\rho}(\beta) = |\gamma\rangle\langle\gamma|. \quad (4.55)$$

If we multiply this equation by an arbitrary state  $|\psi\rangle$  from the right, we find

$$\lim_{\beta \rightarrow \infty} \hat{\rho}(\beta)|\psi\rangle = |\gamma\rangle\langle\gamma|\psi\rangle. \quad (4.56)$$

---

<sup>13</sup>For simplicity we assume here that the ground state is non-degenerate.

Assuming that  $\langle \gamma | \psi \rangle \neq 0$  (which is true for almost all states  $|\psi\rangle$ ), the ground state is therefore

$$|\gamma\rangle = \frac{\lim_{\beta \rightarrow \infty} \hat{\rho}(\beta) |\psi\rangle}{\langle \gamma | \psi \rangle} = \frac{1}{\langle \gamma | \psi \rangle} \lim_{\beta \rightarrow \infty} \frac{1}{Z(\beta)} \times e^{-\beta \hat{\mathcal{H}}} |\psi\rangle. \quad (4.57)$$

This means that if we take almost any state  $|\psi\rangle$  and calculate  $\lim_{\beta \rightarrow \infty} \frac{1}{Z(\beta)} e^{-\beta \hat{\mathcal{H}}} |\psi\rangle$ , we find a state that is proportional to the ground state (the prefactors  $\frac{1}{\langle \gamma | \psi \rangle}$  and  $\frac{1}{Z(\beta)}$  are merely scalar prefactors). But we already know how to do this: the wavefunction  $e^{-\beta \hat{\mathcal{H}}} |\psi\rangle$  is calculated from  $|\psi\rangle$  by *imaginary-time propagation*. In fact the split-step algorithm of Sect. 4.1.9 remains valid if we replace  $i(t - t_0)/\hbar \mapsto \beta$ . The advantage of Eq. 4.57 over the matrix method of Sect. 2.2 is that the former can be implemented even if the Hamiltonian depends on the wavefunction, as in Eq. 4.51. The only caveat is that, while regular time propagation (Sect. 4.1.9) is unitary, imaginary-time propagation is not. The wavefunction must therefore be re-normalized after each imaginary-time evolution step (with the `Normalize` function).

To implement this method of calculating the ground state by imaginary-time propagation, we set  $\beta = M \cdot \delta\beta$  and modify Eq. 4.57 to

$$|\gamma\rangle \propto \lim_{M \cdot \delta\beta \rightarrow \infty} e^{-M \delta\beta \hat{\mathcal{H}}} |\psi\rangle = \lim_{M \cdot \delta\beta \rightarrow \infty} \left[ e^{-\delta\beta \hat{\mathcal{H}}} \right]^M |\psi\rangle \stackrel{\text{Trotter Eq. 4.39}}{=} \lim_{\delta\beta \rightarrow 0} \lim_{M \cdot \delta\beta \rightarrow \infty} \left[ e^{-\frac{\delta\beta}{2} \hat{V}} e^{-\delta\beta \hat{T}} e^{-\frac{\delta\beta}{2} \hat{V}} \right]^M |\psi\rangle. \quad (4.58)$$

In practice we choose a small but finite “imaginary-time” step  $\delta\beta$ , and keep multiplying the wavefunction by  $e^{-\frac{\delta\beta}{2} \hat{V}} e^{-\delta\beta \hat{T}} e^{-\frac{\delta\beta}{2} \hat{V}}$  until the normalized wavefunction no longer changes and the infinite- $\beta$  limit ( $M \cdot \delta\beta \rightarrow \infty$ ) has effectively been reached.

```

1 In[502]:=groundstate[g_?NumericQ, δβ_?NumericQ, tolerance_: 10^-10] :=
2   Module[{Ke, propKin, propPot2, v0, γ},
3     (* compute the diagonal elements of exp[-δβ*T] *)
4     Ke = Exp[-δβ*Range[nmax]^2*π^2/2] / N;
5     (* propagate by a full imaginary-time-step with T *)
6     propKin[v_] := Normalize[FourierDST[Ke*FourierDST[v,1,1]]];
7     (* propagate by a half imaginary-time-step with V *)
8     propPot2[v_] := Normalize[Exp[-δβ/2*(Wgrid + g*(nmax+1)*Abs[v]^2)]*v];
9     (* propagate by a full imaginary-time-step by *)
10    (* H=T+V using the Trotter approximation *)
11    prop[v_] := propPot2[propKin[propPot2[v]]];
12    (* random starting point *)
13    v0 = Normalize@RandomComplex[{-1-I,1+I}, nmax];
14    (* propagation to the ground state *)
15    γ = FixedPoint[prop,v0,SameTest->Function[{v1,v2},Norm[v1-v2]<tolerance]];
16    (* return the ground-state coefficients *)
17    γ]

```

The last argument, `tolerance`, is optional and is given the default value  $10^{-10}$  if not specified (see Sect. 1.6.5). The `FixedPoint` function is used to apply the imaginary-time propagation until the result no longer changes (two consecutive results are considered equal if the function given as `SameTest` returns true when applied to these two results).

Multiplying Eq. 4.51 by  $\psi^*(x)$  and integrating over  $x$  gives

$$\mu = -\underbrace{\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} dx \psi^*(x) \psi''(x)}_{E_{\text{kin}}[\psi]} + \underbrace{\int_{-\infty}^{\infty} dx V(x) |\psi(x)|^2}_{E_{\text{pot}}[\psi]} + g \underbrace{\int_{-\infty}^{\infty} dx |\psi(x)|^4}_{2E_{\text{int}}[\psi]}, \quad (4.59)$$

which is very similar to Eq. 4.49 apart from a factor of two for  $E_{\text{int}}$ . We use this to calculate the total energy and the chemical potential in `In[502]` by replacing lines 16ff with

```

16      (* energy components *)
17      Ekin =  $\pi^2/2$ *Range[nmax]^2.Abs[FourierDST[ $\gamma$ ,1]]^2;
18      Epot = Wgrid.Abs[ $\gamma$ ]^2;
19      Eint = (g/2)(nmax+1)*Total[Abs[ $\gamma$ ]^4];
20      (* total energy *)
21      Etot = Ekin + Epot + Eint;
22      (* chemical potential *)
23       $\mu$  = Ekin + Epot + 2*Eint;
24      (* return energy, chemical potential, coefficients *)
25      {Etot,  $\mu$ ,  $\gamma$ }]

```


and adding the local variables `Ekin`, `Epot`, `Eint`, `Etot`, and  `$\mu$`  on line 2.

As an example we calculate the ground-state density for the gravity well of Sect. 4.1.7 with three different values of the interaction strength  $\kappa$  [in units of  $(g\hbar^4/m)^{1/3}$ ]:

```

1  In[503]:=With[{ $\kappa$  = 3 * (g* $\hbar^4$ /m)^(1/3),  $\delta\beta$  = 10^-4},
2      {Etot,  $\mu$ ,  $\gamma$ } = groundstate[ $\delta\beta$ ,  $\kappa$ ];
3      ListLinePlot[Join[{0, 0}], Transpose[{xgrid, Abs[ $\gamma$ ]^2/ $\Delta$ }], {{a, 0}}],
4      PlotRange -> All, PlotLabel -> {Etot,  $\mu$ }]

```



Note that for  $\kappa = 0$  the Gross–Pitaevskii equation is the Schrödinger equation, and the chemical potential is equal to the total energy, matching the exact result of `Out[450]`.

### Exercises

**Q4.12** Dimensionless problem ( $a=m=\hbar=1$ ): Given a particle moving in the range  $x \in [0, 1]$  with the non-linear Hamiltonian

$$\hat{\mathcal{H}} = -\frac{1}{2} \frac{d^2}{dx^2} + 2500 \left( x - \frac{1}{2} \right)^2 + \kappa |\psi(x)|^2, \quad (4.60)$$

do the following calculations:

1. For  $\kappa = 0$  calculate the exact ground state  $|\zeta\rangle$  (assuming that the particle can move in the whole domain  $x \in \mathbb{R}$ ) and its energy eigenvalue. *Hint:* assume  $\zeta(x) = \exp\left[-\left(\frac{x-\frac{1}{2}}{2\sigma}\right)^2\right] / \sqrt{\sigma\sqrt{2\pi}}$  and find the value of  $\sigma$  that minimizes  $\langle\zeta|\hat{\mathcal{H}}|\zeta\rangle$ .
2. Calculate the ground state  $\lim_{\beta \rightarrow \infty} e^{-\beta\hat{\mathcal{H}}}|\zeta\rangle$  and its chemical potential by imaginary-time propagation (with normalization of the wavefunction after each propagation step), using the code given above.
3. Plot the ground-state density for different values of  $\kappa$ .
4. Plot the total energy and the chemical potential as functions of  $\kappa$ .

### 4.3 Several Particles in One Dimension: Interactions

In Sect. 4.2 we have studied a simple mean-field description of many-particle systems, with the advantage of simplicity and the disadvantage of not describing inter-particle correlations. Here we use a different approach that captures the full quantum mechanics of many-particle systems (including correlations), with the disadvantage of much increased calculation size.

We have seen in Sect. 2.4.2 how to describe quantum-mechanical systems with more than one degree of freedom. This method can be used for describing several particles moving in one dimension. In the following we look at two examples of interacting particles.

When more than one particle is present in a system, we must distinguish between bosons and fermions. Whenever the Hamiltonian is symmetric under particle exchange (which is the case in this section), each one of its eigenstates can be associated with an irreducible representation of the particle permutation group. For two particles, the only available choices are the symmetric and the antisymmetric irreducible representations, and therefore every numerically calculated eigenstate can be labeled as either bosonic (symmetric) or fermionic (antisymmetric). For more particles, however, other irreducible representations exist,<sup>14</sup> meaning that some numerically calculated eigenstates of the Hamiltonian may not be physical at all because they are neither bosonic (fully symmetric) nor fermionic (fully antisymmetric).

#### 4.3.1 Two Identical Particles in One Dimension with Contact Interaction [Supplementary Material 5]

We first look at two identical particles moving in a one-dimensional square well of width  $a$  and interacting through a contact potential  $V_{\text{int}}(x_1, x_2) = \kappa \times \delta(x_1 - x_2)$ .

---

<sup>14</sup>See [https://en.wikipedia.org/wiki/Irreducible\\_representation](https://en.wikipedia.org/wiki/Irreducible_representation).

Such potentials are a good approximation of the  $s$ -wave scattering interactions taking place in cold dilute gases. The Hamiltonian of this system is<sup>15</sup>

$$\hat{\mathcal{H}} = \underbrace{-\frac{\hbar^2}{2m} \left[ \frac{\partial^2}{\partial x_1^2} + \frac{\partial^2}{\partial x_2^2} \right]}_{\hat{T}} + \underbrace{V(x_1) + V(x_2)}_{\hat{V}} + \underbrace{\kappa \delta(x_1 - x_2)}_{\hat{\mathcal{H}}_{\text{int}}}, \quad (4.61)$$

where  $V(x)$  is the single-particle potential (as in Sect. 4.1) and  $\kappa$  is the interaction strength, often related to the  $s$ -wave scattering length  $a_s$ . For the time being we do not need to specify whether the particles are bosons or fermions.

We describe this system with the tensor-product basis constructed from two finite-resolution position basis sets:

$$|j_1, j_2\rangle = |j_1\rangle \otimes |j_2\rangle \quad \text{for } j_1, j_2 \in \{1, 2, 3, \dots, n_{\text{max}}\}. \quad (4.62)$$

Most of the matrix representations of the terms in Eq. 4.61 are constructed as tensor products of the matrix representations of the corresponding single-particle representations since  $\hat{T} = \hat{T}_1 \otimes \mathbb{1} + \mathbb{1} \otimes \hat{T}_2$  and  $\hat{V} = \hat{V}_1 \otimes \mathbb{1} + \mathbb{1} \otimes \hat{V}_2$ . The only new element is the interaction Hamiltonian  $\hat{\mathcal{H}}_{\text{int}}$ . Remembering that its formal operator definition is

$$\hat{\mathcal{H}}_{\text{int}} = \kappa \int_{-\infty}^{\infty} dx_1 dx_2 \left[ |x_1\rangle \otimes |x_2\rangle \right] \delta(x_1 - x_2) \left[ \langle x_1| \otimes \langle x_2| \right] \quad (4.63)$$

(while Eq. 4.61 is merely a shorthand notation), we calculate its matrix elements in the finite-precision position basis with

$$\langle j_1, j_2 | \hat{\mathcal{H}}_{\text{int}} | j'_1, j'_2 \rangle = \kappa \int_{-\infty}^{\infty} dx_1 dx_2 \langle j_1 | x_1 \rangle \langle j_2 | x_2 \rangle \delta(x_1 - x_2) \langle x_1 | j'_1 \rangle \langle x_2 | j'_2 \rangle = \kappa \int_0^a dx \vartheta_{j_1}(x) \vartheta_{j_2}(x) \vartheta_{j'_1}(x) \vartheta_{j'_2}(x). \quad (4.64)$$

These quartic overlap integrals can be calculated by a four-dimensional type-I discrete sine transform (see Eq. 4.20 and Q4.13),

$$\begin{aligned} \int_0^a dx \vartheta_{j_1}(x) \vartheta_{j_2}(x) \vartheta_{j_3}(x) \vartheta_{j_4}(x) = \\ \frac{1}{2a} \sum_{n_1, n_2, n_3, n_4=1}^{n_{\text{max}}} X_{n_1 j_1} X_{n_2 j_2} X_{n_3 j_3} X_{n_4 j_4} \left[ \delta_{n_1+n_2, n_3+n_4} + \delta_{n_1+n_3, n_2+n_4} + \delta_{n_1+n_4, n_2+n_3} \right. \\ \left. - \delta_{n_1, n_2+n_3+n_4} - \delta_{n_2, n_1+n_3+n_4} - \delta_{n_3, n_1+n_2+n_4} - \delta_{n_4, n_1+n_2+n_3} \right], \end{aligned} \quad (4.65)$$

<sup>15</sup>Notice that we write this Hamiltonian in an abbreviated form. The full operator form, with terms similar to Eq. 4.2 but containing double integrals over space, is cumbersome to write (see Eq. 4.63).

which we evaluate in Mathematica very efficiently and all at once with

```

1 In[504]:=overlap4 = FourierDST[Table[KroneckerDelta[n1+n2,n3+n4]
2   +KroneckerDelta[n1+n3,n2+n4]+KroneckerDelta[n1+n4,n2+n3]
3   -KroneckerDelta[n1,n2+n3+n4]-KroneckerDelta[n2,n1+n3+n4]
4   -KroneckerDelta[n3,n1+n2+n4]-KroneckerDelta[n4,n1+n2+n3],
5   {n1,nmax},{n2,nmax},{n3,nmax},{n4,nmax}],1]/(2*a);

```

**Mathematica code** As before, we assume that the quantities  $a$ ,  $m$ , and  $\hbar$  are expressed in a suitable set of units (see Sect. 4.1.1). First we define the grid size and the unit operator `id` acting on a single particle:

```

1 In[505]:=m = h = 1; (* for example *)
2 In[506]:=a = 1; (* for example *)
3 In[507]:=nmax = 50; (* for example *)
4 In[508]:=Δ = a/(nmax+1);
5 In[509]:=xgrid = Range[nmax]*Δ;
6 In[510]:=id = IdentityMatrix[nmax, SparseArray];

```

The total kinetic Hamiltonian is assembled via a Kronecker product (tensor product) of the two single-particle kinetic Hamiltonians:

```

1 In[511]:=T1M = SparseArray[Band[{1,1}]->Range[nmax]^2*π^2*ħ^2/(2*m*a^2)];
2 In[512]:=T1P = FourierDST[T1M, 1];
3 In[513]:=TP = KroneckerProduct[T1P, id] + KroneckerProduct[id, T1P];

```

The same for the potential Hamiltonian (here we assume no potential, that is, a square well; but you may modify this):

```

1 In[514]:=W[x_] = 0;
2 In[515]:=Wgrid = W /@ xgrid;
3 In[516]:=V1P = SparseArray[Band[{1,1}]->Wgrid];
4 In[517]:=VP = KroneckerProduct[V1P, id] + KroneckerProduct[id, V1P];

```

The interaction Hamiltonian is constructed from `In[504]` with the `ArrayFlatten` command, which flattens the combination basis set  $|j_1\rangle \otimes |j_2\rangle$  into a single basis set  $|j_1, j_2\rangle$ , or in other words, which converts the  $n_{\max} \times n_{\max} \times n_{\max} \times n_{\max}$ -matrix `overlap4` into a  $n_{\max}^2 \times n_{\max}^2$ -matrix:

```

1 In[518]:=HintP = ArrayFlatten[overlap4];

```

The full Hamiltonian, in which the amplitude of the potential can be adjusted with the prefactor  $\Omega$  and the interaction strength with  $g$ , is

```

1 In[519]:=HP[Ω_, κ_] = TP + Ω*VP + κ*HintP;

```

We calculate eigenstates (the ground state, for example) with the methods already described previously. The resulting wavefunctions are in the tensor-product basis of Eq. 4.62, and they can be plotted with

```

1 In[520]:=plot2Dwf[ψ_] := Module[{ψ1,ψ2},
2   (* make a square array of wavefunction values *)
3   ψ1 = ArrayReshape[ψ, {nmax,nmax}];
4   (* add a frame of zeros at the edges *)
5   (* representing the boundary conditions *)
6   ψ2 = ArrayPad[ψ1, 1];
7   (* plot *)
8   ArrayPlot[Reverse[Transpose[ψ2]]]

```


Assuming that a given wavefunction  $\psi$  is purely real-valued,<sup>16</sup> we can plot it with

```

1 In[521]:=plot2Dwf[v/Δ]


```

Here we plot the four lowest-energy wavefunctions for  $\Omega = 0$  (no potential, the particles move in a simple infinite square well) and  $\kappa = +25$  (repulsive interaction), using  $n_{\max} = 50$  grid points, with the title of each panel showing the energy and the symmetry (see below). White corresponds to zero wavefunction, red is positive  $\psi(x_1, x_2) > 0$ , and blue is negative  $\psi(x_1, x_2) < 0$ .



We can see that in the ground state for  $g > 0$  the particles avoid each other, i.e., the ground-state wavefunction  $\psi(x_1, x_2)$  is reduced whenever  $x_1 = x_2$ .

And here are the lowest four energy eigenstate wavefunctions for  $\kappa = -10$ :



We can see that in the ground state for  $\kappa < 0$  the particles attract each other, i.e., the ground-state wavefunction  $\psi(x_1, x_2)$  is increased whenever  $x_1 = x_2$ . We also notice that the second-lowest state for  $\kappa = +25$  is exactly equal to the fourth-lowest

<sup>16</sup>The eigenvectors of Hermitian operators can always be chosen to have real coefficients. Proof: Suppose that  $\mathbf{H} \cdot \vec{\psi} = E\vec{\psi}$  for a vector  $\vec{\psi}$  with complex entries. Complex-conjugate the eigenvalue equation,  $\mathbf{H}^\dagger \cdot \vec{\psi}^* = E^* \vec{\psi}^*$ ; but  $\mathbf{H}^\dagger = \mathbf{H}$  and  $E^* = E$ , and hence  $\vec{\psi}^*$  is also an eigenvector of  $\mathbf{H}$  with eigenvalue  $E$ . Thus we can introduce two real-valued vectors  $\vec{\psi}_r = \vec{\psi} + \vec{\psi}^*$  and  $\vec{\psi}_i = i(\vec{\psi} - \vec{\psi}^*)$ , representing the real and imaginary parts of  $\vec{\psi}$ , respectively, which are both eigenvectors of  $\mathbf{H}$  with eigenvalue  $E$ . Mathematica (as well as most other matrix diagonalization algorithms) automatically detect Hermitian matrices and return eigenvectors with real coefficients.

state for  $\kappa = -10$ : its wavefunction vanishes whenever  $x_1 = x_2$  and thus the contact interaction has no influence on this state.

In the above plots we have noted the symmetry of each eigenstate (symmetric or antisymmetric with respect to particle exchange), which is calculated with the integral

$$S[\psi] = \int_0^a dx_1 dx_2 \psi^*(x_1, x_2) \psi(x_2, x_1) = \begin{cases} +1 & \text{for symmetric states } \psi(x_2, x_1) = \psi(x_1, x_2), \\ -1 & \text{for antisymmetric states } \psi(x_2, x_1) = -\psi(x_1, x_2). \end{cases} \quad (4.66)$$


In Mathematica, the mirrored wavefunction  $\psi(x_2, x_1)$  is calculated with the particle interchange operator  $\hat{\Xi}$  defined as

```
1 In[522]:=Ξ = ArrayFlatten[SparseArray[{i_>j_>j_>i_} -> 1, {nmax, nmax, nmax, nmax}]];
```

such that  $\psi(x_2, x_1) = \langle x_2, x_1 | \psi \rangle = \langle x_1, x_2 | \hat{\Xi} | \psi \rangle$ . The symmetry of a state, defined in Eq. 4.66, is therefore the expectation value of the  $\hat{\Xi}$  operator:

```
1 In[523]:=symmetry[v_] := Re[Conjugate[v].(Ξ.v)]
```

Here we show the numerical energy eigenvalues of the contact interaction Hamiltonian, colored according to their symmetry: red dots indicate symmetric states ( $S = +1$ ), whereas blue dots indicate antisymmetric states ( $S = -1$ ).



In this representation it becomes even clearer that antisymmetric states are independent of the contact interaction because their wavefunction vanishes whenever  $x_1 = x_2$  (see Q4.16).

### Bosons and Fermions

The reason why every state in the above calculation is either symmetric or antisymmetric with respect to particle interchange is that the Hamiltonian In[519] commutes with the particle interchange operator In[522] (see Q4.15). As a result,  $\hat{\mathcal{H}}$  and  $\hat{\Xi}$  can be diagonalized simultaneously.

We notice that  $\hat{\Xi}$  has only eigenvalues  $\pm 1$ :



```

1 In[524]:=Ξ //Eigenvalues //Counts
2 Out[524]=<| -1 -> 1225, 1 -> 1275 |>

```

The  $n_{\max}(n_{\max} + 1)/2$  eigenvalues  $+1$  correspond to eigenvectors that are symmetric under particle interchange and form a basis of the symmetric subspace of the full Hilbert space (bosonic states); the  $n_{\max}(n_{\max} - 1)/2$  eigenvalues  $-1$  correspond to eigenvectors that are antisymmetric under particle interchange and form a basis of the antisymmetric subspace of the full Hilbert space (fermionic states). By constructing a matrix whose rows are the symmetric eigenvectors, we construct an operator  $\hat{\Pi}_s$  that projects from the full Hilbert space onto the space of symmetric states,

```

1 In[525]:=ε = Transpose[Eigensystem[Normal[Ξ]]];
2 In[526]:=Πs = Select[ε, #[[1]] == 1 &][[All, 2]] //Orthogonalize //SparseArray;

```

Similarly we construct a projector  $\hat{\Pi}_a$  onto the space of antisymmetric states,

```

1 In[527]:=Πa = Select[ε, #[[1]] == -1 &][[All, 2]] //Orthogonalize //SparseArray;

```

With the help of these projectors, we define the Hamiltonians of the system restricted to the symmetric or antisymmetric subspace, respectively:

```

1 In[528]:=HPS[Ω_, κ_] = Πs.HP[Ω, κ].Transpose[Πs];
2 In[529]:=HPa[Ω_, κ_] = Πa.HP[Ω, κ].Transpose[Πa];

```

If the two particles in the present problem are indistinguishable bosons, then they can only populate the symmetric states (red dots in the above eigenvalue plot). We calculate the  $m$  lowest energy eigenstates of this symmetric subspace with the restricted Hamiltonian **HPS**:

```

1 In[530]:=Clear[sgs];
2 In[531]:=sgs[Ω_?NumericQ, κ_?NumericQ, m_Integer /; m >= 1] := sgs[Ω, κ, m] =
3   {-#[[1]], #[[2]].Πs} &[Eigensystem[-HPS[N[Ω], N[κ]], m,
4     Method -> {"Arnoldi", "Criteria" -> "RealPart", MaxIterations -> 10^6}]]

```

Notice that we convert the calculated eigenstates back into the full Hilbert space by multiplying the results with **Πs** from the right.

In the same way, if the two particles in the present problem are indistinguishable fermions, then they can only populate the antisymmetric states (blue dots in the above eigenvalue plot). We calculate the  $m$  lowest energy eigenstates of this antisymmetric subspace with the restricted Hamiltonian **HPa**:

```

1 In[532]:=Clear[ags];
2 In[533]:=ags[Ω_?NumericQ, κ_?NumericQ, m_Integer /; m >= 1] := ags[Ω, κ, m] =
3   {-#[[1]], #[[2]].Πa} &[Eigensystem[-HPa[N[Ω], N[κ]], m,
4     Method -> {"Arnoldi", "Criteria" -> "RealPart", MaxIterations -> 10^6}]]

```

As an example, here we calculate the six lowest energy eigenvalues of the full Hamiltonian for  $\Omega = 0$  and  $\kappa = 5$ :

```

1 In[534]:=gs[0, 5, 6][[1]] //Sort
2 Out[534]={15.2691, 24.674, 32.3863, 45.4849, 49.348, 58.1333}

```

The six lowest *symmetric* energy eigenvalues are

```

1 In[535]:=sgs[0, 5, 6][[1]] //Sort
2 Out[535]={15.2691, 32.3863, 45.4849, 58.1333, 72.1818, 93.1942}

```

The six lowest *antisymmetric* energy eigenvalues are

```

1 In[536]:=ags[0, 5, 6][[1]] //Sort
2 Out[536]={24.674, 49.348, 64.1524, 83.8916, 98.696, 123.37}

```

From `Out[535]` and `Out[536]` we can see which levels of `Out[534]` are symmetric or antisymmetric.

### Exercises

**Q4.13** Show that Eq. 4.65 is plausible by setting `nmax=3`, evaluating `In[504]`, and then comparing its values to explicit integrals from Eq. 4.65 for several tuples  $(j_1, j_2, j_3, j_4)$ . *Hint:* use `a=1` for simplicity.

**Q4.14** In the problem of Sect. 4.3.1, calculate the expectation value of the inter-particle distance  $\langle x_1 - x_2 \rangle$ , and its variance  $\langle (x_1 - x_2)^2 \rangle - \langle x_1 - x_2 \rangle^2$ , in the ground state as a function of  $\kappa$  (still keeping  $\Omega = 0$ ). *Hint:* Using Eq. 4.24, the position operators `x1` and `x2` are approximately

```

1 In[537]:=x = SparseArray[Band[{1,1}]>->xgrid];
2 In[538]:=x1 = KroneckerProduct[x, id];
3 In[539]:=x2 = KroneckerProduct[id, x];

```

**Q4.15** Show in Mathematica (by explicit calculation) that the Hamiltonian `In[519]` commutes with the particle interchange operator `In[522]`. *Hint:* use the `Norm` function to calculate the matrix norm of the commutator.

**Q4.16** Show in Mathematica (by explicit calculation) that the antisymmetric Hamiltonian `In[529]` does not depend on  $\kappa$ .

**Q4.17** The contact-interaction problem of this section can be solved analytically if  $W(x) = 0$ , which allows us to check the accuracy of the presented numerical calculations. We will study the dimensionless (`a=m=h=1`) Hamiltonian  $\hat{\mathcal{H}} = -\frac{1}{2} \left[ \frac{\partial^2}{\partial x_1^2} + \frac{\partial^2}{\partial x_2^2} \right] + \kappa \delta(x_1 - x_2)$ .

1. The ground-state wavefunction will be of the form

$$\psi(x_1, x_2) = A \times \begin{cases} \cos[\alpha(x_1 + x_2 - 1)] \cos[\beta(x_1 - x_2 + 1)] \\ - \cos[\alpha(x_1 - x_2 + 1)] \cos[\beta(x_1 + x_2 - 1)] & \text{if } 0 \leq x_1 \leq x_2 \leq 1, \\ \cos[\alpha(x_2 + x_1 - 1)] \cos[\beta(x_2 - x_1 + 1)] \\ - \cos[\alpha(x_2 - x_1 + 1)] \cos[\beta(x_2 + x_1 - 1)] & \text{if } 0 \leq x_2 \leq x_1 \leq 1. \end{cases} \quad (4.67)$$

Check that this wavefunction satisfies the boundary conditions  $\psi(x_1, 0) = \psi(x_1, 1) = \psi(0, x_2) = \psi(1, x_2) = 0$ , that it is continuous across the boundary  $x_1 = x_2$  (i.e., that the two pieces of the wavefunction match up), and that it satisfies the symmetries of the calculation box:  $\psi(x_1, x_2) = \psi(x_2, x_1) = \psi(1 - x_1, 1 - x_2)$ .

2. Insert this wavefunction into the time-independent Schrödinger equation. Find the energy eigenvalue by assuming  $x_1 \neq x_2$ . You should find  $E = \alpha^2 + \beta^2$ .
3. Express the Hamiltonian and the wavefunction in terms of the new coordinates  $R = (x_1 + x_2)/\sqrt{2}$  and  $r = (x_1 - x_2)/\sqrt{2}$ . *Hints:*  $\frac{\partial^2}{\partial x_1^2} + \frac{\partial^2}{\partial x_2^2} = \frac{\partial^2}{\partial R^2} + \frac{\partial^2}{\partial r^2}$  and  $\delta(ux) = u^{-1}\delta(x)$ .
4. Integrate the Schrödinger equation, expressed in the  $(R, r)$  coordinates, over  $r \in [-\epsilon, \epsilon]$  and take the limit  $\epsilon \rightarrow 0^+$ . Verify that the resulting expression is satisfied if  $\alpha \tan(\alpha) = \beta \tan(\beta) = \kappa/2$ . *Hint:* Do the integration analytically and use  $\int_a^b dr \frac{\partial^2}{\partial r^2} f(r) = f'(b) - f'(a)$ .
5. The ground state is found by numerically solving  $\alpha \tan(\alpha) = \beta \tan(\beta) = \kappa/2$ . Out of the many solutions of these equations, we choose the correct ones for the ground state by specifying the starting point of the numerical root solver:

```

1 In[540]:=Clear[a,b];
2 In[541]:=a[-∞] = π/2;
3 In[542]:=a[0] = π;
4 In[543]:=a[∞] = 3π/2;
5 In[544]:=a[κ_?NumericQ] := a[κ] = u /.
6           FindRoot[u*Tan[u]==κ/2, {u,π+ArcTan[κ/(2π)]}]
7 In[545]:=b[-∞] = I*∞;
8 In[546]:=b[0] = 0;
9 In[547]:=b[∞] = π/2;
10 In[548]:=b[κ_ /; κ >= 0] := b[κ] = u /. FindRoot[u*Tan[u] == κ/2,
11           {u, If[κ<π, 1, π/2 - π/κ + 2π/κ^2]}]
12 In[549]:=b[κ_ /; κ < 0] := b[κ] = I*u /. FindRoot[u*Tanh[u] == -κ/2, {u,-κ/2}]

```

Compare the resulting  $\kappa$ -dependent ground state energy to the numerically calculated ground-state energies from [In\[519\]](#).

### 4.3.2 Two Particles in One Dimension with Arbitrary Interaction

Two particles in one dimension interacting via an arbitrary potential have a Hamiltonian very similar to Eq. 4.61, except that the interaction is now

$$\hat{\mathcal{H}}_{\text{int}} = V_{\text{int}}(x_1, x_2), \quad (4.68)$$

or, more explicitly as an operator in the Dirac position basis,

$$\hat{\mathcal{H}}_{\text{int}} = \int_0^a dx_1 dx_2 |x_1\rangle \otimes |x_2\rangle V_{\text{int}}(x_1, x_2) \langle x_1| \otimes \langle x_2|. \quad (4.69)$$

As an example, for the Coulomb interaction we have  $V_{\text{int}}(x_1, x_2) = \frac{Q_1 Q_2}{4\pi\epsilon_0|x_1 - x_2|}$  with  $Q_1$  and  $Q_2$  the electric charges of the two particles. For many realistic potentials  $V_{\text{int}}$  only depends on  $|x_1 - x_2|$ .

In the finite-resolution position basis, the matrix elements of this interaction Hamiltonian can be approximated with a method similar to what we have already seen, for example in Sect. 4.1.4:

$$\begin{aligned} \langle j_1, j_2 | \hat{\mathcal{H}}_{\text{int}} | j'_1, j'_2 \rangle &= \int_0^a \vartheta_{j_1}(x_1) \vartheta_{j_2}(x_2) V_{\text{int}}(x_1, x_2) \vartheta_{j'_1}(x_1) \vartheta_{j'_2}(x_2) dx_1 dx_2 \\ &\approx V_{\text{int}}(x_{j_1}, x_{j_2}) \int_0^a \vartheta_{j_1}(x_1) \vartheta_{j_2}(x_2) \vartheta_{j'_1}(x_1) \vartheta_{j'_2}(x_2) dx_1 dx_2 = \delta_{j_1, j'_1} \delta_{j_2, j'_2} V_{\text{int}}(x_{j_1}, x_{j_2}). \end{aligned} \quad (4.70)$$

This approximation is easy to evaluate without the need for integration over basis functions. But realistic interaction potentials are usually singular for  $x_1 = x_2$  (consider, for example, the Coulomb potential), and therefore the approximate Eq. 4.70 fails for the evaluation of the matrix elements  $\langle j, j | \hat{\mathcal{H}}_{\text{int}} | j, j \rangle$ . This problem cannot be solved in all generality, and we can either resort to more accurate integration (as in Sect. 4.3.1) or we can replace the true interaction potential with a less singular version: for the Coulomb potential, we could for example use a truncated singularity for  $|x| < \delta$  for some small distance  $\delta$ :

$$V_{\text{int}}(x) = \frac{Q_1 Q_2}{4\pi\epsilon_0} \times \begin{cases} \frac{1}{|x|} & \text{if } |x| \geq \delta \\ \frac{1}{\delta} & \text{if } |x| < \delta \end{cases} \quad (4.71)$$

As long as the particles move at energies much smaller than  $V_{\text{int}}(\pm\delta) = \frac{Q_1 Q_2}{4\pi\epsilon_0\delta}$  they cannot distinguish the true Coulomb potential from this truncated form.

### Exercises

**Q4.18** Consider two indistinguishable bosons in an infinite square well, interacting via the truncated Coulomb potential of Eq. 4.71. Calculate the expectation value of the inter-particle distance,  $\langle x_1 - x_2 \rangle$ , and its variance,  $\langle (x_1 - x_2)^2 \rangle - \langle x_1 - x_2 \rangle^2$ , in the ground state as a function of the Coulomb interaction strength (attractive and repulsive). *Hint:* set  $\delta = \Delta = a/(n_{\text{max}} + 1)$  in Eq. 4.71.

**Q4.19** Answer **Q4.18** for two indistinguishable fermions.

## 4.4 One Particle in Several Dimensions [Supplementary Material 6]

An important application of the imaginary-time propagation method of Sect. 4.2.1 is the calculation of the shape of a three-dimensional Bose–Einstein condensate. In

this section we use such a calculation as an example of how to extend single-particle lattice quantum mechanics to more spatial dimensions.

The non-linear Hamiltonian describing a three-dimensional Bose–Einstein condensate in a harmonic trap (to use a very common case) is

$$\hat{\mathcal{H}} = -\frac{\hbar^2}{2m} \left( \frac{\partial^2}{\partial x^2} + \frac{\partial^2}{\partial y^2} + \frac{\partial^2}{\partial z^2} \right) + \frac{m}{2} (\omega_x^2 x^2 + \omega_y^2 y^2 + \omega_z^2 z^2) + (N-1) \frac{4\pi\hbar^2 a_s}{m} |\psi(x, y, z)|^2, \quad (4.72)$$

where we have assumed that the single-particle wavefunction  $\psi(x, y, z)$  is normalized:  $\int |\psi(x, y, z)|^2 dx dy dz = 1$ . As before, the contact interaction is described by the  $s$ -wave scattering length  $a_s$ . We will call  $\kappa = \frac{4\pi\hbar^2 a_s}{m}$  the interaction constant, as in previous sections.

We perform this calculation in a square box, where  $|x| \leq \frac{a}{2}$ ,  $|y| \leq \frac{a}{2}$ , and  $|z| \leq \frac{a}{2}$ ; we will need to choose  $a$  large enough so that the BEC fits into this box, but small enough so that we do not need an unreasonably large  $n_{\max}$  for the description of its wavefunction. Notice that this box is shifted by  $\frac{a}{2}$  compared to the  $[0 \dots a]$  boxes used so far; this does not influence the calculations in any way.

The ground state of the non-linear Hamiltonian of Eq. 4.72 can be found by three-dimensional imaginary-time propagation, starting from (almost) any arbitrary state. Here we assemble a Mathematica function `groundstate` that, given an imaginary time step  $\delta\beta$ , propagates a random initial state until the state is converged to the ground state.

The units of the problem are dealt with as in Sect. 4.1.1, differing from `In[409]` ff in that here we choose the length and time units freely:

```

1 In[550]:=LengthUnit = Quantity["Micrometers"]; (* choose freely *)
2 In[551]:=TimeUnit = Quantity["Seconds"]; (* choose freely *)
3 In[552]:=MassUnit = Quantity["ReducedPlanckConstant"]*TimeUnit/LengthUnit^2;
4 In[553]:=EnergyUnit = Quantity["ReducedPlanckConstant"]/TimeUnit;
5 In[554]:=h = Quantity["ReducedPlanckConstant"]/(EnergyUnit*TimeUnit);

```

We will be considering  $N = 1000$   $^{87}\text{Rb}$  atoms in a magnetic trap with trap frequencies  $\omega_x = 2\pi \times 115\text{Hz}$  and  $\omega_y = \omega_z = 2\pi \times 540\text{Hz}$ . The  $^{87}\text{Rb}$  atoms are assumed to be in the  $|F = 1, M_F = -1\rangle$  hyperfine ground state, where their  $s$ -wave scattering length is  $a_s = 100.4a_0$  (with  $a_0 = 52.9177\text{pm}$  the Bohr radius).

```

1 In[555]:=m = Quantity[86.909187, "AtomicMassUnit"]/MassUnit;
2 In[556]:=a = Quantity[10, "Micrometers"]/LengthUnit;
3 In[557]:=wx = 2*\pi*Quantity[115, "Hertz"]*TimeUnit;
4 In[558]:=wy = 2*\pi*Quantity[540, "Hertz"]*TimeUnit;
5 In[559]:=wz = 2*\pi*Quantity[540, "Hertz"]*TimeUnit;
6 In[560]:=as = Quantity[100.4, "BohrRadius"]/LengthUnit;
7 In[561]:=kappa = 4*\pi*\hbar^2*as/m;

```

Next we define the grid on which the calculations will be done. In each Cartesian direction there are  $n_{\max}$  grid points  $x_j = \text{xgrid}[[j]]$  on the interval  $[-a/2, +a/2]$ :

```

1 In[562]:=nmax = 50;
2 In[563]:=Delta = a/(nmax+1);
3 In[564]:=xgrid = a*(Range[nmax]/(nmax+1) - 1/2);

```

We define the dimensionless harmonic-trap potential: the potential has its minimum at the center of the calculation box, i.e., at  $x = y = z = 0$ .

```
1 In[565]:=W[x_,y_,z_] = m/2 * (ωx^2*x^2 + ωy^2*y^2 + ωz^2*z^2);
```

We only need the values of this potential on the grid points. To evaluate this, we build a three-dimensional array whose element `Wgrid[[jx,jy,jz]]` is given by the grid-point value `W[xgrid[[jx]],xgrid[[jy]],xgrid[[jz]]]`:

```
1 In[566]:=Wgrid=Table[W[xgrid[[jx]],xgrid[[jy]],xgrid[[jz]]],{jx,nmax},{jy,nmax},{jz,nmax}];
```

We could also define this more efficiently through functional programming:

```
1 In[567]:=Wgrid = Outer[W, xgrid, xgrid, xgrid];
```

The structure of the three-dimensional `Wgrid` array of potential values mirrors the structure of the wavefunction that we will be using: any wavefunction  $\mathbf{v}$  will be a  $n_{\max} \times n_{\max} \times n_{\max}$  array of coefficients in our finite-resolution position basis:

$$\psi(x, y, z) = \sum_{j_x, j_y, j_z=1}^{n_{\max}} \mathbf{v}[[j_x, j_y, j_z]] \vartheta_{j_x}(x) \vartheta_{j_y}(y) \vartheta_{j_z}(z). \quad (4.73)$$

From Eq. 4.13 we find that on the three-dimensional grid points the wavefunction takes the values

$$\psi(x_{j_x}, x_{j_y}, x_{j_z}) = \mathbf{v}[[j_x, j_y, j_z]] / \Delta^{3/2}. \quad (4.74)$$

The norm of a wavefunction is

$$\int_{-\infty}^{\infty} |\psi(x, y, z)|^2 dx dy dz = \sum_{j_x, j_y, j_z=1}^{n_{\max}} |\mathbf{v}[[j_x, j_y, j_z]]|^2 = \text{Norm}[\text{Flatten}[\mathbf{v}]]^2, \quad (4.75)$$

from which we define a wavefunction normalization function

```
1 In[568]:=nn[v_] := v/Norm[Flatten[v]]
```

The ground state calculation then proceeds by imaginary-time propagation, with step size  $\delta\beta$  corresponding to an evolution  $e^{-\delta\beta\hat{\mathcal{H}}}$  per step. The calculation is done for  $N = n$  particles. Remember that the `FourierDST` function can do multi-dimensional discrete sine transforms, and therefore the kinetic-energy propagator can still be evaluated very efficiently. The last argument, `tolerance`, is optional and is given the value  $10^{-6}$  if not specified (see Sect. 1.6.5).

```

1 In[569]:=groundstate[n_?NumericQ,  $\delta\beta$ _?NumericQ, tolerance_:10^(-6)] :=
2   Module[{Kn, Ke, propKin, propPot2, prop, v0,  $\gamma$ , Ekin, Epot, Eint, Etot,  $\mu$ },
3     (* compute the diagonal elements of  $\exp[-\delta\beta*T]$  *)
4     Kn =  $\pi^2*\hbar^2/(2*m*a^2)*Table[nx^2+ny^2+nz^2,$ 
5       {nx,nmax}, {ny,nmax}, {nz,nmax}];
6     Ke = Exp[- $\delta\beta*Kn$ ] //N;
7     (* propagate by a full imaginary-time-step with T *)
8     propKin[v_] := nn[FourierDST[Ke*FourierDST[v, 1], 1]];
9     (* propagate by a half imaginary-time-step with V *)
10    propPot2[v_] := nn[Exp[-( $\delta\beta/2$ )*(Wgrid+ $\kappa$ (n-1)*Abs[v]^2/ $\Delta^3$ )]*v];
11    (* propagate by a full imaginary-time-step by *)
12    (* H=T+V using the Trotter approximation *)
13    prop[v_] := propPot2[propKin[propPot2[v]]]
14    (* random starting point *)
15    v0 = nn @ RandomVariate[NormalDistribution[], {nmax, nmax, nmax}];
16    (* propagation to the ground state *)
17     $\gamma$  = FixedPoint[prop, v0,
18      SameTest -> Function[{v1,v2}, Norm[Flatten[v1-v2]]<tolerance]];
19    (* energy components *)
20    Ekin = Flatten[Kn].Flatten[Abs[FourierDST[ $\gamma$ , 1]]^2];
21    Epot = Flatten[Wgrid].Flatten[Abs[ $\gamma$ ]^2];
22    Eint = ( $\kappa/2$ )*(n-1)*Total[Flatten[Abs[ $\gamma$ ]^4]]/ $\Delta^3$ ;
23    (* total energy *)
24    Etot = Ekin + Epot + Eint;
25    (* chemical potential *)
26     $\mu$  = Ekin + Epot + 2*Eint;
27    (* return energy, chemical potential, coefficients *)
28    {Etot,  $\mu$ ,  $\gamma$ }]

```

As an example, we calculate the ground state for  $N = 1000$  atoms and a time step of  $\delta\beta = 10^{-5}$  time units, using the default convergence tolerance:

```

1 In[570]:= {Etot,  $\mu$ ,  $\gamma$ } = groundstate[1000, 10^(-4)];
2 In[571]:= {Etot,  $\mu$ } * UnitConvert[EnergyUnit, "Joules"]
3 Out[571]= {6.88125*10^-31 J, 8.68181*10^-31 J}

```

A more common energy unit is the Hertz, arrived at via Planck's constant:

```

1 In[572]:= {Etot,  $\mu$ } * UnitConvert[EnergyUnit/Quantity["PlanckConstant"], "Hertz"]
2 Out[572]= {1038.51 Hz, 1310.25 Hz}

```


One way of plotting the ground-state density in 3D is as an iso-density surface. We plot the surface at half the peak density with

```

1 In[573]:=  $\rho$  = Abs[ $\gamma$ ]^2/ $\Delta^3$ ;
2 In[574]:= ListContourPlot3D[ $\rho$ ,
3   DataRange -> a*(1/(nmax+1)-1/2)*{{-1,1},{-1,1},{-1,1}},
4   Contours -> {Max[ $\rho$ ]/2}, BoxRatios -> Automatic]

```

Here we show several such iso-density surfaces:



For more quantitative results we can, for example, calculate the expectation values  $X = \langle x \rangle$ ,  $Y = \langle y \rangle$ ,  $Z = \langle z \rangle$ ,  $XX = \langle x^2 \rangle$ ,  $YY = \langle y^2 \rangle$ ,  $ZZ = \langle z^2 \rangle$ . We could define coordinate arrays as

```
1 In[575]:=xc = Table[xgrid[[jx]], {jx,nmax}, {jy,nmax}, {jz,nmax}];
2 In[576]:=yc = Table[xgrid[[jy]], {jx,nmax}, {jy,nmax}, {jz,nmax}];
3 In[577]:=zc = Table[xgrid[[jz]], {jx,nmax}, {jy,nmax}, {jz,nmax}];
```

but we define them more efficiently as follows:

```
1 In[578]:=ones = ConstantArray[1, nmax];
2 In[579]:=xc = Outer[Times, xgrid, ones, ones];
3 In[580]:=yc = Outer[Times, ones, xgrid, ones];
4 In[581]:=zc = Outer[Times, ones, ones, xgrid];
```

The desired expectation values are then computed with

```
1 In[582]:=X = Total[Flatten[xc * ρ]];
2 In[583]:=Y = Total[Flatten[yc * ρ]];
3 In[584]:=Z = Total[Flatten[zc * ρ]];
4 In[585]:=XX = Total[Flatten[xc^2 * ρ]];
5 In[586]:=YY = Total[Flatten[yc^2 * ρ]];
6 In[587]:=ZZ = Total[Flatten[zc^2 * ρ]];
```

The root-mean-square size of the BEC is calculated from these as the standard deviations of the position operators in the three Cartesian directions:

```
1 In[588]:= {Sqrt[XX-X^2], Sqrt[YY-Y^2], Sqrt[ZZ-Z^2]} * LengthUnit
2 Out[588]= {1.58829 μm, 0.417615 μm, 0.417615 μm}
```

### 4.4.1 Exercises

**Q4.20** Take the BEC Hamiltonian of Eq. 4.72 in the absence of interactions ( $a_s = 0$ ) and calculate analytically the expectation values  $\langle x^2 \rangle$ ,  $\langle y^2 \rangle$ ,  $\langle z^2 \rangle$  in the ground state.



**Q4.21** Take the BEC Hamiltonian of Eq. 4.72 in the limit of strong interactions (Thomas–Fermi limit), where the kinetic energy can be neglected. The Gross–Pitaevskii equation is then

$$\left[ \frac{m}{2} (\omega_x^2 x^2 + \omega_y^2 y^2 + \omega_z^2 z^2) + (N-1) \frac{4\pi\hbar^2 a_s}{m} |\psi(x, y, z)|^2 \right] \psi(x, y, z) = \mu \psi(x, y, z), \quad (4.76)$$

which has two solutions:

$$|\psi(x, y, z)|^2 = \begin{cases} 0 & \text{or} \\ \frac{\mu - \frac{m}{2} (\omega_x^2 x^2 + \omega_y^2 y^2 + \omega_z^2 z^2)}{(N-1) \frac{4\pi\hbar^2 a_s}{m}}. \end{cases} \quad (4.77)$$

Together with the conditions that  $|\psi(x, y, z)|^2 \geq 0$ , that  $\psi(x, y, z)$  should be continuous, and that  $\int |\psi(x, y, z)|^2 dx dy dz = 1$ , this gives us the Thomas–Fermi “inverted parabola” density

$$|\psi(x, y, z)|^2 = \begin{cases} \rho_0 \left[ 1 - \left( \frac{x}{R_x} \right)^2 - \left( \frac{y}{R_y} \right)^2 - \left( \frac{z}{R_z} \right)^2 \right] & \text{if } \left( \frac{x}{R_x} \right)^2 + \left( \frac{y}{R_y} \right)^2 + \left( \frac{z}{R_z} \right)^2 \leq 1, \\ 0 & \text{if not,} \end{cases} \quad (4.78)$$

which is nonzero only inside an ellipsoid with Thomas–Fermi radii

$$R_x = \left[ \frac{15\hbar^2 a_s (N-1) \omega_y \omega_z}{m^2 \omega_x^4} \right]^{\frac{1}{5}} = \left[ \frac{15\kappa (N-1) \omega_y \omega_z}{4\pi m \omega_x^4} \right]^{\frac{1}{5}}, \quad (4.79a)$$

$$R_y = \left[ \frac{15\hbar^2 a_s (N-1) \omega_z \omega_x}{m^2 \omega_y^4} \right]^{\frac{1}{5}} = \left[ \frac{15\kappa (N-1) \omega_x \omega_z}{4\pi m \omega_y^4} \right]^{\frac{1}{5}}, \quad (4.79b)$$

$$R_z = \left[ \frac{15\hbar^2 a_s (N-1) \omega_x \omega_y}{m^2 \omega_z^4} \right]^{\frac{1}{5}} = \left[ \frac{15\kappa (N-1) \omega_x \omega_y}{4\pi m \omega_z^4} \right]^{\frac{1}{5}}. \quad (4.79c)$$

The density at the origin of the ellipsoid is

$$\rho_0 = \frac{1}{8\pi} \left[ \frac{225m^6 \omega_x^2 \omega_y^2 \omega_z^2}{\hbar^6 a_s^3 (N-1)^3} \right]^{\frac{1}{5}} = \left[ \frac{225m^3 \omega_x^2 \omega_y^2 \omega_z^2}{512\pi^2 \kappa^3 (N-1)^3} \right]^{\frac{1}{5}} \quad (4.80)$$

and the chemical potential is

$$\mu = \frac{1}{2} \left[ 225m\hbar^4 a_s^2 (N-1)^2 \omega_x^2 \omega_y^2 \omega_z^2 \right]^{\frac{1}{5}} = \left[ \frac{225}{512\pi^2} m^3 \kappa^2 (N-1)^2 \omega_x^2 \omega_y^2 \omega_z^2 \right]^{\frac{1}{5}}. \quad (4.81)$$


Using this Thomas–Fermi density profile, calculate the expectation values  $\langle x^2 \rangle$ ,  $\langle y^2 \rangle$ ,  $\langle z^2 \rangle$  in the ground state of the Thomas–Fermi approximation.

*Hints:* Calculate  $\langle x^2 \rangle$  using Eq. 4.78 without substituting Equations 4.79 and Eq. 4.80; do these substitutions only after having found the result. You can find  $\langle y^2 \rangle$  and  $\langle z^2 \rangle$  by analogy, without repeating the calculation.

- Q4.22** Compare the numerical expectation values  $\langle x^2 \rangle$ ,  $\langle y^2 \rangle$ ,  $\langle z^2 \rangle$  of our Mathematica code to the analytic results of Q4.20 and Q4.21. What is the maximum  $^{87}\text{Rb}$  atom number  $N$  which allows a reasonably good description (in this specific trap) with the non-interacting solution? What is the minimum atom number which allows a reasonably good description with the Thomas–Fermi solution?

## Chapter 5

# Combining Spatial Motion and Spin



In this chapter we put together all the techniques studied so far: internal-spin degrees of freedom (Chap. 3) and spatial (motional) degrees of freedom (Chap. 4) are combined with the tensor-product formalism (Chap. 2). We arrive at a complete numerical

---

**Electronic supplementary material** The online version of this chapter ([https://doi.org/10.1007/978-981-13-7588-0\\_5](https://doi.org/10.1007/978-981-13-7588-0_5)) contains supplementary material, which is available to authorized users.

description of interacting spin-ful particles moving through space. To showcase these powerful tools, we study Rashba coupling as well as the Jaynes–Cummings model.

## 5.1 One Particle in 1D with Spin

### 5.1.1 Separable Hamiltonian

The simplest problem combining a spatial and a spin degree of freedom in a meaningful way consists of a single spin-1/2 particle moving in one dimension in a state-selective potential:

$$\hat{\mathcal{H}} = -\frac{\hbar^2}{2m} \frac{d^2}{dx^2} + V_0(x) + V_z(x) \hat{S}_z, \quad (5.1)$$

where  $\hat{S}_z = \frac{1}{2} \hat{\sigma}_z$  is given by the Pauli matrix. As was said before, Eq. (5.1) is a short-hand notation of the full Hamiltonian

$$\hat{\mathcal{H}} = -\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} dx |x\rangle \frac{d^2}{dx^2} \langle x| \otimes \mathbb{1} + \int_{-\infty}^{\infty} dx |x\rangle V_0(x) \langle x| \otimes \mathbb{1} + \int_{-\infty}^{\infty} dx |x\rangle V_z(x) \langle x| \otimes \hat{S}_z, \quad (5.2)$$

where it is more evident that the first two terms act only on the spatial part of the wavefunction, while the third term couples the two degrees of freedom.

The Hilbert space of this particle consists of a one-dimensional degree of freedom  $x$ , which we had described in Chap. 4 with a basis built from square-well eigenstates, and a spin-1/2 degree of freedom  $\hat{\mathbf{S}} = \frac{1}{2} \hat{\boldsymbol{\sigma}}$  described in the Dicke basis (Chap. 3). This tensor-product structure of the Hilbert space allows us to simplify the matrix elements of the Hamiltonian by factoring out the spin degree of freedom,

$$\begin{aligned} \langle \phi, \uparrow | \hat{\mathcal{H}} | \psi, \uparrow \rangle &= -\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} \phi^*(x) \psi''(x) dx \langle \uparrow | \uparrow \rangle + \int_{-\infty}^{\infty} \phi^*(x) V_0(x) \psi(x) dx \langle \downarrow | \downarrow \rangle \\ &\quad + \frac{1}{2} \int_{-\infty}^{\infty} \phi^*(x) V_z(x) \psi(x) dx \langle \uparrow | \hat{\sigma}_z | \uparrow \rangle \\ &= -\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} \phi^*(x) \psi''(x) dx + \int_{-\infty}^{\infty} \phi^*(x) V_0(x) \psi(x) dx + \frac{1}{2} \int_{-\infty}^{\infty} \phi^*(x) V_z(x) \psi(x) dx \\ \langle \phi, \uparrow | \hat{\mathcal{H}} | \psi, \downarrow \rangle &= -\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} \phi^*(x) \psi''(x) dx \langle \uparrow | \downarrow \rangle + \int_{-\infty}^{\infty} \phi^*(x) V_0(x) \psi(x) dx \langle \uparrow | \downarrow \rangle \\ &\quad + \frac{1}{2} \int_{-\infty}^{\infty} \phi^*(x) V_z(x) \psi(x) dx \langle \uparrow | \hat{\sigma}_z | \downarrow \rangle \\ &= 0 \end{aligned}$$

$$\begin{aligned}
\langle \phi, \downarrow | \hat{\mathcal{H}} | \psi, \uparrow \rangle &= -\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} \phi^*(x) \psi''(x) dx \langle \downarrow | \uparrow \rangle + \int_{-\infty}^{\infty} \phi^*(x) V_0(x) \psi(x) dx \langle \downarrow | \uparrow \rangle \\
&\quad + \frac{1}{2} \int_{-\infty}^{\infty} \phi^*(x) V_z(x) \psi(x) dx \langle \downarrow | \hat{\sigma}_z | \uparrow \rangle \\
&= 0 \\
\langle \phi, \downarrow | \hat{\mathcal{H}} | \psi, \downarrow \rangle &= -\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} \phi^*(x) \psi''(x) dx \langle \downarrow | \downarrow \rangle + \int_{-\infty}^{\infty} \phi^*(x) V_0(x) \psi(x) dx \langle \downarrow | \downarrow \rangle \\
&\quad + \frac{1}{2} \int_{-\infty}^{\infty} \phi^*(x) V_z(x) \psi(x) dx \langle \downarrow | \hat{\sigma}_z | \downarrow \rangle \\
&= -\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} \phi^*(x) \psi''(x) dx + \int_{-\infty}^{\infty} \phi^*(x) V_0(x) \psi(x) dx - \frac{1}{2} \int_{-\infty}^{\infty} \phi^*(x) V_z(x) \psi(x) dx.
\end{aligned} \tag{5.3}$$

We see that this Hamiltonian does not mix states with different spin states (since all matrix elements where the spin state differs between the left and right side are equal to zero). We can therefore solve the two disconnected problems of finding the particle's behavior with spin up or with spin down, with effective Hamiltonians

$$\hat{\mathcal{H}}_{\uparrow} = -\frac{\hbar^2}{2m} \frac{d^2}{dx^2} + V_0(x) + \frac{1}{2} V_z(x), \quad \hat{\mathcal{H}}_{\downarrow} = -\frac{\hbar^2}{2m} \frac{d^2}{dx^2} + V_0(x) - \frac{1}{2} V_z(x). \tag{5.4}$$

These Hamiltonians now only describe the spatial degree of freedom, and the methods of Chap. 4 can be used without further modifications.

### 5.1.2 Non-separable Hamiltonian

A more interesting situation arises when the Hamiltonian is not separable as in Sect. 5.1.1. Take, for example, the Hamiltonian of Eq. (5.1) in the presence of a uniform transverse magnetic field  $B_x$ ,

$$\hat{\mathcal{H}} = -\frac{\hbar^2}{2m} \frac{d^2}{dx^2} + V_0(x) + V_z(x) \hat{S}_z + B_x \hat{S}_x. \tag{5.5}$$

The interaction Hamiltonian with the magnetic field is not separable:

$$\begin{aligned}
\langle \phi, \uparrow | B_x \hat{S}_x | \psi, \uparrow \rangle &= \frac{1}{2} B_x \int_{-\infty}^{\infty} \phi^*(x) \psi(x) dx \langle \uparrow | \hat{\sigma}_x | \uparrow \rangle = 0 \\
\langle \phi, \uparrow | B_x \hat{S}_x | \psi, \downarrow \rangle &= \frac{1}{2} B_x \int_{-\infty}^{\infty} \phi^*(x) \psi(x) dx \langle \uparrow | \hat{\sigma}_x | \downarrow \rangle = \frac{1}{2} B_x \int_{-\infty}^{\infty} \phi^*(x) \psi(x) dx \\
\langle \phi, \downarrow | B_x \hat{S}_x | \psi, \uparrow \rangle &= \frac{1}{2} B_x \int_{-\infty}^{\infty} \phi^*(x) \psi(x) dx \langle \downarrow | \hat{\sigma}_x | \uparrow \rangle = \frac{1}{2} B_x \int_{-\infty}^{\infty} \phi^*(x) \psi(x) dx \\
\langle \phi, \downarrow | B_x \hat{S}_x | \psi, \downarrow \rangle &= \frac{1}{2} B_x \int_{-\infty}^{\infty} \phi^*(x) \psi(x) dx \langle \downarrow | \hat{\sigma}_x | \downarrow \rangle = 0.
\end{aligned} \tag{5.6}$$

Therefore we can no longer study separate Hamiltonians as in Eq. (5.4), and we must instead study the joint system of spatial motion and spin. In what follows we study a simple example of such a Hamiltonian, both analytically and numerically. We take the trapping potential to be harmonic,

$$V_0(x) = \frac{1}{2}m\omega^2 x^2 \quad (5.7)$$

and the state-selective potential as a homogeneous force,

$$V_z(x) = -Fx. \quad (5.8)$$

### Ground state for $B_x = 0$

For  $B_x = 0$  we know that the ground states of the two spin sectors are the ground states of the effective Hamiltonians of Eq. (5.4), which are Gaussians:

$$\langle x|\gamma_\uparrow\rangle = \frac{e^{-\left(\frac{x-\mu}{2\sigma}\right)^2}}{\sqrt{\sigma\sqrt{2\pi}}} \otimes |\uparrow\rangle \quad \langle x|\gamma_\downarrow\rangle = \frac{e^{-\left(\frac{x+\mu}{2\sigma}\right)^2}}{\sqrt{\sigma\sqrt{2\pi}}} \otimes |\downarrow\rangle \quad (5.9)$$

with  $\mu = \frac{F}{2m\omega^2}$  and  $\sigma = \sqrt{\frac{\hbar}{2m\omega}}$ . These two ground states are degenerate, with energy  $E = \frac{1}{2}\hbar\omega - \frac{F^2}{8m\omega^2}$ . In both of these ground states the spatial and spin degrees of freedom are entangled: the particle is more likely to be detected in the  $|\uparrow\rangle$  state on the right side ( $x > 0$ ), and more likely to be detected in the  $|\downarrow\rangle$  state on the left side ( $x < 0$ ) of the trap. This results in a positive expectation value of the operator  $\hat{x} \otimes \hat{S}_z$ :

$$\langle \gamma_\uparrow | \hat{x} \otimes \hat{S}_z | \gamma_\uparrow \rangle = \langle \gamma_\downarrow | \hat{x} \otimes \hat{S}_z | \gamma_\downarrow \rangle = \frac{\mu}{2} = \frac{F}{4m\omega^2}. \quad (5.10)$$

### Perturbative ground state for $B_x > 0$

For small  $|B_x|$  the ground state can be described by a linear combination of the states in Eq. (5.9). If we set

$$|\gamma_p\rangle = \alpha \times |\gamma_\uparrow\rangle + \beta \times |\gamma_\downarrow\rangle \quad (5.11)$$

with  $|\alpha|^2 + |\beta|^2 = 1$ , we find that the expectation value of the energy is

$$\begin{aligned} \langle \gamma_p | \hat{\mathcal{H}} | \gamma_p \rangle &= |\alpha|^2 \langle \gamma_\uparrow | \hat{\mathcal{H}} | \gamma_\uparrow \rangle + \alpha^* \beta \langle \gamma_\uparrow | \hat{\mathcal{H}} | \gamma_\downarrow \rangle + \beta^* \alpha \langle \gamma_\downarrow | \hat{\mathcal{H}} | \gamma_\uparrow \rangle + |\beta|^2 \langle \gamma_\downarrow | \hat{\mathcal{H}} | \gamma_\downarrow \rangle \\ &= \frac{1}{2}\hbar\omega - \frac{F^2}{8m\omega^2} + \frac{1}{2}B_x(\alpha^*\beta + \beta^*\alpha)e^{-\frac{F^2}{4m\hbar\omega^3}} \end{aligned} \quad (5.12)$$

For  $B_x > 0$  this energy is minimized for  $\alpha = 1/\sqrt{2}$  and  $\beta = -1/\sqrt{2}$ , and the perturbative ground state is therefore the anti-symmetric combination of the states in Eq. (5.9)

$$\langle x | \gamma_p \rangle = \frac{e^{-\left(\frac{x-\mu}{2\sigma}\right)^2}}{\sqrt{2\sigma\sqrt{2\pi}}} \otimes |\uparrow\rangle - \frac{e^{-\left(\frac{x+\mu}{2\sigma}\right)^2}}{\sqrt{2\sigma\sqrt{2\pi}}} \otimes |\downarrow\rangle. \quad (5.13)$$

with energy

$$\langle \gamma_p | \hat{\mathcal{H}} | \gamma_p \rangle = \frac{1}{2} \hbar \omega - \frac{F^2}{8m\omega^2} - \frac{1}{2} B_x e^{-\frac{F^2}{4m\hbar\omega^3}}. \quad (5.14)$$

The energy splitting between this ground state and the first excited state,

$$\langle x | \epsilon_p \rangle = \frac{e^{-\left(\frac{x-\mu}{2\sigma}\right)^2}}{\sqrt{2\sigma\sqrt{2\pi}}} \otimes |\uparrow\rangle + \frac{e^{-\left(\frac{x+\mu}{2\sigma}\right)^2}}{\sqrt{2\sigma\sqrt{2\pi}}} \otimes |\downarrow\rangle. \quad (5.15)$$

is  $\Delta E = \langle \epsilon_p | \hat{\mathcal{H}} | \epsilon_p \rangle - \langle \gamma_p | \hat{\mathcal{H}} | \gamma_p \rangle = B_x e^{-\frac{F^2}{4m\hbar\omega^3}}$ , which can be very small for large exponents  $\frac{F^2}{4m\hbar\omega^3}$ .

### Numerical Calculation of the Ground State [\[Supplementary Material 1\]](#)

For a numerical description of this particle we use dimensionless units such that  $a = m = \hbar = 1$ ; other units can be used in the same way as presented in Sect. 4.1.1. We describe the spatial degree of freedom with the finite-resolution position basis of Sect. 4.1.2, centered at  $x = 0$  as in Sect. 4.4:

```
1 In[589]:=a = m = ħ = 1;
2 In[590]:=nmax = 100;
3 In[591]:=Δ = a/(nmax+1);
4 In[592]:=xgrid = a*(Range[nmax]/(nmax+1)-1/2);
```

The operator  $\hat{x}$  is approximately diagonal in this representation (see Eq. (4.24)):

```
1 In[593]:=xop = SparseArray[Band[{1,1}] -> xgrid];
```

The identity operator on the spatial degree of freedom is

```
1 In[594]:=idx = IdentityMatrix[nmax, SparseArray];
```

The identity and Pauli operators for the spin degree of freedom are

```
1 In[595]:=ids = IdentityMatrix[2, SparseArray];
2 In[596]:= {sx,sy,sz}=Table[SparseArray[PauliMatrix[i]/2], {i,3}];
```

The kinetic energy operator is constructed via a discrete sine transform, as before:

```
1 In[597]:=TM = SparseArray[Band[{1,1}]->Range[nmax]^2*π^2*ħ^2/(2*m*a^2)];
2 In[598]:=TP = FourierDST[TM, 1];
```

From these we assemble the Hamiltonian, assuming that  $F$  and  $B_x$  are expressed in matching units:

```

1 In[599]:=H[ω_, F_, Bx_] =
2   KroneckerProduct[TP, ids]
3   + m*ω^2/2 * KroneckerProduct[xop.xop, ids]
4   - F * KroneckerProduct[xop, sz]
5   + Bx * KroneckerProduct[idx, sx];

```

We compute the ground state of this Hamiltonian with

```

1 In[600]:=Clear[gs];
2 In[601]:=gs[ω_?NumericQ, F_?NumericQ, Bx_?NumericQ] :=
3   gs[ω, F, Bx] = -Eigensystem[-H[N[ω],N[F],N[Bx]], 1,
4   Method -> {"Arnoldi", "Criteria" -> "RealPart", MaxIterations -> 10^6}]

```

Once a ground state  $|\gamma\rangle$  has been calculated, for example with

```

1 In[602]:=γ = gs[100, 5000, 500][[2, 1]];

```

the usual problem arises of how to display and interpret the wavefunction. Instead of studying the coefficients of  $\gamma$  directly, we calculate several specific properties of the ground state in what follows.

**Operator expectation values:** The mean spin direction (magnetization)  $\langle \hat{\mathbf{S}} \rangle = \{\langle \hat{S}_x \rangle, \langle \hat{S}_y \rangle, \langle \hat{S}_z \rangle\}$  is calculated directly from the ground-state coefficients list with

```

1 In[603]:=mx = Re[Conjugate[γ].(KroneckerProduct[idx,sx].γ)];
2 In[604]:=my = Re[Conjugate[γ].(KroneckerProduct[idx,sy].γ)];
3 In[605]:=mz = Re[Conjugate[γ].(KroneckerProduct[idx,sz].γ)];
4 In[606]:= {mx,my,mz}
5 Out[606]= {-0.233037, 0., -2.08318*10^-12}

```

The mean position  $\langle \hat{x} \rangle$  and its standard deviation are calculated with

```

1 In[607]:=X = Re[Conjugate[γ].(KroneckerProduct[xop,ids].γ)];
2 In[608]:=XX = Re[Conjugate[γ].(KroneckerProduct[xop.xop,ids].γ)];
3 In[609]:= {X, Sqrt[XX-X^2]}
4 Out[609]= {1.2178*10^-11, 0.226209}

```

Even though we found  $\langle \hat{x} \rangle = 0$  and  $\langle \hat{S}_z \rangle = 0$  above, these coordinates are correlated: calculating  $\langle \hat{x} \otimes \hat{S}_z \rangle$ ,

```


1 In[610]:=Xz = Re[Conjugate[γ].(KroneckerProduct[xop,sz].γ)]
2 Out[610]= 0.0954168

```



**Reduced density matrix of the spatial degree of freedom:** Using [In \[257\]](#) we trace out the spin degree of freedom (the last two dimensions) to find the density matrix in the spatial coordinate:

```
1 In[611]:= px = traceout[γ, -2];
2 In[612]:= ArrayPlot[Reverse[Transpose[ArrayPad[px/Δ, 1]]]]
```



**Reduced density matrix of the spin degree of freedom:** We can do the same for the reduced matrix of the spin degree of freedom, using [In \[256\]](#), and find a 2 spin density matrix:

```
1 In[613]:= ps = traceout[γ, nmax]
2 Out[613]= {{0.5, -0.233037}, {-0.233037, 0.5}}
```


**Spin-specific spatial densities:** The reduced density matrix of particles in the spin-up state is found by projecting the ground state  $|\gamma\rangle$  onto the spin-up sector with the projector  $\hat{\Pi}_\uparrow = |\uparrow\rangle\langle\uparrow| = \frac{1}{2}\mathbb{1} + \hat{S}_z$ .<sup>1</sup> Thus,  $|\gamma_\uparrow\rangle = \hat{\Pi}_\uparrow|\gamma\rangle$  only describes the particles that are in the spin-up state:

```
1 In[614]:= γup = KroneckerProduct[idx, ids/2+sz].γ;
2 In[615]:= pxup = traceout[γup, -2];
```

In the same way the reduced density matrix of particles in the spin-down state  $|\gamma_\downarrow\rangle = \hat{\Pi}_\downarrow|\gamma\rangle$  is calculated with the down-projector  $\hat{\Pi}_\downarrow = |\downarrow\rangle\langle\downarrow| = \frac{1}{2}\mathbb{1} - \hat{S}_z$ :

```
1 In[616]:= γdn = KroneckerProduct[idx, ids/2-sz].γ;
2 In[617]:= pxdn = traceout[γdn, -2];
```

<sup>1</sup>Remember that  $\mathbb{1} = |\uparrow\rangle\langle\uparrow| + |\downarrow\rangle\langle\downarrow|$  and  $\hat{S}_z = \frac{1}{2}|\uparrow\rangle\langle\uparrow| - \frac{1}{2}|\downarrow\rangle\langle\downarrow|$ .



The positive correlation between the spin and the mean position,  $\langle \hat{x} \otimes \hat{S}_z \rangle > 0$ , is clearly visible in these plots.

Since  $\hat{\Pi}_\uparrow + \hat{\Pi}_\downarrow = \mathbb{1}$ , these two spin-specific spatial density matrices add up to the total density shown previously. This also means that the spin-specific density matrices do not have unit trace:

```
1 In[618]:= {Tr[ρxup], Tr[ρxdown]}
2 Out[618]= {0.5, 0.5}
```

Hence we have 50% chance of finding the particle in the up or down spin states.

**Space-dependent spin expectation value:** Similarly, we can calculate the reduced density matrix of the spin degree of freedom at a specific point in space by using projection operators  $\hat{\Pi}_j = |j\rangle\langle j|$  onto single position-basis states  $|j\rangle$ :


```
1 In[619]:= γx[j_Integer /; 1 <= j <= nmax] :=
2           KroneckerProduct[SparseArray[{j, j} -> 1, {nmax, nmax}], ids].γ
3 In[620]:= psx[j_Integer /; 1 <= j <= nmax] := traceout[γx[j], nmax]
```

We notice that, as before, these spatially-local reduced density matrices do not have unit trace, but their traces sum up to 1:

```
1 In[621]:= Sum[Tr[psx[j]], {j, nmax}]
2 Out[621]= 1.
```

In fact, the traces of these local reduced density matrices give the probability of finding the particle at the given position. We can use this interpretation to calculate the mean spin expectation value of a particle measured at a given grid point:

```
1 In[622]:= meansx[j_Integer /; 1 <= j <= nmax] := Tr[psx[j].sz]/Tr[psx[j]]
2 In[623]:= ListLinePlot[Transpose[{xgrid, Table[meansx[j], {j, nmax}]}]]
```



This graph confirms the observation that particles detected on the left side are more likely to be in the  $|\downarrow\rangle$  state, while particles detected on the right side are more likely to be in the  $|\uparrow\rangle$  state.

### 5.1.3 Exercises

**Q5.1** In the problem described by the Hamiltonian of Eq. (5.5), calculate the following expectation values (numerically) for several parameter sets  $\{\omega, F, B_x\}$ :

1.  $\langle x \rangle$  for particles detected in the  $|\uparrow\rangle$  state
2.  $\langle x \rangle$  for particles detected in the  $|\downarrow\rangle$  state
3.  $\langle x \rangle$  for particles detected in any spin state
4. the mean and variance of  $\hat{x} \otimes \hat{S}_z$

## 5.2 One Particle in 2D with Spin: Rashba Coupling [Supplementary Material 2]

A particularly interesting kind of interaction is the Rashba coupling between a particle's momentum and its spin.<sup>2</sup> In general, this interaction is proportional to a component of the vector product  $\hat{\kappa} = \hat{\mathbf{p}} \times \hat{\mathbf{S}}$ . For a particle moving in two dimensions  $(x, y)$ , the coupling involves the  $z$ -component  $\hat{\kappa}_z = \hat{p}_x \otimes \hat{S}_y - \hat{p}_y \otimes \hat{S}_x$ .

In this section we study the 2D Rashba Hamiltonian

$$\hat{\mathcal{H}} = \frac{\hat{p}_x^2 + \hat{p}_y^2}{2m} + V(x, y) + \delta \hat{S}_z + \alpha(\hat{p}_x \otimes \hat{S}_y - \hat{p}_y \otimes \hat{S}_x) \quad (5.16)$$

in a square box where  $-\frac{a}{2} \leq x, y \leq \frac{a}{2}$  as before. With a Hilbert space composed as the tensor product of the  $x$ ,  $y$ , and spin coordinates, in this order, the full Hamiltonian thus becomes

<sup>2</sup>See [https://en.wikipedia.org/wiki/Rashba\\_effect](https://en.wikipedia.org/wiki/Rashba_effect).

$$\begin{aligned} \hat{\mathcal{H}} = & \left[ -\frac{\hbar^2}{2m} \int_{-a/2}^{a/2} dx |x\rangle \frac{\partial^2}{\partial x^2} \langle x| \right] \otimes \mathbb{1} \otimes \mathbb{1} + \mathbb{1} \otimes \left[ -\frac{\hbar^2}{2m} \int_{-a/2}^{a/2} dy |y\rangle \frac{\partial^2}{\partial y^2} \langle y| \right] \otimes \mathbb{1} \\ & + \left[ \int_{-a/2}^{a/2} dx dy |x\rangle \langle y| V(x, y) \langle x| \langle y| \right] \otimes \mathbb{1} + \delta(\mathbb{1} \otimes \mathbb{1} \otimes \hat{S}_z) + \alpha(\hat{p}_x \otimes \mathbb{1} \otimes \hat{S}_y - \mathbb{1} \otimes \hat{p}_y \otimes \hat{S}_x). \end{aligned} \quad (5.17)$$

For simplicity we will set  $V(x, y) = 0$ ; but any nonzero potential can be used with the techniques introduced previously. Further, we use  $a = m = \hbar = 1$  to simplify the units; but as usual, any system of units may be used (see Sect. 4.1.1).

Since both the kinetic and the interaction operator are most easily expressed in the momentum representation, we use the momentum representation (see Sect. 4.1.2) to express the spatial degrees of freedom of the Hamiltonian. The identity operator is

```
1 In[624]:=nmax = 50;
2 In[625]:=Δ = a/(nmax+1);
3 In[626]:=idM = IdentityMatrix[nmax, SparseArray];
```

We use the exact form of the kinetic operator from In[423] and the exact form of the momentum operator from In[444]. As discussed previously, these two forms do not exactly satisfy  $\hat{T} = \hat{p}^2/(2m)$ . They are, however, the best available low-energy forms.

For the spin degree of freedom, we assume  $S = 1/2$ , giving us the usual spin operators and the identity operator,

```
1 In[627]:= {sx,sy,sz} = Table[SparseArray[PauliMatrix[i]/2], {i, 3}];
2 In[628]:= idS = IdentityMatrix[2, SparseArray];
```

With these definitions, we assemble the Rashba Hamiltonian of Eq. (5.17) in the momentum representation with

```
1 In[629]:= HM[δ_, α_] = KroneckerProduct[TM, idM, idS]
2 + KroneckerProduct[idM, TM, idS]
3 + δ*KroneckerProduct[idM, idM, sz]
4 + α*(KroneckerProduct[pM, idM, sy] - KroneckerProduct[idM, pM, sx]);
```

Given a state  $\gamma$ , for example the ground state of In[629] for specific values of  $\delta = 1$  and  $\alpha = 20$ , we calculate the mean value  $\langle \hat{x}^2 \rangle = \langle \hat{x}^2 \otimes \mathbb{1} \otimes \mathbb{1} \rangle$  with the position operator  $xM$  expressed in the momentum basis:

```
1 In[630]:= xgrid = a*(Range[nmax]/(nmax + 1) - 1/2);
2 In[631]:= xP = SparseArray[Band[{1, 1}] -> xgrid];
3 In[632]:= xM = FourierDST[xP, 1];
4 In[633]:= Conjugate[γ].(KroneckerProduct[xM.xM, idM, idS].γ) //Re
5 Out[633]=0.0358875
```

In the same way, we calculate the mean value  $\langle \hat{y}^2 \rangle = \langle \mathbb{1} \otimes \hat{y}^2 \otimes \mathbb{1} \rangle$ :

```
1 In[634]:= Conjugate[γ].(KroneckerProduct[idM, xM.xM, idS].γ) //Re
2 Out[634]=0.0358875
```


In order to study the spatial variation of the spin (the expectation value of the spin degree of freedom if the particle is detected at a specific spatial location), we calculate the reduced density matrix of the spin degree of freedom at a specific grid point  $(x_i, y_j)$  of the position grid.<sup>3</sup> For this, we first project the ground-state wavefunction  $\gamma$  onto the spatial grid point at  $x = x_i$  and  $y = y_j$  using the projector  $|i\rangle\langle i| \otimes |j\rangle\langle j|$  in the momentum representation:

```
1 In[635] := PP[j_] := SparseArray[{j, j} -> 1, {nmax, nmax}]
2 In[636] := PM[j_] := FourierDST[PP[j], 1]
3 In[637] := gP[i_, j_] := KroneckerProduct[PM[i], PM[j], idS].γ
```

Tracing out the spatial degrees of freedom with the procedure of Sect. 2.4.3 gives the  $2 \times 2$  spin density matrix at the desired grid point,

```
1 In[638] := RsP[i_, j_] := traceout[gP[i, j], nmax^2]
```

The trace  $\text{Tr}[\text{RsP}[i, j]]$  of such a reduced density matrix gives the probability of finding the particle at grid point  $(x_i, y_j)$ :




We can extract more information from these reduced spin density matrices: the magnetization (mean spin direction) at a grid point has the Cartesian components

```
1 In[639] := mxP[i_, j_] := Re[Tr[RsP[i, j].sx]/Tr[RsP[i, j]]]
2 In[640] := myP[i_, j_] := Re[Tr[RsP[i, j].sy]/Tr[RsP[i, j]]]
3 In[641] := mzP[i_, j_] := Re[Tr[RsP[i, j].sz]/Tr[RsP[i, j]]]
```

<sup>3</sup>Naturally, the following calculations would be simpler if we had represented the ground state in the position basis; however, we use this opportunity to show how to calculate in the momentum basis.

Plotting these components over the entire grid shows interesting patterns of the mean spin orientation (magnetization) in the ground state:



### 5.2.1 Exercises

**Q5.2** While the Hamiltonian of Eqs. (5.16) and (5.17) contains the distinct operators  $\hat{p}_x$  and  $\hat{p}_y$ , the Mathematica form of this Hamiltonian assembled in `In[629]` contains the same matrix `pM` representing both  $\hat{p}_x$  and  $\hat{p}_y$ . Why is this so? What distinguishes the Mathematica representations of these two operators?

## 5.3 Phase-Space Dynamics in the Jaynes–Cummings Model [Supplementary Material 3]

As a final example, we study the interaction of an atom with the light field in an optical cavity. The atom is assumed to have only two internal states: the ground state  $|g\rangle$  and some excited state  $|e\rangle$ . The atomic state is described as a (pseudo-)spin-1/2 system with the operators (see Q5.3)

$$\hat{S}_x = \frac{|e\rangle\langle g| + |g\rangle\langle e|}{2}, \quad \hat{S}_y = \frac{|e\rangle\langle g| - |g\rangle\langle e|}{2i}, \quad \hat{S}_z = \frac{|e\rangle\langle e| - |g\rangle\langle g|}{2}, \quad (5.18)$$

as well as  $\hat{S}^\pm = \hat{S}_x \pm i\hat{S}_y$  (see Sect. 3.2). The cavity field is assumed to consist of only one mode, described with creation and annihilation operators  $\hat{a}^\dagger$  and  $\hat{a}$ , respectively; all other cavity modes are assumed to be so far off-resonant that they are not coupled to the atom.

The Jaynes–Cummings Hamiltonian<sup>4</sup> describing the combined system, as well as the coupling between the atom and the cavity field, is

<sup>4</sup>See [https://en.wikipedia.org/wiki/Jaynes-Cummings\\_model](https://en.wikipedia.org/wiki/Jaynes-Cummings_model).

$$\hat{\mathcal{H}}_{\text{JC}} = \underbrace{\hbar\omega_a \hat{S}_z}_{\text{atom}} + \underbrace{\hbar\omega_c (\hat{a}^\dagger \hat{a} + \frac{1}{2})}_{\text{cavity field}} + \underbrace{\hbar g (\hat{S}^+ \hat{a} + \hat{a}^\dagger \hat{S}^-)}_{\text{coupling}}. \quad (5.19)$$

- The atomic Hamiltonian describes the energy difference  $\hbar\omega_a$  between the two internal states of the atom.
- The cavity field Hamiltonian describes the energy of  $\hat{n} = \hat{a}^\dagger \hat{a}$  photons in the cavity mode, each photon carrying an energy  $\hbar\omega_c$ .
- The coupling term describes the deexcitation of the field  $\hat{a}$  together with the excitation of the atom  $\hat{S}^+$ , as well as the reverse process of the excitation of the field  $\hat{a}^\dagger$  together with the deexcitation of the atom  $\hat{S}^-$  (see [Q5.4](#)).

The cavity mode of the Jaynes–Cummings model is usually studied in the Fock basis of definite photon number, using harmonic-oscillator eigenfunctions as basis states. Here we take an alternative approach and look at the  $X - P$  phase space spanned by the dimensionless quadrature operators  $\hat{X}$  and  $\hat{P}$ ,<sup>5</sup> which are related to the creation and annihilation operators  $\hat{a}^\dagger$  and  $\hat{a}$  via

$$\begin{aligned} \hat{X} &= \frac{\hat{a} + \hat{a}^\dagger}{\sqrt{2}} & \hat{a} &= \frac{\hat{X} + i\hat{P}}{\sqrt{2}} = \frac{\hat{X} + \frac{\partial}{\partial \hat{X}}}{\sqrt{2}} \\ \hat{P} &= -i\frac{\partial}{\partial \hat{X}} = \frac{\hat{a} - \hat{a}^\dagger}{i\sqrt{2}} & \hat{a}^\dagger &= \frac{\hat{X} - i\hat{P}}{\sqrt{2}} = \frac{\hat{X} - \frac{\partial}{\partial \hat{X}}}{\sqrt{2}} \end{aligned} \quad (5.20)$$

with the commutators  $[\hat{a}, \hat{a}^\dagger] = 1$  and  $[\hat{X}, \hat{P}] = i$  (see [Q5.5](#)). We note that the quadrature  $\hat{X}$  is the amplitude of the electromagnetic field of the cavity mode, and  $\hat{P}$  its conjugate momentum; there is no motion in real space in this problem, only in amplitude space. Using these quadrature operators, we write the Jaynes–Cummings Hamiltonian as (see [Q5.6](#))

$$\begin{aligned} \hat{\mathcal{H}}_{\text{JC}} &= \hbar\omega_a \hat{S}_z + \hbar\omega_c \left( \frac{1}{2} \hat{P}^2 + \frac{1}{2} \hat{X}^2 \right) + \sqrt{2}\hbar g (\hat{X} \hat{S}_x - \hat{P} \hat{S}_y) \\ &= \hbar\omega_a \mathbb{1} \otimes \hat{S}_z + \hbar\omega_c \left( \frac{1}{2} \hat{P}^2 + \frac{1}{2} \hat{X}^2 \right) \otimes \mathbb{1} + \sqrt{2}\hbar g (\hat{X} \otimes \hat{S}_x - \hat{P} \otimes \hat{S}_y), \end{aligned} \quad (5.21)$$

where we have made its tensor-product structure explicit in the second line. To assemble this Hamiltonian in Mathematica, we define the Hilbert space to be the tensor product of the  $X - P$  phase space and the spin-1/2 space, in this order.

The phase space is defined as before (see Chap. 4) in a calculation box  $X \in [-\frac{a}{2}, \frac{a}{2}]$  divided into a grid of  $n_{\text{max}} + 1$  intervals. We choose  $a$  such that the state fits well into the box (considering that the ground state of the cavity field has a

---

<sup>5</sup>In a harmonic oscillator of mass  $m$  and angular frequency  $\omega$ , we usually introduce the position operator  $\hat{x} = \sqrt{\frac{\hbar}{m\omega}} \hat{X}$  and the momentum operator  $\hat{p} = \sqrt{\hbar m \omega} \hat{P}$ . Here we restrict our attention to the dimensionless quadratures  $\hat{X}$  and  $\hat{P}$ .

size  $\langle \hat{X}^2 \rangle^{1/2} = \langle \hat{P}^2 \rangle^{1/2} = 1/\sqrt{2}$ , and we choose  $n_{\max}$  such that the Wigner quasi-probability distribution plots have equal ranges in  $X$  and  $P$ . Naturally, any other values of  $a$  and  $n_{\max}$  can be chosen.

```

1 In[642]:=h = 1;    (* natural units *)
2 In[643]:=a = 10;
3 In[644]:=nmax = Round[a^2/π]
4 Out[644]=32
5 In[645]:=Δ = a/(nmax+1);

```

We represent the phase space in the position basis. The  $\hat{X}$  quadrature operator is defined as in Eq. (4.24),

```

1 In[646]:=xgrid = a*(Range[nmax]/(nmax+1) - 1/2);
2 In[647]:=X = SparseArray[Band[{1, 1}] -> xgrid];

```

The definition of the  $\hat{P}$  quadrature operator follows In[444], with  $\hat{P}^2$  defined directly through In[423] for better accuracy at finite  $n_{\max}$ :

```

1 In[648]:=P = FourierDST[SparseArray[{n1_, n2_} /; OddQ[n1-n2] ->
2 4*I*n1*n2]/(a*(n2^2-n1^2)), {nmax, nmax}], 1];
3 In[649]:=P2 = FourierDST[SparseArray[Band[{1, 1}] -> Range[nmax]^2*π^2/a^2], 1];

```

Finally, the phase-space identity operator is

```

1 In[650]:=idX = IdentityMatrix[nmax, SparseArray];

```

The operators on the pseudo-spin degree of freedom are defined directly from the Pauli matrices instead of using the general definitions of Eqs. (3.1) and (3.2):

```

1 In[651]:= {Sx, Sy, Sz} = Table[SparseArray[PauliMatrix[i]/2], {i, 3}];
2 In[652]:= idS = IdentityMatrix[2, SparseArray];

```

The Hamiltonian of Eq. (5.21) is assembled from three parts:

```

1 In[653]:=Ha = KroneckerProduct[idX, Sz];
2 In[654]:=Hc = KroneckerProduct[X.X/2 + P2/2, idS];
3 In[655]:=Hint = Sqrt[2]*(KroneckerProduct[X,Sx]-Re[KroneckerProduct[P,Sy]]);
4 In[656]:=HP[ωa_, ωc_, g_] = h*ωa*Ha + h*ωc*Hc + h*g*Hint;

```

Remember that we use `P2` instead of `P.P` for the operator  $\hat{P}^2$  for better accuracy. We use the `Re` operator in In[655] to eliminate the imaginary parts, which are zero by construction but render the expression `Complex`-valued nonetheless.

In the Mathematica notebook attached to this section, the dynamics induced by this time-independent Hamiltonian is studied in the weak and strong coupling regimes, using the technique of Sect. 2.3.4 to propagate the initial wavefunction.

Given a calculated space⊗spin wavefunction  $\psi$  (a vector of  $2n_{\max}$  complex numbers), we calculate the  $n_{\max} \times n_{\max}$  reduced density matrix of the phase-space degree of freedom (cavity field) with In[257], tracing out the spin degree of freedom (the last 2 dimensions):



```
1 In[657] := pX = traceout[ψ, -2];
```


Similarly, we calculate the  $2 \times 2$  reduced density matrix of the spin degree of freedom (atomic state) with [In\[257\]](#), tracing out the phase-space degree of freedom (the first  $n_{\max}$  dimensions):

```
1 In[658] := pS = traceout[ψ, nmax];
```

Expectation values in the field or spin degrees of freedom are then easily calculated from these reduced density matrices.

To illustrate these techniques, we calculate the time-dependent wavefunction in the resonant weak-coupling regime ( $\omega_a = \omega_c = 1$ ,  $g = 0.1$ ; initial state: coherent field state at  $\langle \hat{X} \rangle = \sqrt{2}$  and  $\langle \hat{P} \rangle = 0$ , spin down). First we show the time-dependence of the atomic spin expectation values, calculated from a reduced spin density matrix with

```
1 In[659] := {Tr[pS.Sx], Tr[pS.Sy], Tr[pS.Sz]}
```




Observations:

- At  $t = 0$  we recognize the initial spin-down state:  $\langle \hat{S}_x \rangle = \langle \hat{S}_y \rangle = 0$  and  $\langle \hat{S}_z \rangle = -\frac{1}{2}$ .
- The  $S_x$  and  $S_y$  spin components rotate rapidly due to the pseudo-spin excitation energy  $\hbar\omega_a$  (phase factor  $e^{-i\omega_a t}$ ). They are  $90^\circ$  out of phase.
- The  $S_z$  spin component has a complicated time dependence. Since the atomic energy is  $\hbar\omega_a \langle \hat{S}_z \rangle$ , this curve shows the energy flowing between the atom and the cavity light field.

The phase-space Wigner quasi-probability distribution of the cavity field, calculated using [In\[481\]](#) from the reduced phase space density matrix of [In\[657\]](#), using the same weak-coupling conditions as above, is plotted here at two evolution times:

```
1 In[660] := WignerDistributionPlot[pX, {-a/2, a/2}]
```



Observations:

- At  $t = 0$  we recognize the initial state: a coherent state (circular Gaussian of minimal area  $\langle \hat{X}^2 \rangle = \langle \hat{P}^2 \rangle = \frac{1}{2}$ ) displaced by  $\delta = \sqrt{2}$  in the  $X$ -direction, implying  $\delta^2/2 = 1$  photon present initially.
- At  $t = 100$  the structure of the Wigner distribution has taken on a qualitatively different shape, including a significant negative-valued region. Such negative regions are forbidden in classical phase-space distributions and hence indicate an essentially quantum-mechanical state.

### 5.3.1 Exercises

**Q5.3** Show that the operators of Eq. (5.18) represent a pseudo-spin-1/2, like in Q3.2.

**Q5.4** Express  $\hat{S}^\pm$  in terms of  $|g\rangle$ ,  $|e\rangle$ ,  $\langle g|$ , and  $\langle e|$ .

**Q5.5** Show that  $[\hat{X}, \hat{P}] = i$  using Eq. (5.20) and assuming that  $[\hat{a}, \hat{a}^\dagger] = 1$ .

**Q5.6** Show that Eq. (5.21) follows from Eq. (5.19) using Eq. (5.20).

# List of Notebooks in Supplementary Material

1. <b>MathematicaExample.nb</b> —a simple example .....	vii
2. <b>FactorialDefinitions.nb</b> —many ways to define the factorial function.....	19
3. <b>ReducedDensityMatrix.nb</b> —reduced density matrices .....	46
4. <b>SpinOperators.nb</b> —spin and angular momentum operators .....	52
5. <b>Electron.nb</b> —spin-1/2 electron in a dc magnetic field .....	54
6. <b>Rubidium87.nb</b> — <sup>87</sup> Rb hyperfine structure.....	56
7. <b>IsingModel.nb</b> —Ising model in a transverse field .....	68
8. <b>QuantumCircuits.nb</b> —quantum gates and quantum circuits .....	80
9. <b>GravityWell.nb</b> —gravity well .....	102
10. <b>WignerDistribution.nb</b> —the Wigner quasi-probability distribution .....	107
11. <b>ParticleDynamics.nb</b> —single-particle dynamics in 1D .....	109
12. <b>GroundState.nb</b> —non-linear Schrödinger equation and imaginary-time propagation in 1D .....	119
13. <b>ContactInteraction.nb</b> —two particles in 1D with contact interaction....	122
14. <b>3DBEC.nb</b> —Bose-Einstein condensate in 3D .....	130
15. <b>ParticleMotionWithSpin.nb</b> —one particle in 1D with spin .....	141
16. <b>RashbaCoupling.nb</b> —one particle in 2D with Rashba coupling .....	145
17. <b>JaynesCummingsModel.nb</b> —phase-space dynamics in the Jaynes-Cummings model .....	148

# Solutions to Exercises

## Chapter 1 Wolfram Language Overview

### Q1.1

```
1 In[661]:=N[Zeta[3]]
2 Out[661]=1.20206
```

### Q1.2

```
1 In[662]:=2
2 Out[662]=1.44494
```

### Q1.3

```
1 In[663]:=Integrate[Sin[x]*Exp[-x], {x, 0, Infinity}]
2 Out[663]=1/2
```

### Q1.4

```
1 In[664]:=N[ $\pi$ , 1000]
2 Out[664]=3.141592653589793238462643383279502884197169399375105820974944592307816406286
3 20899862803482534211706798214808651328230664709384460955058223172535940812848
4 11174502841027019385211055596446229489549303819644288109756659334461284756482
5 33786783165271201909145648566923460348610454326648213393607260249141273724587
6 00660631558817488152092096282925409171536436789259036001133053054882046652138
7 41469519415116094330572703657595919530921861173819326117931051185480744623799
8 62749567351885752724891227938183011949129833673362440656643086021394946395224
9 73719070217986094370277053921717629317675238467481846766940513200056812714526
10 35608277857713427577896091736371787214684409012249534301465495853710507922796
11 89258923542019956112129021960864034418159813629774771309960518707211349999998
12 37297804995105973173281609631859502445945534690830264252230825334468503526193
13 11881710100031378387528865875332083814206171776691473035982534904287554687311
14 59562863882353787593751957781857780532171226806613001927876611195909216420199
```

### Q1.5


```
1 In[665]:=ClebschGordan[{100, 10}, {200, -12}, {110, -2}]
2 Out[665]=8261297798499109361013742279092521767681*
3 Sqrt[769248995636473]/297224869222895274740285232180446271746289127347456291479
4 57669733897130076853320942746928207329]/14
5 In[666]:=2
6 Out[666]=0.0949317
```

### Q1.6

```
1 In[667]:=Limit[Sin[x]/x, x -> 0]
2 Out[667]=1
```


### Q1.7

```
1 In[668]:=Plot[Sin[x]/x, {x, -20, 20}, PlotRange -> All]
```




### Q1.8

```
1 In[669]:=F[c_, imax_] := Abs[NestWhile[#^2+c&, 0., Abs[#] <= 2 &, 1, imax]] <= 2
2 In[670]:=With[{n = 100, imax = 1000},
3 Graphics[Raster[Table[Boole[!F[x+I*y, imax]], {y, -2, 2, 1/n}, {x, -2, 2, 1/n}]]]]
```



### Q1.9

```
1 In[671]:=MandelbrotSetPlot[]
```



**Q1.10** In general, the definition of  $x$  depends on the values of  $u$  and  $v$  at the time of the definition of  $x$ , whereas  $y$  depends on the values at the time of using the symbol  $y$ . The second case below, however, needs special attention since the values of  $u$  and  $v$  are not defined at the time when  $x$  is defined.

- When  $u$  and  $v$  are already defined before  $x$  and  $y$  are defined, then  $x$  and  $y$  return the same value:

```

1 In[672]:=Clear[x, y, u, v];
2 In[673]:=u = 3; v = 7;
3 In[674]:=x = u+v; y := u+v;
4 In[675]:= {x, y}
5 Out[675]= {10, 10}
6 In[676]:= ?x
7          x=10
8 In[677]:= ?y
9          y:=u+v

```

- When  $u$  and  $v$  are defined after  $x$  and  $y$  are defined, then  $x$  and  $y$  also return the same value. Notice, however, that the definition of  $x$  is not static and thus depends on the values of  $u$  and  $v$  at the time of usage:

```

1 In[678]:=Clear[x, y, u, v];
2 In[679]:=x = u+v; y := u+v;
3 In[680]:=u = 3; v = 7;
4 In[681]:= {x, y}
5 Out[681]= {10, 10}
6 In[682]:= ?x
7          x=u+v
8 In[683]:= ?y
9          y:=u+v

```

- When **u** and **v** change values after **x** and **y** are defined, then **x** and **y** differ since only **y** reflects the new values of **u** and **v**:

```

1 In[684]:=Clear[x, y, u, v];
2 In[685]:=u = 3; v = 7;
3 In[686]:=x = u+v; y := u+v;
4 In[687]:=u = 8; v = 9;
5 In[688]:= {x, y}
6 Out[688]= {10, 17}
7 In[689]:= ?x
8          x=10
9 In[690]:= ?y
10         y:=u+v

```

### Q1.11

```

1 In[691]:=N[E]
2 Out[691]=2.71828
3 In[692]:=N0E
4 Out[692]=2.71828
5 In[693]:=E //N
6 Out[693]=2.71828

```

### Q1.12

```

1 In[694]:=Total[Range[123, 9968]]
2 Out[694]=49677993

```

### Q1.13

```

1 In[695]:=Module[{i},
2           i = 123;
3           s = 0;
4           While[s <= 10000, s += i; i++];
5           i - 1]
6 Out[695]=187

```

### Q1.14

```

1 In[696]:=f = #1*#2*#3 &;

```

### Q1.15

```

1 In[697]:=a = {0.1, 0.9, 2.25, -1.9};
2 In[698]:=sa = Map[Sin[#]^2 &, a]
3 Out[698]= {0.00996671, 0.613601, 0.605398, 0.895484}

```

**Q1.16** The **Total** function is the same as applying **Plus** to a list:

```

1 In[699]:=Apply[Plus, sa]
2 In[700]:=2.12445
3 In[701]:=Plus@@sa
4 In[702]:=2.12445
5 In[703]:=Total[sa]
6 In[704]:=2.12445

```

**Q1.17** All built-in symbols, like **Echo**, are protected in order to prevent accidental modification. Trying to modify **Echo** without unprotecting it first gives an error:

```

1 In[705]:=Echo = #1 &
2          Set:: Symbol Echo is Protected.
3 Out[705]=#1 &

```

**Q1.18** See `In[128]` and `In[129]`: the full forms of `a/b` and `x_/y_` are similar and match,

```
1 In[706]:=FullForm[a/b]
2 Out[706]=Times[a, Power[b, -1]]
3 In[707]:=FullForm[x_/y_]
4 Out[707]=Times[Pattern[x, Blank[]], Power[Pattern[y, Blank[]], -1]]
```

while the full form of `2/3` is different and does not match the pattern for replacements,

```
1 In[708]:=FullForm[2/3]
2 Out[708]=Rational[2, 3]
```

**Q1.19** Not all delayed assignments can be replaced by immediate ones. Whenever an immediate assignment can be used, it tends to be faster.

1. `=` and `:=` work equally well.
2. `=` and `:=` work equally well.
3. `=` and `:=` work equally well.
4. `=` and `:=` work equally well. There is a significant difference though: while the delayed assignment executes as a product, the immediate assignment is simplified at the moment of definition to a factorial, which then executes much faster:

```
1 In[709]:=f[n_] = Product[i, {i, n}]
2 Out[709]=n!
```

5. Immediate assignment breaks the recursion, which cannot be executed at definition time.
6. `=` and `:=` work equally well.
7. `=` and `:=` work equally well.
8. Immediate assignment breaks the `Do` loop, which cannot be executed at definition time.
9. Immediate assignment breaks the `For` loop: since `n` is not defined at definition time, the comparison `i<=n` fails at the first iteration and the result is always `f[n_]=1`.
10. Immediate assignment breaks the `Range` command since `n` is not defined at definition time.
11. Immediate assignment breaks the `Range` command since `n` is not defined at definition time.
12. Immediate assignment breaks the `Array` command since `n` is not defined at definition time.
13. Immediate assignment breaks the `Range` command since `n` is not defined at definition time.
14. Immediate assignment always gives `f[n_]=1` since the repeated replacement fails.
15. Immediate assignment breaks the `Range` command since `n` is not defined at definition time.
16. Immediate assignment always gives `f[n_]=1` since the repeated replacement fails.



17. `=` and `:=` work equally well.

**Q1.20** Not all delayed rules can be replaced by immediate ones. Whenever an immediate rule can be used, it tends to be faster.

14. `->` and `:>` work equally well.

16. Immediate rule (`->`) breaks the `Table` command since `m` is not defined at definition time.

**Q1.21** In the recursive definitions 5 and 6, memoization gives a dramatic speedup, as it remembers intermediate results in the recursion. In the other examples, memoization only helps when the function is called repeatedly with the same argument.

**Q1.22** Using a built-in function:

```
1 In[710]:=Table[Fibonacci[n], {n, 100}]
```

Even more directly, by using the `Listable` attribute of the `Fibonacci` function:

```
1 In[711]:=Fibonacci[Range[100]]
```

Recursive with memoization:

```
1 In[712]:=g[1] = g[2] = 1;
2 In[713]:=g[_] := g[n] = g[n-1] + g[n-2]
3 In[714]:=Table[g[n], {n, 100}]
```

Iterative construction of the list:

```
1 In[715]:=L = {1, 1};
2 In[716]:=Do[AppendTo[L, L[[-1]] + L[[-2]]], {98}];
3 In[717]:=L
```

**Q1.23** The eigenvectors are orthogonal, but not necessarily normalized. Eigensystem of  $\hat{\sigma}_x$ :

```
1 In[718]:={eval, evec} = Eigensystem[PauliMatrix[1]]
2 Out[718]:={{-1, 1}, {{-1, 1}, {1, 1}}}
3 In[719]:=Normalize /@ evec
4 Out[719]:={{-1/Sqrt[2], 1/Sqrt[2]}, {1/Sqrt[2], 1/Sqrt[2]}}
```

Eigensystem of  $\hat{\sigma}_y$ :

```
1 In[720]:={eval, evec} = Eigensystem[PauliMatrix[2]]
2 Out[720]:={{-1, 1}, {{I, 1}, {-I, 1}}}
3 In[721]:=Normalize /@ evec
4 Out[721]:={{I/Sqrt[2], 1/Sqrt[2]}, {-I/Sqrt[2], 1/Sqrt[2]}}
```

Eigensystem of  $\hat{\sigma}_z$ :

```
1 In[722]:={eval, evec} = Eigensystem[PauliMatrix[3]]
2 Out[722]:={{-1, 1}, {{0, 1}, {1, 0}}}
```

**Q1.24** The tensor index dimensions do not match:

```
1 In[723]:=TensorContract[u, {3, 4}]
2 TensorContract: Contraction levels {3,4} have different dimensions
3 {3,2}.
```

## Chapter 2 Quantum Mechanics: States and Operators

**Q2.1** We use the computational basis  $\{|\uparrow\rangle, |\downarrow\rangle\}$ , in which the two given basis functions are

```

1 In[724]:=up[θ_, φ] = {Cos[θ/2], E^(I*φ)*Sin[θ/2]};
2 In[725]:=dn[θ_, φ] = {-E^(-I*φ)*Sin[θ/2], Cos[θ/2]};

```

The corresponding  $\langle\uparrow_{\vartheta,\varphi}|$  and  $\langle\downarrow_{\vartheta,\varphi}|$  are calculated with `Conjugate` (see Sect. 1.11).

1. Calculate  $\langle\uparrow_{\vartheta,\varphi}|\uparrow_{\vartheta,\varphi}\rangle = 1$ ,  $\langle\uparrow_{\vartheta,\varphi}|\downarrow_{\vartheta,\varphi}\rangle = 0$ ,  $\langle\downarrow_{\vartheta,\varphi}|\uparrow_{\vartheta,\varphi}\rangle = 0$ ,  $\langle\downarrow_{\vartheta,\varphi}|\downarrow_{\vartheta,\varphi}\rangle = 1$ :

```

1 In[726]:=Conjugate[up[θ, φ]].up[θ, φ] //ComplexExpand //FullSimplify
2 Out[726]=1
3 In[727]:=Conjugate[up[θ, φ]].dn[θ, φ] //ComplexExpand //FullSimplify
4 Out[727]=0
5 In[728]:=Conjugate[dn[θ, φ]].up[θ, φ] //ComplexExpand //FullSimplify
6 Out[728]=0
7 In[729]:=Conjugate[dn[θ, φ]].dn[θ, φ] //ComplexExpand //FullSimplify
8 Out[729]=1

```

2. Construct the ket-bra products with `KroneckerProduct`:

```

1 In[730]:=KroneckerProduct[up[θ, φ], Conjugate[up[θ, φ]]] +
2 KroneckerProduct[dn[θ, φ], Conjugate[dn[θ, φ]]] //
3 ComplexExpand //FullSimplify
4 Out[730]={1, 0}, {0, 1}

```

3.  $|\uparrow\rangle = |\uparrow_{\vartheta,\varphi}\rangle\langle\uparrow_{\vartheta,\varphi}| + |\downarrow_{\vartheta,\varphi}\rangle\langle\downarrow_{\vartheta,\varphi}| = \cos(\vartheta/2)|\uparrow_{\vartheta,\varphi}\rangle - e^{i\varphi}\sin(\vartheta/2)|\downarrow_{\vartheta,\varphi}\rangle$ :

```

1 In[731]:=Cos[θ/2]*up[θ, φ] - E^(I*φ)*Sin[θ/2]*dn[θ, φ] //FullSimplify
2 Out[731]={1, 0}

```

$$|\downarrow\rangle = |\uparrow_{\vartheta,\varphi}\rangle\langle\uparrow_{\vartheta,\varphi}|\downarrow\rangle + |\downarrow_{\vartheta,\varphi}\rangle\langle\downarrow_{\vartheta,\varphi}|\downarrow\rangle = e^{-i\varphi}\sin(\vartheta/2)|\uparrow_{\vartheta,\varphi}\rangle + \cos(\vartheta/2)|\downarrow_{\vartheta,\varphi}\rangle$$

```

1 In[732]:=E^(-I*φ)*Sin[θ/2]*up[θ, φ] + Cos[θ/2]*dn[θ, φ] //FullSimplify
2 Out[732]={0, 1}

```

4. The Pauli operators are defined in Mathematica in our computational basis with the `PauliMatrix` command.

The matrix elements of the Pauli operator  $\hat{\sigma}_x$  are

```

1 In[733]:=sx = PauliMatrix[1];
2 In[734]:=Conjugate[up[θ, φ]].sx.up[θ, φ] //ComplexExpand //FullSimplify
3 Out[734]=Sin[θ]*Cos[φ]
4 In[735]:=Conjugate[up[θ, φ]].sx.dn[θ, φ] //ComplexExpand //FullSimplify
5 Out[735]=Exp[-I*φ]*(Cos[θ]*Cos[φ]+I*Sin[φ])
6 In[736]:=Conjugate[dn[θ, φ]].sx.up[θ, φ] //ComplexExpand //FullSimplify
7 Out[736]=Exp[I*φ]*(Cos[θ]*Cos[φ]-I*Sin[φ])
8 In[737]:=Conjugate[dn[θ, φ]].sx.dn[θ, φ] //ComplexExpand //FullSimplify
9 Out[737]=-Sin[θ]*Cos[φ]
10 In[738]:=sx == Sin[θ]*Cos[φ] * KroneckerProduct[up[θ, φ], Conjugate[up[θ, φ]]] +
11 E^(-I*φ)*(Cos[θ]*Cos[φ]+I*Sin[φ]) *
12 KroneckerProduct[up[θ, φ], Conjugate[dn[θ, φ]]] +
13 E^(I*φ)*(Cos[θ]*Cos[φ]-I*Sin[φ]) *
14 KroneckerProduct[dn[θ, φ], Conjugate[up[θ, φ]]] -
15 Sin[θ]*Cos[φ] * KroneckerProduct[dn[θ, φ], Conjugate[dn[θ, φ]]] //
16 ComplexExpand //FullSimplify
17 Out[738]=True

```

The matrix elements of the Pauli operator  $\hat{\sigma}_y$  are

```

1 In[739]:=sy = PauliMatrix[2];
2 In[740]:=Conjugate[up[0,φ]].sy.up[0,φ] //ComplexExpand //FullSimplify
3 Out[740]=Sin[0]*Sin[φ]
4 In[741]:=Conjugate[up[0,φ]].sy.dn[0,φ] //ComplexExpand //FullSimplify
5 Out[741]=Exp[-I*φ]*(Cos[0]*Sin[φ]-I*Cos[φ])
6 In[742]:=Conjugate[dn[0,φ]].sy.up[0,φ] //ComplexExpand //FullSimplify
7 Out[742]=Exp[I*φ]*(Cos[0]*Sin[φ]+I*Cos[φ])
8 In[743]:=Conjugate[dn[0,φ]].sy.dn[0,φ] //ComplexExpand //FullSimplify
9 Out[743]=-Sin[0]*Sin[φ]
10 In[744]:=sy == Sin[0]*Sin[φ] * KroneckerProduct[up[0,φ], Conjugate[up[0,φ]]] +
11      E^(-I*φ)*(Cos[0]*Sin[φ]-I*Cos[φ]) *
12      KroneckerProduct[up[0,φ], Conjugate[dn[0,φ]]] +
13      E^(I*φ)*(Cos[0]*Sin[φ]+I*Cos[φ]) *
14      KroneckerProduct[dn[0,φ], Conjugate[up[0,φ]]] -
15      Sin[0]*Sin[φ] * KroneckerProduct[dn[0,φ], Conjugate[dn[0,φ]]] //
16      ComplexExpand //FullSimplify
17 Out[744]=True

```

The matrix elements of the Pauli operator  $\hat{\sigma}_z$  are

```

1 In[745]:=sz = PauliMatrix[3];
2 In[746]:=Conjugate[up[0,φ]].sz.up[0,φ] //ComplexExpand //FullSimplify
3 Out[746]=Cos[0]
4 In[747]:=Conjugate[up[0,φ]].sz.dn[0,φ] //ComplexExpand //FullSimplify
5 Out[747]=-Exp[-I*φ]*Sin[0]
6 In[748]:=Conjugate[dn[0,φ]].sz.up[0,φ] //ComplexExpand //FullSimplify
7 Out[748]=-Exp[I*φ]*Sin[0]
8 In[749]:=Conjugate[dn[0,φ]].sz.dn[0,φ] //ComplexExpand //FullSimplify
9 Out[749]=-Cos[0]
10 In[750]:=sz == Cos[0] * KroneckerProduct[up[0,φ], Conjugate[up[0,φ]]] -
11      E^(-I*φ)*Sin[0] * KroneckerProduct[up[0,φ], Conjugate[dn[0,φ]]] -
12      E^(I*φ)*Sin[0] * KroneckerProduct[dn[0,φ], Conjugate[up[0,φ]]] -
13      Cos[0] * KroneckerProduct[dn[0,φ], Conjugate[dn[0,φ]]] //
14      ComplexExpand //FullSimplify
15 Out[750]=True

```

5. We check the eigenvalue equations with eigenvalues  $\pm 1$ :

```

1 In[751]:=s = sx*Sin[0]*Cos[φ] + sy*Sin[0]*Sin[φ] + sz*Cos[0];
2 In[752]:=Eigenvalues[s]
3 Out[752]={-1, 1}
4 In[753]:=s.up[0,φ] == up[0,φ] //FullSimplify
5 Out[753]=True
6 In[754]:=s.dn[0,φ] == -dn[0,φ] //FullSimplify
7 Out[754]=True

```


## Q2.2

1. Since  $\sum_{n=1}^{\infty} |n\rangle\langle n| = \mathbb{1}$ , we have  $P_{\infty}(x, y) = \langle x | \mathbb{1} | y \rangle = \langle x | y \rangle = \delta(x - y)$ .
2.  $P_{n_{\max}}(x, y) = \langle x | \left[ \sum_{n=1}^{n_{\max}} |n\rangle\langle n| \right] | y \rangle = \sum_{n=1}^{n_{\max}} \langle x | n \rangle \langle n | y \rangle = 2 \sum_{n=1}^{n_{\max}} \sin(n\pi x) \sin(n\pi y)$ :

```

1 In[755]:=With[{nmax = 10},
2   P[x_, y_] = 2*Sum[Sin[n*π*x]*Sin[n*π*y], {n, nmax}];
3   DensityPlot[P[x, y], {x, 0, 1}, {y, 0, 1},
4   PlotRange -> All, PlotPoints -> 2*nmax]]

```



3. The operator  $\hat{\Pi}_{n_{\max}} = \sum_{n=1}^{n_{\max}} |n\rangle\langle n|$  is the projector onto the computational subspace (see Sect. 2.1.1). The function  $P_{n_{\max}}(x, y) = \langle x | \hat{\Pi}_{n_{\max}} | y \rangle$  is its real-space representation. Since the plot of  $P_{n_{\max}}(x, y)$  has a finite spatial resolution (*i.e.*, no structure at length scales smaller than  $1/n_{\max}$ ), we see that this projection operator  $\hat{\Pi}_{n_{\max}}$  is associated with a spatial smoothing operation.

**Q2.3** See Q2.1.

**Q2.4** Inserting Eq. (2.32) into Eq. (2.17) gives the quantum state

$$|\psi(t)\rangle = \exp\left[-\frac{i}{\hbar} \int_{t_0}^t \hat{\mathcal{H}}(s) ds\right] |\psi(t_0)\rangle \quad (1)$$

We calculate its time-derivative with the chain rule and  $\frac{d}{dx} \left[ \int_{f(x)}^{g(x)} h(x, y) dy \right] = h(x, g(x))g'(x) - h(x, f(x))f'(x) + \int_{f(x)}^{g(x)} \frac{\partial h(x, y)}{\partial x} dy$ :

$$\frac{d}{dt} |\psi(t)\rangle = -\frac{i}{\hbar} \hat{\mathcal{H}}(t) \exp\left[-\frac{i}{\hbar} \int_{t_0}^t \hat{\mathcal{H}}(s) ds\right] |\psi(t_0)\rangle = -\frac{i}{\hbar} \hat{\mathcal{H}}(t) |\psi(t)\rangle, \quad (2)$$

which is the Schrödinger equation (2.16).

**Q2.5** The Hamiltonian is

```

1 In[756]:= {sx,sy,sz} = Table[PauliMatrix[i], {i, 3}];
2 In[757]:= H = Sin[θ]*Cos[φ]*sx + Sin[θ]*Sin[φ]*sy + Cos[θ]*sz //FullSimplify
3 Out[757]= {{Cos[θ], E^(-I*φ)*Sin[θ]}, {E^(I*φ)*Sin[θ], -Cos[θ]}}
```

and the propagator is calculated from Eq. (2.34)

```

1 In[758]:=U = MatrixExp[-I*(t-t0)/h*H] //FullSimplify
2 Out[758]={{Cos[(t-t0)/h]-I*Cos[θ]*Sin[(t-t0)/h], -I*E^(-I*φ)*Sin[θ]*Sin[(t-t0)/h]},
3           {-I*E^(I*φ)*Sin[θ]*Sin[(t-t0)/h], Cos[(t-t0)/h]+I*Cos[θ]*Sin[(t-t0)/h]}}
```

**Q2.6** The definitions are ordered with decreasing specificity:

```

1 In[759]:= ?U
2           Global`U
3           U[τ_?NumericQ] := MatrixExp[-I H N[τ]]
4           U[τ_] = MatrixExp[-I H τ]
```

In this way, the more general definition In[234] does not override the more specific definition In[235].

## Q2.7

1. The Hamiltonian is

$$\hat{\mathcal{H}} = -\frac{\hbar^2}{2m} \left( \frac{\partial}{\partial x_1^2} + \frac{\partial}{\partial y_1^2} + \frac{\partial}{\partial z_1^2} + \frac{\partial}{\partial x_2^2} + \frac{\partial}{\partial y_2^2} + \frac{\partial}{\partial z_2^2} \right) + \frac{1}{2} m \omega^2 (x_1^2 + y_1^2 + z_1^2 + x_2^2 + y_2^2 + z_2^2) + g \delta(x_1 - x_2) \delta(y_1 - y_2) \delta(z_1 - z_2) \quad (3)$$

2. For example, we could use the harmonic-oscillator basis functions that diagonalize the six degrees of freedom in the absence of coupling ( $g = 0$ ): the states  $|n\rangle$  for which

$$\left[ -\frac{\hbar^2}{2m} \frac{\partial}{\partial x^2} + \frac{1}{2} m \omega^2 x^2 \right] |n\rangle = \hbar \omega \left( n + \frac{1}{2} \right) |n\rangle, \quad (4)$$

with  $n \in \mathbb{N}$ . Explicitly, the position representations of these states are

$$\langle x | n \rangle = \phi_n(x) = x_0^{-1/2} \frac{H_n(x/x_0)}{\sqrt{2^n n! \sqrt{\pi}}} e^{-\frac{x^2}{2x_0^2}}. \quad (5)$$

For the six degrees of freedom we therefore propose the basis functions  $|n_{x_1}, n_{y_1}, n_{z_1}, n_{x_2}, n_{y_2}, n_{z_2}\rangle$ .

## 3. The matrix elements are

$$\begin{aligned}
& \langle n_{x_1}, n_{y_1}, n_{z_1}, n_{x_2}, n_{y_2}, n_{z_2} | \hat{\mathcal{H}} | n'_{x_1}, n'_{y_1}, n'_{z_1}, n'_{x_2}, n'_{y_2}, n'_{z_2} \rangle \\
&= \hbar \omega \delta_{n_{x_1}, n'_{x_1}} \delta_{n_{y_1}, n'_{y_1}} \delta_{n_{z_1}, n'_{z_1}} \delta_{n_{x_2}, n'_{x_2}} \delta_{n_{y_2}, n'_{y_2}} \delta_{n_{z_2}, n'_{z_2}} (n_{x_1} + n_{y_1} + n_{z_1} + n_{x_2} + n_{y_2} + n_{z_2} + 3) \\
&+ g \langle n_{x_1}, n_{y_1}, n_{z_1}, n_{x_2}, n_{y_2}, n_{z_2} | \int_{-\infty}^{\infty} dx_1 dy_1 dz_1 dx_2 dy_2 dz_2 |x_1\rangle \langle x_1| \otimes |y_1\rangle \langle y_1| \otimes |z_1\rangle \langle z_1| \otimes |x_2\rangle \langle x_2| \otimes |y_2\rangle \\
&\quad \langle y_2| \otimes |z_2\rangle \langle z_2| \delta(x_1 - x_2) \delta(y_1 - y_2) \delta(z_1 - z_2) |n'_{x_1}, n'_{y_1}, n'_{z_1}, n'_{x_2}, n'_{y_2}, n'_{z_2} \rangle \\
&= \hbar \omega \delta_{n_{x_1}, n'_{x_1}} \delta_{n_{y_1}, n'_{y_1}} \delta_{n_{z_1}, n'_{z_1}} \delta_{n_{x_2}, n'_{x_2}} \delta_{n_{y_2}, n'_{y_2}} \delta_{n_{z_2}, n'_{z_2}} (n_{x_1} + n_{y_1} + n_{z_1} + n_{x_2} + n_{y_2} + n_{z_2} + 3) \\
&\quad + g \int_{-\infty}^{\infty} dx_1 dy_1 dz_1 dx_2 dy_2 dz_2 \delta(x_1 - x_2) \delta(y_1 - y_2) \delta(z_1 - z_2) \\
&\quad \times \phi_{n_{x_1}}(x_1) \phi_{n'_{x_1}}(x_1) \phi_{n_{y_1}}(y_1) \phi_{n'_{y_1}}(y_1) \phi_{n_{z_1}}(z_1) \phi_{n'_{z_1}}(z_1) \phi_{n_{x_2}}(x_2) \phi_{n'_{x_2}}(x_2) \phi_{n_{y_2}}(y_2) \phi_{n'_{y_2}}(y_2) \phi_{n_{z_2}}(z_2) \phi_{n'_{z_2}}(z_2) \\
&= \hbar \omega \delta_{n_{x_1}, n'_{x_1}} \delta_{n_{y_1}, n'_{y_1}} \delta_{n_{z_1}, n'_{z_1}} \delta_{n_{x_2}, n'_{x_2}} \delta_{n_{y_2}, n'_{y_2}} \delta_{n_{z_2}, n'_{z_2}} (n_{x_1} + n_{y_1} + n_{z_1} + n_{x_2} + n_{y_2} + n_{z_2} + 3) \\
&+ g \left[ \int_{-\infty}^{\infty} dx \phi_{n_{x_1}}(x) \phi_{n'_{x_1}}(x) \phi_{n_{x_2}}(x) \phi_{n'_{x_2}}(x) \right] \left[ \int_{-\infty}^{\infty} dy \phi_{n_{y_1}}(y) \phi_{n'_{y_1}}(y) \phi_{n_{y_2}}(y) \phi_{n'_{y_2}}(y) \right] \\
&\quad \left[ \int_{-\infty}^{\infty} dz \phi_{n_{z_1}}(z) \phi_{n'_{z_1}}(z) \phi_{n_{z_2}}(z) \phi_{n'_{z_2}}(z) \right] \\
&= \hbar \omega \delta_{n_{x_1}, n'_{x_1}} \delta_{n_{y_1}, n'_{y_1}} \delta_{n_{z_1}, n'_{z_1}} \delta_{n_{x_2}, n'_{x_2}} \delta_{n_{y_2}, n'_{y_2}} \delta_{n_{z_2}, n'_{z_2}} (n_{x_1} + n_{y_1} + n_{z_1} + n_{x_2} + n_{y_2} + n_{z_2} + 3) \\
&\quad + \frac{g}{x_0^3} R_{n_{x_1}, n'_{x_1}, n_{x_2}, n'_{x_2}} R_{n_{y_1}, n'_{y_1}, n_{y_2}, n'_{y_2}} R_{n_{z_1}, n'_{z_1}, n_{z_2}, n'_{z_2}}.
\end{aligned} \tag{6}$$

The required dimensionless integrals over products of four harmonic-oscillator eigenstates,

$$R_{a,b,c,d} = x_0 \int_{-\infty}^{\infty} dx \phi_a(x) \phi_b(x) \phi_c(x) \phi_d(x) = \int_{-\infty}^{\infty} d\xi \frac{H_a(\xi) H_b(\xi) H_c(\xi) H_d(\xi)}{\pi \sqrt{2^{a+b+c+d} a! b! c! d!}} e^{-2\xi^2}, \tag{7}$$

can either be calculated by analytic integration,

```

1 In[760] := φ[n_, x_] = HermiteH[n, x]/Sqrt[2^n*n!*Sqrt[π]]*E^(-x^2/2);
2 In[761] := R[a_Integer;/a>=0, b_Integer;/b>=0, c_Integer;/c>=0, d_Integer;/d>=0] :=
3 Integrate[φ[a,x]*φ[b,x]*φ[c,x]*φ[d,x], {x, -∞, ∞}]

```

or by an explicit but hypergeometric formula<sup>1</sup> (much faster),

```

1 In[762] := R[a_Integer;/a>=0, b_Integer;/b>=0, c_Integer;/c>=0, d_Integer;/d>=0] :=
2 If[OddQ[a+b+c+d], 0,
3 1/π*(-1)^((a+b-c+d)/2)*Sqrt[c!/(2a!b!d!)]*
4 Gamma[(1+a-b+c-d)/2]*Gamma[(1-a+b+c-d)/2]*
5 HypergeometricPFQRegularized[{(1+a-b+c-d)/2, (1-a+b+c-d)/2, -d},
6 {1+c-d, (1-a-b+c-d)/2, 1}]

```

## Q2.8

```

1 In[763] := ψ = Flatten[Table[ψ1[[i1]]*ψ2[[i2]]*ψ3[[i3]],
2 {i1, Length[ψ1]}, {i2, Length[ψ2]}, {i3, Length[ψ3]}]]

```

<sup>1</sup>See <http://www.ph.unimelb.edu.au/~jnnnewn/cm-seminar-results/report/AnalyticIntegralOfFourHermite.pdf>.

## Q2.9

```

1 In[764]:=A = Flatten[Table[a1[[i1,j1]]*a2[[i2,j2]]*a3[[i3,j3]],
2   {i1, Length[a1]}, {i2, Length[a2]}, {i3, Length[a3]}],
3   {j1, Length[Transpose[a1]]}, {j2, Length[Transpose[a2]]},
4   {j3, Length[Transpose[a3]]}], {{1,2,3}, {4,5,6}}]
```

## Q2.10 Manual calculation:

$$|\psi\rangle = [0.8|\uparrow\rangle - 0.6|\downarrow\rangle] \otimes [0.6i|\uparrow\rangle + 0.8|\downarrow\rangle] = 0.48i|\uparrow\uparrow\rangle + 0.64|\uparrow\downarrow\rangle - 0.36i|\downarrow\uparrow\rangle - 0.48|\downarrow\downarrow\rangle, \quad (8)$$

where  $|\uparrow\downarrow\rangle = |\uparrow\rangle \otimes |\downarrow\rangle$  etc. In Mathematica, using the computational basis  $\{|\uparrow\uparrow\rangle, |\uparrow\downarrow\rangle, |\downarrow\uparrow\rangle, |\downarrow\downarrow\rangle\}$ , in this order:

```

1 In[765]:=ψ1 = {0.8, -0.6};
2 In[766]:=ψ2 = {0.6*I, 0.8};
3 In[767]:=ψ = Flatten[KroneckerProduct[ψ1, ψ2]]
4 Out[767]={0.+0.48*I, 0.64+0.*I, 0.-0.36*I, -0.48+0.*I}
```

The ordering of the joint basis in the `KroneckerProduct` result is therefore  $\{|\uparrow\uparrow\rangle, |\uparrow\downarrow\rangle, |\downarrow\uparrow\rangle, |\downarrow\downarrow\rangle\}$ .

**Q2.11** We calculate the reduced density matrices with the `traceout` command of `In[256]` and `In[257]`:

```

1 In[768]:=p1 = traceout[ψ, -2]
2 Out[768]={{0.64+0.*I, -0.48+0.*I}, {-0.48+0.*I, 0.36+0.*I}}
3 In[769]:=p2 = traceout[ψ, 2]
4 Out[769]={{0.36+0.*I, 0.+0.48*I}, {0.-0.48*I, 0.64+0.*I}}
```

Since  $|\psi\rangle$  is a product state, these reduced density matrices are equal to the pure states of the subsystems:

```

1 In[770]:=p1 == KroneckerProduct[ψ1, Conjugate[ψ1]]
2 Out[770]=True
3 In[771]:=p2 == KroneckerProduct[ψ2, Conjugate[ψ2]]
4 Out[771]=True
```

## Chapter 3 Spin and Angular Momentum

### Q3.1

```

1 In[772]:=sx[1/2] == 1/2*PauliMatrix[1]
2 Out[772]=True
3 In[773]:=sy[1/2] == 1/2*PauliMatrix[2]
4 Out[773]=True
5 In[774]:=sz[1/2] == 1/2*PauliMatrix[3]
6 Out[774]=True
```

### Q3.2 We only check up to $S = 10$ :

#### 1. commutators:

```

1 In[775]:=Table[sx[S].sy[S]-sy[S].sx[S] == I*sz[S], {S, 0, 10, 1/2}]
2 Out[775]={True,True,True,True,True,True,True,True,True,True,True,True,
3           True,True,True,True,True,True,True,True}
4 In[776]:=Table[sy[S].sz[S]-sz[S].sy[S] == I*sx[S], {S, 0, 10, 1/2}]
5 Out[776]={True,True,True,True,True,True,True,True,True,True,True,True,
6           True,True,True,True,True,True,True,True}
7 In[777]:=Table[sz[S].sx[S]-sx[S].sz[S] == I*sy[S], {S, 0, 10, 1/2}]
8 Out[777]={True,True,True,True,True,True,True,True,True,True,True,True,
9           True,True,True,True,True,True,True,True}

```

#### 2. spin length:

```

1 In[778]:=Table[sx[S].sx[S]+sy[S].sy[S]+sz[S].sz[S] == S*(S+1)*id[S], {S,0,10,1/2}]
2 Out[778]={True,True,True,True,True,True,True,True,True,True,True,True,
3           True,True,True,True,True,True,True,True}

```

#### 3. Make sure to quit the Mathematica kernel before loading the spin-operator definitions and executing the following commands. On a MacBook Pro (Retina, 13-inch, Early 2015) with a 3.1 GHz Intel Core i7 CPU and 16GB 1867MHz DDR3 RAM, the limit is around $S = 10^5$ for all verifications:

```

1 In[779]:=s=100000;
2 In[780]:=sx[s].sy[s]-sy[s].sx[s] == I*sz[s] //Timing
3 Out[780]={54.3985, True}
4 In[781]:=sy[s].sz[s]-sz[s].sy[s] == I*sx[s] //Timing
5 Out[781]={58.4917, True}
6 In[782]:=sz[s].sx[s]-sx[s].sz[s] == I*sy[s] //Timing
7 Out[782]={57.8856, True}
8 In[783]:=sx[s].sx[s]+sy[s].sy[s]+sz[s].sz[s] == s*(s+1)*id[s] //Timing
9 Out[783]={33.5487, True}

```

### Q3.3 The expressions rapidly increase in complexity with increasing $S$ :

```

1 In[784]:=n = {Sin[0]*Cos[φ], Sin[0]*Sin[φ], Cos[0]};
2 In[785]:=With[{S=0}, MatrixExp[-I*α*n.{sx[S],sy[S],sz[S]}] //FullSimplify]
3 Out[785]={1}
4 In[786]:=With[{S=1/2}, MatrixExp[-I*α*n.{sx[S],sy[S],sz[S]}] //FullSimplify]
5 Out[786]={{Cos[α/2]-I*Cos[0]*Sin[α/2], Sin[α/2]*Sin[0]*(-I*Cos[φ]-Sin[φ])},
6           {Sin[α/2]*Sin[0]*(-I*Cos[φ]+Sin[φ]), Cos[α/2]+I*Cos[0]*Sin[α/2]}}
7 In[787]:= % /. α -> 0
8 Out[787]={{1, 0}, {0, 1}}
9 In[788]:=With[{S=1}, MatrixExp[-I*α*n.{sx[S],sy[S],sz[S]}] //FullSimplify]
10 Out[788]={{(Cos[α/2]-I*Cos[0]*Sin[α/2])^2,
11           E^(-I*φ)*((-1+Cos[α])*Cos[0]-I*Sin[α])*Sin[0]/Sqrt[2],
12           -E^(-2I*φ)*Sin[α/2]^2*Sin[0]^2,
13           {Sqrt[2]*E^(-I*α)*Sin[α/2]*(Cos[α/2]-I*Cos[0]*Sin[α/2])
14            *Sin[0]*(-I*Cos[α+φ]+Sin[α+φ]),
15           Cos[α/2]^2+Cos[2θ]*Sin[α/2]^2,
16           E^(-I*φ)*(Cos[0]-Cos[α])*Cos[0]-I*Sin[α])*Sin[0]/Sqrt[2]},
17           {-E^(2I*φ)*Sin[α/2]^2*Sin[0]^2,
18           -E^(I*φ)*((-1+Cos[α])*Cos[0]+I*Sin[α])*Sin[0]/Sqrt[2],
19           (Cos[α/2]+I*Cos[0]*Sin[α/2])^2}}
20 In[789]:= % /. α -> 0
21 Out[789]={{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}

```

### Q3.4 In the unit system of In [275] we have

```

1 In[790]:= {eval, evec} = Eigensystem[H[Quantity[1, "Teslas"]/MagneticFieldUnit, 0, 0]]
2 Out[790]={{-14012.476, 14012.476}, {{-0.7071068, 0.7071068}, {0.7071068, 0.7071068}}}

```





**Q3.7** The Clebsch–Gordan coefficients  $\langle I, M_I, J, M_J | I, J, F, M_F \rangle$  serve to construct the states that simultaneously diagonalize  $\hat{I}^2$ ,  $\hat{J}^2$ ,  $\hat{F}^2$ , and  $\hat{F}_z$  from those that simultaneously diagonalize  $\hat{I}^2$ ,  $\hat{I}_z$ ,  $\hat{J}^2$ , and  $\hat{J}_z$ : for  $|I - J| \leq F \leq I + J$  and  $|M_F| \leq F$ ,

$$|I, J, F, M_F\rangle = \sum_{M_I=-I}^I \sum_{M_J=-J}^J \langle I, M_I, J, M_J | I, J, F, M_F \rangle |I, M_I\rangle \otimes |J, M_J\rangle \quad (9)$$

In Mathematica,  $S[i, j, F, MF] = |I, J, F, M_F\rangle$ :

```
1 In[805]:=S[i_,j_,F_,MF_] := Sum[ClebschGordan[{i,Mi},{j,Mj}},{F,MF}]*
2 Flatten[KroneckerProduct[SparseArray[i-Mi+1->1, 2i+1],
3 SparseArray[j-Mj+1->1, 2j+1]]],
4 {Mi,-i,i}, {Mj,-j,j}]
```

Check that these diagonalize  $\hat{I}^2$ ,  $\hat{J}^2$ ,  $\hat{F}^2$ , and  $\hat{F}_z$  simultaneously:

```
1 In[806]:=With[{i=3, j=5},
2 Table[{Ix.Ix+Iy.Iy+Iz.Iz}.S[i,j,F,MF] == i(i+1)*S[i,j,F,MF] &&
3 (Jx.Jx+Jy.Jy+Jz.Jz).S[i,j,F,MF] == j(j+1)*S[i,j,F,MF] &&
4 (Fx.Fx+Fy.Fy+Fz.Fz).S[i,j,F,MF] == F(F+1)*S[i,j,F,MF] &&
5 Fz.S[i,j,F,MF] == MF*S[i,j,F,MF],
6 {F,Abs[i-j],i+j}, {MF,-F,F}]
```

(disregard the warnings about `ClebschGordan::phy`).

When we use the basis of product states  $|I, M_I\rangle \otimes |J, M_J\rangle$  to calculate the eigenvectors of the matrices  $Fx.Fx+Fy.Fy+Fz.Fz$  and  $Fz$ , using either `Eigenvectors` or `Eigensystem`, these Clebsch–Gordan coefficients appear naturally as coefficients of the resulting eigenvectors.

### Q3.8

```
1 In[807]:=With[{S = 100},
2 ψ = SparseArray[1->1, 2S+1];
3 {x,y,z, xx,yy,zz} = Conjugate[ψ].(#.ψ)& /@
4 {sx[S],sy[S],sz[S], sx[S].sx[S],sy[S].sy[S],sz[S].sz[S]};
5 {{x,y,z}, {xx-x^2,yy-y^2,zz-z^2}}
6 Out[807]={{{0, 0, 100}, {50, 50, 0}}}
```

In general,

$$\begin{aligned} \langle \hat{S}_x \rangle &= 0 & \langle \hat{S}_y \rangle &= 0 & \langle \hat{S}_z \rangle &= S \\ \langle \hat{S}_x^2 \rangle &= S/2 & \langle \hat{S}_y^2 \rangle &= S/2 & \langle \hat{S}_z^2 \rangle &= S^2 \\ \langle \hat{S}_x^2 \rangle - \langle \hat{S}_x \rangle^2 &= S/2 & \langle \hat{S}_y^2 \rangle - \langle \hat{S}_y \rangle^2 &= S/2 & \langle \hat{S}_z^2 \rangle - \langle \hat{S}_z \rangle^2 &= 0 \end{aligned} \quad (10)$$

### Q3.9

```
1 In[808]:=Δ, Ω /.
2 {Ei -> eval[[2]], Ej -> eval[[7]], Tij -> T[[2,7]], Tji -> T[[7,2]]} /.
3 hfc /. {Bz->3.22895,Bacx->0.1,Bacy->0,Bacc->0,ω->2π*6827.9}
4 Out[808]={0.00476766, 0.762616}
```

The oscillation period is  $2\pi/\Omega = 8.238\,99\,\mu\text{s}$ , which matches the full oscillation periods of the plots of [In\[321\]](#) and [In\[322\]](#).

**Q3.10**  $^{23}\text{Na}$  has the same nuclear spin  $I = 3/2$  as  $^{87}\text{Rb}$ ; they differ only in the constants:

- $A_{\text{hfs}} = 885.813\,064\,40\,\text{MHz}$
- $g_I = 0.000\,804\,610\,80$
- $g_L = -0.999\,976\,13$

As a result, there is a magic field between the same states as for  $^{87}\text{Rb}$ , but at a field strength  $B_z = 0.676\,851\,\text{G} \approx \frac{16Ag_I}{3\mu_B g_S^2}$ .

**Q3.11**  $^{85}\text{Rb}$  has a nuclear spin  $I = 5/2$ , which means that we must re-define the spin operators; all operators are now  $12 \times 12$  matrices. Further, the constants to be used are

- $A_{\text{hfs}} = 1.011\,910\,813\,0\,\text{GHz}$
- $g_I = 0.000\,293\,640\,00$
- $g_L = -0.999\,993\,54$

There are two magic fields:

- $B_z = 0.357\,312\,\text{G} \approx \frac{27Ag_I}{4\mu_B g_S^2}$ : the energy difference between  $|F = 2, M_F = -1\rangle$  and  $|F = 3, M_F = 1\rangle$  is stationary.
- $B_z = 1.143\,42\,\text{G} \approx \frac{108Ag_I}{5\mu_B g_S^2}$ : the energy difference between  $|F = 2, M_F = -2\rangle$  and  $|F = 3, M_F = 2\rangle$  is stationary.

**Q3.12**  $^{133}\text{Cs}$  has a nuclear spin  $I = 7/2$ , which means that we must re-define the spin operators; all operators are now  $16 \times 16$  matrices. Further, the constants to be used are

- $A_{\text{hfs}} = 2.298\,157\,942\,5\,\text{GHz}$
- $g_I = 0.000\,398\,853\,95$
- $g_L = -0.999\,995\,87$

There are three magic fields:

- $B_z = 1.393\,34\,\text{G} \approx \frac{128Ag_I}{15\mu_B g_S^2}$ : the energy difference between  $|F = 3, M_F = -1\rangle$  and  $|F = 4, M_F = 1\rangle$  is stationary.
- $B_z = 3.483\,38\,\text{G} \approx \frac{64Ag_I}{3\mu_B g_S^2}$ : the energy difference between  $|F = 3, M_F = -2\rangle$  and  $|F = 4, M_F = 2\rangle$  is stationary.
- $B_z = 8.9572\,\text{G} \approx \frac{384Ag_I}{7\mu_B g_S^2}$ : the energy difference between  $|F = 3, M_F = -3\rangle$  and  $|F = 4, M_F = 3\rangle$  is stationary.

**Q3.13** In the result from

```
1 In[809]:=Assuming[A>0, FullSimplify[T/.{Bx->0,By->0,Bz->0,Bacx->B,Bacy->I*B,Bacz->0}]]
```

we can see that the transitions  $1 \leftrightarrow 5$ ,  $1 \leftrightarrow 6$ ,  $3 \leftrightarrow 7$ ,  $3 \leftrightarrow 8$ ,  $4 \leftrightarrow 7$ ,  $4 \leftrightarrow 8$  are allowed. Using Eq. (3.9) we identify these transitions as  $|2, 2\rangle \leftrightarrow |1, 1\rangle$ ,  $|2, 2\rangle \leftrightarrow |2, 1\rangle$ ,  $|1, 0\rangle \leftrightarrow |1, -1\rangle$ ,  $|1, 0\rangle \leftrightarrow |2, -1\rangle$ ,  $|2, 0\rangle \leftrightarrow |1, -1\rangle$ ,  $|2, 0\rangle \leftrightarrow |2, -1\rangle$ . These transitions are all  $\Delta M_F = \pm 1$ .

**Q3.14** In the result from

```
1 In[810]:=Assuming[A>0, FullSimplify[T/.{Bx->0,By->0,Bz->0,Bacx->0,Bacy->0,Bacz->B}]]
```

we can see that the transitions  $3 \leftrightarrow 4$ ,  $5 \leftrightarrow 6$ ,  $7 \leftrightarrow 8$  are allowed. Using Eq. (3.9) we identify these transitions as  $|1, 0\rangle \leftrightarrow |2, 0\rangle$ ,  $|1, 1\rangle \leftrightarrow |2, 1\rangle$ ,  $|1, -1\rangle \leftrightarrow |2, -1\rangle$ . These transitions are all  $\Delta M_F = 0$ . Further, the energy of levels 1, 2, 5, 6, 7, 8 (*i.e.*, all levels with  $M_F \neq 0$ ) will be shifted by the non-zero diagonal elements of  $\mathbf{T}$ .

**Q3.15** On a MacBook Pro (Retina, 13-inch, Early 2015) with a 3.1 GHz Intel Core i7 CPU and 16 GB 1867 MHz DDR3 RAM, it takes around 20 min (`AbsoluteTiming[γ=gs[1,1];]`) to calculate the ground state `gs[1,1]` with the definition of `In[350]` and  $N = 22$ . This calculation uses over 24 GB of compressed RAM (`MaxMemoryUsed[]`) and is the upper limit on  $N$  for this computer. For  $N = 23$  the calculation runs out of memory.

**Q3.16** With the Mathematica code of Sect. 3.4, setting  $S = 1$  and  $N = 12$ , we find a phase transition around  $b = \pm 2$ . Similar to the  $S = 1/2$  case, the Ising model is gapless for  $|b| < 2$  and gapped for  $|b| > 2$ . The correlations look qualitatively similar to the ones found for  $S = 1/2$ .

**Q3.17**

1. For  $b \rightarrow \pm\infty$  the ground states are analogous to those of the transverse Ising model, Eq. (3.28), along the  $\pm z$  axis:

$$|\psi_{+\infty}\rangle = |+\rangle^{\otimes N}, \quad |\psi_{-\infty}\rangle = |-\rangle^{\otimes N}. \quad (11)$$

Notice that, unlike the transverse Ising model, these asymptotic ground states are the exact ground states for  $|b| > 2$ , not just in the limits  $b \rightarrow \pm\infty$ .

2. There are phase transitions at  $b = \pm 2$ , recognizable in the ground-state gap.
3. Since the states of Eq. (11) are product states, there are absolutely no correlations between the states of the spins for  $|b| > 2$ . For  $|b| < 2$  the magnetization, spin-spin correlations, and entanglement entropy are qualitatively similar to those of the transverse Ising model. For  $b = 0$  the spin-spin correlations do not reach the full uniform 0.25 as for the Ising model, but rather they still decay with distance.

**Q3.18**

1. For  $b \rightarrow \pm\infty$  the ground states are the same as Eq. (11).
2. At  $b = 0$  the ground-state degeneracy is  $N + 1$ .

3. For any  $b > 0$ ,  $|\psi_{+\infty}\rangle$  is the exact ground state; for any  $b < 0$ ,  $|\psi_{-\infty}\rangle$  is the exact ground state. There is a phase transition at  $b = 0$ .
4. Since the states of Eq. (11) are product states, there are absolutely no correlations between the states of the spins for any  $b \neq 0$ .

### Q3.19

1.  $\hat{\rho}_{AB} = |\psi\rangle\langle\psi| = |\uparrow\uparrow\rangle\langle\uparrow\uparrow|$ :

```

1 In[811]:=ψ = Flatten[KroneckerProduct[{1,0}, {1,0}]]
2 Out[811]= {1, 0, 0, 0}
3 In[812]:=ρAB = KroneckerProduct[ψ, Conjugate[ψ]]
4 Out[812]= {{1,0,0,0}, {0,0,0,0}, {0,0,0,0}, {0,0,0,0}}
```

2.  $\hat{\rho}_A = \text{Tr}_B \hat{\rho}_{AB} = |\uparrow\rangle\langle\uparrow|$  is a pure state:

```

1 In[813]:=ρA = traceout[ρAB, -2]
2 Out[813]= {{1,0}, {0,0}}
3 In[814]:=Tr[ρA.ρA]
4 Out[814]= 1
```

3.  $\hat{\rho}_B = \text{Tr}_A \hat{\rho}_{AB} = |\uparrow\rangle\langle\uparrow|$  is a pure state:

```

1 In[815]:=ρB = traceout[ρAB, 2]
2 Out[815]= {{1,0}, {0,0}}
3 In[816]:=Tr[ρB.ρB]
4 Out[816]= 1
```

4. Using In[353] and In[354], we see that the entropy of entanglement is  $S_A - S_{AB} = S_B - S_{AB} = 0$  (no entanglement):

```

1 In[817]:=SAB = Total[s /@ Eigenvalues[ρAB]]
2 Out[817]= 0
3 In[818]:=SA = Total[s /@ Eigenvalues[ρA]]
4 Out[818]= 0
5 In[819]:=SB = Total[s /@ Eigenvalues[ρB]]
6 Out[819]= 0
```

### Q3.20

1.  $\hat{\rho}_{AB} = |\psi\rangle\langle\psi| = \frac{|\uparrow\downarrow\rangle - |\downarrow\uparrow\rangle}{\sqrt{2}} \frac{\langle\uparrow\downarrow| - \langle\downarrow\uparrow|}{\sqrt{2}} = \frac{1}{2}(|\uparrow\downarrow\rangle\langle\uparrow\downarrow| - |\uparrow\downarrow\rangle\langle\downarrow\uparrow| - |\downarrow\uparrow\rangle\langle\uparrow\downarrow| + |\downarrow\uparrow\rangle\langle\downarrow\uparrow|)$ :

```

1 In[820]:=ψ = Flatten[KroneckerProduct[{1,0}, {0,1}]
2 - KroneckerProduct[{0,1}, {1,0}]]/Sqrt[2]
3 Out[820]= {0, 1/Sqrt[2], -1/Sqrt[2], 0}
4 In[821]:=ρAB = KroneckerProduct[ψ, Conjugate[ψ]]
5 Out[821]= {{0,0,0,0}, {0,1/2,-1/2,0}, {0,-1/2,1/2,0}, {0,0,0,0}}
```

2.  $\hat{\rho}_A = \text{Tr}_B \hat{\rho}_{AB} = \frac{1}{2}(|\uparrow\rangle\langle\uparrow| + |\downarrow\rangle\langle\downarrow|)$  is a mixed state:

```

1 In[822]:=ρA = traceout[ρAB, -2]
2 Out[822]= {{1/2,0}, {0,1/2}}
3 In[823]:=Tr[ρA.ρA]
4 Out[823]= 1/2
```

3.  $\hat{\rho}_B = \text{Tr}_A \hat{\rho}_{AB} = \frac{1}{2}(|\uparrow\rangle\langle\uparrow| + |\downarrow\rangle\langle\downarrow|)$  is a mixed state:

```

1 In[824]:=ρB = traceout[ρAB, 2]
2 Out[824]={{1/2,0}, {0,1/2}}
3 In[825]:=Tr[ρB.ρB]
4 Out[825]=1/2

```

4. Using In[353] and In[354], we see that the entropy of entanglement is  $S_A - S_{AB} = S_B - S_{AB} = 1$  (maximal entanglement):

```

1 In[826]:=SAB = Total[s /@ Eigenvalues[ρAB]]
2 Out[826]=0
3 In[827]:=SA = Total[s /@ Eigenvalues[ρA]]
4 Out[827]=1
5 In[828]:=SB = Total[s /@ Eigenvalues[ρB]]
6 Out[828]=1

```

**Q3.21** Don't forget to complex-conjugate  $\psi_{\text{out}}$  on the left, for generality:

```

1 In[829]:=ψout = {1/Sqrt[2], 0, 0, 1/Sqrt[2]};
2 In[830]:=Table[Conjugate[ψout].(KroneckerProduct[PauliMatrix[i],PauliMatrix[j]].ψout),
3   {i,0,3}, {j,0,3}]
4 Out[830]={{1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, -1, 0}, {0, 0, 0, 1}}

```

We find  $\langle\psi_{\text{out}}|\mathbb{1} \otimes \mathbb{1}|\psi_{\text{out}}\rangle = 1$  (normalization),  $\langle\psi_{\text{out}}|\hat{\sigma}_x \otimes \hat{\sigma}_x|\psi_{\text{out}}\rangle = 1$ ,  $\langle\psi_{\text{out}}|\hat{\sigma}_y \otimes \hat{\sigma}_y|\psi_{\text{out}}\rangle = -1$ ,  $\langle\psi_{\text{out}}|\hat{\sigma}_z \otimes \hat{\sigma}_z|\psi_{\text{out}}\rangle = 1$ , and all others equal to zero. The density matrix is therefore

$$\begin{aligned}
 \hat{\rho} &= \frac{1}{4} (\mathbb{1} \otimes \mathbb{1} + \hat{\sigma}_x \otimes \hat{\sigma}_x - \hat{\sigma}_y \otimes \hat{\sigma}_y + \hat{\sigma}_z \otimes \hat{\sigma}_z) \\
 &= \frac{1}{2} (|00\rangle\langle 00| + |00\rangle\langle 11| + |11\rangle\langle 00| + |11\rangle\langle 11|) = \frac{|00\rangle + |11\rangle}{\sqrt{2}} \frac{\langle 00| + \langle 11|}{\sqrt{2}}
 \end{aligned} \tag{12}$$

as expected.

**Q3.22** Replace In[390] and In[391] with

```

1 In[831]:=u = {1,1}/Sqrt[2];
2 In[832]:=U[φ_] = Exp[2π*I*φ] * {{1,0},{0,1}};

```

and re-evaluate the attached Mathematica notebook. All results remain unchanged.

**Q3.23** Replace In[390] and In[391] with

```

1 In[833]:=u = {1,1}/Sqrt[2];
2 In[834]:=U[φ_] = {{Exp[2π*I*φ], 0}, {0, Exp[4π*I*φ]}};


```

and re-evaluate the attached Mathematica notebook. The probabilities for the different estimates of  $\varphi$  show both frequencies simultaneously, and there is no cross-talk between them:

```

1 In[835]:=ListDensityPlot[Transpose[Table[prob[φ], {φ,0,1,1/256}]],
2   PlotRange->All, DataRange->{{0,1},{0,1-2^-t}},
3   FrameLabel->{"setting φ","estimated φ"}]

```



## Chapter 4 Quantum Motion in Real Space

### Q4.1 Starting with the Schrödinger equation

$$\left[ -\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} dx |x\rangle \frac{d^2}{dx^2} \langle x| + \int_{-\infty}^{\infty} dx |x\rangle V(x) \langle x| \right] |\psi\rangle = E |\psi\rangle, \quad (13)$$

we (i) leave away the bracket and (ii) multiply by  $\langle y|$  from the left ( $y \in \mathbb{R}$ ):

$$-\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} dx \langle y|x\rangle \frac{d^2}{dx^2} \langle x|\psi\rangle + \int_{-\infty}^{\infty} dx \langle y|x\rangle V(x) \langle x|\psi\rangle = E \langle y|\psi\rangle \quad (14)$$

Remembering that  $\langle y|x\rangle = \delta(x - y)$  and  $\langle x|\psi\rangle = \psi(x)$ :

$$-\frac{\hbar^2}{2m} \int_{-\infty}^{\infty} dx \delta(x - y) \psi''(x) + \int_{-\infty}^{\infty} dx \delta(x - y) V(x) \psi(x) = E \psi(y) \quad (15)$$

Simplify the integrals with the Dirac  $\delta$ -functions:

$$-\frac{\hbar^2}{2m} \psi''(y) + V(y) \psi(y) = E \psi(y) \quad (16)$$

Since this is valid for any  $y \in \mathbb{R}$ , it concludes the proof.

## Q4.2

$$\begin{aligned}
\langle \psi | \chi \rangle &= \left[ \int_{-\infty}^{\infty} dx \psi^*(x) \langle x | \right] \left[ \int_{-\infty}^{\infty} dy \chi(y) | y \rangle \right] \\
&= \int_{-\infty}^{\infty} dx dy \psi^*(x) \chi(y) \langle x | y \rangle \\
&= \int_{-\infty}^{\infty} dx dy \psi^*(x) \chi(y) \delta(x - y) \\
&= \int_{-\infty}^{\infty} dx \psi^*(x) \chi(x)
\end{aligned} \tag{17}$$

## Q4.3

```

1 In[836]:=a = m = ħ = 1; (* natural units *)
2 In[837]:=nmax = 100;
3 In[838]:=TM = SparseArray[Band[{1,1}] -> Range[nmax]^2*π^2*ħ^2/(2*m*a^2)];
4 In[839]:=pM = SparseArray[{n1_,n2_};OddQ[n1-n2]->(4*I*ħ*n1*n2)/(a*(n2^2-n1^2)),
5 {nmax,nmax}];
6 In[840]:=TM //N //Eigenvalues //Sort
7 Out[840]={4.9348, 19.7392, 44.4132, 78.9568, 123.37, 177.653, ..., 48366., 49348.}
8 In[841]:=pM.pM/(2m) //N //Eigenvalues //Sort
9 Out[841]={4.8183, 4.8183, 43.3646, 43.3646, 120.457, 120.457, ..., 47257.4, 47257.4}

```

The eigenvalues of  $\hat{T}$  are quadratically spaced, whereas those of  $\hat{p}^2/(2m)$  come in degenerate pairs (one involving only states of even  $n$  and one only states of odd  $n$ ) and thus never converge to the eigenvalues of  $\hat{T}$ , even in the limit  $n_{\max} \rightarrow \infty$ .

**Q4.4** We use the more accurate form of the position operator from In[434]:

```

1 In[842]:=a = m = ħ = 1; (* natural units *)
2 In[843]:=nmax = 20;
3 In[844]:=xM = SparseArray[{
4   Band[{1,1}] -> a/2,
5   {n1_,n2_} /; OddQ[n1-n2] -> -8*a*n1*n2/(π^2*(n1^2-n2^2)^2)},
6 {nmax,nmax}];
7 In[845]:=pM = SparseArray[{n1_,n2_};OddQ[n1-n2]->4*I*ħ*n1*n2/(a*(n2^2-n1^2)),
8 {nmax,nmax}];
9 In[846]:=coM = xM.pM - pM.xM; (* commutator [x,p] in the momentum basis *)
10 In[847]:=coM/ħ //N //MatrixForm

```

In the upper-left corner (low values of  $n$ ) the result looks like the unit matrix multiplied by the imaginary unit  $i$ ; but towards the lower-right corner (large values of  $n$ ) it deviates dramatically from the correct expression. This is to be expected from the problematic nature of the momentum operator; see Sect. 4.1.6.

**Q4.5** The exact probability is about 37.1%:

```

1 In[848]:=Integrate[Ψ[1,x]^2, {x, 0, 1}] //N
2 Out[848]=0.37087

```

Using In[460], the first numerical method gives a good approximation of 37.0%:

```

1 In[849]:=Integrate[Ψ0[x]^2, {x, 0, 1}]
2 Out[849]=0.369801

```



Using `In[477]`, the second numerical method gives an approximation of 36.2%:

```
1 In[850]:=Integrate[ψ0[x]^2, {x, 0, 1}]
2 Out[850]=0.362126
```

Alternatively, we set up an interpolating function from the data of `In[474]`, and integrate it numerically. The result depends on the interpolation order: higher-order interpolations tend to yield more accurate results.

```
1 In[851]:=ψ0i1 = Interpolation[γ, InterpolationOrder -> 1];
2 In[852]:=NIntegrate[ψ0i1[x]^2, {x, 0, 1}]
3 Out[852]=0.302899
4 In[853]:=ψ0i2 = Interpolation[γ, InterpolationOrder -> 2];
5 In[854]:=NIntegrate[ψ0i2[x]^2, {x, 0, 1}]
6 Out[854]=0.358003
7 In[855]:=ψ0i3 = Interpolation[γ, InterpolationOrder -> 3];
8 In[856]:=NIntegrate[ψ0i3[x]^2, {x, 0, 1}]
9 Out[856]=0.3812
```

**Q4.6** From Eq. (4.31) the average height in state  $k$  is

$$\langle k|\hat{x}|k\rangle = \int_0^\infty dx |\psi_k(x)|^2 x = -\alpha_k \cdot \left( \frac{4\hbar^2}{27m^2g} \right)^{1/3}, \quad (18)$$

which you can verify with `In[448]` and

```
1 In[857]:=Assuming[m>0&&g>0&&ħ>0, Table[Integrate[ψ[k,x]^2*x, {x, 0, ∞}], {k, 1, 5}]]
```

For a neutron in earth's gravitational field this gives an average height of about 9  $\mu\text{m}$ :

```
1 In[858]:=With[{k = 1,
2             m = Quantity["NeutronMass"],
3             g = Quantity["StandardAccelerationOfGravity"],
4             ħ = Quantity["ReducedPlanckConstant"]},
5             UnitConvert[-AiryAiZero[k]*(4*ħ^2/(27*m^2*g))^(1/3), "Micrometers"]]
6 Out[858]=9.147654 μm
```

**Q4.7** The exact energy levels are  $E_n = \hbar\omega(n + 1/2)$  with  $n \in \mathbb{N}_0$ .

In the given unit system, the mass is  $m = 1$ , Planck's constant is  $\hbar = 1$ , and the angular frequency is  $\omega = 1/\hbar = 1$ .

We set up a calculation in the position basis with the mixed-basis numerical method:

```
1 In[859]:=a = 10; (* calculation box size *)
2 In[860]:=m = ħ = ω = 1; (* natural units *)
3 In[861]:=nmax = 100;
4 In[862]:=Δ = a/(nmax+1); (* grid spacing *)
5 In[863]:=xgrid = Range[nmax]*Δ; (* the computational grid *)
6 In[864]:=TP = FourierDST[SparseArray[Band[{1,1}]->Range[nmax]^2*π^2*ħ^2/(2*m*a^2)], 1];
7 In[865]:=W[x_] = m*ω^2*(x-a/2)^2/2; (* the potential function, centered *)
8 In[866]:=Wgrid = Map[W, xgrid]; (* the potential on the computational grid *)
9 In[867]:=VP = SparseArray[Band[{1,1}] -> Wgrid];
10 In[868]:=HP = TP + VP;
```


We find the energy eigenvalues (in units of  $\mathcal{E} = \hbar\omega$ ) with

```
1 In[869]:=Eigenvalues[HP] //Sort
2 Out[869]={0.5, 1.5, 2.5, 3.5, 4.50001, 5.5001, 6.5006, 7.50293, 8.51147, 9.53657, ...}
```

and see that at least the lowest eigenvalues match the analytic expression. Using a larger value of  $n_{\max}$  will give more accurate eigenstates and eigenvalues.

**Q4.8** The excited-state Wigner distribution has a significant negative region around its center:

```
1 In[870]:=gsP = Transpose[Sort[Transpose[-Eigensystem[-N[HP], 2,
2 Method->{"Arnoldi", "Criteria"->"RealPart", MaxIterations->10^6}]]];
3 In[871]:=WignerDistributionPlot[gsP[[2, 2]], {0, a}]
```



#### Q4.9

```
1 In[872]:=X = RandomReal[{0,1}, {2,2}]
2 Out[872]={{0.580888, 0.80848}, {0.218175, 0.979598}}
3 In[873]:=Y = RandomReal[{0,1}, {2,2}]
4 Out[873]={{0.448364, 0.774595}, {0.490198, 0.310169}}
5 In[874]:=X.Y - Y.X
6 Out[874]={{0.227318, -0.420567}, {0.225597, -0.227318}}
7 In[875]:=MatrixExp[X + Y]
8 Out[875]={{4.68326, 6.06108}, {2.71213, 5.68068}}
9 In[876]:=MatrixExp[X].MatrixExp[Y]
10 Out[876]={{5.0593, 5.38209}, {3.10936, 5.31705}}
```


**Q4.10** We use the split-step propagation code of Sect. 4.1.9 with the potential

```
1 In[877]:=a = m = h = 1;
2 In[878]:=With[{W0 = 0 * h^2/(m*a^2)},
3 W[x_] = W0*Sin[10*π*x/a];]
```


and the initial wavefunction

```
1 In[879]:=With[{x0=a/2, σ=0.05*a, k=100/a},
2 v0=Normalize[Function[x, E^-((x-x0)^2/(4*σ^2))]*E^(I*k*x)] /@ xgrid;]
```

For  $W_0 = 0$  the Gaussian wavepacket bounces back and forth between the simulation boundaries and disperses slowly; the self-interference at the reflection points is clearly visible:



For  $W_0 = 5000 \frac{\hbar^2}{ma^2}$  the Gaussian wavepacket remains mostly trapped:




#### Q4.11

1.  $W(x)$  is a double-well potential with minima at  $x = \frac{1}{2} \pm \delta$  and a barrier height of  $\Omega$ :

```

1 In[880]:=W[{Ω_, δ_}, x_] = Ω*((x-1/2)/δ)^2-1)^2;
2 In[881]:=Table[W[{Ω,δ},x], {x,1/2-δ,1/2+δ,δ}]
3 Out[881]={0, Ω, 0}
4 In[882]:=Table[D[W[{Ω,δ},y],y] /. y->x, {x,1/2-δ,1/2+δ,δ}]
5 Out[882]={0, 0, 0}
6 In[883]:=Plot[W[{1, 1/4}, x], {x, 0, 1}]

```



2. As in **Q4.10**, we use the split-step propagation code of Sect.4.1.9 with the potential

```

1 In[884]:=With[{Ω = 250, δ = 1/4},
2   W[x_] = W[{Ω, δ}, x];]


```

With the initial state from [In\[879\]](#) ([Q4.10](#)) with  $x_0 = 0.2694$ ,  $\sigma = 0.0554$ ,  $k = 0$ , the time-dependent density is seen to oscillate between the wells:

```

1 In[885]:=With[{Δt = 20, M = 10^4},
2   V = propApprox[Δt, M, v0];]
3 In[886]:=ρ = ArrayPad[(nmax+1)*Abs[#[[2]]]^2& /@ V, {{0,0},{1,1}}];
4 In[887]:=ArrayPlot[Reverse[Transpose[ρ]]]

```




This oscillation is apparent in the left/right probabilities:

```

1 In[888]:=ListLinePlot[{{#[[1]], Norm[#[[2,;;(nmax/2)]]]^2& /@ V,
2   {#[[1]], Norm[#[[2,nmax/2+1;]]]^2& /@ V}}]

```




- Now we use [In\[500\]](#) and observe that the attractive interactions prevent the particle from tunneling between the wells:

```

1 In[889]:=With[{κ = 0.5, Δt = 20, M = 10^4},
2   V = propApprox[W[#1]&, κ, Δt, M, v0];]
3 In[890]:=ρ = ArrayPad[(nmax+1)*Abs[#[[2]]]^2& /@ V, {{0,0},{1,1}}];
4 In[891]:=ArrayPlot[Reverse[Transpose[ρ]]]


```



```

1 In[892]:=ListLinePlot[{Norm[#[[2,;;(nmax/2)]]]^2& /@ V,
2             Norm[#[[2,nmax/2+1;]]]^2& /@ V]}

```



#### Q4.12

1. We do this calculation in Mathematica, for the more general potential  $W(x) = \frac{1}{2}k(x - \frac{1}{2})^2$ :

```

1 In[893]:=ζ[x_] = E^(-(x-1/2)/(2*σ))^2/Sqrt[σ*Sqrt[2*π]];
2 In[894]:=Assuming[σ>0, Integrate[ζ[x]^2, {x, -∞, ∞}]]
3 Out[894]=1
4 In[895]:=Assuming[σ>0,
5           e = Integrate[ζ[x]*(-1/2*ζ'[x] + 1/2*k*(x-1/2)^2*ζ[x]), {x, -∞, ∞}]]
6 Out[895]=(1+4*k*σ^4)/(8*σ^2)
7 In[896]:=Solve[D[e, σ] == 0, σ]
8 Out[896]={{σ -> -1/(Sqrt[2]*k^(1/4))}, {σ -> -I/(Sqrt[2]*k^(1/4))},
9           {σ -> I/(Sqrt[2]*k^(1/4))}, {σ -> 1/(Sqrt[2]*k^(1/4))}}

```

Of these four solutions, we choose  $\sigma = (4k)^{-1/4}$  because it is real and positive:

```

1 In[897]:=e /. σ -> (4*k)^(-1/4)
2 Out[897]=Sqrt[k]/2

```

For  $k = 5000$ , the ground state is therefore  $\zeta(x)$  with  $\sigma = 20000^{-1/4} \approx 0.0840896$  and energy  $E = \sqrt{5000}/2 \approx 35.3553$ .

2. We use the same code as in Sect. 4.2.1 but with the potential

```

1 In[898]:=With[{k = 5000},
2           W[x_] = 1/2*k*(x-1/2)^2;]


```

Further, we use `nmax = 1000` to describe the wavefunction with strongly attractive interactions better. The result matches the Gaussian approximation: both the energy and the chemical potential are approximately  $\sqrt{k}/2$ ,

```
1 In[899]:=groundstate[10^-4, 0][[;;2]]
2 Out[899]= {35.3553, 35.3553}
```


### 3. Ground-state density for repulsive interactions:

```
1 In[900]:=With[{κ = 100, δβ = 10^-4},
2   {Etot, μ, γ} = groundstate[δβ, κ];
3   ListLinePlot[Join[{0, 0}], Transpose[{xgrid, Abs[γ]^2/Δ}], {{a, 0}}],
4   PlotRange -> All, PlotLabel -> {Etot, μ}]
```




### Ground-state density for no interactions:

```
1 In[901]:=With[{κ = 0, δβ = 10^-4},
2   {Etot, μ, γ} = groundstate[δβ, κ];
3   ListLinePlot[Join[{0, 0}], Transpose[{xgrid, Abs[γ]^2/Δ}], {{a, 0}}],
4   PlotRange -> All, PlotLabel -> {Etot, μ}]
```




### Ground-state density for attractive interactions:

```
1 In[902]:=With[{κ = -100, δβ = 10^-4},
2   {Etot, μ, γ} = groundstate[δβ, κ];
3   ListLinePlot[Join[{0, 0}], Transpose[{xgrid, Abs[γ]^2/Δ}], {{a, 0}}],
4   PlotRange -> All, PlotLabel -> {Etot, μ}]
```



4. The energy and chemical potential differ for  $\kappa \neq 0$ :

```
In[903]:=With[{delta beta = 10^-4},
  ListLinePlot[Transpose[Table[{kappa, groundstate[delta beta, kappa][[1]]},
    {kappa, groundstate[delta beta, kappa][[2]]}], {kappa, -100, 100, 10}]]]
```



**Q4.13** We do this calculation for  $a = 1$ ; the prefactor  $a^{-1}$  of the right-hand side of Eq.(4.65) can be found by a variable substitution  $x \mapsto ax'$ . From the momentum basis functions

```
1 In[904]:=phi[n_, x_] = Sqrt[2]*Sin[n*pi*x];
```

we define the position basis functions

```
1 In[905]:=theta[nmax_, j_, x_] := 1/Sqrt[nmax+1]*Sum[phi[n, j/(nmax+1)]*phi[n, x], {n, nmax}]
```

The exact overlap integrals are

```
1 In[906]:=J[nmax_, {j1_, j2_, j3_, j4_}] :=
2 Integrate[theta[nmax, j1, x]*theta[nmax, j2, x]*theta[nmax, j3, x]*theta[nmax, j4, x], {x, 0, 1}]
```

We make a table of overlap integrals, calculated both exactly and approximately through In[504], and show that the difference is zero (up to numerical inaccuracies):

```

1 In[907]:=With[{nmax = 3},
2   A = Table[J[nmax, {j1,j2,j3,j4}], {j1,nmax},{j2,nmax},{j3,nmax},{j4,nmax}];
3   B = FourierDST[Table[KroneckerDelta[n1+n2,n3+n4]
4     +KroneckerDelta[n1+n3,n2+n4]+KroneckerDelta[n1+n4,n2+n3]
5     -KroneckerDelta[n1,n2+n3+n4]-KroneckerDelta[n2,n1+n3+n4]
6     -KroneckerDelta[n3,n1+n2+n4]-KroneckerDelta[n4,n1+n2+n3],
7     {n1,nmax},{n2,nmax},{n3,nmax},{n4,nmax}],1]/2;]
8   A - B //Abs //Max
9 Out[907]=8.88178*10^-16

```

**Q4.14** We define memoizing functions that calculate  $\langle \hat{x}_1 - \hat{x}_2 \rangle$  and  $\langle (\hat{x}_1 - \hat{x}_2)^2 \rangle$  with

```

1 In[908]:=Clear[Δ1,Δ2];
2 In[909]:=Δ1[κ_?NumericQ] := Δ1[κ] =
3   With[{γ=gs[0,κ,1][[2,1]]}, Re[Conjugate[γ].((x1-x2).γ)]]
4 In[910]:=Δ2[κ_?NumericQ] := Δ2[κ] =
5   With[{γ=gs[0,κ,1][[2,1]]}, Re[Conjugate[γ].((x1-x2).(x1-x2).γ)]]


```

The mean distance in the ground state is zero for symmetry reasons: (notice the numerical inaccuracies)

```

1 In[911]:=ListLinePlot[Table[{κ, Δ1[κ]}, {κ, -25, 25, 1}]]

```




The variance of the distance in the ground state increases with  $\kappa$ :

```

1 In[912]:=ListLinePlot[Table[{κ, Δ2[κ]-Δ1[κ]^2}, {κ, -25, 25, 1}]]

```





**Q4.15** We show that all three terms of the Hamiltonian commute with the particle interchange operator:

```

1 In[913]:=Ξ.TP - TP.Ξ //Norm
2 Out[913]=0.
3 In[914]:=Ξ.VP - VP.Ξ //Norm
4 Out[914]=0
5 In[915]:=Ξ.HintP - HintP.Ξ //Norm
6 Out[915]=1.15903*10^-14

```

### Q4.16

```

1 In[916]:=D[Normal[HPa[Ω, κ]], κ] //Abs //Max
2 Out[916]=1.03528*10^-15

```

**Q4.17** We define the wavefunctions  $\psi_1(x_1, x_2)$  for  $x_1 < x_2$  and  $\psi_2(x_1, x_2)$  for  $x_1 > x_2$ :

```

1 In[917]:=ψ1[x1_, x2_] = A*(Cos[α*(x1+x2-1)]*Cos[β*(x1-x2+1)]
2               -Cos[α*(x1-x2+1)]*Cos[β*(x1+x2-1)]);
3 In[918]:=ψ2[x1_, x2_] = ψ1[x2, x1];

```

1. Check the boundary conditions  $\psi(x_1, 0) = \psi(x_1, 1) = \psi(0, x_2) = \psi(1, x_2) = 0$ :

```

1 In[919]:= {ψ2[x1, 0], ψ1[x1, 1], ψ1[0, x2], ψ2[1, x2]} //FullSimplify
2 Out[919]= {0, 0, 0, 0}

```

Check that the two pieces match up for  $x_1 = x_2$ :

```

1 In[920]:= ψ1[x, x] == ψ2[x, x]
2 Out[920]= True

```

Check the symmetries of the wavefunction:

```

1 In[921]:= ψ1[x1, x2] == ψ2[x2, x1] == ψ2[1-x1, 1-x2] //FullSimplify
2 Out[921]= True

```

2. Check that the two pieces of the wavefunction satisfy the Schrödinger equation whenever  $x_1 \neq x_2$ , with energy value  $E = \alpha^2 + \beta^2$ :

```

1 In[922]:= -1/2*D[ψ1[x1, x2], {x1, 2}] + D[ψ1[x1, x2], {x2, 2}] ==
2   (α^2+β^2)*ψ1[x1, x2] //FullSimplify
3 Out[922]= True
4 In[923]:= -1/2*D[ψ2[x1, x2], {x1, 2}] + D[ψ2[x1, x2], {x2, 2}] ==
5   (α^2+β^2)*ψ2[x1, x2] //FullSimplify
6 Out[923]= True

```

3. The transformed Hamiltonian is

$$\hat{\mathcal{H}} = \left[ -\frac{1}{2} \frac{\partial^2}{\partial R^2} \right] + \left[ -\frac{1}{2} \frac{\partial^2}{\partial r^2} + \frac{\kappa}{\sqrt{2}} \delta(r) \right] \quad (19)$$

and the transformed wavefunctions are

```

1 In[924]:=ψ1[(R+r)/Sqrt[2],(R-r)/Sqrt[2]] //FullSimplify
2 Out[924]=A*(Cos[α*(R*Sqrt[2]-1)]*Cos[β*(r*Sqrt[2]+1)]
3           -Cos[β*(R*Sqrt[2]-1)]*Cos[α*(r*Sqrt[2]+1)])
4 In[925]:=ψ2[(R+r)/Sqrt[2],(R-r)/Sqrt[2]] //FullSimplify
5 Out[925]=A*(Cos[α*(R*Sqrt[2]-1)]*Cos[β*(r*Sqrt[2]-1)]
6           -Cos[β*(R*Sqrt[2]-1)]*Cos[α*(r*Sqrt[2]-1)])

```

with

$$\psi(R, r) = \begin{cases} \psi_1(R, r) & \text{if } r < 0 \\ \psi_2(R, r) & \text{if } r > 0 \end{cases} \quad (20)$$

4. The Schrödinger equation in  $(R, r)$  coordinates is

$$\left[ -\frac{1}{2} \frac{\partial^2}{\partial R^2} \right] \psi(R, r) + \left[ -\frac{1}{2} \frac{\partial^2}{\partial r^2} + \frac{\kappa}{\sqrt{2}} \delta(r) \right] \psi(R, r) = (\alpha^2 + \beta^2) \psi(R, r) \quad (21)$$

We integrate Eq. (21) over  $r \in [-\epsilon, \epsilon]$ :

$$-\frac{1}{2} \int_{-\epsilon}^{\epsilon} dr \frac{\partial^2 \psi(R, r)}{\partial R^2} - \frac{1}{2} \int_{-\epsilon}^{\epsilon} dr \frac{\partial^2 \psi(R, r)}{\partial r^2} + \frac{\kappa}{\sqrt{2}} \int_{-\epsilon}^{\epsilon} dr \delta(r) \psi(R, r) = (\alpha^2 + \beta^2) \int_{-\epsilon}^{\epsilon} dr \psi(R, r) \quad (22)$$

Using partial integration on the second term of the left-hand side:

$$-\frac{1}{2} \int_{-\epsilon}^{\epsilon} dr \frac{\partial^2 \psi(R, r)}{\partial R^2} - \frac{1}{2} \left[ \frac{\partial \psi(R, r)}{\partial r} \Big|_{r=\epsilon} - \frac{\partial \psi(R, r)}{\partial r} \Big|_{r=-\epsilon} \right] + \frac{\kappa}{\sqrt{2}} \psi(R, 0) = (\alpha^2 + \beta^2) \int_{-\epsilon}^{\epsilon} dr \psi(R, r) \quad (23)$$

In the limit  $\epsilon \rightarrow 0^+$  this equation becomes

$$-\frac{1}{2} \left[ \frac{\partial \psi_2(R, r)}{\partial r} \Big|_{r=0} - \frac{\partial \psi_1(R, r)}{\partial r} \Big|_{r=0} \right] + \frac{\kappa}{\sqrt{2}} \psi(R, 0) = 0 \quad (24)$$

Inserting the definitions of  $\psi_1$  and  $\psi_2$ :

```

1 In[926]:=-1/2*(D[ψ2[(R+r)/Sqrt[2],(R-r)/Sqrt[2]],r]/.r->0)
2           - (D[ψ1[(R+r)/Sqrt[2],(R-r)/Sqrt[2]],r]/.r->0))
3           + κ/Sqrt[2]*ψ1[R/Sqrt[2],R/Sqrt[2]] //FullSimplify
4 Out[926]=A*(Cos[β*(R*Sqrt[2]-1)]*(2α*Sin[α]-κ*Cos[α])
5           -Cos[α*(R*Sqrt[2]-1)]*(2β*Sin[β]-κ*Cos[β]))/Sqrt[2]

```

The only way that this expression can be zero for all values of  $R \in [0, \sqrt{2}]$  is if  $2\alpha \sin(\alpha) - \kappa \cos(\alpha) = 2\beta \sin(\beta) - \kappa \cos(\beta) = 0$ , and hence if  $\alpha \tan(\alpha) = \beta \tan(\beta) = \kappa/2$ .

5. See the attached Mathematica notebook `ContactInteraction.nb`.

**Q4.18** We solve this problem with the code of Sect. 4.3.1, in the same way as **Q4.14**. The interaction potential is, according to Eq. (4.71),

```

1 In[927]:=With[{δ=Δ},
2   Q[x_] = Piecewise[{{1/Abs[x],Abs[x]>δ}, {1/δ,Abs[x]<=δ}}];]

```


and the interaction Hamiltonian  $\hat{\mathcal{H}}_{\text{int}} \approx \kappa/|x|$  is approximately

```


1 In[928]:=HintP = SparseArray[{j1_,j1_,j2_,j2_} :> Q[xgrid[[j1]]-xgrid[[j2]]],
2   {nmax,nmax,nmax,nmax}] //ArrayFlatten;

```

With these definitions, the energy levels are (with  $a = m = \hbar = 1$ )



We see that the lowest energy level is always symmetric under particle exchange (colored in red); the bosonic ground state is therefore just the lowest energy level. The expectation value  $\langle \hat{x}_1 - \hat{x}_2 \rangle$  is zero by symmetry; its variance is




**Q4.19** We see in the answer of **Q4.18** that the lowest fermionic state (blue) depends on the coupling strength  $\kappa$ . In the spirit of Sect. 4.3.1 we define the fermionic Hamiltonian with [In \[529\]](#) and calculate the fermionic ground state with [In \[533\]](#). The expectation values of  $\hat{x}_1 - \hat{x}_2$  and  $(\hat{x}_1 - \hat{x}_2)^2$  are calculated from the antisymmetric ground state with

```

1 In[929]:=Clear[FΔx, FΔx2];
2 In[930]:=FΔx[κ_?NumericQ] := FΔx[κ] =
3   With[{γ=ags[0,κ,1][[2,1]]}, Re[Conjugate[γ].((x1-x2).γ)]]
4 In[931]:=FΔx2[κ_?NumericQ] := FΔx2[κ] =
5   With[{γ=ags[0,κ,1][[2,1]]}, Re[Conjugate[γ].((x1-x2).(x1-x2).γ)]]

```

The expectation value  $\langle \hat{x}_1 - \hat{x}_2 \rangle$  is zero by symmetry; its variance is larger than that for bosons:



**Q4.20** The expectation values are the usual ones of the harmonic oscillator, given by

$$\langle x^2 \rangle = \frac{\hbar}{2m\omega_x}, \quad \langle y^2 \rangle = \frac{\hbar}{2m\omega_y}, \quad \langle z^2 \rangle = \frac{\hbar}{2m\omega_z}. \quad (25)$$

They are independent in the three Cartesian directions.

**Q4.21** We calculate the integral over the density in Cartesian coordinates by integrating only over the ellipsoid in which the density is nonzero:

```

1 In[932]:= A = Assuming[Rx>0 && Ry>0 && Rz>0,
2   Integrate[p0*(1-(x/Rx)^2-(y/Ry)^2-(z/Rz)^2),
3     {x, -Rx, Rx},
4     {y, -Ry*Sqrt[1-(x/Rx)^2], Ry*Sqrt[1-(x/Rx)^2]},
5     {z, -Rz*Sqrt[1-(x/Rx)^2-(y/Ry)^2], Rz*Sqrt[1-(x/Rx)^2-(y/Ry)^2]}]]
6 Out[932]= 8/15*π*Rx*Ry*Rz*p0

```

Similarly, we calculate the integral of the density times  $x^2$  with

```

1 In[933]:= B = Assuming[Rx>0 && Ry>0 && Rz>0,
2   Integrate[x^2 * p0*(1-(x/Rx)^2-(y/Ry)^2-(z/Rz)^2),
3     {x, -Rx, Rx},
4     {y, -Ry*Sqrt[1-(x/Rx)^2], Ry*Sqrt[1-(x/Rx)^2]},
5     {z, -Rz*Sqrt[1-(x/Rx)^2-(y/Ry)^2], Rz*Sqrt[1-(x/Rx)^2-(y/Ry)^2]}]]
6 Out[933]= 8/105*π*Rx^3*Ry*Rz*p0

```

The expectation value  $\langle x^2 \rangle$  is the ratio of these two integrals,

```

1 In[934]:= B/A
2 Out[934]= Rx^2/7

```

With the value of  $R_x$  given in Eq. (4.79a), this becomes


$$\langle x^2 \rangle = \frac{1}{7} \left[ \frac{15\hbar^2 a_s (N-1) \omega_y \omega_z}{m^2 \omega_x^4} \right]^{\frac{2}{5}} = \frac{1}{7} \left[ \frac{15\kappa (N-1) \omega_y \omega_z}{4\pi m \omega_x^4} \right]^{\frac{2}{5}}, \quad (26a)$$

$$\langle y^2 \rangle = \frac{1}{7} \left[ \frac{15\hbar^2 a_s (N-1) \omega_x \omega_z}{m^2 \omega_y^4} \right]^{\frac{2}{5}} = \frac{1}{7} \left[ \frac{15\kappa (N-1) \omega_x \omega_z}{4\pi m \omega_y^4} \right]^{\frac{2}{5}}, \quad (26b)$$

$$\langle z^2 \rangle = \frac{1}{7} \left[ \frac{15\hbar^2 a_s (N-1) \omega_x \omega_y}{m^2 \omega_z^4} \right]^{\frac{2}{5}} = \frac{1}{7} \left[ \frac{15\kappa (N-1) \omega_x \omega_y}{4\pi m \omega_z^4} \right]^{\frac{2}{5}}. \quad (26c)$$

We see that, in contrast to **Q4.20**, the expectation values of the three Cartesian directions are not independent of each other's trapping frequencies.

**Q4.22** We plot the second moments of Eqs. (25) and (26a) as functions of the particle number  $N$ :



The values of  $\langle z^2 \rangle$  are equal to those of  $\langle y^2 \rangle$  because of the cylindrical symmetry of the problem. The crossover point where the Thomas–Fermi second moment is equal to the noninteracting second moment is at

$$\bar{N}_x = \frac{49}{60a_s\omega_y\omega_z} \sqrt{\frac{7\hbar\omega_x^3}{2m}} + 1, \quad \bar{N}_y = \frac{49}{60a_s\omega_x\omega_z} \sqrt{\frac{7\hbar\omega_y^3}{2m}} + 1, \quad \bar{N}_z = \frac{49}{60a_s\omega_x\omega_y} \sqrt{\frac{7\hbar\omega_z^3}{2m}} + 1, \quad (27)$$

indicated with vertical lines in the above plot. The noninteracting limit, Eq. (25), is good for  $N \lesssim 10$ . The Thomas–Fermi limit, Eq. (26a), is good for  $N \gtrsim 5000$ . Notice that for  $N \gtrsim 3000$  the numeric value of  $\langle x^2 \rangle$  deviates from the Thomas–Fermi limit because of the finite size of the calculation box.

## Chapter 5 Combining Spatial Motion and Spin

**Q5.1** The operators for these expectation values are

1.  $A1 = \text{KroneckerProduct}[xop, \Pi_\uparrow] = \text{KroneckerProduct}[xop, ids/2+sz]$
2.  $A2 = \text{KroneckerProduct}[xop, \Pi_\downarrow] = \text{KroneckerProduct}[xop, ids/2-sz]$
3.  $A3 = \text{KroneckerProduct}[xop, ids] = A1+A2$
4.  $A4 = \text{KroneckerProduct}[xop, sz] = (A1-A2)/2$

With these we evaluate the quantities

1.  $\text{Re}[\text{Conjugate}[\gamma] \cdot (\text{A1} \cdot \gamma)]$
2.  $\text{Re}[\text{Conjugate}[\gamma] \cdot (\text{A2} \cdot \gamma)]$
3.  $\text{Re}[\text{Conjugate}[\gamma] \cdot (\text{A3} \cdot \gamma)]$
4.  $\text{Re}[\text{Conjugate}[\gamma] \cdot (\text{A4} \cdot \gamma)]$  for the mean  
 $\text{Re}[\text{Conjugate}[\gamma] \cdot (\text{A4} \cdot \text{A4} \cdot \gamma) - (\text{Conjugate}[\gamma] \cdot (\text{A4} \cdot \gamma))^2]$  for the variance

**Q5.2** The ordering of the subspaces of the Hilbert space is what matters here. We have defined the Hilbert space to be a tensor product of the  $x$ ,  $y$ , and spin degrees of freedom, in this order. In [In \[629\]](#) the operators  $\hat{p}_x$  and  $\hat{p}_y$  are distinguished by the position in the Kronecker product in which **PM** appears.

**Q5.3** We do the first two checks of **Q3.2**:

$$\begin{aligned}
 1. \quad [\hat{S}_x, \hat{S}_y] &= \left[ \frac{|e\rangle\langle g| + |g\rangle\langle e|}{2}, \frac{|e\rangle\langle g| - |g\rangle\langle e|}{2i} \right] \\
 &= \frac{(|e\rangle\langle g| + |g\rangle\langle e|)(|e\rangle\langle g| - |g\rangle\langle e|) - (|e\rangle\langle g| - |g\rangle\langle e|)(|e\rangle\langle g| + |g\rangle\langle e|)}{4i} \\
 &= \frac{(-|e\rangle\langle e| + |g\rangle\langle g|) - (|e\rangle\langle e| - |g\rangle\langle g|)}{4i} \\
 &= \frac{|g\rangle\langle g| - |e\rangle\langle e|}{2i} = i \frac{|e\rangle\langle e| - |g\rangle\langle g|}{2} = i\hat{S}_z \quad (28)
 \end{aligned}$$

$$\begin{aligned}
 [\hat{S}_y, \hat{S}_z] &= \left[ \frac{|e\rangle\langle g| - |g\rangle\langle e|}{2i}, \frac{|e\rangle\langle e| - |g\rangle\langle g|}{2} \right] \\
 &= \frac{(|e\rangle\langle g| - |g\rangle\langle e|)(|e\rangle\langle e| - |g\rangle\langle g|) - (|e\rangle\langle e| - |g\rangle\langle g|)(|e\rangle\langle g| - |g\rangle\langle e|)}{4i} \\
 &= \frac{(-|e\rangle\langle g| + |g\rangle\langle e|) - (|e\rangle\langle g| + |g\rangle\langle e|)}{4i} \\
 &= \frac{-|e\rangle\langle g| - |g\rangle\langle e|}{2i} = i \frac{|e\rangle\langle g| + |g\rangle\langle e|}{2} = i\hat{S}_x \quad (29)
 \end{aligned}$$

$$\begin{aligned}
 [\hat{S}_z, \hat{S}_x] &= \left[ \frac{|e\rangle\langle e| - |g\rangle\langle g|}{2}, \frac{|e\rangle\langle g| + |g\rangle\langle e|}{2} \right] \\
 &= \frac{(|e\rangle\langle e| - |g\rangle\langle g|)(|e\rangle\langle g| + |g\rangle\langle e|) - (|e\rangle\langle g| + |g\rangle\langle e|)(|e\rangle\langle e| - |g\rangle\langle g|)}{4} \\
 &= \frac{(|e\rangle\langle g| - |g\rangle\langle e|) - (-|e\rangle\langle g| + |g\rangle\langle e|)}{4} \\
 &= \frac{|e\rangle\langle g| - |g\rangle\langle e|}{2} = i \frac{|e\rangle\langle g| - |g\rangle\langle e|}{2i} = i\hat{S}_y \quad (30)
 \end{aligned}$$

$$\begin{aligned}
 2. \quad \hat{S}_x^2 + \hat{S}_y^2 + \hat{S}_z^2 &= \left( \frac{|e\rangle\langle g| + |g\rangle\langle e|}{2} \right)^2 + \left( \frac{|e\rangle\langle g| - |g\rangle\langle e|}{2i} \right)^2 + \left( \frac{|e\rangle\langle e| - |g\rangle\langle g|}{2} \right)^2 \\
 &= \frac{|e\rangle\langle e| + |g\rangle\langle g|}{4} + \frac{|e\rangle\langle e| + |g\rangle\langle g|}{4} + \frac{|e\rangle\langle e| + |g\rangle\langle g|}{4} \\
 &= \frac{3}{4}(|g\rangle\langle g| + |e\rangle\langle e|) = \frac{3}{4}\mathbb{1} \text{ and hence } S = 1/2. \quad (31)
 \end{aligned}$$

**Q5.4**  $\hat{S}^+ = \hat{S}_x + i\hat{S}_y = \frac{|e\rangle\langle g| + |g\rangle\langle e|}{2} + i\frac{|e\rangle\langle g| - |g\rangle\langle e|}{2i} = \frac{|e\rangle\langle g| + |g\rangle\langle e|}{2} + \frac{|e\rangle\langle g| - |g\rangle\langle e|}{2} = |e\rangle\langle g|$ .  $\hat{S}^- = \hat{S}_x - i\hat{S}_y = \frac{|e\rangle\langle g| + |g\rangle\langle e|}{2} - i\frac{|e\rangle\langle g| - |g\rangle\langle e|}{2i} = \frac{|e\rangle\langle g| + |g\rangle\langle e|}{2} - \frac{|e\rangle\langle g| - |g\rangle\langle e|}{2} = |g\rangle\langle e|$ . We can see that  $\hat{S}^+$  is the operator that excites the atom ( $\hat{S}^+|g\rangle = |e\rangle$ ) and  $\hat{S}^-$  is the operator that deexcites the atom ( $\hat{S}^-|e\rangle = |g\rangle$ ).

**Q5.5**  $[\hat{X}, \hat{P}] = \hat{X}\hat{P} - \hat{P}\hat{X} = \frac{\hat{a} + \hat{a}^\dagger}{\sqrt{2}} \frac{\hat{a} - \hat{a}^\dagger}{i\sqrt{2}} - \frac{\hat{a} - \hat{a}^\dagger}{i\sqrt{2}} \frac{\hat{a} + \hat{a}^\dagger}{\sqrt{2}}$   
 $= \frac{(\hat{a}\hat{a} - \hat{a}\hat{a}^\dagger + \hat{a}^\dagger\hat{a} - \hat{a}^\dagger\hat{a}^\dagger) - (\hat{a}\hat{a} + \hat{a}\hat{a}^\dagger - \hat{a}^\dagger\hat{a} - \hat{a}^\dagger\hat{a}^\dagger)}{2i} = i[\hat{a}, \hat{a}^\dagger] = i$ .

**Q5.6** Cavity field:  $\hat{a}^\dagger\hat{a} = \frac{\hat{X} - i\hat{P}}{\sqrt{2}} \frac{\hat{X} + i\hat{P}}{\sqrt{2}} = \frac{\hat{X}^2 + \hat{P}^2 + i[\hat{X}, \hat{P}]}{2} = \frac{\hat{X}^2 + \hat{P}^2 - 1}{2}$  and hence  $\hat{a}^\dagger\hat{a} + \frac{1}{2} = \frac{1}{2}\hat{P}^2 + \frac{1}{2}\hat{X}^2$ . Coupling:  $\hat{S}^+\hat{a} + \hat{a}^\dagger\hat{S}^- = (\hat{S}_x + i\hat{S}_y) \frac{\hat{X} + i\hat{P}}{\sqrt{2}} + \frac{\hat{X} - i\hat{P}}{\sqrt{2}} (\hat{S}_x - i\hat{S}_y) = \frac{\hat{S}_x\hat{X} + i\hat{S}_x\hat{P} + i\hat{S}_y\hat{X} - \hat{S}_y\hat{P} + \hat{X}\hat{S}_x - i\hat{X}\hat{S}_y - i\hat{P}\hat{S}_x - \hat{P}\hat{S}_y}{\sqrt{2}}$ . Since the operators on the field and atom degrees of freedom commute (for example,  $[\hat{X}, \hat{S}_x] = [\hat{X} \otimes \mathbb{1}, \mathbb{1} \otimes \hat{S}_x] = 0$ ), this becomes  $\hat{S}^+\hat{a} + \hat{a}^\dagger\hat{S}^- = \frac{\hat{X}\hat{S}_x + i\hat{P}\hat{S}_x + i\hat{X}\hat{S}_y - \hat{P}\hat{S}_y + \hat{X}\hat{S}_x - i\hat{X}\hat{S}_y - i\hat{P}\hat{S}_x - \hat{P}\hat{S}_y}{\sqrt{2}} = \sqrt{2}(\hat{X}\hat{S}_x - \hat{P}\hat{S}_y)$ .

# Index

## A

Airy function, 103  
Angular momentum, 51, 52  
Arnoldi algorithm, 24

## B

Basis set, 34, 44  
    construction, 42  
    finite-resolution position basis, 96  
    incomplete, 35  
    momentum basis, 94  
    position basis, 92  
Bohr magneton, 54, 57  
Boltzmann constant, 119  
Bose–Einstein condensate, 117, 130  
Boson, 122, 126

## C

C, 7, 8, 21  
Chemical potential, 117  
Clebsch–Gordan coefficient, 3, 67  
Completeness relation, 34, 35, 92  
Contact interaction, 122  
Correlations, 75  
Coulomb interaction, 130  
    truncated, 130

## D

Decoherence, 61  
Density matrix, 108  
    reduced, *see* partial trace  
Detuning, 64  
Dicke states, 42, 43, 52  
Discrete Fourier transform, 85

Discrete sine transform, 98, 132  
Double slit experiment, 33

## E

Electron, 54  
Energy gap, 72  
Entanglement, 77  
Entropy of entanglement, 77

## F

Fast Fourier transform, 85, 98  
Fermion, 122, 126  
Fock basis, 149  
Fortran, 21

## G

$g$ -factor, 54, 57  
Gravitational acceleration, 103  
Gravity well, 102  
Gross–Pitaevskii equation, *see* Schrödinger equation, non-linear

## H

Harmonic oscillator, 43  
Heisenberg model, 79  
Heisenberg principle, *see* uncertainty principle  
Hilbert space, 34, 35, 37, 42, 56, 72, 138  
Hydrogen, 43  
Hyperfine interaction, 57

## I

Imaginary-time propagation, 119



Interaction, 43, 122  
 Interaction picture, 40  
 Ising model, 68, 79

## J

Java, 7, 8, 21  
 Jaynes–Cummings model, 148

## K

Kinetic energy, *see* operator, kinetic

## L

Lagrange multiplier, 117  
 Lanczos algorithm, 24  
 Level shift, 67  
 Light shift, 67

## M

Magnetic field, 54, 56  
 Magnetization, 74  
 Magnus expansion, 38  
 Mandelbrot set, 3  
 Mathematica, 1  
   anonymous function, 9, 19, 108  
   assumptions, 29  
   brackets, 5, 21, 22  
   complex number, 28  
   conditional execution, 8  
   debugging, 17  
   delayed assignment, 5, 12  
   differential equation, 63  
   evaluation, 17  
   factorial, 19  
   fixed point of a map, 120  
   front end, 2  
   full form, 18, 29  
   function, 6, 11  
   functional programming, 9, 19, 20, 132  
   immediate assignment, 4, 11  
   kernel, 2  
   Kronecker product, 44, 45, 57, 70, 124  
   list, 6, 21  
   loop, 7, 19, 20  
   matrix, 22, 35  
     eigenvalues, 24, 55, 58  
     eigenvectors, 24, 55, 58  
     exponential, 39, 41, 42  
     identity matrix, 46, 53  
     matrix exponential, 110  
     printing, 7, 22, 53

  sparse matrix, 23, 53  
   minimization, 61  
   module, 9  
   nesting function calls, 112  
   numerical evaluation, 7  
   outer product, 132  
   pattern, 11, 13, 20, 23  
     alternative, 78  
   physical units, 29  
   plotting, 58, 63, 124  
   postfix notation, 6  
   prefix notation, 6  
   random number, 4, 12  
   recursion, 19, *see also* recursion  
   remembering results, 12  
   replacements, 15  
   rules, 15  
   saving definitions, 13  
   timing a calculation, 7, 53  
   tracing, 17  
   units, *see* physical units  
   variable, 4  
   vector, 21, 35  
     normalize, 120  
     orthogonal vectors, 58

Matlab, 21

Mean-field interaction, 117

Memoization, 12

Momentum operator, *see* operator, momentum

Moore's law, 72

## N

Nuclear spin, 57

Nyquist frequency, 108

## O

Operator, 34, 45

  kinetic, 43, 95, 146  
   momentum, 101, 146, 150  
   position, 150  
   potential, 43

Oscillating field, 61

## P

Partial trace, 46, 77, 143

Path integral, 72

Pauli matrices, 36, 53, 54

Planck's constant, 57, 131

Plane wave, 43

Potential energy, *see* operator, potential

Product state, 45, 70  
Propagator, 38, 110  
Pseudospin, 152  
Pseudovector, 52  
Python, 7, 21

## Q

Quantum circuit, 80  
Quantum Fourier transform, 85  
Quantum gate, 80  
Quantum information, 61  
Quantum phase estimation, 87  
Quantum phase transition, 73  
Quantum state, 34, 44  
Quantum state tomography, 84  
Qubit, 61, 80

## R

Rabi frequency, 65  
Rashba coupling, 145  
Real-space dynamics, 92, 137  
Reciprocal lattice, 43  
Reduced density matrix, *see* partial trace  
Rotating-wave approximation, 64  
Rotation, 53  
Rubidium-87, 56, 131  
    magic field, 61

## S

$s$ -wave scattering, 117, 123, 131  
Schrödinger equation  
    non-linear, 116, 117  
    time-dependent, 38, 40, 62, 109, 115  
    time-independent, 36, 55  
Spherical harmonics, 43  
Spin, 43, 51, 56, 137

Split-step method, 99, 112, 115, 118  
Square well, 43  
Stark shift  
    ac, 67  
Stern–Gerlach experiment, 33  
Sturm–Liouville theorem, 95

## T

Tensor, 26  
    contraction, 27, 47  
    product, 43, 57, 69, 123, 137  
Tensor networks, 72  
Thomas–Fermi approximation, 135  
Transition matrix elements, 62  
Trotter expansion, 110, 116

## U

Uncertainty principle, 33

## V

Von Neumann entropy, 77

## W

Wigner distribution, 107, 151  
Wolfram language, 1

## X

XY model, 79

## Z

Zeeman shift  
    ac, 67  
    dc, 60